

John M. Nevison

Programmazione in Basic per l'uomo d'affari



Il piacere del computer





Il piacere del computer

Il piacere del computer serie diretta da Mauro Boscarol

- 1 *Tom Rugg e Phil Feldman* 32 programmi con il PET
- 2 *Rich Didday* Intervista sul personal computer, hardware
- 3 *Tom Rugg e Phil Feldman* 32 programmi con l'Apple
- 4 *Ken Knecht* Microsoft Basic
- 5 *Paul M. Chirlian* Pascal
- 6 *Tom Rugg e Phil Feldman* 32 programmi con il TRS-80
- 7 *Rich Didday* Intervista sul personal computer, software
- 8 *Herbert D. Peckham* Imparate il Basic con il PET/CBM
- 9 *Karl Townsend e Merl Miller* Il personal computer come professione
- 10 *Karen Billings e David Moursund* Te ne intendi di computer?
- 11 *Thomas Dwyer e Margot Critchfield* Il Basic e il personal computer, uno: introduzione
- 12 *Don Inman e Kurt Inman* Imparate il linguaggio dell'Apple
- 13 *Thomas Dwyer e Margot Critchfield* Il Basic e il personal computer, due: applicazioni
- 14 *Luigi Pierro* Il manuale del CP/M
- 15 *Carlo Sintini* A scuola con il PET/CBM
- 16 *David-Johnson-Davies* Il manuale dell'Atom
- 17 *David E. Schultz* Il libro del Commodore VIC 20
- 18 *Him Huffman e Robert Bruce* Il "debug" nei personal computer
- 19 *John M. Nevison* Programmazione in Basic per l'uomo d'affari

John M. Nevison

*Programmazione
in Basic
per l'uomo d'affari*



franco muzzio & c. editore

Titolo originale *Executive computing*
Traduzione di Flavio Santini
Copertina di Elda Danese

Prima edizione: ottobre 1983
ISBN 88-7021-232-7

© 1983 franco muzzio & c. editore
Via Bonporti 36, 35141 Padova, tel. 049/661147-661873
© 1981 Addison Wesley Publishing Company, Inc.
Tutti i diritti sono riservati.

Indice generale

PARTE PRIMA: L'UFFICIO DELL'UOMO D'AFFARI

- 9 Introduzione: uno sguardo ai numeri**
Crescita composta I prezzi durante l'inflazione Il programma
INFLAZIONE Grafici delle vendite L'ufficio Il programma
VENDITE
- 20 La pianificazione a lungo termine**
Cominciando con le ipotesi Tassi di crescita Il programma PIA-
NO Flusso di cassa Il programma CASSA Aiuto nel budget
Il programma TABELLA
- 37 La revisione della strategia**
Uno sguardo ai costi Analisi Il programma RAGGI-X Il valo-
re dell'esperienza Il programma QUOTA Crescita concorrenzia-
le Il programma CRESCITA Risultati finali
- 64 Valore attuale e rischio futuro**
Quattro metodi Una nuova politica Il programma ATTUALE
Rischio Il programma RISCHIO
- 80 I dati societari**
Struttura Il data base della società Utenti di data base Il pro-
gramma RICHIESTA


 PARTE SECONDA: PROGRAMMARE LA PRODUZIONE

- 93 Pianificazione dei progetti, programmazione e controllo**
 Contributo alla cena Il programma GANTT Dalle operazioni alle reti di operazioni Il nuovo diagramma di GANTT La tabella dei predecessori Il programma CPM-I Pianificazione con scarsità di risorse Il programma CPM-II Il controllo del progetto Il programma CPM-III Stime dei tempi La previsione dei lavori Tutti i lavori assieme Il programma TEMPI
- 134 La gestione delle scorte**
 Un modello di magazzino Il programma MAGAZZINO Scorta per il rifornimento del magazzino Il programma MAG Ritardo nei rifornimenti Domanda Il programma MAGAZZ Esaurimento delle scorte Costi bilanciati Sconti sulla quantità e variazioni nelle forniture
- 154 Il problema della dieta: programmazione lineare**
 Il programma DIETA Riflessione La tabella di Tucker Il metodo del semplice a otto passi I risultati Un altro problema Misurare un utile Applicazioni Uguaglianze Il programma SEMPLICE Problemi di trasporto
- 183 Il problema dell'ombrello: analisi delle decisioni**
 Un programma Il programma DECISIONE Una decisione di marketing Un giochetto finale
- 194 Previsioni: vendite del prossimo mese, riunioni della prossima settimana**
 Il programma REGRESSIONE Previsione dei mesi seguenti Il programma PROSMESE Riunioni della prossima settimana Il programma TEMPOLIBERO

 APPENDICI

- 215 Introduzione al Basic ed alla strutturazione dei programmi**
 Il Basic in azione Stile e struttura Paragrafi di programma Il programma più grande
- 233 Programmi d'utilità**
- 249 L'acquisto**
 Verifica delle esigenze Buona assistenza Buona marca
- 252 Il VisiCalc**

Parte prima:
l'ufficio dell'uomo d'affari



Introduzione: uno sguardo ai numeri

Steven Caldwell sapeva che c'erano molti vantaggi nello scrivere da sé i propri programmi, ma non aveva idea di quale fosse il più importante. Egli sapeva che la scrittura di un programma lo obbligava ad esprimere le proprie idee in un modo differente e che proprio lo sforzo di riesprimerle spesso gli faceva capire qualcosa di nuovo. Per esempio, aveva imparato come rielaborare la semplice idea dell'interesse composto per trattare i prezzi nei periodi di inflazione.

CRESCITA COMPOSTA

Di per sé la crescita composta era un'idea molto semplice. La crescita al 15 per cento equivaleva a moltiplicare per un fattore 1.15. In altre parole, la nuova somma doveva essere 1.15 volte la vecchia. In Basic si faceva così:

$$S=1.15*S$$

Questa linea si legge al contrario. La S iniziale è alla destra del segno di uguale. La nuova S sulla sinistra è il risultato. Quando egli scrisse un programma di quattro linee e numerò ogni istruzione, il computer eseguì le istruzioni nell'ordine:

```

10 S=100
20 S=1.15*S
30 PRINT S
40 END

```

e la risposta fu

115

L'idea iniziale ebbe una semplice conseguenza. Per comporre l'interesse di cinque anni, si poteva ripetere la stessa operazione cinque volte. Usando l'istruzione NEXT per rimandare il computer all'istruzione FOR, egli fece ciclare il computer riutilizzando le stesse istruzioni:

```

10 S=100
20 REM
30 FOR Y=1 TO 5
40 S=1.15*S
50 PRINT Y,S
60 NEXT Y
70 REM
80 END

```

(Le linee 20 e 70 possono contenere dei commenti. Non fanno niente ma rendono il programma più semplice da leggere.) Il computer usò le vecchie istruzioni con i nuovi valori ogni volta che esegui il ciclo. I risultati furono:

1	115
2	132.25
3	152.088
4	174.901
5	201.136

I PREZZI DURANTE L'INFLAZIONE

Calcolare i prezzi in un periodo d'inflazione era un'elaborazione dell'idea dell'interesse composto. Ma costi differenti avevano tassi d'inflazione differenti. Il costo del lavoro aumentava al tasso dell'1.15 all'anno, mentre il materiale aumentava al tasso dell'1.03 all'anno. Se per ogni dollaro c'erano 52 cents di materiale e 25 cents di lavoro, cosa succedeva al margine di profitto quando i prezzi aumentavano ad un tasso dell'1.12 all'anno?

Il programma per rispondere alla domanda era:

```

270 P=100
280 R=56
290 L=21
300 PRINT O:R,L,R+L,P,(P-(R+L))/P*100
310 FOR Y=1 TO 5
315 P=1.12*P
320 R=1.03*R
330 L=1.15*L
340 PRINT Y:R,L,R+L,P,(P-(R+L))/P*100
350 NEXT Y
999 END

```

I risultati furono:

ANNO	MAT.	LAVORO	COSTO	PREZZO	MARGINE (%)
0	56	21	77	100	23
1	57.68	24.15	81.83	112	26.9375
2	59.4104	27.7725	87.1829	125.44	30.4983
3	61.1927	31.9384	93.1311	140.493	33.7111
4	63.0285	36.7291	99.7576	157.352	36.6022
5	64.9193	42.2385	107.158	176.234	39.1958

Il tasso di crescita del prezzo 1.12 era chiaramente alto, così Cauldwell ribatté il tasso a 1.10 ed eseguì nuovamente il programma. Il margine era sempre troppo alto. Egli provò con 1.08, 1.06 ed infine con 1.07 per trovare il valore che conservava il margine.

Nel passare dalla crescita composta ai prezzi, Cauldwell aveva imparato che un'idea espressa come programma di calcolatore è quasi sempre suscettibile di essere perfezionata. Egli poteva, se l'avesse voluto, andare avanti e realizzare un modello con cinque o sei componenti di costo. Così il primo vantaggio del computer era la libertà di riesprimere e perfezionare un'idea.

Il programma illustrava un secondo vantaggio: facilità di calcolo. Era più facile e più veloce per Cauldwell scrivere e rivedere il programma che trovare la risposta in qualsiasi altro modo. Con i numeri egli temeva sempre di fare qualche errore; con il suo calcolatore questo pericolo non c'era.

C'era un terzo vantaggio: facilità di riutilizzazione. La più ovvia era quando i tassi cambiavano: e cambiavano sempre. Ora che il programma era scritto, poteva salvarlo e riutilizzarlo velocemente quando le circostanze richiedevano la revisione del piano.

C'era un quarto vantaggio collegato con la facilità di riutilizzazione: la capacità di rispondere alla domanda "cosa succede se?". Un programma di calcolatore poteva essere visto come un insieme di ipotesi, e quello che il computer faceva era trarre le necessarie conclusioni in una frazione di secondo. Così se egli desiderava usare INFLAZIONE per trovare quanto influiva sul suo margine un differente saggio di inflazione, egli lo poteva fare. In effetti, provando alcuni casi, egli poteva sviluppare la

propria intuizione su quello che il futuro poteva riservare, sia in minacce che in promesse. Questo era certo un grosso vantaggio dei computer.

IL PROGRAMMA INFLAZIONE

```

100 REM ***INFLAZIONE***
110 REM
114 REM
120 REM CALCOLA IL MARGINE DI PROFITTO IN UN PERIODO DI INFLAZIONE
130 REM IN CUI I COSTI DELLE MATERIE PRIME E DEL LAVORO E I
140 REM PREZZI AUMENTANO A TASSI DIFFERENTI.
145 REM
150 REM VARIABILI:
160 REM L.....COSTO DEL LAVORO
170 REM P.....PREZZO
180 REM R.....COSTO DELLE MATERIE PRIME
190 REM
200 PRINT"ANNO MAT.", "LAVORO", "COSTO", "PREZZO", "MARGINE(%) ""
210 PRINT
220 REM
270 P=100
280 R=56
290 L=21
300 PRINT O:R,L,R+L,P, (P-(R+L))/P*100
310 FOR Y=1 TO 5
315 P=1.12*P
320 R=1.03*R
330 L=1.15*L
340 PRINT Y:R,L,R+L,P, (P-(R+L))/P*100
350 NEXT Y
999 END

```

Se non avete mai visto un programma Basic, dovete fermarvi qui e leggere l'Appendice A. Se conoscete già il Basic, questo programma non dovrebbe riservare sorprese per voi.

Il programma usa tre variabili, R per le materie prime, L per il lavoro e P per i prezzi. Prima il programma stampa l'intestazione. Quindi cominciano i calcoli, e dopo aver stampato i valori iniziali al tempo zero, il programma stampa i valori di ogni anno per cinque anni successivi.

Per provare un differente incremento dei prezzi, dovete solo ribattere la linea $P = 1.12 * P$ con un nuovo tasso.

ESERCIZI

1. Verificate che un incremento del prezzo di 1.07 mantiene il margine costante. Si tratta esattamente della risposta giusta?
2. Supponete che il materiale grezzo costi 40 lire, l'energia 16 lire e il lavoro 21 lire. Se i costi aumentano come prima e l'energia cresce ad un tasso di 1.35, cosa succede al margine se il prezzo incrementa a 1.15? Rivedete INFLAZIONE includendo i costi dell'energia.

Steven Cauldwell era il direttore generale della Chordata, e le sue giornate erano dense di attività. Proprio prima di lasciare il suo ufficio, una sera, seppe che un meeting che doveva tenersi alla fine della settimana era stato spostato alle otto e trenta del mattino dopo. Il meeting era con Frank Bradshaw, il direttore della divisione Bear. La divisione Bear produceva elettrodomestici. Era la divisione più grossa della compagnia e le sue vendite non andavano bene. Erano le nove di sera. Il problema di Cauldwell era che aveva solo il resto della notte per esaminare le cifre e capire cosa era successo.

Le cifre che aveva di fronte erano le vendite mensili degli ultimi cinque anni. Egli era anche in possesso di cifre simili per ognuna delle altre tre divisioni della Chordata. Cauldwell ricordò un grafico che aveva visto in un recente articolo, e che gli aveva dato un mezzo veloce e completo per esaminare le cifre delle vendite. Il grafico era composto di due parti: un grafico sommario a barre delle vendite medie di diversi anni passati, ed un grafico a linea spezzata delle vendite degli ultimi dodici mesi. Era simile a quello di figura 1.1.



Fig. 1.1 Andamento delle vendite

Un tale grafico indicava velocemente quello che serviva conoscere. Quella sera, Cauldwell si preparò a realizzare un programma per disegnare il grafico. Lavorò al programma in senso inverso, scrivendo prima la sezione che doveva stampare il grafico a linea spezzata mensile.

```

800 REM GRAFICO MENSILE
810 REM
820 DIM S(12)
830 FOR M=1 TO 12
840 READ S(M)
850 DATA 14,20,22,24,26,15,16,20,26,23,19,17

```

```

860      PRINT TAB(S(M)); "*"
870      NEXT M
999      END

```

Il programma stampava una vendita mensile su ogni linea, in modo che l'anno correva lungo la pagina. Per fare il grafico, egli girò la carta dalla sua parte e unì con una matita gli asterischi. Il risultato era come questo:




La prossima parte che scrisse era la routine per stampare il grafico a barre delle vendite medie mensili degli anni passati. Il programma leggeva la media annuale Y e stampava una linea di asterischi lunga Y . Egli stimò alcuni dati approssimati per un'esecuzione di prova.

```

500 REM GRAFICO A BARRE
510 REM
520   FOR I=1 TO 4
530     READ Y
540     DATA 15,18,19,21
545     FOR J=1 TO Y
550       PRINT "*";
560     NEXT J
570     PRINT
580   NEXT I
590 REM
600 REM GRAFICO MENSILE
610 REM
620   DIM S(12)
630   FOR M=1 TO 12
640     READ S(M)
650     DATA 14,20,22,24,26,15,16,20,26,23,19,17
660     PRINT TAB(S(M)); "*"
670   NEXT M
999   END

```

Quando esegui l'intero programma, ne risultò questo:



Dieci minuti dopo aver iniziato, Caldwell aveva completato il programma essenziale. Egli desiderava un grafico facile da leggere, e così continuò a rivedere e migliorare il suo sforzo iniziale. Dopo aver migliorato l'estetica della stampa, rivolse la sua attenzione ai dati. Arrivavano mensilmente, e quindi il suo programma doveva calcolare le vendite degli anni passati per avere le medie e i totali annui. Egli voleva che il programma stampasse le cifre dei più recenti dodici mesi, così che la sua segretaria potesse riutilizzare il programma il mese dopo aggiungendo il dato del nuovo mese e aumentando di uno il numero del mese corrente.

Egli continuò il suo cammino all'indietro dai risultati alle condizioni iniziali. Quando aggiunse la sezione che leggeva i nuovi dati scartò i dati di prova iniziali. Quindi aggiunse i nomi dei mesi in modo che potessero essere stampati sul grafico. Infine, aggiunse i commenti ad ogni passo così che la versione finale fosse semplice da leggere. (Un commento a questo programma è alla fine del capitolo.)

```

100 REM ***VENDITE***
110 REM
120 REM
140 REM QUESTO PROGRAMMA STAMPA UN GRAFICO (*) DELL'ANDAMENTO DELLE
150 REM VENDITE MENSILI DELLA COMPAGNIA. L'ANDAMENTO NEI QUATTRO
160 REM ANNI PRECEDENTI E' VISUALIZZATO MEDIANTE UN GRAFICO A BARRE.
180 REM
190 REM VARIABILI:
200 REM I,J.....VARIABILI INDICE
210 REM L*( ).....ETICHETTE PER I MESI
220 REM M.....INDICE DEI MESI
230 REM S( ).....VENDITE PER MESE
240 REM Y( ).....MEDIA DELLE VENDITE ANNUALI
250 REM
260 REM DIMENSIONI:
270 DIM L*(60),S(60)
280 REM
290 REM COSTANTI:
300 REM
310 FOR M=37 TO 60
320 READ L*(M)
330 NEXT M
340 DATA GEN,FEB,MAR,APR,MAG,GIU,LUG,AGO,SET,OTT,NOV,DIC
350 DATA GEN,FEB,MAR,APR,MAG,GIU,LUG,AGO,SET,OTT,NOV,DIC


```

```

360 REM
370 FOR M=1 TO 48
380 READ S(M)
390 NEXT M
400 DATA 10,14,16,18,20,16,15,13,11,13,14,15
410 DATA 10,15,19,20,21,20,19,15,17,20,22,21
420 DATA 12,17,19,23,22,19,18,15,18,20,23,25
430 DATA 15,17,21,27,25,20,19,14,20,22,24,26
440 REM
450 C=7
460 FOR M=48+1 TO 48+C
470 READ S(M)
480 NEXT M
490 DATA 15,16,20,26,23,19,17
500 REM
510 REM
520 REM PROGRAMMA PRINCIPALE
530 REM
540 REM CALCOLA LA MEDIA DELLE VENDITE ANNUALI
550 REM
560 Y(1)=S(1)+S(2)+S(3)+S(4)+S(5)+S(6)+S(7)
570 Y(1)=Y(1)+S(8)+S(9)+S(10)+S(11)+S(12)
580 Y(1)=Y(1)/12
590 REM
600 Y(2)=S(13)+S(14)+S(15)+S(16)+S(17)+S(18)+S(19)
610 Y(2)=Y(2)+S(20)+S(21)+S(22)+S(23)+S(24)
620 Y(2)=Y(2)/12
630 REM
640 Y(3)=S(25)+S(26)+S(27)+S(28)+S(29)+S(30)+S(31)
650 Y(3)=Y(3)+S(32)+S(33)+S(34)+S(35)+S(36)
660 Y(3)=Y(3)/12
670 REM
680 Y(4)=S(37)+S(38)+S(39)+S(40)+S(41)+S(42)+S(43)
690 Y(4)=Y(4)+S(44)+S(45)+S(46)+S(47)+S(48)
700 Y(4)=Y(4)/12
710 REM
720 REM STAMPA IL GRAFICO
730 REM
740 REM INTRESTAZIONE
750 REM
760 PRINT" 0 10 20";
770 PRINT" 30 40";
775 PRINT" +-----+";
776 PRINT"-----+";
780 REM
790 REM GRAFICO A BARRE ANNUALE
800 REM
810 FOR I=1 TO 4
815 PRINT" I"
820 PRINT I*75;" I";
830 FOR J=1 TO INT(Y(I)+.5)
840 PRINT"*";
850 NEXT J
860 PRINT
870 NEXT I
880 REM
890 REM GRAFICO A PUNTI DEGLI ULTIMI 12 MESI
900 REM
910 FOR M=48+C-11 TO 48+C
915 PRINT" I"
920 PRINT L$(M);" I";
930 PRINT TAB(S(M)+6);"*";
940 NEXT M
950 REM
960 END

```

Quando esegui il programma finale, il risultato fu questo:



Steven Cauldwell lavorò per conto proprio a casa producendo un rapporto aggiornato in circa un'ora. Più esattamente, egli produsse un dispositivo che produceva rapporti aggiornati. Il dispositivo, cioè il programma, avrebbe prodotto d'ora in poi un rapporto ogni mese. La sua segretaria avrebbe eseguito il lavoro. Così Cauldwell ebbe i vantaggi del rapporto immediato e stabilì un modo per creare nuovi rapporti nel futuro.

Egli trovò anche un modo per conoscere da vicino le vendite della sua divisione in difficoltà. Dopo aver completato il programma, lo modificò per visualizzare il grafico degli stessi dodici mesi in ognuno degli anni precedenti. Egli era completamente preparato per il meeting del giorno dopo.

L'UFFICIO

Dopo la sua promozione circa un anno fa, Cauldwell si assicurò che l'intero staff fosse in grado di scrivere semplici programmi. Essi seguirono un corso di due giorni che rinfrescò quello che avevano imparato all'università e diede ai principianti nozioni sufficienti per procedere per conto proprio.

La sua seconda innovazione fu l'acquisto, con l'aiuto del direttore dell'elaborazione dati, di diversi piccoli computer per i vari uffici. Egli portò uno di questi computer nel suo studio a casa. Il piccolo computer doveva fare due cose: eseguire programmi e servire come terminale per interrogare il grosso computer della compagnia sui dati degli archivi centrali.

Il giorno dopo, al lavoro, Cauldwell usò i suoi nuovi grafici per studiare i problemi della divisione Bear con il suo direttore, Frank Bradshaw. Assieme, stabilirono molte cause che avevano portato al declino estivo. La divisione Bear ebbe il sospirato incremento nel budget per le attrezzature per il prossimo anno.

Dopo il meeting, Cauldwell entrò nell'ufficio di Peter Bates, il suo direttore della pianificazione a lungo termine. Egli fece cadere una copia del programma e dei risultati sul suo tavolo. "D'ora in avanti vorrei vedere queste cifre sotto forma di grafico", cominciò. "Puoi dare i dati delle altre divisioni alla mia segretaria e mostrarle come aggiornarli quando ci sono i nuovi dati? Oltre a ogni grafico ci deve essere una tavola che riporti le cifre. E sarebbe un grosso aiuto se riunissimo le quattro divisioni in un unico grafico della compagnia."

Peter Bates si chinò in avanti ed esaminò il programma. Egli sapeva che quello che doveva fare gli prendeva solo poche ore di lavoro. "Quando ne hai bisogno?" chiese.

"Dopodomani."

"Bene," disse Bates, "lo farò."

IL PROGRAMMA VENDITE

Questo programma illustra parecchie caratteristiche del Basic. Tutte le istruzioni REM (*remark*) indicano un commento e servono all'autore del programma.

Le normali variabili a cui possono venir assegnati valori sono singole lettere dell'alfabeto come I, J o M. Una variabile che rappresenta una stringa di caratteri è indicata con una lettera seguita da un segno di dol-

laro (\$) come L\$. Ad L\$ si può dare il valore "G" o "GEN" o "FEB" o ogni altra stringa di caratteri ed è chiamata variabile stringa.

Una variabile può anche essere una lettera seguita da una cifra (vedi S9 nel programma CASSA del capitolo 2).

Alcune volte una lettera viene usata per creare molte variabili con indice, per esempio S(34) o Y(3). In questo programma, usiamo L\$(I) con un indice che va da 1 a 60 e Y(I) con indice da 1 a 4. Se usate una variabile con un indice più grande di 10, l'istruzione DIM (dimensione) dice al Basic di riservare spazio sufficiente per la variabile.

Il programma legge i valori delle variabili da L\$(37) a L\$(60) (le prime 36 sono lasciate vuote: perché?). Quindi legge le vendite mensili da S(1) a S(48) e legge S(49) per il mese attuale.

Quindi, calcola quattro medie annuali, Y(1), Y(2), Y(3) e Y(4). Infine, il programma usa l'istruzione di stampa (PRINT) per fare il grafico desiderato. Se non capite queste tecniche, provate sul vostro computer e vedete come funzionano. Se non avete un computer sottomano, prendetele per buone. I loro dettagli non sono importanti.

ESERCIZI

1. Cambiate il programma per stampare un grafico delle vendite del 1978.
2. Aiutate Peter Bates ad ampliare il programma VENDITE per fare in modo che stampi una tavola di dati di supporto al grafico.
3. Cambiate il programma in modo che legga i dati di altre due divisioni e stampi un grafico ed una tavola per ognuno.
4. Cambiate il programma in modo che faccia un grafico ed una tavola della compagnia dopo quelli delle singole divisioni.
5. Prendete un vostro problema ed imitate il metodo di Steven Cauldwell per scrivere un programma:
 - (a) tracciate un esempio dei risultati;
 - (b) scrivete l'ultima parte del programma con le istruzioni PRINT che danno la figura quando avete i numeri;
 - (c) scrivete la prima parte del programma che legge i dati;
 - (d) scrivete la parte intermedia che calcola i risultati necessari;
 - (e) scrivete l'intero programma con stile, in modo che possa essere letto facilmente.

La pianificazione a lungo termine

Rose Thompson pensò che stava facendo bene. La divisione Wolf costruiva apparecchi industriali, e per gli ultimi due anni aveva investito in nuovi impianti per rendere possibile l'uso dei circuiti integrati negli apparecchi. I prossimi anni promettevano bene. Rose Thompson non prevedeva più alcuna spesa di investimento ma le vendite dovevano aumentare del 15 per cento all'anno. Le sue vendite per il prossimo anno ammontavano probabilmente a 107 miliardi, e i suoi debiti erano di 28 miliardi.

Le preoccupazioni di Rose Thompson riguardavano il fatto che ella desiderava ben più di un presentimento: voleva la certezza. La sua reputazione e il suo stipendio dipendevano dai successi della divisione. Inoltre, durante la distribuzione dei budget, la compagnia poteva chiedere cosa sarebbe successo se le sue ipotesi sulle vendite fossero state alte o basse. Ella chiamò George Lee, un analista di mercato che aveva recentemente completato i suoi studi in economia e commercio prima di unirsi alla divisione.

"George, queste cifre rappresentano quello che io penso succederà il prossimo anno. Prendile e sviluppa un rapporto sulle entrate nei prossimi cinque anni. Devi fare in modo che sia possibile variare i tassi di crescita di anno in anno.

"Ho parlato con Simon Wilson del centro di calcolo e possiamo avere un piccolo computer come quello che hanno negli uffici della compagnia. Ce lo prestano per un mese.

“Vorrei che tu vedessi se possiamo fare questo lavoro, e se possiamo farlo nel prossimo mese. Wilson dice che se ci serve un computer per un lungo periodo, potrebbe provvedere all’acquisto. Questa è un’altra possibilità. Pensiamo di aver bisogno di un computer per un lungo periodo?”

Dopo che Lee se ne andò, Rose Thompson si sentì molto meglio. Ora si poteva concentrare sul suo budget per il prossimo anno. Sapeva che il piano di lungo periodo era in buone mani.

COMINCIANDO CON LE IPOTESI

Due giorni più tardi, George Lee aveva già risposto a molte delle domande di Rose Thompson. Aveva provato il computer e aveva visto che riusciva a comprendere il Basic, che aveva studiato all’università.

Le richieste di Rose Thompson erano chiare: basandosi sull’aumento delle vendite e sulle cifre iniziali del prossimo anno, estendere il rapporto sulle entrate della divisione per altri quattro anni. Lee sapeva che poteva scrivere il programma e che un programma era il modo più appropriato di trattare il problema. Il programma lo avrebbe messo in grado di esaminare le conseguenze di diversi scenari.

Egli mise assieme le equazioni del suo memorandum, scrivendo le istruzioni Basic vicino ad ognuna.

1. I debiti attuali sono 28 milioni di dollari e costanti

$$D=28000 \quad (\text{I numeri sono in migliaia.})$$

2. Le vendite saranno 107 milioni

$$S=107000$$

3. Il costo delle merci è il 48 per cento delle vendite

$$C=.48*S$$

4. Il profitto lordo sono le vendite meno i costi

$$G=S-C$$

5. Il deprezzamento è costante a 3.5 milioni

$$B=3500$$

6. I costi di vendita, generali e amministrativi sono il 32.7 per cento delle vendite

$$A = .327 * S$$

7. Il totale dei costi fissi è il deprezzamento più i costi di vendita, generali e amministrativi

$$F = B + A$$

8. L'interesse sui debiti è 14.5 per cento all'anno

$$I = .145 * D$$

9. Il profitto prima delle tasse è il profitto lordo meno i costi fissi e gli interessi

$$P = G - (F + I)$$

10. Le tasse sono il 42 per cento dei profitti

$$X = .42 * P$$

11. Il profitto netto è il profitto prima delle tasse meno le tasse

$$N = P - X$$

George tirò fuori le variabili iniziali dalla lista:

$$\begin{aligned} S &= 107000 \\ B &= 3500 \\ D &= 28000 \end{aligned}$$

L'ipotesi di crescita era che le vendite incrementavano al 15 per cento.

$$S = 1.15 * S$$

Lee guardò i suoi appunti per alcuni minuti e pensò di saper scrivere il programma. Scrivere le equazioni iniziali. Calcolare le variabili dipendenti del rapporto sulle entrate. Stampare i risultati dell'anno. Applicare il tasso di crescita annuale alle vendite per avere l'inizio del prossimo anno, tornare indietro e calcolare il rapporto sulle entrate del secondo anno, e così via. Continuare a fare questo ciclo per tutti i cinque anni. Il cuore del suo programma era come questo (senza la stampa):


```

375 REM VARIABILI PRINCIPALI:
380   S(1)=107000
385   B(1)=3500
390   D(1)=28000
395 REM
400   FOR Y=1 TO 5
405 REM VARIABILI DIPENDENTI:
410     C(Y)=.48*S(Y)
415     G(Y)=S(Y)-C(Y)
420     A(Y)=.327*S(Y)
425     F(Y)=A(Y)+B(Y)
435     I(Y)=.145*D(Y)
445     P(Y)=G(Y)-F(Y)-I(Y)
450     X(Y)=.42*P(Y)
455     N(Y)=P(Y)-X(Y)
520 REM
525 REM VARIABILI PRINCIPALI DEL PROSSIMO ANNO
530   S(Y+1)=H(Y+1)*S(Y)
535   B(Y+1)=B(Y)
540   D(Y+1)=D(Y)
545   NEXT Y
550 REM

```

I risultati erano:

	(MIGLIAIA DI \$)				
	1980	1981	1982	1983	1984
VENDITE	107,000	123,050	141,508	162,734	187,144
COSTO DELLE MERCI	51,360	59,064	67,924	78,112	89,829
PROFITTO LORDO	55,640	63,986	73,584	84,621	97,315
DEPREZZAMENTO	3,500	3,500	3,500	3,500	3,500
SPESE GEN. E AMM.	34,989	40,237	46,273	53,214	61,196
COSTI FISSI	38,489	43,737	49,773	56,714	64,696
INTERESSI	4,060	4,060	4,060	4,060	4,060
PROF. SENZA TASSE	13,091	16,189	19,751	23,848	28,559
TASSE	5,498	6,799	8,295	10,016	11,995
PROFITTO NETTO	7,593	9,389	11,456	13,832	16,564
DEBITI	28,000	28,000	28,000	28,000	28,000

I risultati risposero alle richieste di base di Thompson. Ora Lee volse l'attenzione alla seconda parte delle richieste di Thompson. Doveva fare un piano che poteva essere adattato ai tassi di crescita previsti dalla divisione.

TASSI DI CRESCITA

Nel modello iniziale le vendite salivano al tasso dell'1.15 nel secondo anno e nei successivi. Il programma iniziale poteva essere cambiato sempli-

cemente ribattendo l'equazione di crescita. Questa variazione standard non era un problema. Se tuttavia si volevano variare i tassi di crescita di anno in anno, il programma doveva essere più flessibile.

Un modo per dargli flessibilità, pensò George, sarebbe stato quello di memorizzare tutti i tassi di crescita in una tabella.

<i>Tabella dei tassi di crescita</i>				
	Anno			
	2	3	4	5
Crescita delle vendite	1.15	1.15	1.15	1.15

Scrisse quindi le due istruzioni Basic

```
READ H(2),H(3),H(4),H(5)
DATA 1.15,1.15,1.15,1.15
```

Il tasso di crescita H(3) era il tasso di crescita per le vendite del terzo anno. Il valore di H(3) era 1.15. Lee cambiò l'equazione di crescita in modo da usare i tassi di crescita della tabella:

$$S(Y+1)=H(Y+1)*S(Y)$$

Poiché la tabella aveva esattamente gli stessi valori della prima volta, Lee si aspettava che i risultati fossero gli stessi. Egli eseguì il programma per controllarlo. Se il programma funzionava, Lee sapeva che poteva variare il tasso di crescita di ogni anno ribattendo solo una linea di dati.

IL PROGRAMMA PIANO

```
100 REM ***PIANO***
110 REM
120 REM CALCOLA IL PIANO DELLE ENTRATE
130 REM
135 REM VARIABILI:
140 REM A(Y).....COSTI DI VENDITA GENERALI E COSTI AMMINISTRATIVI
145 REM B(Y).....DEPREZZAMENTO
150 REM C(Y).....COSTO DELLE MERCI VENDUTE
160 REM D(Y).....DEBITI
170 REM F(Y).....SPESE FISSE
180 REM G(Y).....PROFITTO LORDO
182 REM H(Y).....TASSO DI CRESCITA DELLE VENDITE
185 REM I(Y).....INTERESSI SUI DEBITI
190 REM N(Y).....PROFITTO NETTO
200 REM P(Y).....PROFITTO
```

```

205 REM P*.....FORMATO DI STAMPA
210 REM R*.....INDICE DELLA RIGA
215 REM R*.....ETICHETTA DELLA RIGA
225 REM S(Y).....VENDITE
235 REM X(Y).....TASSE
240 REM Y*.....INDICE DELL'ANNO
245 REM
250 REM DIMENSIONI:
255   DIN R*(15)
260 REM
265 REM TABELLA DELLA CRESCITA DELLE VENDITE
270   READ H(2),H(3),H(4),H(5)
275   DATA 1.15,1.15,1.15,1.15
280 REM
285 REM PROGRAMMA PRINCIPALE
290 REM
295 REM CALCOLA IL PIANO DELLE ENTRATE PER CINQUE ANNI
310 REM
315 REM IPOTESI: LE VENDITE PARTONO DA 107.000 E CRESCONO AD UN TASSO
320 REM DI 1,15 ALL'ANNO. IL COSTO DELLE MERCI E LE SPESE GENERALI E
325 REM AMMINISTRATIVE SONO UNA PERCENTUALE DELLE VENDITE. IL DEPREZZAMENTO
327 REM E' COSTANTE. L'ULTIMO ANNO I DEBITI ERANO DI 28.000.
330 REM
375 REM VARIABILI PRINCIPALI:
380   S(1)=107000
385   B(1)=3500
390   D(1)=28000
395 REM
400   FOR Y=1 TO 5
405 REM VARIABILI DIPENDENTI:
410   C(Y)=.48*S(Y)
415   G(Y)=S(Y)-C(Y)
420   A(Y)=.327*S(Y)
425   F(Y)=A(Y)+B(Y)
435   I(Y)=.145*D(Y)
445   P(Y)=G(Y)-F(Y)-I(Y)
450   X(Y)=.42*P(Y)
455   N(Y)=P(Y)-X(Y)
520 REM
525 REM VARIABILI PRINCIPALI DEL PROSSIMO ANNO
530   S(Y+1)=H(Y+1)*S(Y)
535   B(Y+1)=B(Y)
540   D(Y+1)=D(Y)
545   NEXT Y
550 REM
555 REM STAMPA DEI RISULTATI
560   P*=""%           % ###,### ##,###"
565   P*="P*+" ###,### ##,### ##,###"
568   PRINT"                               (MIGLIAIA DI $)"
570   PRINT"                               1980  1981"
575   PRINT"   1982   1983   1984"
580   PRINT
585   FOR R=1 TO 11
590     READ R*(R)
595   NEXT R
600   DATA VENDITE,COSTO DELLE MERCI,PROFITTO LORDO,DEPREZZAMENTO
605   DATA SPESE GEN. E AMM.,COSTI FISSI,INTERESSI
610   DATA PROF. SENZA TASSE,TASSE,PROFITTO NETTO
620   DATA DEBITI
625 REM
630   PRINT USING P*:"R*(1),S(1),S(2),S(3),S(4),S(5)
635   PRINT USING P*:"R*(2),C(1),C(2),C(3),C(4),C(5)
640   PRINT USING P*:"R*(3),G(1),G(2),G(3),G(4),G(5)
645   PRINT
650   PRINT USING P*:"R*(4),B(1),B(2),B(3),B(4),B(5)
655   PRINT USING P*:"R*(5),A(1),A(2),A(3),A(4),A(5)

```

```

660 PRINT USING P#;R#(6),F(1),F(2),F(3),F(4),F(5)
665 PRINT
670 PRINT USING P#;R#(7),I(1),I(2),I(3),I(4),I(5)
675 PRINT
680 PRINT USING P#;R#(8),P(1),P(2),P(3),P(4),P(5)
685 PRINT USING P#;R#(9),X(1),X(2),X(3),X(4),X(5)
690 PRINT USING P#;R#(10),N(1),N(2),N(3),N(4),N(5)
695 PRINT
700 PRINT
735 PRINT
740 PRINT USING P#;R#(11),D(1),D(2),D(3),D(4),D(5)
745 END

```

Il programma usa le variabili con indice per leggere e memorizzare i valori della tavola dei tassi di crescita e del piano quinquennale.

Dopo che il programma ha letto i tassi di crescita e le variabili iniziali, entra in un ciclo che si ripete cinque volte, una per ogni anno. La prima parte del ciclo calcola le variabili che dipendono dai valori iniziali. Alla fine della prima parte i valori del piano delle entrate per un anno sono stati completati. La seconda parte del ciclo prepara le condizioni iniziali per il prossimo anno. Quando il ciclo viene ripetuto per la seconda volta, può calcolare le variabili dipendenti del secondo anno perché i dati iniziali sono predisposti alla fine del primo ciclo. Quando il ciclo viene eseguito per la quinta volta, calcola le variabili del quinto anno e predispone le variabili iniziali del sesto anno, che non vengono mai usate.

Dopo aver completato il ciclo, il programma stampa i risultati accumulati nelle variabili con indice. Se qualcuna delle istruzioni di stampa non funzionasse sul vostro computer, consultate il vostro manuale Basic per trovare delle soluzioni alternative. Il programma qui riportato è stato eseguito su un computer Radio Shack TRS-80.

ESERCIZI

1. Provate differenti vendite iniziali fino a che trovate quella che dà 3.5 milioni di profitto netto alla fine del quinto anno.
2. Cambiate le quantità iniziali in modo che riflettano i vostri affari e introducete tassi di crescita che pensate realistici. Eseguite il programma e controllate i vostri prossimi cinque anni.
3. Esaminate le ipotesi contenute nell'equazione delle entrate. Modificate l'equazione in modo che rifletta le vostre ipotesi personali.
4. Riscrivete il programma iniziale in modo da avere un piano delle entrate dove le vendite variano, essendo i profitti nel primo anno 16 000 dollari con una crescita del 20% all'anno.

FLUSSO DI CASSA

Rose Thompson sapeva che il suo lavoro non era veramente finito. L'ipotesi che i debiti rimanessero costanti la preoccupava. In effetti ella usava i suoi debiti come un tampone per le necessità di cassa. Chiedeva dei prestiti quando aveva bisogno di contante e pagava i debiti quando aveva un surplus di cassa.

Dopo aver compilato il conto profitti e perdite, ella poteva limitare il flusso di cassa e simularlo per cinque anni. Il flusso di cassa operativo poteva essere definito come l'introito netto unito al deprezzamento.




Fig. 2.1 Conto profitti-perdite e flusso di cassa

Una certa quantità di contante doveva coprire l'incremento del capitale di lavoro. Le sue vendite avevano un triplice ciclo annuale (*turnover*), così la variazione nel capitale circolante sarebbe stata un terzo degli incrementi delle vendite. La variazione nelle spese di capitale poteva essere approssimata dal loro deprezzamento regolato con l'inflazione. Il contante che rimaneva poteva essere usato per ridurre i debiti.

La Thompson descrisse il problema a George Lee, che disegnò un diagramma a blocchi per spiegare la cosa (fig. 2.1).

"Le nuove istruzioni Basic potrebbero essere queste", spiegò Lee.

$$\begin{aligned}
 D &= N + B \\
 M &= .33 * (S9 - 1) / S9 * 5 (Y) \\
 E &= 1.09 [Y * B \\
 R &= D - (M + E)
 \end{aligned}$$

"Ma il problema", disse la Thompson, "è che se abbiamo liquidi, possiamo ridurre i debiti. Se riduciamo i debiti, l'interesse sul nostro piano delle entrate è minore. Dobbiamo dunque rimodificare il piano delle entrate, quindi rimodificare il flusso di cassa e così via. Puoi farlo fare al computer?"

"Certo", disse Lee, "dammi un giorno o due."

Due giorni dopo, Lee rispose alla domanda della Thompson con un risultato diverso:

	(MIGLIAIA DI \$)				
	1980	1981	1982	1983	1984
VENDITE	107,000	123,050	141,508	162,734	187,144
COSTO DELLE MERCI	51,360	59,064	67,924	78,112	89,829
PROFITTO LORDO	55,640	63,986	73,584	84,621	97,315
DEPREZZAMENTO	3,500	3,500	3,500	3,500	3,500
SPESE GEN. E AMM.	34,989	40,237	46,273	53,214	61,196
COSTI FISSI	38,489	43,737	49,773	56,714	64,696
INTERESSI	3,637	3,054	2,276	1,260	-46
PROF. SENZA TASSE	13,514	17,194	21,535	26,648	32,665
TASSE	5,676	7,222	9,045	11,192	13,719
PROFITTO NETTO	7,838	9,973	12,490	15,456	18,946
FLUSSO CASSA OP.	11,338	13,473	15,990	18,956	22,446
VARIAZIONI IN C.C.	4,606	5,296	6,091	7,005	8,055
VAR. SPESA IN C.C.	3,815	4,158	4,533	4,941	5,385
AMMORTAMENTI	2,917	4,018	5,367	7,011	9,005
DEBITI	25,083	21,065	15,698	8,687	-318

"È magnifico", disse la Thompson. "Queste cifre sono tutte verosimili. Guarda, più incrementiamo i nostri profitti e il nostro margine, più riduciamo i nostri debiti. Ciò è incoraggiante. Avrei impiegato un sacco di

tempo a fare questi conti con il mio calcolatore. Come sei riuscito a farlo fare al computer?"

"Un'idea veramente molto semplice," disse Lee, "il programma fa quello che faremmo noi a mano. Fa un tentativo, vede se il tentativo funziona, e se non funziona fa un altro tentativo. All'interno del nostro ciclo di tentativi c'è tutto ciò che cambia come risultato del tentativo: le entrate e il flusso di cassa. La variabile di ciclo G è per i tentativi, D per i debiti e R è la riduzione del debito. D1 è il primo tentativo col nuovo debito e D2 è il prossimo tentativo col nuovo debito. Il modo in cui funziona è:

```

425      D1=D(Y)
430      FOR G=1 TO 100
485          D2=D(Y)-R(Y)
490          IF ABS(D2-D1)<=.001 THEN 492 ELSE 500
492          D(Y)=D2
495          GOTO 520
500 REM
505          D1=D2
510      NEXT G
515      STOP
520 REM

```

"Fammi vedere se lo capisco", disse Thompson. "Qui dice che se la differenza tra i due tentativi ($D2-D1$) è minore di un piccolo importo, scegliamo questo tentativo. Cosa significa ABS?"

"È la funzione del valore assoluto. Non sappiamo con sicurezza qual è il numero maggiore nella sottrazione, così prendiamo il valore assoluto per essere sicuri che la risposta sia positiva. Quindi verifichiamo se è abbastanza piccola per uscire dal ciclo."

"E continuiamo a fare tentativi fino a che due tentativi non sono molto vicini?"

"Sì," disse Lee, "è così. E teniamo conto del nostro debito iniziale D così che possiamo calcolare il nostro nuovo tentativo ogni volta."

"Questo programma è molto bello," disse Thompson, "come ci sei arrivato?"

"Ho fatto un po' di prove prima di arrivarci", confessò Lee. "Ma tu lo puoi usare senza preoccupazioni. Ora abbiamo un metodo che possiamo sempre usare."

"È vero", disse Thompson. "Ho visto che hai predisposto cento cicli. Quanti tentativi servono realmente per andarci abbastanza vicino e uscire dai cicli?"

"Generalmente si esce dopo sette o otto tentativi."

"Otto tentativi all'anno", disse Thompson. "Cinque volte otto fa quaranta calcoli. Non ce l'avrei mai fatta a mano. Un buon lavoro, Lee."

IL PROGRAMMA CASSA

```

100 REM ***CASSA***
110 REM
120 REM UN MODELLO DI PIANIFICAZIONE FINANZIARIA DEL TIPO "COSA SUCCEDERÀ SE"
121 REM CHE PRODUCE IL PIANO DELLE ENTRATE E IL FLUSSO DI CASSA PER 5 ANNI.
130 REM
135 REM VARIABILI:
140 REM A(Y).....COSTI DI VENDITA GENERALI E COSTI AMMINISTRATIVI
145 REM B(Y).....DEPREZZAMENTO
150 REM C(Y).....COSTO DELLE MERCI VENDUTE
155 REM D1,D2.....STIMA DEI DEBITI
160 REM D(Y).....DEBITI
165 REM E(Y).....VARIAZIONE SPESA IN CONTO CAPITALE
170 REM F(Y).....SPESE FISSE
175 REM G.....INDICE PER I TENTATIVI (PER I DEBITI)
180 REM G(Y).....PROFITTO LORDO
185 REM I(Y).....INTERESSI SUI DEBITI
190 REM N(Y).....PROFITTO NETTO
195 REM O(Y).....FLUSSO DI CASSA OPERATIVO
200 REM P(Y).....PROFITTO
205 REM P%.....FORMATO DI STAMPA
210 REM R.....INDICE DELLA RIGA
215 REM R%.....ETICHETTA DELLA RIGA
220 REM R(Y).....AMMORTAMENTI
225 REM S(Y).....VENDITE
230 REM W(Y).....VARIAZIONE CAPITALE CIRCOLANTE
235 REM X(Y).....TASSE
240 REM Y.....INDICE DELL'ANNO
245 REM
250 REM DIMENSIONI:
255 DIM R$(15)
260 REM
265 REM COSTANTI:
270 S9=1.15
275 REM
280 REM
285 REM PROGRAMMA PRINCIPALE
290 REM
295 REM CALCOLA IL PIANO DELLE ENTRATE E IL FLUSSO DI CASSA PER CINQUE ANNI.
300 REM APPLICA L'INDEBITAMENTO ANNUO SIA AL FLUSSO DI CASSA CHE AL PIANO
305 REM DELLE ENTRATE.
310 REM
315 REM IPOTESI: LE VENDITE PARTONO DA 107,000 E CRESCONO AD UN TASSO
320 REM DI 1.15 ALL'ANNO, IL COSTO DELLE MERCI E LE SPESE GENERALI E
325 REM AMMINISTRATIVE SONO UNA PERCENTUALE DELLE VENDITE, IL DEPREZZAMENTO
327 REM E' COSTANTE. L'ULTIMO ANNO I DEBITI ERANO DI 28,000.
330 REM
335 REM LA CASSA E' USATA PER IL CAPITALE CIRCOLANTE, LE SPESE DI CAPITALE E
340 REM PER L'AMMORTAMENTO. LA VARIAZIONE NEL CAPITALE CIRCOLANTE E' UN TERZO
345 REM DELLA VARIAZIONE NELLE VENDITE, PERCHE' IL CAPITALE CIRCOLANTE VIENE
350 REM MESSO IN CIRCOLAZIONE TRE VOLTE ALL'ANNO.
355 REM LA VARIAZIONE DELLA SPESA IN CONTO CAPITALE E' IL DEPREZZAMENTO AL
360 REM TASSO DI INFLAZIONE (1.09). LA CASSA RESIDUA VIENE USATA PER RIDURRE
365 REM I DEBITI O PER FARE PRESTITI.
370 REM
375 REM VARIABILI PRINCIPALI:
380 S(1)=107000
385 B(1)=3500
390 D(1)=28000
395 REM
400 FOR Y=1 TO 5
405 REM VARIABILI DIPENDENTI:
410 C(Y)=.48*S(Y)

```



```

415      G(Y)=S(Y)-C(Y)
420      A(Y)=.327*S(Y)
422      F(Y)=A(Y)+B(Y)
425      D1=D(Y)
430      FOR B=1 TO 100
435          I(Y)=.145*D1
445          P(Y)=G(Y)-F(Y)-I(Y)
450          X(Y)=.42*P(Y)
455          N(Y)=P(Y)-X(Y)
460      REM
465          O(Y)=N(Y)+B(Y)
470          W(Y)=.33*(S9-1)/S9*S(Y)
475          E(Y)=1.09*(Y*B(Y)
480          R(Y)=O(Y)-(W(Y)+E(Y))
485          D2=D(Y)-R(Y)
490          IF ABS(D2-D1)<=.001 THEN 492 ELSE 500
492          D(Y)=D2
495          GOTO 520
500      REM
505          D1=D2
510      NEXT G
515      STOP
520      REM
525      REM VARIABILI PRINCIPALI DEL PROSSIMO ANNO
530          S(Y+1)=S9*S(Y)
535          B(Y+1)=B(Y)
540          D(Y+1)=D(Y)
545      NEXT Y
550      REM
555      REM STAMPA DEI RISULTATI
560      P$=""
565      P$=P$+"          % ###,### ###,###"
568      PRINT "          ###,### ###,### ###,###"
570      PRINT "          (MIGLIAIA DI %)"
575      PRINT "          1982          1983          1984"
580      PRINT "          1981"
585      FOR R=1 TO 15
590          READ R$(R)
595      NEXT R
600      DATA VENDITE,COSTO DELLE MERCI,PROFITTO LORDO,DEPREZZAMENTO
605      DATA SPESE GEN. E AMM.,COSTI FISSI,INTERESSI
610      DATA PROF. SENZA TASSE,TASSE,PROFITTO NETTO
615      DATA FLUSSO CASSA OP.,VARIAZIONI IN C.C.
620      DATA VAR. SPESA IN C.C.,AMMORTAMENTI,DEBITI
625      REM
630      PRINT USING P$;R$(1),S(1),S(2),S(3),S(4),S(5)
635      PRINT USING P$;R$(2),C(1),C(2),C(3),C(4),C(5)
640      PRINT USING P$;R$(3),G(1),G(2),G(3),G(4),G(5)
645      PRINT
650      PRINT USING P$;R$(4),B(1),B(2),B(3),B(4),B(5)
655      PRINT USING P$;R$(5),A(1),A(2),A(3),A(4),A(5)
660      PRINT USING P$;R$(6),F(1),F(2),F(3),F(4),F(5)
665      PRINT
670      PRINT USING P$;R$(7),I(1),I(2),I(3),I(4),I(5)
675      PRINT
680      PRINT USING P$;R$(8),P(1),P(2),P(3),P(4),P(5)
685      PRINT USING P$;R$(9),X(1),X(2),X(3),X(4),X(5)
690      PRINT USING P$;R$(10),N(1),N(2),N(3),N(4),N(5)
695      PRINT
700      PRINT
715      PRINT USING P$;R$(11),O(1),O(2),O(3),O(4),O(5)
720      PRINT USING P$;R$(12),W(1),W(2),W(3),W(4),W(5)
725      PRINT USING P$;R$(13),E(1),E(2),E(3),E(4),E(5)
730      PRINT USING P$;R$(14),R(1),R(2),R(3),R(4),R(5)
735      PRINT
740      PRINT USING P$;R$(15),D(1),D(2),D(3),D(4),D(5)
745      END

```

Il programma è una revisione del programma originario PIANO. Questo è un buon esempio di come si possa perfezionare un'idea primitiva per fornire risultato supplementare.

ESERCIZI

1. Cambiate il debito iniziale da 28 000 a 40 000 dollari. Come influisce questo cambio sul profitto netto dopo le tasse? Cosa succede se i debiti sono solo 18 000 dollari?
2. Trovate un rapporto annuale di una compagnia in cui siete interessati e provate i suoi dati in CASSA.

AUTO NEL BUDGET

George Lee esaminò il suo lavoro. Egli era sorpreso di constatare che il programma lo liberava dalla schiavitù dei calcoli. In precedenza, se aveva tempo di fare tre prove per trovare la soluzione, era anche troppo. Ma ora, con un programma di computer, poteva provare 20 o 30 insieme di numeri nel tempo in cui ne faceva tre a mano. Il risultato era che egli sapeva qual era il miglior intervallo di valori. Allora il mondo dietro i numeri gli ritornava a fuoco e si sentiva più sicuro.

Poiché i numeri venivano trattati facilmente, divenivano meno importanti. E poiché venivano trattati velocemente, aveva più tempo da dedicare alla ricerca. Così il suo lavoro migliorava sia in qualità dei risultati numerici sia nella qualità della ricerca dietro i numeri.

Egli vedeva un altro grosso vantaggio: l'intuito analitico. Ogni problema che potesse essere quantificato poteva essere simulato con un piccolo programma di computer.

Quando egli costruiva il modello, capiva la struttura del problema in modo nuovo. Vedeva come le variabili reagivano le une con le altre. Sviluppò un intuito analitico che non sarebbe stato possibile senza un computer.

"E tutto questo", pensò Lee, "manipolando una semplice tabella." Ebbe un'idea. Sarebbe stato utile avere un semplice programma che sommasse le righe e le colonne di una tabella. Tutti i manager lo fanno quando si tratta di preparare il budget del prossimo anno per il loro gruppo. Essi delineavano i loro progetti sui quattro trimestri per vedere dove si raggruppavano i soldi, e quindi tentavano di ridistribuire i costi in modo che i quattro trimestri fossero bilanciati.

Il vantaggio di un semplice programma per una tabella non sarebbe stato nell'esecuzione iniziale, ma in quelle ripetute. Un utente poteva modificare una linea semplicemente ribattendola e rieseguendo il programma per vedere la nuova tabella. Questa capacità di giocare con le cifre velo-

cemente poteva risparmiare loro del tempo nella preparazione dei budget.

Egli schizzò l'idea del programma. Ci sarebbero state quattro colonne, una per ogni trimestre. I numeri delle colonne potevano essere cambiati dall'utente. Il programma visualizzava la tabella con i totali per riga e per colonna e con il totale generale. I dati del programma erano:

```

400 REM
410 DATA 0, 1234, 1234, 1234
420 DATA 40000,60000, 0,30000
430 DATA 333, 444, 333, 444
440 DATA 1200, 1824, 1560, 1960
450 DATA 8825, 9998, 4444, 7777
460 DATA 32111,42222,52333,62444
470 DATA 44444,33333,22222,11111
480 REM
  
```

e dovevano dar luogo a una tabella come questa:

RIGA	TRIMESTRI				TOTALE
	1	2	3	4	
1	0	1,234	1,234	1,234	3,702
2	40,000	30,000	30,000	30,000	130,000
3	333	444	333	444	1,554
4	1,200	1,824	1,560	1,960	6,544
5	8,825	9,998	4,444	7,777	31,044
6	32,111	42,222	52,333	62,444	189,110
7	44,444	33,333	22,222	11,111	111,110
TOT	126,913	119,055	112,126	114,970	473,064

che poteva essere cambiata così:

RIGA	TRIMESTRI				TOTALE
	1	2	3	4	
1	0	1,234	1,234	1,234	3,702
2	40,000	60,000	0	30,000	130,000
3	333	444	333	444	1,554
4	1,200	1,824	1,560	1,960	6,544
5	8,825	9,998	4,444	7,777	31,044
6	32,111	42,222	52,333	62,444	189,110
7	44,444	33,333	22,222	11,111	111,110
TOT	126,913	149,055	82,126	114,970	473,064

Lee fece un breve discorso al meeting mensile dei manager della divisione Wolf. Egli spiegò brevemente come poteva essere usato il programma TABELLA e diede una dimostrazione dal vivo su un grande schermo allestito per l'occasione. Dopo la dimostrazione, Lee rispose per alcuni minuti alle domande e quindi sedette.

Alla fine di novembre, sei persone avevano eseguito il programma con le proprie cifre. TABELLA si rivelò un successo. Era semplice, facile da

usare e pratico. Faceva quello che Lee sperava facesse. Dava ai manager la capacità di giudicare sulle cifre dei loro budget facilmente e velocemente. Due dei sei utenti ci tennero a dirgli che il programma aveva risparmiato loro qualche ora di calcoli. Uno disse che aveva provato diverse combinazioni prima di arrivare al budget del prossimo anno. George fu contento che un programma così semplice potesse essere così utile. Egli decise di comperare un computer per il proprio ufficio.

IL PROGRAMMA TABELLA

```

100 REM ***TABELLA***
110 REM
120 REM UN SEMPLICE PROGRAMMA DI UTILITA' PER LEGGERE UNA TABELLA A QUATTRO
130 REM COLONNE, PER ESEMPIO PER QUATTRO TRIMESTRI.
140 REM LA TABELLA PUO' ESSERE COSTITUITA DA 50 RIGHE.
150 REM IL NUMERO DI RIGHE ATTUALE E' N.
160 REM
170 REM PER USARE LA TABELLA STABILITE IL NUMERO DI RIGHE CHE DESIDERATE.
180 REM BATTETE LA VOSTRA TABELLA CON LE ISTRUZIONI DATA. E FATE GIRARE IL
190 REM PROGRAMMA. QUESTO STAMPERA' I TOTALI DELLE RIGHE E DELLE COLONNE
200 REM E IL TOTALE COMPLESSIVO.
210 REM
220 REM PER FARE DIVERSI TENTATIVI E' SUFFICIENTE RIBATTERE LA LINEA DEI DATA
230 REM CHE VOLETE MODIFICARE E FAR GIRARE IL PROGRAMMA UN'ALTRA VOLTA.
240 REM
250 REM VARIABILI:
260 REM C.....CONTATORE DELLE COLONNE
265 REM N.....NUMERO DI RIGHE
270 REM R.....CONTATORE DELLE RIGHE
280 REM S.....SOMMATORE
290 REM T().....TABELLA DEI VALORI
300 REM
310 REM DIMENSIONI:
320 DIM T(50,5)
330 REM
340 REM LEGGE NELLA TABELLA
345 REM
350 N=7
360 FOR R=1 TO N
365 FOR C=1 TO 4
370 READ T(R,C)
380 NEXT C
390 NEXT R
400 REM
410 DATA 0, 1234, 1234, 1234
420 DATA 40000,60000, 0,30000
430 DATA 333, 444, 333, 444
440 DATA 1200, 1824, 1560, 1960
450 DATA 8825, 9998, 4444, 7777
460 DATA 32111,42222,52333,62444
470 DATA 44444,33333,22222,11111
480 REM
490 REM SOMMA OGNI RIGA E DA' LA RISPOSTA IN T(R,4)
495 REM
500 FOR R=1 TO N
510 T(R,5)=T(R,1)+T(R,2)+T(R,3)+T(R,4)
520 NEXT R

```

```

530 REM
540 REM SOMMA OGNI COLONNA E DA' LA RISPOSTA IN T(N+1,C)
550 REM
560   FOR C=1 TO 5
570     S=0
580     FOR R=1 TO N
590       S=S+T(R,C)
600     NEXT R
610     T(N+1,C)=S
620   NEXT C
630 REM
640 REM STAMPA LE TABELLE CON I TOTALI
645 REM
650   PRINT"
660   PRINT"RIGA 1          TRIMESTRI"
670   PRINT"      3          2"
680   PRINT"      4          TOTALE"
690   FOR R=1 TO N
700     PRINT R:
710     FOR C=1 TO 5
720       PRINT USING" ###,###":T(R,C):
730     NEXT C
740     PRINT
750   NEXT R
760 REM
770   PRINT
780   PRINT"TOT":
790   FOR C=1 TO 5
800     PRINT USING" ###,###":T(N+1,C):
810   NEXT C
820   PRINT
830 REM
840   END

```

Il programma legge una tabella sette (o ogni numero che volete) per quattro, somma le righe e le colonne e stampa i risultati. La somma di colonna è realizzata ripetendo la

$$S=S+T(R,C)$$

I calcoli cominciano sulla destra dell'uguale, considerando il vecchio valore di S e sommandolo al valore attuale di T(R,C). Il risultato di questa addizione (un numero) viene assegnato alla variabile a sinistra dell'uguale, in questo caso la nuova S. Se il programma è organizzato in modo che ogni volta che passiamo per questa linea abbiamo un T(R,C) diverso, il programma continuerà ad aggiungere il T(R,C) attuale alla somma corrente. (Se inserissimo una linea immediatamente dopo questa, con l'istruzione PRINT S, vedremmo S crescere ogni volta che viene aggiunto un T(R,C).)

La sezione che stampa la tabella richiede qualche astuzia per facilitarne la lettura. Provate con il vostro Basic ciò che può andar bene per voi. ne per voi.

ESERCIZI

1. Eseguite il programma con i vostri numeri. Modificatelo ed eseguitelo nuovamente.
2. Pensate ad una tabella di numeri che vada bene per voi. Modificate TABELLA in modo che legga la vostra tabella, esegua i calcoli necessari e stampi i risultati. (Diversi programmi in questo libro sono una modifica di TABELLA.)
3. Leggete l'Appendice D per vedere come usare VisiCalc per trattare una tabella di numeri.

La revisione della strategia

Nei primi giorni dell'anno nuovo Frank Bradshaw chiese a Steven Cauldwell di prestargli Peter Bates. La divisione Bear di Bradshaw aveva chiuso l'anno malamente e lui aveva bisogno dell'opera di Bates per fare un'analisi strategica del proprio complesso di produzione. Cauldwell sapeva che il piano per l'anno che stava per cominciare era già a buon punto, così alla fine di gennaio lasciò che Bates gli desse una mano.

Quando Bates arrivò al quartier generale della divisione Bear, Frank Bradshaw era molto agitato.

“Dobbiamo trovare la formula produttiva migliore”, disse il direttore generale. “Abbiamo alcuni prodotti promettenti, parecchi articoli vecchi ed altri che sono una via di mezzo. Il tuo compito è di ordinarli. Siamo la divisione più grossa della Chordata ed è importante che ci mettiamo a posto rapidamente. Siccome nei nostri stabilimenti produciamo molti articoli,” aggiunse Bradshaw, “è difficile assegnare dei costi fissi. La prima cosa da fare è trovare il valore che corrisponde più fedelmente ai nostri costi effettivi, in modo che possiamo assegnare dei costi fissi ai vari prodotti. Possono essere le unità prodotte, le ore di lavoro, il numero di prodotti finiti o le ordinazioni soddisfatte. Torna da me quando hai dei risultati. Io sarò fuori città fino a venerdì pomeriggio.”

"Quali sono gli accordi con i vostri analisti finanziari?" chiese Bates. "Ti ho dato carta bianca. Devono essere in grado di fornirti i costi diretti, ma puoi andare dove vuoi e rivolgerti a chi è necessario. Pensi di farcela per la settimana prossima?"

"Non c'è molto tempo, ma va bene", rispose Bates.

Dopo nove giorni Bates andò da Bradshaw per mostrargli i risultati. Aveva trovato che il dato più rispondente al costo dei prodotti era il numero di ordini. Accanto ad ogni ordinazione c'erano i costi di amministrazione, di montaggio e di mano d'opera. A parte il caso di poche ordinazioni molto grandi, la dimensione effettiva dell'ordinazione aveva poco a che fare con la relativa spesa. Una piccola ordinazione costava quanto una grande. Il numero totale delle ordinazioni della divisione Bear era 213 000. Mostrò a Bradshaw la tabella dei dati con i valori di produzione e le ordinazioni complessive:

380 REM	PRODOTTO	QUOTA (IN \$1000)	ORDINI (IN 100)
385	REM		
390	DATA"FORNO ELETTRICO"	30520,498	
395	DATA"TOSTAPANE	"15870,660	
400	DATA"TRITACARNE	"15870,350	
405	DATA"SPREMIAGRUMI	"8280,182	
410	DATA"CENTRIFUGA	"10350,64	
415	DATA"APRISCATOLE	"4140,116	
420	DATA"FERRO DA STIRO	"3450,78	
425	DATA"ASCIUGACAPELLI	"2760,73	
430	DATA"TEMPERAMATITE	"2760,109	
435	REM		

Scrivendo su un pezzo di carta, Bates spiegò come aveva calcolato i profitti. "Consideriamo il valore di produzione, per esempio nel caso del forno elettrico è 30 520 000. Per ottenere il suo costo fisso consideriamo il rapporto delle ordinazioni evase sul totale 49 800/213 000, per il costo fisso complessivo, 39 000 000. Si ottiene 9 118 000. Il profitto è dato dalla differenza tra i due, cioè 21 402 000."

"Sembra ragionevole", disse Bradshaw. "Ed il profitto che deriva da questo valore rispecchia l'efficienza di produzione dell'articolo?"

"Appunto", aggiunse Bates. "Qui ho inserito il concetto di profitto per dollaro di ricavato in modo che gli articoli possano essere confrontati. Ecco quel che ho ottenuto."

TOTALE VENDITE: 275000
 COSTI DIRETTI: 181000
 VALORE DELLA PRODUZIONE -- 94000
 PROFITTO LORDO DA IMPOSTE: 55000
 COSTI FISSI -- 39000

	VALORE	%	COSTI FISSI	%	PROF.	%	PROF./ RIC.
FORNO ELETTRICO	30,520	32	9,118	23	21,402	39	0.70
TOSTAPANE	15,870	17	12,085	31	3,785	7	0.24
TRITACARNE	15,870	17	6,408	16	9,462	17	0.60
SPREMIAGRUMI	8,280	9	3,332	9	4,948	9	0.60
CENTRIFUGA	10,350	11	1,172	3	9,178	17	0.89
APRISCATOLE	4,140	4	2,124	5	2,016	4	0.49
FERRO DA STIRO	3,450	4	1,428	4	2,022	4	0.59
ASCIUGACAPELLI	2,760	3	1,337	3	1,423	3	0.52
TEMPERAMATITE	2,760	3	1,996	5	764	1	0.28
TOTALE	94,000		39,000		55,000		0.59

"Aspetta un momento", disse Bradshaw, "fammi vedere se ho capito. Tu dici di aver tolto i costi diretti dalle vendite per avere il valore della produzione, e quindi di aver usato il numero di ordini per rapportare i costi fissi ai prodotti. Dopo che i costi fissi sono stati sottratti, abbiamo una migliore idea di quali siano stati i prodotti che nello scorso anno hanno realizzato il profitto".

"Esatto" disse Bates. "Come puoi vedere, i tostapane e i temperamatite causano dei problemi. E probabilmente anche gli apriscatole e gli asciugacapelli.

A questo punto è interessante chiedersi cosa succederebbe se vendessimo più forni elettrici che tostapane. Possiamo giocare al 'cosa succede se?' con i nostri prodotti".

"Okay", disse Bradshaw, "come possiamo chiedere 'cosa succede se?'".

 ANALISI

Peter Bates portò il piccolo schermo televisivo e la tastiera sul tavolo. "Grazie di averlo chiesto. Il programma che ho scritto ha un insieme di quelli che io chiamo "fattori di cambio". Per la nostra tabella iniziale, tutti i fattori erano 1. Se desideriamo raddoppiare una linea di prodotti, possiamo cambiare il fattore in 2.

Se vogliamo eliminare un prodotto cambiamo il suo fattore in 0".

"È difficile cambiare i fattori?".

"Guarda" disse Bates. Accese il computer e caricò il programma. Mostrò a Bradshaw il programma. Quindi scrisse sullo schermo:

DATA 1.5,0,1,1,1,1,2,1,0

"Ho solo riscritto una riga del programma. Ho eliminato il tostapane ed il temperamatite ed aumentato il forno elettrico ed il ferro da stiro." Bates scrisse RUN e sul video apparve la tabella.

"Pazzesco", esclamò Bradshaw. "Guarda come diminuiscono i nostri costi e come crescono i profitti." Si mise ad esaminare attentamente lo schermo. "Cosa succede se ci sbarazziamo anche degli altri due articoli più deboli?"

"Perché non provi tu stesso?" disse Bates lasciando a Bradshaw la sedia di fronte al video.

Bradshaw era titubante. "Non sono molto veloce alla tastiera."

"Non preoccuparti. Se sbagli, puoi tornare indietro e riscrivere."

Bradshaw proseguì con attenzione. Scrisse una riga con i nuovi fattori:

DATA 1.5,0.1.1.1.6,0.2,0,0

E poi RUN. Sullo schermo apparve la nuova tabella:

TOTALE VENDITE: 275000
 COSTI DIRETTI: 181000
 VALORE DELLA PRODUZIONE -- 94000
 PROFITTO LORDO DA IMPOSTE: 55000
 COSTI FISSI -- 39000

	VALORE	%	COSTI FISSI	%	PROF.	%	PROF./RIC.
FORNO ELETTRICO	45,780	49	13,677	49	32,103	49	0.70
TOSTAPANE	0	0	0	0	0	0	
TRITACARNE	15,870	17	6,408	23	9,462	15	0.60
SPREMIAGRUMI	8,280	9	3,332	12	4,948	8	0.60
CENTRIFUGA	16,560	18	1,875	7	14,685	23	0.89
APRISCATOLE	0	0	0	0	0	0	
FERRO DA STIRO	6,900	7	2,856	10	4,044	6	0.59
ASCIUGACAPELLI	0	0	0	0	0	0	
TEMPERAMATITE	0	0	0	0	0	0	
TOTALE	93,390		28,150		65,240		0.70

"Se smettessimo di produrre il tostapane, l'apricatole, l'asciugacapelli ed il temperamatite e ci concentriamo sul forno elettrico, la centrifuga ed il ferro da stiro, in teoria potremmo aumentare i nostri utili del 18%. Niente male." Bradshaw sembrava essersi dimenticato di Bates. Esaminava lo schermo e continuava a pensare ad alta voce. "In realtà però non venderemo mai così tanti forni elettrici. I tritacarne invece sì. Diminuiamo i forni elettrici ed aumentiamo i tritacarne."

Pochi minuti più tardi apparve sullo schermo la nuova versione della tabella. "Anche secondo questi criteri di ragionevolezza otteniamo un aumento di profitto del 15 per cento. E non è uno scherzo; la possibilità c'è davvero. Perché i dati nella parte superiore del video non coincidono con quelli della parte inferiore?" chiese Bradshaw.

"Le prime cinque righe sono le condizioni di partenza. Rimangono sempre uguali in modo che si possano seguire le variazioni."

Bradshaw abbassò il capo e tornò a studiare lo schermo. Peter Bates si

appoggiò sulla scrivania vicino a Bradshaw in silenzio. Bisognava esaminare parecchie altre cose, prima di prender qualsiasi decisione. C'erano da analizzare le quote di mercato, i canali di distribuzione, la potenziale crescita di mercato, i controlli di costo, il personale e gli aspetti finanziari. Ma si era giunti ad un buon punto. Bradshaw era più compiacente di quanto ci si potesse aspettare.

Bradshaw si staccò dal terminale. "Finora ci siamo divertiti. È stato anche utile. Adesso bisogna pensare alle quote di mercato. Entro una settimana, a partire da mercoledì, vorrei avere una idea della situazione dei nostri prodotti. Se non sarà possibile, gradirei almeno dare un'occhiata alla curva di esperienza e vedere come si può applicarla al nostro lavoro sulle quote di mercato. A proposito, perché hai chiamato RAGGI-X il tuo programma?"

"Da una frase di Peter Drucker, che ha definito un'analisi di questo tipo come i raggi X di una azienda. Alcune delle idee che hanno ispirato il programma sono sue, ma la maggior parte sono mie. Ho accettato quel nome come tributo a Drucker, ma soprattutto perché è facile tenerlo a mente."

 IL PROGRAMMA RAGGI-X

```

100 REM ***RAGGI X***
110 REM
120 REM QUESTO PROGRAMMA CONSIDERA LE QUOTE E LE TRANSAZIONI DI PRODOTTO,
125 REM ESTRAE I COSTI FISSI DEI PRODOTTI DAL COSTO FISSO TOTALE E CALCOLA
130 REM I PROFITTI REALI.
135 REM
140 REM INOLTRE, IL PROGRAMMA CALCOLA LE PERCENTUALI SUI TOTALI DELLA COMPAGNIA
145 REM E UNA MISURA DELLA PERFORMANCE DEL PRODOTTO,
150 REM PROFITTO PER QUOTA DI DOLLARI.
155 REM
160 REM PER CAMBIARE LA VARIETA' DI PRODOTTI, CAMBIARE I DATI DEI FATTORI
165 REM DI MODIFICA IN MODO CHE LE QUOTE TOTALI RIMANGANO INVARIATE.
170 REM
180 REM VARIABILI:
185 REM C.....COSTI FISSI ATTRIBUIBILI AI PRODOTTI
190 REM C(I,1)....COSTI FISSI DI PRODUZIONE
195 REM C(I,2)....COSTI FISSI DI PRODUZIONE (% DEL TOTALE)
200 REM D.....COSTI DIRETTI
205 REM F(I).....FATTORI DI MODIFICA
210 REM I.....VARIABILE INDICE
215 REM N#(I)....NOME DEL PRODOTTO
220 REM P.....PROFITTO TOTALE AL LORDO DI IMPOSTE
225 REM P(I,1)....PROFITTO DEL PRODOTTO
230 REM P(I,2)....PROFITTO DEL PRODOTTO (% DEL TOTALE)
235 REM P#().....FORMATO DI STAMPA
240 REM R(I,1)....QUOTA DI PRODOTTO
245 REM R(I,2)....QUOTA DI PRODOTTO (% DEL TOTALE)
250 REM S.....TOTALE DELLE VENDITE
255 REM T(I).....TRANSAZIONI DI PRODOTTO
260 REM T1,T2,T3,T4...VARIABILI DI LAVORO
265 REM
  
```

```

270 REM LEGGE I FATTORI DI MODIFICA (ALL' INIZIO TUTTI 1)
275 REM
280 FOR I=1 TO 9
285 READ F(I)
290 NEXT I
295 DATA 1,1,1,1,1,1,1,1,1
300 REM
305 REM CONDIZIONI INIZIALI
310 READ S,D,P
315 DATA 275000,181000,55000
320 C=S-D-P
325 REM
330 REM LEGGE IL NOME DEL PRODOTTO, LA QUOTA E LE TRANSAZIONI.
335 REM APPLICA ALLA QUOTA IL FATTORE DI VARIAZIONE.
345 REM
350 FOR I=1 TO 9
355 READ N*(I),R(I,1),T(I)
360 R(I,1)=F(I)*R(I,1)
365 T1=T1+T(I)
370 NEXT I
375 REM
380 REM PRODOTTO QUOTA (IN %1000) ORDINI (IN 100)
385 REM
390 DATA"FORNO ELETTRICO",30520,498
395 DATA"TOSTAPANE ",15870,660
400 DATA"TRITACARNE ",15870,350
405 DATA"SPREMIAGRUMI ",8280,182
410 DATA"CENTRIFUGA ",10350,64
415 DATA"APRISCATOLE ",4140,116
420 DATA"FERRO DA STIRO ",3450,78
425 DATA"ASCIUGACAPELLI ",2760,73
430 DATA"TEMPERAMATITE ",2760,109
435 REM
440 REM CALCOLA IL COSTO FISSO DEL PRODOTTO, C(). APPLICA IL FATTORE DI
445 REM MODIFICA AL COSTO FISSO. CALCOLA IL PROFITTO, P().
450 REM COSTI FISSI E PROFITTI.
455 REM
460 FOR I=1 TO 9
465 C(I,1)=T(I)/T1*C
470 C(I,1)=F(I)*C(I,1)
475 P(I,1)=R(I,1)-C(I,1)
480 T2=T2+R(I,1)
485 T3=T3+C(I,1)
490 T4=T4+P(I,1)
495 NEXT I
500 REM
505 REM TROVA LA QUOTA PERCENTUALE, IL COSTO FISSO E IL PROFITTO.
510 REM
515 FOR I=1 TO 9
520 R(I,2)=R(I,1)/T2*100
525 C(I,2)=C(I,1)/T3*100
530 P(I,2)=P(I,1)/T4*100
535 NEXT I
540 REM
545 REM STAMPA LA TABELLA
550 REM ***ATTENZIONE: L'ISTRUZIONE PRINT USING PUO' ESSERE DIVERSA
555 REM ***SUL VOSTRO COMPUTER.
560 REM
565 P$(1)=" ###,### ##"
570 P$(2)=" ###,### "
575 PRINT"TOTALE VENDITE: ";S
580 PRINT"COSTI DIRETTI: ";D
585 PRINT"VALORE DELLA PRODUZIONE --";S-D
590 PRINT"PROFITTO LORDO DA IMPOSTE: ";P
595 PRINT"COSTI FISSI --";S-D-P
600 PRINT

```

```

605 PRINT"
610 PRINT" % PROF./" VALORE % COSTI % PROF.";
611 PRINT"
612 PRINT" RIC." FISSI";
615 PRINT
620 FOR I=1 TO 9
625 PRINT N*(I);
630 PRINT USING P*(1):R(I,1),R(I,2);
635 PRINT USING P*(1):C(I,1),C(I,2);
640 PRINT USING P*(1):F(I,1),F(I,2);
645 IF R(I,1)<>0 THEN 650 ELSE 660
650 PRINT USING" ##.##":F(I,1)/R(I,1)
655 GOTO 670
660 REM
665 PRINT
670 REM
675 NEXT I
680 PRINT
685 PRINT"TOTALE " ";
690 PRINT USING P*(2):T2;
695 PRINT USING P*(2):T3;
700 PRINT USING P*(2):T4;
702 PRINT USING" ##.##":T4/T2
705 END

```

Il programma inizia con un buon numero di commenti sulla sua utilizzazione. La lista delle variabili spiega come queste saranno usate e serve come dizionario per una rapida consultazione.

Il programma vero e proprio inizia con i fattori di modifica F(). Cambiando questi fattori si altereranno i livelli di produzione: il due raddoppierà l'attività; lo zero la sospenderà. Da notare soprattutto la riga dove il fattore di modifica viene applicato ai valori di produzione e quella dove è applicato ai costi. Aumentando o diminuendo questi due elementi, si modifica l'attività di produzione.

Le condizioni iniziali sono i dati di base della società. Da queste condizioni si calcola il costo fisso C. Il dato che viene introdotto dopo le condizioni iniziali è l'informazione sul prodotto.

Il programma estrae i costi, calcola i profitti, controlla le colonne e trova le percentuali.

Dopo aver finito i calcoli, stampa i risultati. La costruzione IF...THEN...ELSE permette di evitare la divisione per zero quando si annulla il fattore di modifica di un prodotto. Troverete ulteriori dettagli su questa costruzione nell'appendice A.

ESERCIZI

1. Alterate i fattori di modifica per seguire i suggerimenti di Frank Bradshaw.
2. Inventate un nuovo insieme di fattori di modifica.
3. Per ottenere il massimo profitto, quale prodotto dovrete realizzare da solo? Perché questa non è in effetti una buona soluzione?
4. Applicare i fattori di modifica F() alle transazioni invece che ai costi. Cosa c'è che non va nei risultati?
5. Eseguite RAGGI-X con un altro insieme di prodotti.

IL VALORE DELL'ESPERIENZA

“Scusa se ti sembrerà una lezione,” disse Peter Bates, “ma questo è quello che facevo per lo studio di consulenza dove lavoravo prima. Fermami se c'è qualche dubbio.”

“Non preoccuparti, è ciò che voglio”, disse Frank Bradshaw.

“Quando scoppiò la seconda guerra mondiale, diventò molto più conveniente costruire aerei da combattimento. Ogni apparecchio costava di meno perché le fabbriche divennero più efficienti — tutta l'industria si sforzava di costruire aerei sempre migliori. Si può rappresentare il fenomeno tramite un grafico.




Fig. 3.1 Una curva di apprendimento

All'inizio degli anni '60 il concetto di curva di apprendimento fu allargato a quella che venne chiamata curva dell'esperienza. Molte industrie denunciavano lo stesso fenomeno: il loro costo per unità (in dollari costanti) aveva una caduta al crescere del numero totale di unità prodotte. Questa caduta su vasta scala riguardava qualcosa di più che lo studio della produzione di un articolo. Comprende l'intero complesso di produzione: finanze, management, ricerca e sviluppo, produzione e marketing. Un insieme di curve in realtà assomiglia a quello rappresentato in figura 3.2. La curva del costo è relativa ad una singola azienda. La curva del prezzo invece riguarda l'intero mercato. Sull'asse delle ascisse c'è il volume cumulativo di unità prodotte; sull'asse verticale i dollari per unità. Una delle conseguenze della curva del costo è che se tre aziende hanno diversi volumi di produzione cumulativi e diversi livelli di esperienza, i loro costi saranno differenti. Ad esempio, nella figura in questione, la ditta A ha il costo minore, mentre la ditta C ha quello maggiore.”

“Ed entrambe praticano lo stesso prezzo?” chiese Bradshaw.




Fig. 3.2 Due curve dell'esperienza

"Nel modello semplificato sì. Il prezzo industriale è determinato dalla somma dei volumi individuali. In questo caso si tratta del punto X."

"L'asse orizzontale delle X non rappresenta la somma dei valori d'ascissa di A, B e C", disse Bradshaw. "Perché?"

"Perché entrambi gli assi sono riportati in scala logaritmica. I valori riportati spianano le curve dell'esperienza ed hanno l'effetto collaterale di rappresentarle in prospettiva. La distanza da 1 a 10 è uguale a quella da 10 a 100. Così i valori di A, B e C, sommati, danno effettivamente il valore di X."

Bates soggiunse: "Siccome X è basato sul valore di mercato, che è la somma di A, B e C, X porterà sempre ad A, B e C. Nota che, a causa dei loro diversi costi unitari, le tre aziende hanno margini di profitto diversi. Come vedi, quello di C è un misero 10 per cento, quello di B è il 25 per cento, e per A c'è addirittura il 40 per cento. Questo succede effettivamente per le aziende leader del mercato."

"Quando la crescita di mercato rallenta, A raggiungerà il prezzo necessario per tenere in gioco abbastanza concorrenti per mantenere le distanze dal gruppo. Grazie al suo grosso vantaggio sia nel numero di unità prodotte che nel profitto per unità, può difendersi da sola contro tutti gli inseguitori."

"Stai dicendo", commentò Bradshaw, "che chi controlla il mercato fa molti affari. Fermiamoci un minuto. Questa curva dell'esperienza aiuta a spiegare perché le grandi aziende hanno i costi più bassi e fanno più affari. Bene. Ma a cosa ci può servire per la nostra produzione? Abbiamo sempre voluto che i nostri prodotti fossero leader dei rispettivi mercati."

"E lo sono?" chiese Bates.

“La maggior parte”, disse Bradshaw, “ha delle ottime quote di mercato.”

“Ho visto”, disse Peter Bates. “È difficile sapere come i nostri costi sono in relazione con quelli dei nostri concorrenti, ma le quote di mercato sembrano essere un buon termine di paragone. Ho anche controllato se i mercati erano nuovi e in crescita o vecchi e stabilizzati. È meglio guadagnare spazio in un mercato piccolo e nuovo che in uno grande e vecchio. Quando il mercato rallenta, le posizioni tendono a congelarsi. Non è una conseguenza della curva dell'esperienza; è una conseguenza dell'inerzia nella distribuzione, nelle vendite e nelle preferenze dei consumatori.”

“E cosa hai trovato?” chiese Bradshaw.

“Guarda questa tabella.

PRODOTTI	QUOTA DI MERCATO	VANTAGGIO	CRESCITA DEL MERC.	QUOTA REL.
FORNO ELETTRICO	0,40	0,25	0,15	1,60
TOSTAPANE	0,10	0,20	0,12	0,50
TRITACARNE	0,20	0,10	0,40	2,00
SPREMIAGRUMI	0,35	0,40	0,35	0,88
CENTRIFUGA	0,50	0,15	0,23	3,33
APRISCATOLE	0,25	0,45	0,30	0,56
FERRO DA STIRO	0,30	0,25	0,35	1,20
ASCIUGACAPELLI	0,30	0,22	0,45	1,36
TEMPERAMATITE	0,05	0,20	0,12	0,25

“Siccome ogni tipo di prodotto ha una quota di mercato diversa con un numero di concorrenti diverso e con differenti grandezze del mercato, mi serviva una misura che mi consentisse di confrontare un tipo di prodotto con un altro. La quota relativa sembra che vada bene. Ora spiego come l'ho trovata. Se il leader siamo noi, allora la quota relativa è il nostro vantaggio sui nostri concorrenti più prossimi; se siamo in posizione intermedia, allora è come ci confrontiamo con il leader. Se non siamo in grado di superare il leader, o di contrastarlo efficacemente, è il caso di pensare ad abbandonare la competizione.”

“Significa”, chiese Bradshaw, “che se spartissimo con un'altra ditta il primo posto avremmo una quota relativa di 1,0?”

“Proprio così”, convenne Bates. “La quota relativa di mercato ci mostra i nostri prodotti dal punto di vista dei leader nei confronti degli inseguitori. Se vogliamo vedere l'andamento nei rispettivi mercati, possiamo misurarli con un'altra scala di crescita del mercato. Una crescita bassa corrisponde ad un mercato vecchio; una crescita alta ad un mercato nuovo. Ho usato come tasso medio il 25 per cento. Un mercato è salito al 45 per cento in un anno e due mercati vecchi riescono appena a tenere il passo dell'inflazione con circa il 12 per cento. Ecco come vanno le cose.” (fig. 3.3).

Frank Bradshaw non disse niente. Fissava attentamente il foglio di car-




Fig. 3.3 Andamento di produzione — Crescita in relazione alla quota di mercato

ta. "La situazione peggiore è avere una quota bassa in un mercato vecchio. Direi che la tua analisi conferma il risultato dei nostri raggi X. Il tostapane ed il temperamatite sono tagliati fuori. Ma si vede che l'asciugacapelli ha un'importanza particolare. E dovremmo incrementare la produzione di apriscatole e spremiagiumi se sembra che vadano bene." Esaminò il grafico ulteriormente. "Questo grafico non l'hai fatto con il calcolatore. Perché?"

Peter Bates sorrise. "Un piccolo programma non può fare tutto. Non è impegnativo fare a mano un grafico con nove dati numerici. Ho fatto col computer la tabella solo perché sono riuscito ad avere un programma scritto da qualcuno della Wolf Division. L'ho modificato per adattarlo al nostro caso."

IL PROGRAMMA QUOTA

```

100 REM ***QUOTA***
105 REM
110 REM UNA MODIFICA DEL PROGRAMMA TABELLA
115 REM
120 REM IL PROGRAMMA LEGGE LA QUOTA DI MERCATO , IL VANTAGGIO SUL CONCORRENTE
125 REM PIU' PROSSIMO O SUL LEADER E LA CRESCITA DEL MERCATO.
130 REM
135 REM IL PROGRAMMA CALCOLA LA QUOTA DI MERCATO RELATIVA E NE STAMPA
140 REM I RISULTATI.

```

```

145 REM
150 REM VARIABILI:
155 REM C.....CONTATORE DELLE COLONNE
160 REM N.....NUMERO DELLE RIGHE
165 REM R.....CONTATORE DELLE RIGHE
170 REM R*( ).....NOME DELLE RIGHE
175 REM T( ).....TABELLA DEI VALORI
180 REM
185 REM DIMENSIONI:
190 DIM T(50,5)
195 REM
200 REM LEGGE NELLA TABELLA
205 REM
210 N=9
215 FOR R=1 TO N
220 READ R*(R)
225 FOR C=1 TO 3
230 READ T(R,C)
240 NEXT C
245 NEXT R
250 REM
260 DATA"FORNO ELETTRICO",40,25,15
265 DATA"TOSTAPANE ",10,20,12
270 DATA"TRITACARNE ",20,10,40
275 DATA"SPREMIAGRUMI ",35,40,35
280 DATA"CENTRIFUGA ",50,15,23
285 DATA"APRISCATOLE ",25,45,30
290 DATA"FERRO DA STIRO ",30,25,35
295 DATA"ASCIUGACAPELLI ",30,22,45
300 DATA"TEMPERAMATITE ",5,20,12
305 REM
310 REM CONVERTE LE PERCENTUALI IN DECIMALI E CALCOLA IL VANTAGGIO MAGGIORE
315 REM D IL NOSTRO VANTAGGIO NEI CONFRONTI DEL CONCORRENTE PIU' PROSSIMO.
320 REM QUOTA DI MERCATO RELATIVA.
325 FOR R=1 TO N
330 FOR C=1 TO 3
335 T(R,C)=T(R,C)/100
340 NEXT C
350 T(R,4)=T(R,1)/T(R,2)
360 NEXT R
365 REM
370 REM STAMPA DELLA TABELLA
375 REM
380 PRINT"PRODOTTI QUOTA DI VANTAGGIO ";
385 PRINT"CRESCITA QUOTA "
387 PRINT" MERCATO ";
388 PRINT"DEL MERC. REL."
390 PRINT
395 FOR R=1 TO N
400 PRINT R*(R);
405 FOR C=1 TO 4
410 PRINT USING" #.##";T(R,C);
415 NEXT C
420 PRINT
425 NEXT R
430 REM
435 END

```

Il programma QUOTA è una versione aggiornata di TABELLA. Legge valori diversi, calcola risultati diversi e stampa una rappresentazione finale ancora differente. In effetti, ciò che rimane di TABELLA è il concetto fondamentale della tripartizione: legge una tavola, effettua alcuni calcoli, e stampa la tavola finale. Se il programma iniziale di Lee fosse

stato scritto male, probabilmente Bates se ne sarebbe scritto un altro. Ma poiché il programma era chiaramente diviso in tre parti distinte, Bates fece prima a modificarlo che a riscriverlo dall'inizio.

L'esperienza di Bates ha messo a fuoco il valore dello sforzo di George Lee per scrivere un programma chiaro e facile da leggere. Siccome Lee ha fatto le cose bene la prima volta, Bates ha risparmiato tempo e fatica.

ESERCIZI

1. Modificate i dati di QUOTA per rappresentare l'insieme dei vostri prodotti.
2. Studiate l'istruzione PRINT USING del vostro Basic e modificate il programma perché funzioni sul vostro computer.

CRESCITA CONCORRENZIALE

Una settimana dopo Peter Bates si incontrò una quarta volta con Frank Bradshaw. Dopo aver localizzato i prodotti della Bear nei rispettivi mercati con le relative quote, vollero sapere quale sarebbe stato il costo di creare una quota in un mercato giovane. Per rispondere dovevano analizzare attentamente il mercato, i concorrenti, e come si muovevano. Per Bates era stata una settimana molto intensa.

Bradshaw era cordiale. Sapeva che Bates aveva a che fare con dati certi ma anche con voci incontrollate. I calcoli erano il pane di Bates e ciò rassicurava Bradshaw. Bates avrebbe ricondotto a numeri qualsiasi cosa. E questa tendenza si adattava particolarmente a questa parte del lavoro. "Ti faccio vedere un altro giochetto col computer", disse Bates.

"Ti piacerà."

"Un giochetto?" esclamò Bradshaw sorpreso. "Con tutto il lavoro che c'era da fare in questa settimana, hai avuto tempo per giocare? Come hai potuto?"

"Beh, lo avevo già pronto. L'ho usato per analizzare le nostre possibilità finanziarie in rapporto alla mia idea delle realtà di mercato. A dir la verità, senza il modello non avrei potuto completare l'analisi in una settimana."

"Prima di mostrarti alcuni dei dettagli," proseguì Bates, "vorrei farti vedere un semplice caso per entrare nello spirito del modello."

"Vediamo." Disse Bradshaw. Si accomodò sulla poltrona.

"Immagina un mercato semplificato con tre ditte concorrenti. Mentre tutte tre le aziende crescono, anche il mercato è in crescita. Supponiamo di sapere che un'azienda aumenterà la sua quota, una la conserverà e l'altra, per disinteresse o incompetenza, perderà terreno.

"Il primo anno tutte tre le aziende erano pari. Ognuna produceva 3000 unità. Il costo era di 10 dollari per unità. Il prezzo per tutte le 9000 unità del mercato era di 15 dollari l'una. La prima annata defini i punti di partenza per due curve di esperienza: la curva del costo per le singole aziende e la curva del prezzo di mercato. La curva del prezzo è un'invenzione utile, ma che dopo lascerò perdere. Ho supposto che entrambe siano curve all'80 per cento."

"Curve all'80 per cento?" chiese Bradshaw. "Cosa sono?"

"Significa che quando il valore cumulativo raddoppia, i dollari per unità scendono all'80 per cento del loro valore iniziale. Una curva al 90 per cento scende molto difficilmente. Una curva al 70 per cento molto facilmente. Per questo esempio ho scelto l'80 per cento, che è proprio in mezzo. Inoltre, molti prodotti della Bear stanno su questa curva. Ad ogni modo, dopo aver scelto i punti iniziali su entrambe le curve e la pendenza dell'80 per cento, ho abbozzato uno schema del mercato con i tre concorrenti. Le istruzioni con i dati mostrano come cresce il mercato e quali sono le quote di mercato per ogni azienda."

```

360      READ G,M(1),M(2),M(3)
365      DATA 1.00,.33,.33,.33
370      DATA 1.40,.43,.33,.24
375      DATA 1.40,.50,.33,.17
380      DATA 1.30,.50,.33,.17
385      DATA 1.20,.50,.33,.17
390      DATA 1.20,.50,.33,.17
395      DATA 1.20,.50,.33,.17
400      DATA 1.20,.50,.33,.17
405      DATA 1.20,.50,.33,.17
410      DATA 1.20,.50,.33,.17
415      DATA 1.20,.50,.33,.17
420      REM

```

"Il primo anno è il nostro punto di partenza. Tutte tre possiedono un terzo del mercato. L'anno dopo, il mercato cresce del 40 per cento (1.40). L'azienda A raggiunge lo .43 del mercato, l'azienda B lo .33 e l'azienda C scende allo .24.

"Nel secondo anno il valore di crescita del mercato è ancora di 1.40. La ditta A si attesta allo .50 del mercato, la ditta B conserva il suo .33 e la ditta C scende allo .17.

"Nel terzo anno il mercato comincia a diminuire (1.30) e le aziende conservano le loro quote. Nel quarto anno il mercato scende al valore 1.20 che conserverà a lungo.

"Modificando questi dati iniziali, si può ottenere qualsiasi modello. Infatti è proprio ciò che ho fatto nel rivedere ogni dato di produzione per giungere ai miei suggerimenti.

"In un certo anno il numero di unità prodotte sarà la crescita annuale per il numero prodotto nell'anno precedente."

$$U(0) = G \cdot U(0)$$

"Il numero totale di articoli sul mercato è $U(0)$. Ogni ditta deve produrre la sua parte:"

$$U(1) = M(1) \cdot U(0)$$

$$U(2) = M(2) \cdot U(0)$$

$$U(3) = M(3) \cdot U(0)$$

"Aspetta", interruppe Bradshaw. "Ciò non significa supporre che le aziende venderanno tutti gli articoli?"

"Sì, è così. Ciò può succedere se le aziende che guadagnano mercato applicano prezzi inferiori a quelli dei loro concorrenti. Ma, prima di cominciare a dare i prezzi ai nostri prodotti, dobbiamo finanziare la loro produzione. Lo facciamo prudentemente. Calcoliamo i nostri costi correnti per produrre le unità degli anni seguenti. Otterremo i prezzi di fine d'anno in modo che i nostri profitti risultino sottostimati se i nostri costi diminuiranno nel corso dell'annata come c'è da aspettarsi.

"Ho chiamato $F()$ la quota di finanziamento. Al primo anno la si calcola come il costo unitario per il numero di articoli per il tasso d'interesse.

$$-U(1) \cdot C(1) \cdot I^r$$

"Ho supposto che i proventi delle vendite del primo anno vengano reinvestiti per la crescita del secondo. La nuova quota di finanziamento è la vecchia quota più le vendite dell'anno precedente, meno il valore necessario a finanziare la produzione dell'anno corrente.

$$F(1) = (F(1) + S(1) - U(1) \cdot C(1)) \cdot I^r$$

$$F(2) = (F(2) + S(2) - U(2) \cdot C(2)) \cdot I^r$$

$$F(3) = (F(3) + S(3) - U(3) \cdot C(3)) \cdot I^r$$

"Dopo un po', se le vendite crescono molto, l'indice di finanziamento diventerà positivo. Si ottiene lo stesso interesse per la ditta che ci ha finanziati.

"Certo, è prudente", intervenne Bradshaw. "Se non riusciamo ad ammortizzare il tasso d'interesse siamo messi male."

"Sì," convenne Bates, "è prudente; almeno così si sottostimano i profitti. Vorrei che niente restasse scoperto.

"Sovvenzionando la nostra produzione annuale costruiamo i nostri volumi cumulativi per l'industria (per il prezzo) e per le tre aziende (per i costi).

$$V(0) = V(0) + U(0)$$

$$V(1) = V(1) + U(1)$$

$$V(2) = V(2) + U(2)$$

$$V(3) = V(3) + U(3)$$

“I costi unitari individuali vengono calcolati dal volume cumulativo di ogni azienda sulla curva costo-esperienza.

$$\begin{aligned}C(1) &= 10 * (V(1) / 3000) \uparrow (X) \\C(2) &= 10 * (V(2) / 3000) \uparrow (X) \\C(3) &= 10 * (V(3) / 3000) \uparrow (X)\end{aligned}$$

“X è il fattore d'esperienza per la curva all'80 percento. I calcoli sono piuttosto noiosi. Vuoi che li salti?”

“Grazie, ma ti ho chiamato proprio per fare i calcoli”, rispose Bradshaw. Bradshaw poteva permettersi di evitarli solo perché c'era Bates, che era un esperto.

“Il prezzo industriale è calcolato usando la curva prezzo-esperienza”, proseguì Bates.

$$P(0) = 15 * (V(0) / 9000) \uparrow (X)$$

“Il prezzo praticato da ogni azienda è calcolato come media ponderata dei valori iniziali ed in seguito come media del cambiamento annuale della quota di mercato. Le equazioni sono un po' complicate, ma hanno due interessanti proprietà. Primo, se vuoi aumentare la quota, devi praticare prezzi inferiori a quelli dei concorrenti. Secondo, se vuoi conservarla, devi far scendere ogni anno il tuo prezzo. La media ponderata dei diversi prezzi è il prezzo industriale.

```

525          P(0)=15*(V(0)/9000)↑(X)
530          IF Y<=0 THEN 535 ELSE 555
535          P(1)=.33/M(1)*P(0)
540          P(2)=.33/M(2)*P(0)
545          P(3)=.33/M(3)*P(0)
550          GOTO 575
555 REM
560          P(1)=L(1)/M(1)*P(0)
565          P(2)=L(2)/M(2)*P(0)
570          P(3)=(P(1)*M(1)+P(2)*M(2)-P(0))/(-M(3))
575 REM
580 REM

```

“Alla fine dell'anno, le vendite delle aziende sono date dal numero di unità vendute per il loro prezzo.

$$\begin{aligned}S(1) &= U(1) * P(1) \\S(2) &= U(2) * P(2) \\S(3) &= U(3) * P(3)\end{aligned}$$

“A questo punto abbiamo tutte le informazioni necessarie per stampare un risultato annuale, così passiamo ad una subroutine che fa proprio questo. Infine, prima di tornare all'inizio del nuovo anno, memorizziamo i valori delle quote correnti di mercato per poterle riutilizzare.

$$L(1) = M(1)$$

$$L(2) = M(2)$$

$$L(3) = M(3)$$

“All’anno seguente, quando avremo bisogno di riferirci alle quote di mercato dell’anno prima, le troveremo nelle L).”

“Quando servono?” chiese Bradshaw.

“Quando calcoliamo il prezzo relativo ad una ditta. Il prezzo è una funzione del cambiamento della quota di mercato. Per ricavarlo, bisogna conoscere sia il valore corrente che quello vecchio.”

“Okay, d’accordo”, annui Bradshaw. “Se l’azienda sta aumentando la sua quota, praticherà prezzi inferiori a quelli dei concorrenti.”

“Bene, c’è tutto,” disse Bates, “ora puoi vedere quali risultati si ottengono su queste curve dell’esperienza”. (fig. 3.4)




Fig. 3.4 La concorrenza fra tre ditte in cinque anni

“Puoi vedere che, dopo cinque anni, la ditta col .50 del mercato ha un margine di profitto di .41, mentre la quota di mercato .17 dà un margine di profitto di .20. Ricorda che entrambe, alla fine del primo anno, avevano un margine di profitto di .33.”

“In effetti è qualcosa.” Bradshaw scossa la testa. “Se fossero partite con un margine più stretto, la ditta C rischierebbe di uscire di scena. È chiaro che A controlla la situazione fin dal quinto anno.”

“C’è un altro dettaglio da aggiungere al nostro esempio”, disse Peter Bates. “La situazione finanziaria. È il motivo per cui non tutti cercano di conquistarsi una quota di mercato. Ecco cosa succede ad A, B e C al passare del tempo.” (fig. 3.5)




Fig. 3.5 La resa della conquista di una quota di mercato

“Il valore netto attuale della situazione delle tre ditte mostra che aumentare la quota è dispendioso in breve periodo, mentre si rivela molto vantaggioso a lungo andare. Il rischio consiste nel dover pianificare tutto con anni di anticipo. Non tutti sono disposti a farlo.”

“Adesso abbiamo un semplice modello che sottostima i nostri utili, se ce ne sono”, disse Bradshaw. “Come lo hai utilizzato per esaminare la nostra produzione?”

“Non è stato facile”, rispose Bates. “Nella realtà non esiste una curva del prezzo dell’industria. Ciò significa che nelle nostre ipotesi iniziali ogni meta per la quota di mercato deve essere accompagnata da una stima del prezzo che consente di raggiungere la meta. Ho raccolto questi valori parlando con i manager di produzione. Per esempio, le effettive istruzioni READ e DATA possono essere così:

```

READ G, M(1),P(1), M(2),P(2), M(3),P(3)
.
.
DATA 1.4, .50,8.40, .33,9.50, .17,14.00
.
.

```

“Dopo di che è stato semplice”, proseguì Bates. “Ho modificato la curva del costo per riflettere l’andamento di ogni prodotto. Ho dato uno

sguardo ai concorrenti e progettato alcune prospettive alternative. Una cosa che si acquisisce rapidamente è il voler raggiungere la quota il più presto possibile. Perciò ho provato a far raggiungere ad un certo prodotto la posizione di vertice in meno di cinque anni, possibilmente in tre. Ho confrontato i costi necessari per far questo con ogni prodotto. Ciò che CRESCITA ha reso possibile è una rapida valutazione delle alternative a partire dalle informazioni necessarie."

IL PROGRAMMA CRESCITA

```

100 REM ***CRESCITA***
110 REM
120 REM MODELLO DI UN'AZIENDA CHE OPERA IN UN MERCATO CON TRE CONCORRENTI
125 REM
130 REM VARIABILI:
135 REM C().....COSTO D'AZIENDA PER UNITA'
140 REM F().....QUOTA DI FINANZIAMENTO (DEBITI O CASSA)
145 REM G().....TASSO DI CRESCITA ANNUALE DEL MERCATO
150 REM L().....QUOTE DI MERCATO DELL'ULTIMO ANNO
155 REM M().....QUOTE DI MERCATO
157 REM P#.....FORMATO DI STAMPA
160 REM P(O).....PREZZO DI MERCATO PER UNITA'
165 REM P().....PREZZO D'AZIENDA PER UNITA'
170 REM S().....VENDITE DELL'ANNO
175 REM U(O).....UNITA' ANNUE SUL MERCATO
180 REM U().....UNITA' ANNUE PRODOTTE DALL'AZIENDA
185 REM V(O).....VOLUME CUMULATIVO DEL MERCATO
190 REM V().....VOLUME CUMULATIVO D'AZIENDA
195 REM Y.....VARIABILE INDICE ANNUALE
197 REM Z1,Z2,Z3.....VARIABILI DEL VALORE CORRENTE
200 REM
205 REM COSTANTI:
210 REM I9=1,10
215 REM X=LOG(.80)/LOG(2)
220 REM
225 REM PROGRAMMA PRINCIPALE
230 REM
235 REM INIZIALIZZAZIONE VARIABILI
240 REM
245 REM FOR I=0 TO 3
250 REM C(I)=10
255 REM F(I)=0
260 REM S(I)=0
265 REM V(I)=0
270 REM U(I)=0
275 REM NEXT I
280 REM
285 REM U(O)=9000
290 REM
295 REM CICLO ANNUALE
300 REM
305 REM LEGGE LA CRESCITA DI MERCATO G E LE QUOTE DI MERCATO DELLE AZIENDE M().
310 REM TROVA LE UNITA' PRODOTTE NELL'ULTIMO ANNO U(O) E LA PRODUZIONE DELLE
320 REM AZIENDE. CALCOLA LE QUOTE DI FINANZIAMENTO NECESSARIE F(), IL VOLUME
330 REM CUMULATIVO V(), I COSTI C() E I PREZZI P() D'AZIENDA, LE VENDITE S()
340 REM E STAMPA I RISULTATI.
350 REM
355 REM FOR Y=0 TO 10

```

```

360 READ G,M(1),M(2),M(3)
365 DATA 1.00,.33,.33,.33
370 DATA 1.40,.43,.33,.24
375 DATA 1.40,.50,.33,.17
380 DATA 1.30,.50,.33,.17
385 DATA 1.20,.50,.33,.17
390 DATA 1.20,.50,.33,.17
395 DATA 1.20,.50,.33,.17
400 DATA 1.20,.50,.33,.17
405 DATA 1.20,.50,.33,.17
410 DATA 1.20,.50,.33,.17
415 DATA 1.20,.50,.33,.17
420 REM
425 U(0)=G*U(0)
430 REM
435 U(1)=M(1)*U(0)
440 U(2)=M(2)*U(0)
445 U(3)=M(3)*U(0)
450 REM
455 F(1)=(F(1)+S(1)-U(1)*C(1))*19
460 F(2)=(F(2)+S(2)-U(2)*C(2))*19
465 F(3)=(F(3)+S(3)-U(3)*C(3))*19
470 REM
475 V(0)=V(0)+U(0)
480 REM
485 V(1)=V(1)+U(1)
490 V(2)=V(2)+U(2)
495 V(3)=V(3)+U(3)
500 REM
505 C(1)=10*(V(1)/3000)↑(X)
510 C(2)=10*(V(2)/3000)↑(X)
515 C(3)=10*(V(3)/3000)↑(X)
520 REM
525 P(0)=15*(V(0)/9000)↑(X)
530 IF Y<=0 THEN 535 ELSE 555
535 P(1)=.33/M(1)*P(0)
540 P(2)=.33/M(2)*P(0)
545 P(3)=.33/M(3)*P(0)
550 GOTO 575
555 REM
560 P(1)=L(1)/M(1)*P(0)
565 P(2)=L(2)/M(2)*P(0)
570 P(3)=(P(1)*M(1)+P(2)*M(2)-P(0))/(-M(3))
575 REM
580 REM
585 S(1)=U(1)*P(1)
590 S(2)=U(2)*P(2)
595 S(3)=U(3)*P(3)
600 REM
605 REM
610 GOSUB 655
615 REM
620 L(1)=M(1)
625 L(2)=M(2)
630 L(3)=M(3)
635 NEXT Y
640 STOP
645 REM
650 REM
655 REM SUBROUTINE: STAMPA UN ANNO
660 REM INPUT: C(),F(),G,M(),P(),S(),U(),V(),Y
665 REM OUTPUT: --
670 REM
675 REM L'ISTRUZIONE PRINT USING PUO' ESSERE DIVERSA NEL VOSTRO BASIC.
680 REM VERIFICATE.
685 REM

```

```

690 PRINT
695 PRINT
/00 P$="%"
702 D$="###,###.## ###,###.## ###,###.##"
705 PRINT"ANNO ";Y;" CRESCITA DEL MERCATO: ";G;" MERCATO: ";U(0)
710 PRINT"VOLUME CUMULATIVO DEL MERCATO":V(0)
715 PRINT
720 PRINT"
725 PRINT USING P$:"QUOTA",M(1),M(2),M(3)
730 PRINT
735 PRINT USING P$:"UNITA' ANNUE",U(1),U(2),U(3)
740 PRINT USING P$:"VOLUME CUM.",V(1),V(2),V(3)
745 PRINT USING P$:"PREZZO UNIT.",P(1),P(2),P(3)
750 PRINT USING P$:"COSTO UNIT.",C(1),C(2),C(3)
755 PRINT USING P$:"MARGINE", (P(1)-C(1))/P(1);
757 PRINT USING D$:(P(2)-C(2))/P(2), (P(3)-C(3))/P(3)
760 PRINT
765 PRINT USING P$:"DEBITI",F(1),F(2),F(3)
770 PRINT USING P$:"VENDITE",S(1),S(2),S(3)
775 PRINT USING P$:"NETTO CASSA",S(1)+F(1),S(2)+F(2),S(3)+F(3)
780 REM
785 REM CALCOLO DEL VALORE ATTUALE
790 Z1=(S(1)+F(1))*(1/19)↑Y
795 Z2=(S(2)+F(2))*(1/19)↑Y
800 Z3=(S(3)+F(3))*(1/19)↑Y
805 PRINT
810 PRINT USING P$:"VALORE ATT.",Z1,Z2,Z3
815 PRINT
820 PRINT
825 RETURN
830 END

```

ANNO 0 CRESCITA DEL MERCATO: 1 MERCATO: 9000
 VOLUME CUMULATIVO DEL MERCATO 9000

	A	B	C
QUOTA	0.33	0.33	0.33
UNITA' ANNUE	2,970.00	2,970.00	2,970.00
VOLUME CUM.	2,970.00	2,970.00	2,970.00
PREZZO UNIT.	15.00	15.00	15.00
COSTO UNIT.	10.03	10.03	10.03
MARGINE	0.33	0.33	0.33
DEBITI	-32,670.00	-32,670.00	-32,670.00
VENDITE	44,550.00	44,550.00	44,550.00
NETTO CASSA	11,880.00	11,880.00	11,880.00
VALORE ATT.	11,880.00	11,880.00	11,880.00

ANNO 1 CRESCITA DEL MERCATO: 1.4 MERCATO: 12600
VOLUME CUMULATIVO DEL MERCATO 21600

	A	B	C
QUOTA	0.43	0.33	0.24
UNITA' ANNUE	5,418.00	4,158.00	3,024.00
VOLUME CUM.	8,388.00	7,128.00	5,994.00
PREZZO UNIT.	8.68	11.32	16.03
COSTO UNIT.	7.18	7.57	8.00
MARGINE	0.17	0.33	0.50
DEBITI	-46,723.10	-32,818.20	-20,303.80
VENDITE	47,051.70	47,051.70	48,477.50
NETTO CASSA	328.53	14,233.50	28,173.70
VALORE ATT.	298.66	12,939.50	25,612.40

ANNO 2 CRESCITA DEL MERCATO: 1.4 MERCATO: 17640
VOLUME CUMULATIVO DEL MERCATO 39240

	A	B	C
QUOTA	0.50	0.33	0.17
UNITA' ANNUE	8,820.00	5,821.20	2,998.80
VOLUME CUM.	17,208.00	12,949.20	8,992.80
PREZZO UNIT.	8.03	9.34	13.18
COSTO UNIT.	5.70	6.25	7.02
MARGINE	0.29	0.33	0.47
DEBITI	-69,318.70	-32,806.10	4,593.11
VENDITE	70,825.50	54,354.50	39,530.50
NETTO CASSA	1,506.84	21,548.30	44,123.60
VALORE ATT.	1,245.33	17,808.50	36,465.80

ANNO 3 CRESCITA DEL MERCATO: 1.3 MERCATO: 22932
VOLUME CUMULATIVO DEL MERCATO 62172

	A	B	C
QUOTA	0.50	0.33	0.17
UNITA' ANNUE	11,466.00	7,567.56	3,898.44
VOLUME CUM.	28,674.00	20,516.80	12,891.20
PREZZO UNIT.	8.05	8.05	8.05
COSTO UNIT.	4.83	5.39	6.25
MARGINE	0.40	0.33	0.22
DEBITI	-70,219.00	-28,282.70	18,420.00
VENDITE	92,319.20	60,930.70	31,388.50
NETTO CASSA	22,100.20	32,648.00	49,808.60
VALORE ATT.	16,604.20	24,528.90	37,421.90

ANNO 4 CRESCITA DEL MERCATO: 1.2 MERCATO: 27518.4
 VOLUME CUMULATIVO DEL MERCATO 89690.4

	A	B	C
QUOTA	0.50	0.33	0.17
UNITA' ANNUE	13,759.20	9,081.07	4,678.13
VOLUME CUM.	42,433.20	29,597.80	17,569.40
FREZZO UNIT.	7.16	7.16	7.16
COSTO UNIT.	4.26	4.79	5.66
MARGINE	0.40	0.33	0.21
DEBITI	-48,867.40	-17,880.00	22,606.30
VENDITE	98,455.10	64,980.40	33,474.70
NETTO CASSA	49,587.70	47,100.40	56,081.00
VALORE ATT.	33,869.10	32,170.20	38,304.10

ANNO 5 CRESCITA DEL MERCATO: 1.2 MERCATO: 33022.1
 VOLUME CUMULATIVO DEL MERCATO 122712

	A	B	C
QUOTA	0.50	0.33	0.17
UNITA' ANNUE	16,511.00	10,897.30	5,613.75
VOLUME CUM.	58,944.20	40,495.10	23,183.10
FREZZO UNIT.	6.47	6.47	6.47
COSTO UNIT.	3.83	4.33	5.18
MARGINE	0.41	0.33	0.20
DEBITI	-22,857.20	-5,557.59	26,733.00
VENDITE	106,805.00	70,491.30	36,313.70
NETTO CASSA	83,947.70	64,933.70	63,046.70
VALORE ATT.	52,125.00	40,318.70	39,147.10

ANNO 6 CRESCITA DEL MERCATO: 1.2 MERCATO: 39626.5
 VOLUME CUMULATIVO DEL MERCATO 162339

	A	B	C
QUOTA	0.50	0.33	0.17
UNITA' ANNUE	19,813.30	13,076.70	6,736.51
VOLUME CUM.	78,757.50	53,571.90	29,919.60
FREZZO UNIT.	5.91	5.91	5.91
COSTO UNIT.	3.49	3.95	4.77
MARGINE	0.41	0.33	0.19
DEBITI	8,783.66	9,193.69	30,986.00
VENDITE	117,124.00	77,302.10	39,822.30
NETTO CASSA	125,908.00	86,495.80	70,808.30
VALORE ATT.	71,071.90	48,824.70	39,969.50

ANNO 7 CRESCITA DEL MERCATO: 1.2 MERCATO: 47551.8
VOLUME CUMULATIVO DEL MERCATO 209891

	A	B	C
QUOTA	0.50	0.33	0.17
UNITA' ANNUE	23,775.90	15,692.10	8,083.81
VOLUME CUM.	102,533.00	69,264.00	38,003.40
PREZZO UNIT.	5.44	5.44	5.44
COSTO UNIT.	3.21	3.64	4.42
MARGINE	0.41	0.33	0.19
DEBITI	47,159.30	26,899.10	35,480.40
VENDITE	129,393.00	85,399.40	43,993.60
NETTO CASSA	176,552.00	112,299.00	79,474.00
VALORE ATT.	90,599.40	57,626.90	40,782.80

ANNO 8 CRESCITA DEL MERCATO: 1.2 MERCATO: 57062.2
VOLUME CUMULATIVO DEL MERCATO 266953

	A	B	C
QUOTA	0.50	0.33	0.17
UNITA' ANNUE	28,531.10	18,830.50	9,700.57
VOLUME CUM.	131,064.00	88,094.50	47,704.00
PREZZO UNIT.	5.04	5.04	5.04
COSTO UNIT.	2.96	3.37	4.10
MARGINE	0.41	0.33	0.19
DEBITI	93,524.70	48,133.30	40,302.10
VENDITE	143,704.00	94,844.80	48,859.40
NETTO CASSA	237,229.00	142,978.00	89,161.50
VALORE ATT.	110,669.00	66,700.40	41,594.60

ANNO 9 CRESCITA DEL MERCATO: 1.2 MERCATO: 68474.6
VOLUME CUMULATIVO DEL MERCATO 335428

	A	B	C
QUOTA	0.50	0.33	0.17
UNITA' ANNUE	34,237.30	22,596.60	11,640.70
VOLUME CUM.	165,302.00	110,691.00	59,344.70
PREZZO UNIT.	4.68	4.68	4.68
COSTO UNIT.	2.75	3.13	3.83
MARGINE	0.41	0.33	0.18
DEBITI	149,314.00	73,542.00	45,524.90
VENDITE	160,224.00	105,748.00	54,476.10
NETTO CASSA	309,537.00	179,290.00	100,001.00
VALORE ATT.	131,274.00	76,036.50	42,410.20

ANNO 10 CRESCITA DEL MERCATO: 1.2 MERCATO: 82169.5
 VOLUME CUMULATIVO DEL MERCATO 417597

	A	B	C
QUOTA	0.50	0.33	0.17
UNITA' ANNUE	41,084.80	27,115.90	13,968.80
VOLUME CUM.	206,387.00	137,807.00	73,313.50
PREZZO UNIT.	4.36	4.36	4.36
COSTO UNIT.	2.56	2.92	3.57
MARGINE	0.41	0.33	0.18
DEBITI	216,170.00	103,859.00	51,218.40
VENDETE	179,174.00	118,255.00	60,919.10
NETTO CASSA	395,343.00	222,114.00	112,137.00
VALORE ATT.	152,422.00	85,634.60	43,233.90

Dopo l'elenco delle variabili, vediamo due costanti. Sono le due variabili il cui valore non cambia durante l'esecuzione del programma. Parecchi altri valori contenuti nel programma possono essere considerati come costanti. Ad esempio 9000, 3000 e lo .33 nelle equazioni del prezzo.

Abbiamo già visto nel testo che precede il programma i dettagli relativi ai calcoli.

In termini di programmazione, il neo del programma CRESCITA è costituito dall'istruzione GOSUB. Quando il computer la incontra, passa alla subroutine che inizia al numero di riga che segue il comando. Poi prosegue finché incontra l'istruzione RETURN e a tal punto ritorna al numero di riga che segue quello dell'istruzione GOSUB. Se da diversi punti del programma principale si passa alla subroutine con più istruzioni GOSUB, il RETURN riesce a riconoscere a quale delle varie istruzioni si riferisce. Le subroutine sono spiegate nell'Appendice A.

ESERCIZI

1. Quali sono le conseguenze dell'inclinazione della curva dell'esperienza sull'aumento della quota, in termini quantitativi e di tempo? E al 70 per cento? E al 90 per cento?
2. Qual è l'effetto di un fattore d'interesse di 1.20 sul valore d'aumento della quota?
3. Pensate ad una diversa pianificazione delle quote di mercato. È meglio aumentare la quota presto o tardi?
4. È facile o difficile aumentare la quota in un mercato che cresce lentamente?
5. Modificate la sezione di stampa perché consideri solo le variabili a cui siete interessati.
6. Qual è l'effetto di avere all'inizio cinque concorrenti invece di tre? Si possono ridurre le dimensioni del programma CRESCITA e, contemporaneamente, modificarlo per trattare il caso di cinque concorrenti?

7. Trovate nel programma CRESCITA qualche aspetto che non vi piace. Pensate ad un modo per migliorarlo. Modificate in questo senso il programma.
8. Inserite tre costanti per 3000, 9000 e .33. Chiamatele C9, P9 ed M9. Elencatele all'inizio del programma nella sezione apposita. Assicuratevi che il programma dia poi gli stessi risultati. Il cambiamento è vantaggioso?

RISULTATI FINALI

Era abbastanza facile trarre le decisioni finali a proposito dei nove tipi di prodotto della Bear. Un genere fu liquidato il più rapidamente possibile. Due erano venduti. La produzione di altri due venne incrementata come non mai. Peter Bates era perplesso. I cambiamenti non erano proprio come risultavano dai calcoli. Frank Bradshaw aveva contribuito con dei ragguagli sulla forza o debolezza dei diversi tipi di prodotto. Ed aveva anche adottato le stime di Bates per alcune previsioni di mercato. Il risultato netto era un complesso di produzione più forte e la previsione di utili molto maggiori per i due anni seguenti.

Il forno elettrico avrebbe mantenuto la quota di mercato. Erano necessari dei controlli di costo e quindi Bradshaw trasferì alcuni manager validi dal gruppo che produceva il tostapane ed assegnò loro il compito di tenere sotto controllo le variazioni dei costi. Il prezzo del tostapane crebbe per dar luogo a profitti in breve termine. Si progettò di trasferire le attrezzature dedicate ad esso alla produzione del forno elettrico. La produzione sarebbe stata distribuita in 24 mesi.

Si individuò nel tritacarne il prodotto trainante dell'intera divisione. Sarebbe stato prodotto in modo tale da guadagnare una porzione nettamente maggioritaria nel suo vasto mercato. Lo spremiagrumi si stava sforzando di raggiungere la leadership di mercato in 18 mesi. Se non ce l'avesse fatta, a quel punto sarebbe stato rivalutato.

La centrifuga era il prodotto disposto meglio. Avrebbe mantenuto la sua quota e procurato fondi per la produzione del tritacarne e dello spremiagrumi.

L'apriscatole sarebbe stato venduto. La divisione non aveva fondi abbastanza per spingerlo ad acquisire la necessaria quota nel suo difficile mercato.

Il ferro da stiro avrebbe conservato la sua quota.

Sull'asciugacapelli si fondavano le speranze a lungo termine della divisione. Le previsioni di marketing dicevano che il suo mercato, in futuro, avrebbe superato quello attuale del forno elettrico. Si sarebbe spinto a

tutti i costi l'asciugacapelli a raggiungere il 60 per cento del mercato entro tre anni.

Il temperamatite sarebbe stato immediatamente ceduto.

Valore attuale e rischio futuro

“Vorrei che tu mi aiutassi a progettare una politica di bilancio preventivo per il capitale di tutta la società, che ci consenta di programmare e controllare gli investimenti di capitale”, disse John Saltman.

Saltman era il vicepresidente della Chordata Corporation. Stava spiegando un nuovo progetto ad Harriet DeAngelo, sua assistente da tre anni. “Per ottenere ciò che vogliamo, dobbiamo cominciare dalla situazione attuale. Adesso abbiamo un diverso metodo di pianificazione del capitale per ognuno dei quattro settori. Ciò che ci serve è un nuovo sistema che si adatti ad ogni settore, ma che vada bene anche per tutta la società.

“Il nostro problema è che abbiamo una disponibilità di talenti distribuita casualmente nei nostri settori. Solo proporre loro un semplice concetto come il valore attuale netto richiederà tempo, non è il caso di pensare a tecniche complesse. Anche i settori più sofisticati sono troppo impegnati dalla concorrenza per riservarci molta attenzione, anche quando hanno bisogno di qualcosa. A loro interessa soprattutto ottenere le autorizzazioni il più presto possibile, e appena possono partono al galoppo per continuare con il loro lavoro.

“Per avvertirti di un cambiamento dovrai passare da tutti i direttori finanziari di settore, vedere di cosa si stanno occupando e raccomandare loro che l'investimento di capitale sia coordinato a livello di società.

“Poi torna qui, componi le varie parti e fai la tua sintesi. Vorrei una relazione entro una settimana.”

Una settimana più tardi la DeAngelo si presentò con una cartella sotto braccio per render conto del suo lavoro.

"C'è proprio una fiera là fuori. Solo la Wolf Division si serve del valore attuale netto, ma anche lì devono alla fine avere la loro stima del costo del capitale. La Bear Division usa il tasso interno di rendimento (TIR). La Deer Division utilizza il tasso contabile di rendimento (TCR), e la Hawk Division il periodo di remunerazione.

"Sebbene la Hawk Division produca sistemi moderni di pilotaggio per l'aviazione, i suoi responsabili finanziari sono gente all'antica. Hanno sempre fatto a modo loro e non vedono ragione di cambiare. Un tipo ha addirittura suggerito che tutti usino l'analisi del periodo di remunerazione.

"Per mettere a fuoco l'argomento, ho fatto un piccolo esempio ed ho chiesto loro come effettuassero la scelta tra due progetti."

"Che esempio hai fatto"? chiese Saltman.

"Supponiamo di avere due progetti, A e B. Quale scegliere per l'investimento? Ecco la mia tabella." DeAngelo aprì la sua cartella e mostrò a Saltman una piccola tavola:

ANNO	A	B
0	-10000	-10000
1	1000	2000
2	2000	3000
3	3000	5000
4	4000	5000
5	15000	8000

"Il tasso contabile di rendimento è la media dei profitti al netto d'imposta divisa per l'investimento iniziale. Per semplicità, ho detto alla Deer Division che tutti i valori sono profitti al netto d'imposta. Hanno fatto i loro calcoli così:

$$\begin{aligned} \text{TCR di A} &= (-10000 + 1000 + 2000 + 3000 + 4000 + 15000) / 5 / 10000 \\ &= 30 \text{ per cento} \end{aligned}$$

$$\begin{aligned} \text{TCR di B} &= (-10000 + 2000 + 3000 + 5000 + 5000 + 8000) / 5 / 10000 \\ &= 26 \text{ per cento} \end{aligned}$$

"Così A è il progetto da preferire per la Deer Division. Non è molto sorprendente se si tiene presente che è una catena al dettaglio di articoli sportivi.

"Ho detto a tutti gli altri settori di considerare questi valori come flussi

di cassa. Così, mostrando queste stesse due proposte alla Hawk Division, si ottengono periodi di remunerazione come questi:

- A—il periodo di remunerazione è di 4 anni.
B—il periodo di remunerazione è di 3 anni.

“Così, se andate alla Hawk, l'investimento preferito è B.

“Per la verità uno di loro era almeno un po' imbarazzato. Non gli andava giù che con A venisse persa la somma di 15 000 dollari al quinto anno.”

“Due settori su quattro usano metodi di confronto dei progetti che sono antiquati”, Saltman scosse la testa. “È proprio un problema. E gli altri due?”

“Tutti gli altri comprendono il concetto di flusso di cassa”, rispose la DeAngelo. “Così quando ho mostrato loro la tabella e ho detto che il costo del capitale era del 13 per cento, hanno iniziato con una tavola come questa:

ANNO	A	B	FATTORE DI ATTUALIZZAZIONE
0	-10000	-10000	1.0
1	1000	2000	.88
2	2000	3000	.78
3	3000	5000	.69
4	4000	5000	.61
5	15000	8000	.54

“Tutto bene per la Wolf Division. Hanno confrontato il valore attuale netto dei futuri flussi di entrata ed i risultati in questo modo:

	PAGAMENTO	FATTORE	VALORE ATTUALE
0	-10000	1	-10000
1	1000	.884956	884.956
2	2000	.783147	1566.29
3	3000	.693051	2079.15
4	4000	.613319	2453.28
5	15000	.54276	8141.41

IL VALORE ATTUALE NETTO E': 5125.08
IL COSTO DEL CAPITALE E': .13

ANNO	PAGAMENTO	FATTORE	VALORE ATTUALE
0	-10000	1	-10000
1	2000	.884956	1769.91
2	3000	.783147	2349.44
3	5000	.693051	3465.25
4	5000	.613319	3066.6
5	8000	.54276	4342.08

IL VALORE ATTUALE NETTO E': 4993.28
IL COSTO DEL CAPITALE E': .13

"Così alla Wolf Division preferiscono A."

"Come hai costruito queste tabelle?"; chiese Saltman. Era stupito del fatto che la DeAngelo sapesse programmare.

"Ho usato uno dei personal computer", disse DeAngelo. "Le equazioni erano molto semplici. Cominci con un costo di capitale C e calcoli il fattore di sconto:

$$C = .15 \\ F = 1 / (1 + C)$$

"Quindi, per un'entrata P, all'anno Y il calcolo è:

$$V = F^Y * P \\ \text{PRINT } Y, P, F^Y, V \\ T = T + V$$

"Con F^Y indichiamo la Y-esima potenza di F.

"Per andare dall'anno 0 all'anno 5, usiamo un ciclo così:

```
260   FOR Y=0 TO 5
270     READ P
280     DATA -10000,2000,3000,5000,5000,8000
290     V=F^Y*P
300     PRINT Y,P,F^Y,V
310     T=T+V
320   NEXT Y
325 REM
```

"Il dato interessante è che per introdurre un altro flusso di entrate, basta variare l'istruzione DATA. Può servire anche per trovare il tasso interno di rendimento, cioè il costo del capitale che annulla il valore attuale netto."

"So cos'è il TIR," disse Saltman, "ma come fai con quel programma?"

"Prendendo un risultato a caso ed eseguendo il programma," rispose DeAngelo. "Può sembrare dispendioso, ma è una tecnica molto rapida per trovare il TIR. Dopo un paio di tentativi, lo si ottiene. Se provi con i miei due progetti A e B, ottieni il 25.5 per cento per A ed il 27.9 per cento per B.

Per questo alla Bear Division scelgono B.

Ora alla Bear quelli che hanno ottenuto questi valori mi hanno voluto parlare di come confrontare due progetti piccoli con un terzo più grande quando ognuno ha un TIR differente. Intuivano vagamente che per trattare un gruppo di progetti dovevano esaminare tutte le possibili combinazioni, ma naturalmente nessuno di loro lo ha fatto.

Chi usa il TIR di solito è facilmente in grado di passare al metodo del valore attuale netto. Per chi utilizza il periodo di remunerazione ciò non è agevole perché non considera alcune delle entrate. Tutti quelli che usa-

no il tasso contabile di rendimento si accorgono di non tener conto dell'inflazione e del valore corrente della moneta. Tutti erano disposti a prendere in considerazione dei cambiamenti.

"Finora alla Wolf hanno usato i valori pubblicati del nostro indebitamento, dei titoli e del rischio di mercato per ottenere il costo del capitale per i loro calcoli del valore attuale netto. Dovremmo fissare il costo del capitale della società ed assicurarci che tutti loro lo conoscano."

DeAngelo si poneva nell'ottica dei responsabili di settore. La sua visita aveva mostrato che sforzi facevano i vari settori per tenere tutto sotto controllo.

"Un'altra cosa", proseguì DeAngelo. "Quasi tutti si lamentavano per il limite di 5000 dollari alle decisioni di impiego del capitale locale. Tutti hanno detto che è una vera seccatura. Uno ha detto che dover questionare con la società per ogni voce di bilancio da 5000 dollari è come chiedere ogni volta il permesso per andare al gabinetto."

"Convengo che sia troppo restrittivo", disse Saltman. "Quel limite rientrava in un vecchio tentativo di controllare tutto dal vertice. Ora non si può far niente con 5000 dollari. Se fissiamo un nuovo limite a 25 000 dollari, per noi potrebbe andar bene e loro sarebbero contenti."

"Probabilmente", aggiunse DeAngelo, "una quantità di spazio ragionevole migliorerà la loro efficienza."

"Bene", disse Saltman, "Possiamo cominciare a lavorare. C'è qualcos'altro?"

"No, è tutto. Ecco una relazione scritta sui miei colloqui."

"Grazie", disse Saltman. "Vorrei chiederti di curare il passaggio dei vari settori al nuovo sistema. Ti va bene?"

DeAngelo sorrise. Le era piaciuto incontrare i dirigenti finanziari di settore. Ciò che loro facevano giorno dopo giorno di riflesso rendeva tranquillo il quartier generale. Le andava l'idea di visitare i settori.

"Sì, grazie, sarei felice di veder applicato il nuovo piano."

"Ancora una cosa", disse Saltman. "Dove hai imparato a programmare i computer?"

"Ho fatto l'errore di chiedere a Peter Bates cosa ci faceva quella televisione nel suo ufficio. In un pomeriggio mi ha fatto un corso accelerato. Ma la cosa migliore è che mi ha prestato il computer per il fine settimana. Ora me ne servo quando devo eseguire calcoli troppo grandi per la mia calcolatrice."

UNA NUOVA POLITICA AZIENDALE

Saltman scrisse il costo del capitale della società. Si prevedeva che il tasso di mercato sarebbe stato più o meno del 24 per cento nell'anno se-

guente. Il *prime rate* si avviava ad assestarsi a circa il 13 per cento. Il rischio sistematico della Chordata nel mercato era del .4 per cento. Saltman calcolò il costo del capitale azionario:

$$\begin{aligned}\text{Costo del capitale azionario} &= .13 + (.24 - .13) \cdot .4 \\ &= .13 + .044 \\ &= .174\end{aligned}$$

Il carico fiscale dell'azienda era del 48 per cento. Il capitale totale era così ripartito: 42 per cento di obbligazioni e 58 per cento detenuto dagli azionisti. Così il costo medio ponderato del capitale era:

$$\begin{aligned}\text{Costo medio ponderato del capitale} &= (1 - .48) \cdot .13 \cdot (.42) \\ &\quad + .174 \cdot (.58) \\ &= .028 + .101 \\ &= .129\end{aligned}$$

Saltman si incontrò con i dirigenti finanziari di settore e descrisse il nuovo sistema. Dopo una lunga discussione furono d'accordo di metterlo in funzione. A distanza di una settimana uscì il comunicato ufficiale:

Secondo le indicazioni del consiglio direttivo questo ufficio ha lavorato alla formulazione di una coerente politica di bilancio. Di recente Harriet DeAngelo ha raccolto i vostri suggerimenti per una migliore realizzazione. Il suo lavoro ha determinato le seguenti modifiche al piano:

1. *Tutte le decisioni sull'investimento del capitale faranno riferimento al metodo del valore attuale netto (NPV).*
2. *A livello di settore si possono autorizzare tutti i progetti al di sotto dei 25 000 dollari.*

Speriamo che queste due modifiche possano portare un miglioramento in termini di rapidità e flessibilità delle decisioni di settore e nel contempo coordinare la politica di investimento del capitale della società.

Come prima tutte le decisioni sull'investimento del capitale devono essere comunicate a questo ufficio entro cinque giorni dalla loro autorizzazione. Inoltre valgono le forme di notifica correnti. Per calcolare il NPV avrete bisogno di conoscere il costo del capitale della società. La percentuale attuale della società è del 12.9 per cento ed ha valore da subito. Se alterazioni improvvise nel mercato suggeriranno qualche variazione, ve la trasmetteremo prontamente. La percentuale del 12.9 per cento contiene la stima migliore della nostra attuale posizione nel mercato, il nostro indebitamento e la posizione fiscale. È l'unico dato necessario per il calcolo del NPV.

Il metodo NPV è stato scelto per diversi motivi:

1. Si usa col flusso di cassa reale, non con valori contabili.
2. Comprende tutte le somme.
3. Tiene conto del valore del denaro.
4. È legato all'effettivo costo del capitale della società.
5. Consente di confrontare e raggruppare i progetti.
6. Tende a massimizzare il valore dei titoli degli azionisti.

Se qualcuno avesse difficoltà dovute al cambiamento di metodo, quest'ufficio sarà pronto ad aiutarlo. Per l'assistenza si parli con me o con Harriet DeAngelo. Colgo l'occasione per ringraziare tutti voi per i suggerimenti e le riflessioni che avete proposto. Questo nuovo sistema è risultato diretto del lavoro comune. Se aveste qualche idea per ulteriori miglioramenti, fatemelo sapere.

*John Saltman
Vicedirettore finanziario*

IL PROGRAMMA ATTUALE

```

100 REM ***ATTUALE***
110 REM
120 REM STAMPA UNA TABELLA DEL VALORE ATTUALE NETTO DI UNA SERIE DI
130 REM PAGAMENTI FUTURI. SE IL RISULTATO E' POSITIVO ALLORA SI TRATTA
140 REM DI UN BUON INVESTIMENTO.
150 REM
160 REM VARIABILI:
170 REM C.....COSTO DEL CAPITALE
180 REM F.....FATTORE DI SCONTO
190 REM P.....PAGAMENTO
200 REM T.....VALORE ATTUALE NETTO TOTALE
210 REM V.....VALORE ATTUALE DEL PAGAMENTO
220 REM Y.....INDICE ANNUALE
222 REM
225 REM PROGRAMMA PRINCIPALE
226 REM
230 C=.13
240 F=1/(1+C)
245 REM
250 PRINT"ANNO", "PAGAMENTO", "FATTORE", "VALORE ATTUALE"
260 FOR Y=0 TO 5
270 READ P
280 DATA -10000,2000,3000,5000,5000,8000
290 V=F*Y*P
300 PRINT Y,P,FTY,V
310 T=T+V
320 NEXT Y
325 REM
327 PRINT
330 PRINT"IL VALORE ATTUALE NETTO E'":T
340 PRINT"IL COSTO DEL CAPITALE E'":C
350 END

```


Questo programma fa proprio ciò che diceva DeAngelo. Modificando i dati del numero di anni e della serie di cifre, l'utente può ottenere un altro problema del valore attuale.

ESERCIZI

- Se un investitore immobiliare avesse due progetti, A e B, con le somme mostrate sotto, quale progetto dovrebbe scegliere se il costo del capitale è del 12 per cento? E se è del 18 per cento? Rifletteteci.
 A: -100 000 10 000 20 000 10 000 10 000 160 000
 B: -110 000 20 000 20 000 20 000 20 000 130 000
- Utilizzate il programma ATTUALE per trovare il tasso interno di rendimento di A e B. Analizzate a fondo l'argomento per scoprire perché non è un buon metodo per prendere decisioni sul bilancio di previsione.

RISCHIO

John Saltman notò che Harriet DeAngelo, durante la riunione del personale, stava scarabocchiando su un foglio. Alcuni tratti assomigliavano ad un programma per computer. Dopo la riunione le chiese cosa stesse facendo.

"Lo vedrai tra un paio di giorni", rispose. "Per ora è solo un abbozzo."

Saltman si rassegnò all'attesa e tre giorni dopo DeAngelo entrò nel suo ufficio con un sorriso raggianti. "Dai un'occhiata a questi grafici", disse.

Il primo era molto semplice. Illustrava tre possibili risultati con le relative probabilità (fig. 4.1).




Fig 4.1 Risultati possibili

"Ebbene?" chiese Saltman. "Cosa c'è di nuovo?"

"Solo un paio di cose, per ora", disse DeAngelo. "Innanzitutto che abbiamo tre risultati possibili con le diverse possibilità."

"Ma cosa c'è di così interessante?" reagì Saltman, quasi stupito.

"Si tratta di stime realistiche", disse DeAngelo. "Non sono simmetrici. E non si tratta solo di numeri. Sono i tre possibili risultati che possiamo stimare. Potremmo anche sviluppare il caso di più risultati con probabilità più precise."

"Non capisco ancora dove vuoi andare a parare", disse Saltman.

"Ecco gli stessi dati sotto forma diversa", proseguì DeAngelo mostrandogli un altro diagramma (fig. 4.2).




Fig. 4.2 Probabilità e numeri casuali tra 0 e 1

"Ci mostra le tre possibilità distribuite nell'intervallo tra zero e uno. È interessante sapere che il Basic possiede un generatore di numeri casuali compresi tra zero e uno. Così possiamo servirci di un programma come quello che segue per simulare gli esiti delle nostre decisioni di investimento:

```

200      R=RND(0)
210      IF R<=.11 THEN 220 ELSE 240
220      S=200
230      GOTO 300
240 REM
250      IF R<=.69 THEN 260 ELSE 280
260      S=500
270      GOTO 300
280 REM
290      S=1000
300 REM
310      PRINT S

```

"In questo modo possiamo vedere ciò che succede in questo caso particolare."

Saltman cominciava a capire perché lei fosse così soddisfatta.

"Ciò significa", disse, "che possiamo ripetere un centinaio di prove e fare la media per avere un valore più attendibile."

"Certo, si potrebbe", annuì DeAngelo.

"Che tipo di programma ne uscirebbe?" chiese Saltman.

"Qualcosa tipo questo." DeAngelo aggiunse alcune righe all'esempio precedente.

```

180 S=0
190 FOR T=1 TO 100
200 R=RND(0)
210 IF R<=.11 THEN 220 ELSE 240
220 S=S+200
230 GOTO 300
240 REM
250 IF R<=.69 THEN 260 ELSE 280
260 S=S+500
270 GOTO 300
280 REM
290 IF R<=1.0 THEN 292 ELSE 296
292 S=S+1000
294 GOTO 300
296 REM
298 PRINT"ERRORE NEL PROGRAMMA"
299 STOP
300 REM
310 NEXT T
320 PRINT"MEDIA DEI VALORI SCELTI":S/100

```

"È molto interessante", disse Saltman. "Mi hai mostrato un modo per considerare le alternative di rischio. È questo il tuo scopo?"

"Di più", rispose DeAngelo. "Avremmo potuto trattare a mano il problema degli investimenti, ma così abbiamo un modo per rappresentare le situazioni di rischio, cosa molto difficile da fare senza il calcolatore. Per esempio, vedere se l'esito di un evento può influenzare il successivo o se può condurre all'interruzione in un momento sbagliato.

"Come esempio di interruzione improvvisa, consideriamo il prezzo di una azione di una ditta con quattro elementi incerti nel suo futuro: l'economia, le offerte di assorbimento, gli effetti di concorrenza e la legislazione. Per ciascuno dei prossimi tre anni tutti quattro gli elementi influenzeranno il valore delle azioni.

"L'economia può crescere normalmente e portare le azioni a 1.10 (aumento del 10 per cento). Potrebbe subire un rallentamento e mantenere la crescita a 1.04. Potrebbe esplodere e spingere il prezzo ad una quota di 1.25. Le possibilità sono del 60 per cento per la crescita normale, del 30 per cento nel caso di recessione, e del 10 per cento nel caso di boom economico.

"Se il nostro valore delle azioni, P1, parte da 10, la prima parte sarà così:

```

2215 P=10
2220 FOR I=1 TO 3
2225 REM COMPORNTAMENTO ECONOMICO
2230 DATA .60,1.10,.30,1.04,.10,1.20,1,1
2232 GOSUB 4000
2234 P=S*P
2236 REM

```

"Il secondo elemento è una offerta di assorbimento. In un dato anno c'è il 20 per cento di probabilità di ricevere un'offerta. Se ciò accade, il

valore delle azioni andrà a 1.30 (30 per cento in più), ma non di più. Si fermerà improvvisamente. Il programma apparirà così:

```

2240 REM ASSORBIMENTO
2242     DATA .20,1,.80,0,1,1
2244     GOSUB 4000
2246     IF S=1 THEN 2247 ELSE 2249
2247     P=1.30*P
2248     GOTO 2315
2249 REM
2250 REM

```

“Il terzo elemento è ciò che avviene a causa della concorrenza. Se le cose procedono normalmente, non ci saranno effetti (1.00). C'è un 40 per cento di possibilità che il concorrente debole venga eliminato e spinga il valore delle azioni alla quota di 1.09. C'è un piccolo 20 per cento di possibilità che due concorrenti vogliano entrare nella nostra ditta. Ciò aumenterebbe di molto il valore delle azioni (1.50). Il programma potrebbe essere così:

```

2260 REM CONCORRENZA
2262     DATA .40,1.00,.40,1.09,.20,1.50,1,1
2264     GOSUB 4000
2266     P=S*P
2268 REM

```

“Il governo potrebbe approvare leggi con tariffe favorevoli che accrescano il prezzo (1.30), non fare niente (1.00), o approvarne altre che causino l'uscita della ditta dal mercato. Il valore delle azioni in caso di fallimento scenderebbe al 50 per cento del prezzo corrente. Questa sarebbe una seconda improvvisa interruzione. Ecco un'ipotesi di programma:

```

2270 REM LEGISLAZIONE
2272     DATA .10,1.30,.86,1.00,.04,.50,1,1
2274     GOSUB 4000
2276     P=S*P
2278     IF S=.50 THEN 2280 ELSE 2282
2280     GOTO 2315
2282 REM
2284 REM
2286     RESTORE
2310     NEXT I
2315 REM

```

“La caratteristica interessante di questo programma è che regoliamo le nostre distribuzioni a seconda della visione più opportuna della realtà. Quindi, abbiamo preso in considerazione due tipi di interruzioni improvvise che rendono il calcolo del rischio quasi impossibile da farsi a mano. Ripetendo 1000 volte l'intera procedura per i tre anni otteniamo una buona stima non solo per il valore delle azioni — la media — ma anche per il rischio ad esso associato — la deviazione standard.

"In effetti, questo programma potrebbe essere la base per una buona serie di altri programmi. Il ciclo con indice I potrebbe rappresentare una qualsiasi situazione con alcune distribuzioni.

La si potrebbe rimpiazzare con qualche situazione diversa. Ad ogni punto della situazione in cui sia nota la distribuzione potremmo inserire le opportune righe di programma."

```

2000 REM ***RISCHIO***
2020 REM
2040 REM CALCOLA LA DEVIAZIONE MEDIA E STANDARD (RISCHIO) DI UN MIGLIAIO
2050 REM DI SITUAZIONI DI RISCHIO.
2060 REM
2070 REM IN QUESTO PROGRAMMA LA SITUAZIONE E' IL PREZZO DI STOCK P.
2090 REM
2100 REM VARIABILI:
2105 REM D.....DEVIAZIONE DEI RISULTATI R()
2110 REM I.....INDICE ANNUALE
2115 REM M.....MEDIA DEI RISULTATI
2120 REM P.....PREZZO DELLO STOCK
2122 REM R().....RISULTATO DEI TENTATIVI
2124 REM R().....SOMMA DEI RISULTATI
2125 REM S.....VALORE SCELTO PER UN ANNO DATO
2126 REM T.....INDICE DEI TENTATIVI
2127 REM V.....VARIANZA DEI RISULTATI
2150 REM
2190 REM DIMENSIONI:
2194 DIM R(1000)
2200 REM
2210 FOR T=1 TO 1000
2215 P=10
2220 FOR I=1 TO 3
2225 REM COMPORTAMENTO ECONOMICO
2230 DATA .60,1.10,.30,1.04,.10,1.20,1,1
2232 GOSUB 4000
2234 P=S*P
2236 REM
2240 REM ASSORBIMENTO
2242 DATA .20,1,.80,0,1,1
2244 GOSUB 4000
2246 IF S=1 THEN 2247 ELSE 2249
2247 P=1.30*P
2248 GOTO 2315
2249 REM
2250 REM
2260 REM CONCORRENZA
2262 DATA .40,1.00,.40,1.09,.20,1.50,1,1
2264 GOSUB 4000
2266 P=S*P
2268 REM
2270 REM LEGISLAZIONE
2272 DATA .10,1.30,.86,1.00,.04,.50,1,1
2274 GOSUB 4000
2276 P=S*P
2278 IF S=.50 THEN 2280 ELSE 2282
2280 GOTO 2315
2282 REM
2284 REM
2286 RESTORE
2310 NEXT I
2315 REM
2320 RESTORE
2330 R(I)=P
2340 R(I)=R(I)+P
2350 NEXT T

```

```

2360 REM
2370 M=R(0)/1000
2375 REM
2380 FOR T=1 TO 1000
2390 V=V+(R(T)-M)*2
2400 NEXT T
2410 REM
2420 LET D=SDR(V/1000)
2430 REM
2440 PRINT"IL RISULTATO MEDIO E' ";M
2450 PRINT"LA DEVIAZIONE STANDARD E' ";D
2460 REM
2470 STOP
2480 REM
2490 REM
4000 REM SUBROUTINE: SCEGLI UN VALORE
4010 REM INPUT: <DATI DAL PROGRAMMA PRINCIPALE>
4020 REM OUTPUT: S
4030 REM
4040 REM LEGGE UNA LISTA DI NUMERI CHE RAPPRESENTA UNA DISTRIBUZIONE DI
4050 REM POSSIBILI USCITE. LA LISTA E' DI QUESTO TIPO:
4060 REM DATA .10,300,.60,500,.30,900,1,1
4070 REM OGNI COPPIA RAPPRESENTA UNA PROBABILITA' P3, E UN VALORE VS.
4080 REM LE PROBABILITA' IN TOTALE NON POSSONO SUPERARE 1.00. L'ULTIMA
4090 REM COPPIA (1,1) AVVISA LA SUBROUTINE DI NON LEGGERE PIU' DATI.
4120 REM
4125 REM VARIABILI:
4130 REM P3.....PROBABILITA' DI UN VALORE
4140 REM P4.....PROBABILITA' CUMULATIVA DELLA DISTRIBUZIONE
4160 REM R3.....UN NUMERO CASUALE TRA 0 E 1
4165 REM S.....IL VALORE SCELTO DALLA SUBROUTINE
4170 REM VS.....UN VALORE NELLA DISTRIBUZIONE
4180 REM
4190 REM INIZIALIZZA LE VARIABILI, SCEGLIE UN VALORE S DALLA DISTRIBUZIONE.
4200 REM FINISCE DI LEGGERE LA DISTRIBUZIONE.
4210 REM SI ASSICURA CHE AD S SIA STATO ATTRIBUITO UN VALORE.
4220 REM
4300 P4=0
4305 S=-.999999
4310 R3=RND(0)
4320 READ P3,V3
4330 P4=P4+P3
4340 IF P4>1.00 THEN 4450
4350 IF R3<=P4 THEN 4360 ELSE 4380
4360 S=V3
4370 GOTO 4400
4380 REM
4390 GOTO 4320
4400 REM
4410 READ P3,V3
4420 P4=P4+P3
4430 IF P4>1.00 THEN 4450
4440 GOTO 4410
4410 READ P3,V3
4420 P4=P4+P3
4430 IF P4>1.00 THEN 4450
4440 GOTO 4410
4450 REM
4460 IF S=-.999999 THEN 4470 ELSE 4500
4470 PRINT"LA SUBROUTINE CHE SCEGLIE UN VALORE"
4480 PRINT"HA FALLITO. CONTROLLARE I VALORI INTORNO A";P3;V3
4490 STOP
4500 REM
4510 RETURN
4515 REM
4530 REM
4540 END

```

IL RISULTATO MEDIO È 18.0329
LA DEVIAZIONE STANDARD È 6.56837

“A cosa serve la subroutine ‘scegli un valore?’” chiese Saltman. “Estrae un numero dalla distribuzione?”

“Esatto”, rispose DeAngelo. “Ho scritto una subroutine che può trattare una distribuzione estesa con molti dettagli, oppure una distribuzione breve con un solo valore. L'unico vincolo è che la somma delle probabilità non sia maggiore di 1.00. L'ultima coppia, la coppia uno-uno, informa la subroutine che è terminata la lettura della distribuzione.

“Utilizzando la subroutine ‘scegli un valore’ ogni volta che ci serve, possiamo rappresentare la maggior parte delle situazioni casuali dove siamo in grado di stimare la distribuzione degli eventi. Eseguendo 1000 volte il ciclo di variabile T dell'intera situazione, possiamo avere un numero di dati sufficiente per delle buone medie.”

“Così”, disse Saltman, “abbiamo uno strumento per trattare le situazioni di rischio di qualsiasi forma e dimensione. Giusto?”

“Già”, convenne DeAngelo.

“È molto per un programma così piccolo”, disse Saltman. “È quasi incredibile.”

“Sì, è sorprendente”, aggiunse DeAngelo. “Ho cominciato con lo scrivere una semplice routine per il calcolo di una distribuzione e mi sono ritrovata uno strumento per simulare il calcolo del rischio futuro in molti casi diversi.

“A proposito di rischio”, proseguì DeAngelo, “il risultato per la questione del valore delle azioni è stato 18.3 con una deviazione standard (rischio) di 7.2. È stato anche interessante vedere come erano distribuiti i risultati. Erano accumulati vicino a 16 con delle lacune attorno a 6 e si estendevano fino a 40” (fig. 4.3).




Fig. 4.3 Una distribuzione di risultati

IL PROGRAMMA RISCHIO

Questo programma offre un vantaggio immediato e uno a lungo termine. Il suo vantaggio immediato è che simula la situazione del valore delle azioni quando i risultati futuri non sono perfettamente conosciuti. Il vantaggio a lungo termine è che, come spiegava DeAngelo a Saltman, può servire come strumento per verificare una situazione non perfettamente nota se risultati diversi hanno diverse probabilità.

Per esaminare un'altra situazione basta rimpiazzare tutta la parte che va dalla riga 2215 alla 2315 e la variabile risultato (ora P). La nuova variabile risultato dovrebbe entrare nel ciclo di prova. Per esempio, se la nuova variabile fosse X, il testo potrebbe essere:

```
R(T)=X
R(O)=R(O)+X
X=0
```

R(O) è la somma complessiva dei risultati. Serve per calcolare la media M.

```
M=R(O)/1000
FOR T=1 TO 1000
  V=V+(R(T)-M)*2
NEXT T
LET D=SGR(V/1000)
```

La parte che accumula il totale V calcola anche la deviazione standard dei risultati. Questi calcoli sono spiegati su qualsiasi libro di statistica elementare.

La subroutine "scegli un valore" è una versione più generale del primo elaborato di DeAngelo. La subroutine estrae un numero casuale, continua la lettura della distribuzione finché non trova l'intervallo in cui sta il numero in questione, e poi sceglie quel valore come risposta. Ma, prima di finire il suo compito, la routine deve leggere il resto della distribuzione in modo che i dati non rientrino nelle future istruzioni READ. Se, per qualche ragione, nel corso della subroutine non viene assegnato un valore alla variabile S, l'ultima istruzione IF...THEN identifica l'errore e ferma il programma.

NOTA: La simulazione presenta molti aspetti particolari e non dovrebbe essere usata nella realtà per prendere decisioni d'affari senza la consulenza di un professionista competente.

ESERCIZI

1. Modificate RISCHIO sostituendo nelle istruzioni DATA le informazioni relative al vostro problema sul valore delle azioni.
2. Hafer scava pozzi petroliferi. Per ogni pozzo ha il 10 per cento di probabilità di guadagnare 1 000 000 di dollari, il 20 per cento di guadagnare 500 000 dollari, il 60 per cento di perdere l'intero costo della piattaforma di trivellazione (-140 000 dollari) ed il 10 per cento di perderne metà (-70 000 dollari). Trattate il problema di Hafer con RISCHIO e vedete come dovrà fare se parte con 500 000 dollari.
3. Modificate il programma ISTOGRAMMA dell'Appendice B ed aggiungetelo come subroutine di RISCHIO per la stampa dei risultati.

I dati societari

Simon Wilson era uno specialista in informatica alto e magro; aveva lavorato per quindici anni nel campo dell'elaborazione dati ed era direttore EDP alla Chordata. Pochi sapevano che aveva seguito gli studi universitari. Teneva ancora molto alle sue radici nell'Oklahoma, dove era cresciuto, e tutti lo conoscevano come "Slim" (lo smilzo).

Ben presto Wilson si era accorto che i microcomputer stavano entrando nelle case di tutti, volenti o nolenti. Decise di studiare il fenomeno: si informò sui modelli disponibili, ottenne uno sconto all'ingrosso per un buon modello, e diffuse la notizia che era disposto ad andare incontro a chi volesse comperare una macchina a prezzo favorevole.

Il commercio dei piccoli calcolatori si realizzava per le vie più informali. Wilson cercava di raccogliere informazioni in qualsiasi momento. Un giorno incontrò Peter Bates al bar e non si lasciò sfuggire l'occasione. "Ciao Peter. Sembra che abbia più a che fare coi computer tu all'ufficio programmazione che noi al centro elaborazione dati."

"Mah", disse Bates sogghignando. "Non è poi molto. Ma quel po' che ci vuole nel posto giusto può dare buoni risultati. Ho lavorato per un mese intero con Frank Bradshaw. Abbiamo realizzato un paio di programmi diversi per studiare la nostra gamma di prodotti. Uno dei due è risultato piuttosto lungo."

"Più di una pagina?"

"Sì, due o forse tre."

"Non sarebbe il caso", disse Wilson, "di cominciare a programmare in modo strutturato?"

“Ehi!” rispose Bates, “Io sono un analista economico, non un programmatore.”

“Non lasciarti intimorire dalle parole”, disse Wilson. “Programmazione strutturata è solo un termine ricercato per esprimere alcuni concetti derivanti dal buon senso per mantenere le dimensioni dei programmi entro limiti ragionevoli.”

“Ragionevoli?” Bates si fece serio. “Non capisco cosa vuoi dire.”

“Voglio dire che se i programmi diventano troppo complicati è facile non riuscire più a controllarli. Se ti capita di riutilizzarli dopo qualche modifica, potresti trovare nuovi inconvenienti. Seguendo alcune semplici norme nella stesura dei programmi, si possono evitare queste possibilità. Se fai un salto nel mio ufficio, ti posso dare le dispense che distribuiamo agli studenti dei nostri corsi.” (Vedi i paragrafi “Stile e struttura” e “Il programma più grande” nell'Appendice A).

 STRUTTURA

Due settimane più tardi Peter Bates incontrò di nuovo Slim Wilson al bar.

“Finalmente sono riuscito a leggere le tue dispense sulla programmazione strutturata, Slim. Poi ho riletto alcuni miei programmi e mi sono vergognato.”

“Non è il caso, Pete. Io ho violato ognuna di quelle norme almeno una dozzina di volte. Noi programmatori cominciamo ora ad applicare delle tecniche che ogni buon manager usa da anni.”

“Cioè?”

“Guarda”, Wilson disegnò qualcosa su un pezzo di carta:




Fig. 5.1 Lavoro diviso in subroutine

“Questa è l’idea: c’è un programma principale che affida il lavoro ad una subroutine. Puoi immaginare un manager che incarica un dipenden-

te di svolgere un certo compito. E il paragone vale anche per altri aspetti. Il manager fornisce le informazioni necessarie al dipendente, lui se ne serve e poi restituisce i risultati. Così si possono coordinare più lavori allo stesso modo. Il programma principale fa la stessa cosa. Si limita a raccogliere i risultati forniti dalle varie subroutine. Sulla base di essi, il programma principale può decidere quali altre routine richiamare, e così via.

"A loro volta le subroutine possono richiamare altre subroutine. A vantaggio del programma principale, raggruppando le funzioni in subroutine, si limita il numero di variabili da trattare."

"È così importante?" chiese Bates.

"Certo, se un certo linguaggio (e alcune versioni di Basic lo consentono) restringe il numero dei parametri di passaggio, automaticamente limita le possibilità che un errore si diffonda dalle subroutine al programma principale. In poche parole, così è più facile mantenere localizzati gli errori."

"È un vantaggio abbastanza ovvio", convenne Bates.

"Se hai a che fare con un programma di 10 000 righe che contiene un errore ti accorgi che è essenziale. Molte associazioni di persone sono collegate tra loro molto più strettamente delle subroutine di un programma per calcolatore, ma c'è un'organizzazione nel mondo reale che ha collegamenti ridotti all'essenziale come le strutture che stiamo esaminando."

"Quale?"

"Una rete di spionaggio. I canali di comunicazione sono limitati per lo stesso motivo: per limitare il danno se qualcosa non funziona."

"Capisco", disse Bates. "Poiché nessuno ha piena fiducia negli altri, si tratta di ridurre le conoscenze di ciascuno per minimizzare il pericolo."

"Giusto", commentò Wilson. "In un programma lungo, strutturato in questo modo, è molto più semplice rintracciare un errore passando con metodo da una subroutine all'altra. Una volta, prima che imparassi a programmare in questo modo, se c'era un errore in un programma lungo, impiegavo tre settimane per tornare dall'output sbagliato al punto del programma dove c'era l'errore."

"All'idea devono essere state tre pessime settimane. Perché dai così importanza alla modifica ed alla correzione dei programmi? A me capita di spendere molto più tempo a scrivere il programma che a modificarlo."

"Questo perché sei relativamente nuovo del mestiere e perché quei programmi servono solo a te", disse Wilson. "Che percentuale del tempo di lavoro su un programma dedichi alle modifiche?"

"Dopo che è ritenuto ufficialmente finito?"

"Sì. Completamente provato e funzionante."

"Mah," disse Bates, "forse il 10 o il 15 per cento di tempo in più."

"Negli ambienti professionali", disse Wilson, "si parla dell'80 per cento. L'ottanta per cento del lavoro su un programma è di modifica."

Bates fu sorpreso. "Ecco perché nello scrivere i programmi curi così tanto il fatto che siano facili da modificare."

"Amen, fratello", disse Wilson.

"Grazie per i consigli", disse Bates. "Me ne andrò e cercherò di non peccare più."

IL DATA BASE DELLA SOCIETA

Parecchi proprietari di microcomputer avevano chiesto a Wilson se potevano collegarsi al grande calcolatore della società e, se avessero potuto, cosa potevano fare. Wilson decise di convocare una riunione per informare tutti su ciò che il dipartimento EDP stava facendo e cosa potesse riguardare coloro che accedevano ai dati dei file del calcolatore grosso. "Molti di voi hanno chiesto cosa intendiamo quando diciamo che stiamo convertendo il nostro sistema a file in un data base. Oggi proverò a rispondere a questa domanda.

"Stiamo operando il consolidamento della contabilità dell'intero gruppo: registrazione fatture, acquisti, ordini, libro giornale, paghe. Ma il fatto che stiamo parlando di contabilità ora non è importante come il metodo che usiamo per farlo.

"In passato con i dati contenuti in file è stato possibile avere allo stesso tempo diverse copie della stessa informazione nel sistema. Per esempio, tutti e tre i nostri programmi che stampavano estratti del registro fatture avevano una copia del file corrispondente.

Il problema sorge quando si corregge un errore o si inserisce un nuovo elemento in un file e non in tutti tre. Finita l'operazione, due dei tre file contengono dati vecchi o sbagliati. Se viene inserita una nuova colonna in una riga di dati (ogni riga è un singolo record), allora ogni programma che legge quel file deve essere aggiornato per tener conto del nuovo inserimento. Se la colonna nuova è inserita al terzo posto, la colonna quattro diventa la cinque, e così via.

"Il metodo per evitare questi problemi consiste nell'inserire tutte le informazioni in un data base, una grande biblioteca a cui il pubblico non può accedere. Un impiegato è seduto dietro una scrivania con un catalogo pubblico degli argomenti ed uno schedario riservato. Voi potete utilizzare il catalogo per sapere cosa c'è nella biblioteca, ma l'impiegato è l'unico che può accedere allo schedario ed andare a prendere l'informazione che vi serve.

"Se volete aggiungere nuove informazioni (o modificare o cancellare informazioni) dovete mostrare il permesso speciale all'impiegato. Può darsi che dobbiate mostrarlo anche per leggere dati particolarmente importanti.

“L'impiegato all'interno del computer è un sistema che regola il data base e si chiama sistema di gestione del data base. Limitando l'accesso ai dati solo a questo “gestore”, il sistema distingue l'informazione (che è nel catalogo) dal luogo dove il dato è immagazzinato (che è nello schedario). Un utente può leggere il catalogo per vedere cosa c'è a disposizione, per esempio le fatture dell'ultimo mese per ogni settore commerciale, ma non occorre che l'utente sappia dove si trova, perché il “gestore” lo cercherà per lui.

“Il primo vantaggio del data base è che risolve il problema delle copie multiple, che si presentava spesso usando i file. Nel data base esiste una e una sola copia autorizzata dei dati. Tutte le modifiche vengono effettuate su quell'unica copia e tutti i programmi che si servono di essa sono perciò aggiornati. Siccome il catalogo assegna ad ogni colonna un nome specifico, ci si riferisce ai dati tramite questo nome, non col numero di colonna, ed è così risolto il secondo grande problema per la nostra riorganizzazione dei dati.

“Entrambe le parti in causa hanno vantaggi che non avevano in precedenza. Gli utenti possono ignorare la denominazione dei dati. Non occorre più che sappiano dove o come sono immagazzinati nella biblioteca. D'altro canto, i tecnici possono organizzare gli scaffali della biblioteca in qualsiasi modo — anche secondo un sistema efficiente, ma difficile da capire. Inoltre possono comperare nuove attrezzature e, sebbene modifichino lo schedario per l'impiegato, dal punto di vista dell'utente tutto rimarrà immutato.

“La conseguenza naturale di questa facilitazione dei compiti di entrambe le parti è che diventa molto più complicato il lavoro di collegamento. L'impiegato è una persona molto laboriosa che deve curare il catalogo per il pubblico e lo schedario e deve essere in grado di soddisfare tutte le richieste che vengono fatte.

“Così la vita si complica notevolmente per il sistema di gestione del data base, ma giacché si tratta di un programma per computer con questo scopo specifico, esso funziona abbastanza rapidamente da soddisfare le esigenze di tutti.”

UTENTI DI DATA BASE

“Ora è il caso di spendere alcuni minuti per guardare quali sono i diversi tipi di utente del nostro sistema data base. L'utente più comune è ancora un programma per computer; ad esempio, i tre programmi che producevano estratti dal registro fatture. Ognuno di questi programmi, invece di leggere i dati da un file, farà una “chiamata” al data base servendosi di un nuovo comando del linguaggio che usa. La chiamata dice

al gestore del data base di andare a cercare l'informazione e la passa al programma.

"La conseguenza operativa più rilevante di questo modo di procedere è che non c'è bisogno di modificare i programmi ogni volta che viene aggiornato il data base. Certo se è stato cancellato un pezzo d'informazione ed il programma cerca di leggerlo, sarà il caso di modificare il programma. Ma se i cambiamenti sono effettuati su dati che il programma non utilizza o sui metodi di memorizzazione di un elemento effettivamente utilizzato, il programma continuerà a funzionare bene senza bisogno di modifiche.

"Un secondo utente è il responsabile del data base con i suoi mezzi di accesso. Questo responsabile è un uomo in carne ed ossa. Le sue responsabilità sono molte: la sicurezza del sistema, l'integrità dei dati, l'efficienza del servizio all'utente e della memorizzazione dei dati. Soprattutto l'integrità dei dati è una responsabilità pesante. È come se il direttore di una biblioteca fosse responsabile di tutto ciò che è scritto nei libri che essa contiene.

"Si può scaricare parte di questa responsabilità affidando ad altre persone che conoscono meglio i dati il compito di correggere le informazioni. In effetti, il responsabile del data base cerca di delegare a persone competenti la maggior parte possibile di responsabilità. L'amministrazione continua del data base è un lavoro sufficientemente impegnativo comunque.

"Un terzo utente di data base può essere un linguaggio conversazionale. Senza di esso ci ritroviamo una biblioteca dove possono richiedere dei dati solo gli esperti, cioè quelli che sanno scrivere programmi, come tutti voi. Un tale linguaggio facilita le cose nel caso che un utente occasionale debba avere accesso diretto ai dati. Il linguaggio conversazionali sono stati scritti per andare incontro a questi utenti occasionali. Li aiutano ad ottenere facilmente e velocemente le informazioni dal data base. Guardate un momento questo gruppo di dati." Wilson proiettò una diapositiva su uno schermo.

530	DATA VERDE, COLORE, 37, 30, 39, 23, 24, 20, 33, 23
540	DATA BLU, COLORE, 32, 62, 44, 69, 58, 63, 49, 25
550	DATA BIALLO, COLORE, 32, 48, 32, 39, 55, 39, 48, 44
560	DATA MARRONE, COLORE, 45, 55, 27, 38, 56, 44, 41, 35
570	DATA ROSSO, COLORE, 37, 47, 77, 82, 85, 40, 41, 45
580	DATA MARGHERITA, FIORE, 70, 72, 63, 35, 60, 54, 56, 57
590	DATA NARCISO, FIORE, 8, 7, 6, 6, 4, 7, 6
600	DATA TULIPANO, FIORE, 58, 42, 66, 69, 57, 37, 50, 39
610	DATA CICUTA, PIANTA, 36, 44, 41, 15, 18, 35, 33, 42
620	DATA ABEETE, ALBERO, 83, 90, 86, 93, 91, 57, 114, 53
630	DATA PINO, ALBERO, 26, 29, 46, 49, 29, 50, 37, 31
640	DATA ACERO, ALBERO, 10, 10, 8, 10, 14, 9, 11, 8
650	DATA NOCCILO, ALBERO, 60, 41, 44, 71, 64, 28, 20, 45
660	DATA BETULLA, ALBERO, 39, 31, 24, 22, 27, 47, 43, 22
670	DATA QUERCIA, ALBERO, 70, 81, 42, 66, 29, 96, 45, 43
680	DATA PASSERINO, PESCE, 9, 8, 13, 12, 10, 5, 12, 9

```

690 DATA SALMONE,PESCE,79,45,51,53,43,46,55,63
700 DATA MERLUZZO,PESCE,5,4,4,4,5,4,3
710 DATA PERSICO,PESCE,12,55,37,55,41,43,18,14
720 DATA TROTA,PESCE,43,119,68,42,47,53,27,74
730 DATA BLACKBASS,PESCE,6,5,4,4,6,6,5,5
740 REM

```

“Se le otto cifre di ogni riga rappresentano le vendite previste, un buon linguaggio conversazionale potrebbe consentire un paio di operazioni immediate come queste.” Mostrò un'altra dispositiva:

- Definire una nuova colonna, “Somme di riga”, come $Q5 + Q6 + Q7 + Q8$
- Stampare il nome del prodotto, $Q5$, $Q6$, $Q7$, $Q8$ e la somma di riga quando questa è maggiore di 200

“Ecco un programma Basic che simula questi due comandi. Ve lo faccio vedere in modo che possiate vedere cosa fa il gestore del data base. E anche perché impariate ad usare i vostri file di dati che non verranno mai riuniti in data base.”

“Perché no?” chiese Harriet DeAngelo. “Non sarebbe più comodo averli in un data base?”

“Consentitemi per ora di rispondervi in un modo non del tutto esatto”, Wilson sorrise. “No. È più facile trattarli sotto forma di file. Specialmente quando sono presenti nei vostri computer. Tornerò su questa questione tra breve.

“Ecco il programma Basic con la corrispondente istruzione conversazionale.

```

360 REM TOTALI DELLE VENDITE DEGLI ULTIMI 4 TRIMESTRI
370 REM
380 FOR R=1 TO 21
390 S(R)=Q(R,5)+Q(R,6)+Q(R,7)+Q(R,8)
400 NEXT R
410 REM

```

“Notate che assegniamo il risultato a $S(R)$, la somma di riga. Questa rappresenta la nostra variabile conversazionale, Somma di riga.

“Poi simuliamo il nostro comando di stampa conversazionale.

```

420 REM STAMPA DEI RISULTATI
430 REM
440 PRINT"PRODOTTO TR15 TR16 TR17 TR18 PROD.TOT."
450 PRINT
460 FOR R=1 TO 21
470 IF S(R)>=200 THEN 480 ELSE 500
480 PRINT P*(R);TAB(12);Q(R,5);TAB(18);Q(R,6);
490 PRINT TAB(24);Q(R,7);TAB(30);Q(R,8);TAB(37);S(R)
500 REM
510 NEXT R
520 REM

```


“La stampa è prodotta dalle quattro istruzioni PRINT. L'istruzione IF...THEN...ELSE limita la stampa al caso in cui Somma di riga sia maggiore di 200.

“Ovviamente prima di fare tutto questo dobbiamo introdurre tutti i dati nel computer. Per questo utilizziamo una prima parte di input.

```

280 REM LETTURA DEI DATI
290 REM
300 FOR R=1 TO 21
310 READ P*(R), D*(R), O(R, 1), O(R, 2), O(R, 3), O(R, 4)
320 READ O(R, 5), O(R, 6), O(R, 7), O(R, 8)
330 NEXT R
340 REM
  
```

“La caratteristica interessante di questo programma è che può essere modificato per effettuare i calcoli relativamente a qualsiasi tabella di dati e che la stampa può avvenire nella forma che voi desiderate. Da questo esempio banale potete capire che potreste avere una notevole potenza anche senza un data base ed il relativo linguaggio conversazionale.”

“Ah si?” chiese DeAngelo, “E questo cosa significa?”

“Lo avete già intuito”, disse Wilson. “Quando il nostro sistema sarà pronto e funzionante faremo alcune lezioni introduttive per vedere come progettare i vostri data base personali ed ottenere da essi qualsiasi informazione.”

“È terribile”, commentò DeAngelo.

“Ma imparando questi semplici concetti di programmazione sarete già pronti al passo seguente”, proseguì Wilson. “Voglio dire che molti dirigenti, quando hanno i loro dati, non sanno poi cosa farsene. Dopo aver estratto i dati dal file della società, potete elaborarli con i vostri programmi Basic.”

“Così, imparando a programmare i nostri piccoli calcolatori, facciamo il secondo passo prima del primo”, intervenne Peter Bates. “Già”, disse Wilson. “E contemporaneamente avete ingranato la marcia di molti vostri concorrenti.”

IL PROGRAMMA RICHIESTA

```

100 REM ***RICHIESTA***
110 REM
120 REM
130 REM LEGGE UN DATA BASE, AGGIUNGE LE VENDITE DEL SECONDO ANNO
140 REM E STAMPA LE DIVISIONI SCELTE.
150 REM
160 REM VARIABILI:
170 REM C.....INDICE DI COLONNA
180 REM D*(.....NOME DELLA DIVISIONE DEL PRODOTTO
190 REM P*(.....PRODOTTO (O NOME DELLA RIGA)
  
```

```

200 REM  Q().....VENDITE PER TRIMESTRE
210 REM  R.....INDICE DI RIGA
220 REM  S().....SOMMA DI RIGA
230 REM
240 REM  DIMENSIONI:
250     DIM P*(25),D*(25),Q(25,8),S(25)
260 REM
270 REM
280 REM  LETTURA DEI DATI
290 REM
300     FOR R=1 TO 21
310         READ P*(R),D*(R),Q(R,1),Q(R,2),Q(R,3),Q(R,4)
320         READ Q(R,5),Q(R,6),Q(R,7),Q(R,8)
330     NEXT R
340 REM
360 REM  TOTALI DELLE VENDITE DEGLI ULTIMI 4 TRIMESTRI
370 REM
380     FOR R=1 TO 21
390         S(R)=Q(R,5)+Q(R,6)+Q(R,7)+Q(R,8)
400     NEXT R
410 REM
420 REM  STAMPA DEI RISULTATI
430 REM
440     PRINT"PRODOTTO   TRI5  TRI6  TRI7  TRI8  PROD.TOT."
450     PRINT
460     FOR R=1 TO 21
470         IF S(R)>=200 THEN 480 ELSE 500
480             PRINT P*(R);TAB(12);Q(R,5);TAB(18);Q(R,6);
490             PRINT TAB(24);Q(R,7);TAB(30);Q(R,8);TAB(37);S(R)
500 REM
510     NEXT R
520 REM
530     DATA VERDE,COLORE,37,30,39,23,24,20,33,23
540     DATA BLU,COLORE,32,62,44,69,58,63,49,25
550     DATA GIALLO,COLORE,32,48,32,39,55,39,48,44
560     DATA MARRONE,COLORE,45,55,27,38,56,44,41,35
570     DATA ROSSO,COLORE,37,47,77,82,85,40,41,45
580     DATA MARGHERITA,FIORE,70,72,63,35,60,54,56,57
590     DATA NARCISO,FIORE,8,7,6,6,4,7,7,6
600     DATA TULIPANO,FIORE,58,42,66,69,57,37,50,39
610     DATA CICUTA,PIANTA,36,44,41,15,18,35,33,42
620     DATA ABETE,ALBERO,83,90,86,93,91,57,114,53
630     DATA PINO,ALBERO,26,29,46,49,29,50,37,31
640     DATA ACERO,ALBERO,10,10,8,10,14,9,11,8
650     DATA NOCCIOLO,ALBERO,60,41,44,71,64,28,20,45
660     DATA BETULLA,ALBERO,39,31,24,22,27,47,43,22
670     DATA QUERCIA,ALBERO,70,81,42,66,29,96,45,43
680     DATA PASSERINO,PESCE,9,8,13,12,10,5,12,9
690     DATA SALMONE,PESCE,79,45,51,53,43,46,55,63
700     DATA MERLUZZO,PESCE,5,4,4,4,4,5,4,3
710     DATA PERSICO,PESCE,12,55,37,55,41,43,18,14
720     DATA TROTA,PESCE,43,119,68,42,47,53,27,74
730     DATA BLACKBASS,PESCE,6,5,4,4,6,6,5,5
740 REM
750     END

```

PRODOTTO	TRI5	TRI6	TRI7	TRI8	PROD. TOT.
ROSSO	85	40	41	45	211
MARGHERITA	60	54	56	57	227
ABETE	91	57	114	53	315
QUERCIA	29	96	45	43	213
SALMONE	43	46	55	63	207
TROTA	47	53	27	74	201

Il programma effettua le operazioni descritte da Wilson nella riunione. È composto da tre parti: lettura della tabella, calcolo e stampa dei risultati. Nella fase finale, la stampa è organizzata per mezzo dei TAB.

ESERCIZI

1. Scoprite che disponibilità, in termini di data base, offre il vostro centro di calcolo. Provate ad usare il linguaggio conversazionale, se c'è.
2. Modificate il programma RICHIESTA per leggere dati che si riferiscano a una delle vostre operazioni.
3. Modificate l'output di RICHIESTA per limitare la quantità di dati stampati.
4. Cambiate la sezione di calcolo di RICHIESTA perché faccia qualcosa di diverso.

Parte seconda:
programmare la produzione

Pianificazione dei progetti, programmazione e controllo

I tre manager di produzione della Deer Division, Jean Grant, Robert Grasso e Martin Graves, provenivano tutti dal gruppo di vendita. La general manager del settore, Eileen Randall, apprezzava le loro capacità nel marketing ma era preoccupata delle loro lacune in fase di programmazione, pianificazione e controllo dei progetti. Su suggerimento di Steve Cauldwell, aveva fatto in modo che George Lee della Wolf Division venisse a tenere un seminario di una giornata per i suoi tre manager di produzione.

Lee aveva lavorato duramente per preparare la sua relazione. Sapeva che avrebbe dovuto parlare a tre manager esperti e voleva che tutto filasse liscio. Il gruppo lo aspettava in una piccola aula per conferenze.

“Stiamo per parlare di una tecnica per la pianificazione di progetti composti da più operazioni. Per essere più chiaro mi servirò di un esempio molto semplice, ma affronteremo anche alcuni metodi che possono essere utilizzati con progetti molto più grandi. Lo faremo in modo informale. Siete pregati di interrompermi in qualsiasi momento se vi servono dei chiarimenti. Il nostro primo esempio è il problema dell'ospite-cuoco.

“Supponiamo che un gruppo di persone si stia preparando ad una cena di sabato sera e che uno degli ospiti insista per poter aiutare in cucina. Il

suo apporto di solito è più di danno che d'aiuto nella preparazione. Il problema consiste nell'utilizzare l'apporto dell'ospite senza che la cena subisca dei ritardi.

"Una buona base di partenza è una rappresentazione grafica del progetto di preparazione della cena. Il diagramma che segue è noto come diagramma di Gantt. Il nome è quello di Henry Gantt, che agli inizi di questo secolo lavorò molto alla stesura di diagrammi utili per la pianificazione ed il controllo dei progetti.



LEGENDA: XXX PREVISTO

"Il diagramma mostra cinque lavori, la loro durata ed il momento in cui vengono compiuti. I lavori sono: fare la spesa, preparare le pietanze, condire l'insalata, cuocere le pietanze e servire in tavola. Il lavoro che dura di più è fare la spesa (SPESA); il più veloce è SERVIRE. Il primo da fare è SPESA, l'ultimo è SERVIRE. (Solo per coincidenza sono il più lungo ed il più breve.) Due operazioni possono essere svolte contemporaneamente: per esempio CONDIRE e PREPARARE. Non è stabilito quale dev'essere la successione dei lavori, ma possiamo intuire che SPESA preceda CONDIRE e PREPARARE, PREPARARE preceda CUOCERE, e CUOCERE e CONDIRE precedano SERVIRE.

"Per costruire un programma che faccia tracciare questo grafico al computer, dobbiamo passargli alcune informazioni."

"Scusa," disse Grant, "perché c'è bisogno di un computer? È così semplice che lo si può fare a mano."

"Certo, si può", convenne Lee. "Ma la tecnica che sviluppiamo per questo esempio banale potrebbe essere usata in casi molto più complicati."

"Okay, vediamo", ammise Grant.

"Ecco la prima informazione", disse Lee.

```

330 DATA INIZIO,1,0
340 DATA SPESA,12,1
350 DATA PREPARARE,6,13
360 DATA CONDIRE,8,19
370 DATA CUOCERE,8,19
380 DATA SERVIRE,2,27
390 DATA FINE,0,29
400 REM

```


“Il primo numero è la durata in unità di cinque minuti. Il secondo numero è il momento in cui si può iniziare il lavoro. Per stampare la prima riga del diagramma di Gantt dovremmo scrivere così:

```

480     PRINT J*(J):TAB(11):"I";
500     FOR K=S(J) TO (S(J)+L(J)-1)
510         PRINT TAB(11+K):"X";
520     NEXT K
530     PRINT
540 REM

```

“Si otterrà:

```

          SPESA      IXXXXXXXXXXXX

```

dove ogni X corrisponde a cinque minuti di tempo.”

“L'intero programma per stampare il nostro diagramma di Gantt completo è questo:

IL PROGRAMMA GANTT

```

100 REM ***DIAGRAMMA DI GANTT***
110 REM
120 REM QUESTO PROGRAMMA STAMPA UN DIAGRAMMA DI GANTT DI UN INSIEME
130 REM DI LAVORI IN UN PROGETTO.
140 REM
150 REM
160 REM VARIABILI:
170 REM J,K.....VARIABILI INDICE
180 REM J*(J).....LAVORO
190 REM L().....DURATA DI UN LAVORO
200 REM N.....NUMERO DI LAVORI
210 REM S().....TEMPO DI PARTENZA
220 REM
230 REM
240 REM LEGGE IL NOME DELLAVORO J*(J), LA DURATA L()
250 REM E IL TEMPO DI PARTENZA S().
260 REM
270     READ N
280     DATA 7
290 REM
300     FOR J=1 TO N
310         READ J*(J),L(J),S(J)
320     NEXT J
322 REM
325 REM TEMPO (INTERVALLI DI 5 MINUTI)
330     DATA INIZIO,1,0
340     DATA SPESA,12,1
350     DATA PREPARARE,6,13
360     DATA CONDIRE,8,19
370     DATA CUOCERE,8,19
380     DATA SERVIRE,2,27
390     DATA FINE,0,29
400 REM
410 REM STAMPA UN DIAGRAMMA DI GANTT

```

```

420 REM
430 PRINT TAB(25);"TEMPO (MINUTI)"
440 PRINT"LAVORD";TAB(11);"0"           50           100           150           200"
450 PRINT TAB(11);"+-----+-----+-----+-----+
460 REM
470 FOR J=2 TO N-1
480   PRINT J*(J);TAB(11);"I";
500   FOR K=S(J) TO (S(J)+L(J)-1)
510     PRINT TAB(11+K);"X";
520   NEXT K
530   PRINT
540 REM
560 NEXT J
570 PRINT
580 PRINT
590 PRINT"LEGENDA: XXX PREVISTO"
660 REM
670 END

```

“Nel nostro caso, il diagramma di Gantt è un modo per insegnare a qualcuno come si prepara una cena.”

“Non è vero”, intervenne Grasso. “Non si impara come cuocere dei cibi guardando quel diagramma.”

“È vero”, ammise Lee. “Ho sbagliato. Ciò che il diagramma insegna è come programmare la preparazione della cena. Elenca le operazioni che si devono fare. Mostra come le varie componenti del progetto si distendono nel tempo. La spesa è la prima cosa da fare ed impiega un'ora di tempo. Mentre si preparano le pietanze da cuocere si può condire l'insalata. Quando tutto è pronto, si serve la cena.

“Possiamo servirvi di questo diagramma per seguire il progetto per controllarlo. Quando un'operazione è compiuta, facciamo un segno sul diagramma. Oppure possiamo tracciare un'altra riga che rappresenta lo stato effettivo dell'operazione, dall'inizio alla fine. Potremmo modificare il diagramma in modo che tenga conto dei cambiamenti di programma. Sul progetto aggiornato potrebbe apparire la nuova situazione.

“Infine, siccome un vecchio diagramma riporta una storia del progetto e la sua conclusione, possiamo utilizzarlo per migliorare le nostre previsioni per il prossimo progetto. Una persona può impiegare più tempo di quanto si è stabilito, per completare un'operazione. Un'altra può fare tutto molto più velocemente. Un'altra ancora impiegare la metà del tempo previsto. Tutti questi dati dell'esperienza possono servire per redarre nel modo migliore il prossimo progetto.

“In effetti, aggiungendo l'inizio e la fine di ogni operazione alla nostra tabella, e rivedendo un po' il nostro programma, possiamo ottenere un diagramma così:



“Guardando il diagramma, vediamo che la cena era pronta un po’ prima di quanto era stato previsto, perché si è impiegato meno tempo in fase di cottura. Inoltre il diagramma ha parzialmente risposto al problema dell’ospite-cuoco. Poiché solo due operazioni possono essere effettuate contemporaneamente, l’ospite aiuterà in una delle due. Il problema ora è decidere quale.”

ESERCIZI

1. Modificate il programma GANTT perché stampi anche i tempi effettivi di svolgimento delle operazioni.

DALLE OPERAZIONI ALLE RETI DI OPERAZIONI

“Per rispondere alla domanda precedente dobbiamo sapere precisamente che relazione c’è tra un lavoro e tutti gli altri. Ci serve un grafo delle operazioni. Proviamo a tracciare il grafo che riguarda la nostra cena.” (fig. 6.1)

“Perché nel disegno ci sono l’INIZIO e la FINE?” chiese Graves.

“Perché può darsi che all’inizio i progetti si ramifichino subito o che alla fine parecchie operazioni si svolgano in parallelo. Per iniziare e finire chiaramente la rete, conviene avere un unico punto d’inizio e un’unico punto di fine. Per esempio, se avessimo aggiunto un’operazione OFFRIRE-UN-DRINK, quest’operazione non avrebbe avuto precedenti tranne INIZIO, né successori tranne FINE.

“Guardando la rete potete calcolare quanto ci vuole per andare dall’inizio alla fine. Ad ogni operazione associamo due numeri, la durata del lavoro e l’orario minimo d’inizio:

	INIZIO	(1,0)	
	SPESA	(12,1)	
PREPARARE	(6,13)		CONDIRE (7,13)
CUOCERE	(8,19)		
	SERVIRE	(2,27)	
	FINE	(0,89)	




Fig. 6.1 Una rete di operazioni

“Il calcolo del tempo minimo d’inizio è facilissimo se si conosce la durata di ogni operazione. Notate che SERVIRE deve attendere finché l’ultimo dei suoi predecessori, CUOCERE, è finito. SERVIRE ha come orario minimo d’inizio 27. Il tempo è misurato ancora in intervalli di cinque minuti.”

“Ciò significa che l’intera cena sarà completata in 28 unità di tempo e si potrà cominciare a mangiare al tempo 29?” chiese Grant.

“Sì, proprio così”, rispose Lee. “Grazie per averlo ricordato.”

“Perché INIZIO ha durata uno?” domandò Grasso.

“È un espediente per far iniziare tutte le prime operazioni all’intervallo numero uno. Non cambia niente. Domanda interessante.”

“Questo nostro esempio è così semplice,” proseguì Lee, “che i calcoli successivi possono sembrare banali, ma abbiate pazienza. Ora comincio dalla fine per tornare indietro sulla rete e calcolare il tempo ultimo d’inizio di ogni operazione. Sarà il terzo numero all’interno delle parentesi:

	FINE	(0,29,29) 0	
	SERVIRE	(2,27,27) 0	
CUOCERE	(8,18,19) 0		
PREPARARE	(6,13,13) 0	CONDIRE	(7,13,20) 7
	SPESA	(12,1,1) 0	
	INIZIO	(1,0,0) 0	

“Tranne che in un caso, i tempi di inizio minimo e ultimo sono uguali. Solo per CONDIRE i due momenti sono diversi. Poiché le operazioni che precedono SERVIRE impiegano così tanto, CONDIRE potrebbe dover aspettare fino al tempo 20. Ma a partire dal tempo 13, si potrebbe iniziare. La differenza tra i tempi d’inizio è detta ‘scarto’. CONDIRE ha uno scarto di sette unità. Questo numero appare stampato a destra delle parentesi.

“Tutti gli altri lavori non hanno scarto. Si notino gli zeri a destra delle parentesi. Le operazioni che non hanno scarto giacciono sul cammino ultimo del progetto. Questo cammino ultimo è il ‘percorso critico’ perché il ritardo di qualsiasi operazione del percorso causa un ritardo nell’intero progetto. Il percorso critico nel caso del nostro progetto è SPESA, PREPARARE, CUOCERE, SERVIRE.

“Se la rete di lavori fosse più grande di quella del nostro esempio, e avesse ad esempio qualche centinaia di elementi, sarebbe difficile controllarli tutti. Il manager prudente riserva la sua attenzione soprattutto ai lavori che giacciono sul percorso critico. Assicurandosi che questi procedano bene, il manager è sicuro che anche l’intero progetto rispetterà le scadenze. Ovviamente, se un’altra operazione supera lo scarto previsto, può finire anch’essa sul percorso critico.”


“Se il progetto è grande,” disse Graves, “le cose si possono fare complicate.”

IL NUOVO DIAGRAMMA DI GANTT

“Grazie Marty”, disse Lee. “La tua domanda ci riporta al nostro diagramma di Gantt.

“Possiamo in qualche modo cambiarlo perché tenga conto di argomenti essenziali come il tempo di scarto ed il percorso critico? Naturalmente ho posto la questione sapendo bene che la risposta è ‘sì’.”

Lee mostrò loro un diagramma di Gantt migliorato:



TEMPO COMPLESSIVO= 145

“Notate come abbiamo fatto in modo che questo diagramma ci dica come sono ordinati i vari lavori. Ogni operazione che sta sul percorso critico è rappresentata con le lettere ‘C’. Se un lavoro non è critico, sarà rappresentato dalle ‘X’ e dai punti che indicano lo scarto disponibile nel caso che quel lavoro inizi il più presto possibile.”

“Come ha potuto il computer sapere quali operazioni fanno parte del percorso critico?” chiese Grasso. “Gli abbiamo dato qualche altra informazione?”

“Gli abbiamo passato qualche informazione nuova e ne abbiamo tolta qualcuna di vecchia”, rispose Lee. “La cosa più importante è che abbiamo scritto una nuova parte di programma che tenga conto di parecchie cose.

“Cominciamo dal fondo. Il diagramma di Gantt riporta il nome di ogni operazione, la sua durata, il suo tempo minimo d’inizio ed il suo scarto.”

“Ciò significa”, disse Graves, “che il programma deve conoscere anche il tempo ultimo d’inizio, perché lo scarto è dato dalla differenza tra il tempo minimo ed il tempo ultimo d’inizio.”

“Esatto”, confermò Lee. “Ma tutto ciò che vogliamo vedere dal diagramma è lo scarto. Lo scarto fornisce anche un indizio sulla sequenza delle operazioni. Esso si estende fino all’inizio dell’operazione seguente, perciò dal diagramma così com’è possiamo dedurre quasi tutto il grafo corrispondente. Non ci serve nient’altro per gestire il progetto. Stampiamo i numeri in una tabella corrispondente al diagramma di Gantt:

OPERAZIONE	DURATA	TEMPI DI INIZIO		
		MINIMO	MASSIMO	SCARTO
SPESA	12	1	1	0
PREPARAZ.	6	13	13	0
CONDIRE	7	13	20	7
CUOCERE	8	19	19	0
SERVIRE	2	27	27	0

“Rivedremo più avanti la parte di programma che stampa la tabella ed il diagramma. Una cosa più interessante da considerare sono i dati iniziali richiesti dal programma.”

LA TABELLA DEI PREDECESSORI

“Dobbiamo informare il programma sui collegamenti tra le varie operazioni. Un modo per fare questo consiste nell'aggiungere i predecessori di una certa operazione, quando la si introduce. Le informazioni relative al progetto per la nostra cena potrebbero essere queste:”

```

2200 DATA INIZIO, BASTA
2205 DATA SPESA, INIZIO, BASTA
2210 DATA PREPARAZ., SPESA, BASTA
2215 DATA CONDIRE, SPESA, BASTA
2220 DATA CUOCERE, PREPARAZ., BASTA
2225 DATA SERVIRE, CUOCERE, CONDIRE, BASTA
2230 DATA FINE, SERVIRE, BASTA
2255 REM

```

“Cosa significa BASTA?” intervenne Grant. “Non ci sono operazioni con questo nome.”

“È un modo per dire al programma che, per quel lavoro, non ci sono altri predecessori. Poiché non sappiamo preventivamente quanti predecessori un'operazione avrà, facciamo che il programma legga i predecessori finché non trova BASTA a indicare la fine della riga.”

“Una domanda”, disse Grasso. “Guardiamo PREPARARE. Ha come predecessore solo SPESA. Perché non c'è anche INIZIO? Viene prima di PREPARARE.”

“Giusto”, disse Lee. “Notate però che INIZIO non è un predecessore immediato. Si può risalire da PREPARARE a SPESA e da SPESA ad INIZIO, ma ciò che ci serve per costruire la rete del progetto è la lista dei predecessori immediati.

“Il programma considera queste informazioni e crea una tabella come questa:

		Lavoro numero J =						
		1	2	3	4	5	6	7
Lavoro numero I =	1	0	0	0	0	0	0	0
	2	1	0	0	0	0	0	0
	3	0	1	0	0	0	0	0
	4	0	1	0	0	0	0	0
	5	0	0	1	0	0	0	0
	6	0	0	0	1	1	0	0
	7	0	0	0	0	0	1	0

Fig. 6.2 Una tabella dei predecessori

“Un 'I' indica che il lavoro I ha il lavoro J come predecessore. Se chiamiamo P(I,J) la tabella, per ogni I il programma può leggere tutti i J, trovare i P(I,J) che hanno valore '1' e riconoscere che il lavoro J precede I.”

“Credo di non aver capito”, disse Graves. “Potresti ripetere?”

“Certo”, rispose Lee. “Cominciando da una riga, un'operazione, le si può passare tutte in rassegna per trovare le colonne che contengono '1' e quindi conoscere i predecessori di quella operazione.”

“Okay, se è questo che intendi, ho capito”, disse Graves. “Ma senti: ciò vuol dire che se io scelgo una colonna, posso scorgerla per trovare i successori del lavoro che rappresenta?”

“Mi hai tolto le parole di bocca”, disse Lee. “Proprio così. Questa tabella consente al programma di percorrere la rete.

“Un vantaggio di avere la rete contenuta nel programma è che si possono visualizzare i tempi d'inizio. Quest'informazione non ci serve più. L'unico dato necessario, oltre ai predecessori, è la durata di ogni operazione. Possiamo introdurre questi dati così:

```
2085   FOR J=1 TO N
2090     READ L(J)
2095   NEXT J
2096   REM
2097   REM TEMPO (INTERVALLI DI 5 MINUTI)
2100   DATA 1,12,6,7,8,2,0
2105   REM
```

“Poi il programma effettua le operazioni che abbiamo fatto a mano. Calcola i tempi minimi ed ultimi d'inizio, e gli scarti. Quando possiede queste informazioni, stampa il diagramma di Gantt perfezionato e la relativa tabella.”

“Possiamo avere una copia del programma?” chiese Grasso.

“Certo”, disse Lee. “Infatti ora analizzeremo un po' questo programma. Ognuno di voi si prepari un progetto composto da più lavori, con la rete dei predecessori e la durata di ogni lavoro. Quando avete scritto tutto su un foglio, inseriamo i dati nel programma e vediamo quali sono i risultati.

“Tra l'altro, il diagramma di Gantt perfezionato risolve il problema dell'ospite-cuoco: per non ritardare la cena bisogna che l'ospite non agisca sul percorso critico. L'ospite dunque condirà l'insalata.”

IL PROGRAMMA CPM-I

```

1000 REM ***CPM-I***
1005 REM METODO DEL PERCORSO CRITICO (I)
1006 REM
1007 REM QUESTO PROGRAMMA PUO' ESSERE USATO PER TROVARE IL PERCORSO CRITICO
1008 REM E IL TEMPO DI PARTENZA MINIMO POSSIBILE PER UNA RETE DI OPERAZIONI.
1009 REM
1010 REM DATA UNA LISTA DI OPERAZIONI, LA LORO DURATA E I LORO PREDECESSORI
1015 REM STAMPA UN DIAGRAMMA DI GANTT CHE MOSTRA OGNI OPERAZIONE "XXX", OGNI
1020 REM PERIODO DI INATTIVITA' "... " E QUALE OPERAZIONE DEL PROGETTO SI
1025 REM TROVA NEL PERCORSO CRITICO "CCC".
1030 REM
1035 REM VARIABILI:
1040 REM C.....TEMPO COMPLESSIVO DEL PROGETTO
1045 REM E.....TEMPO FINALE MASSIMO
1050 REM E().....TEMPO FINALE PER LE OPERAZIONI
1055 REM E#.....FLAG DI ERRORE
1060 REM I,J,K.....VARIABILI INDICE
1065 REM J#().....NOME DELL'OPERAZIONE
1070 REM L.....TEMPO MINIMO DI INIZIO
1075 REM L().....DURATA DI UN'OPERAZIONE
1080 REM N.....NUMERO DI OPERAZIONI
1085 REM P#.....NOME DEL PREDECESSORE
1090 REM P().....TABELLA DEI PREDECESSORI
1095 REM S().....TEMPO DI INIZIO
1100 REM
1105 REM DIMENSIONI:
1110 DIM E(12),J#(12),L(12),P(12,12),S(12,2)
1115 REM
1120 REM
1125 REM PROGRAMMA PRINCIPALE
1130 REM
1135 GOSUB 2000
1140 IF E#(<)"OK" THEN 1142 ELSE 1144
1142 GOTO 1180
1144 REM
1145 GOSUB 3000
1150 C=S(N,1)+L(N)
1155 GOSUB 4000
1160 GOSUB 7000
1165 GOSUB 8000
1170 REM
1175 REM
1180 STOP
1185 REM
1190 REM
2000 REM SUBROUTINE CHE LEGGE I DATI
2005 REM INPUT: --
2010 REM OUTPUT: J#(),L(),N,P()
2015 REM
2020 REM LEGGE IL NUMERO DI OPERAZIONI N, LE DURATE DELLE OPERAZIONI L(J) E
2025 REM I NOMI DELLE OPERAZIONI CON I NOMI DELLE OPERAZIONI IMMEDIATAMENTE
2030 REM PRECEDENTI. CONTRASSEGNA LA TABELLA DEI PREDECESSORI CON UN 1 PER
2035 REM OGNI COLONNA K RELATIVA AD UN'OPERAZIONE CHE PRECEDA UNA DATA RIGA
2040 REM I RELATIVA AD UN'OPERAZIONE.
2045 REM
2050 REM IL LETTORE FA DIVERSI TENTATIVI E RIEMPIE LA TABELLA P() VARIE VOLTE
2055 REM PER VEDERNE I RISULTATI.
2060 REM
2065 E#="OK"
2070 READ N
2075 DATA 7
2080 REM

```

```

2085     FOR J=1 TO N
2090     READ L(J)
2095     NEXT J
2096 REM
2097 REM TEMPO (INTERVALLI DI 5 MINUTI)
2100 DATA 1,12,6,7,8,2,0
2105 REM
2110     FOR J=1 TO N
2115     READ J$(J),P$
2120     IF P$="BASTA" THEN 2160
2125     GOSUB 2280
2130     IF E$<>"OK" THEN 2135 ELSE 2140
2135     GOTO 2265
2140 REM
2145     P(J,K)=1
2150     READ P$
2155     GOTO 2120
2160 REM
2165     NEXT J
2170 REM
2175     IF J$(N)<>"FINE" THEN 2180 ELSE 2190
2180     PRINT"OPERAZIONE NON FINITA"
2185     E$="NON OK"
2190 REM
2195 REM
2200 DATA INIZIO,BASTA
2205 DATA SPESA,INIZIO,BASTA
2210 DATA PREPARAZ.,SPESA,BASTA
2215 DATA CONDIRE,SPESA,BASTA
2220 DATA CUOCERE,PREPARAZ.,BASTA
2225 DATA SERVIRE,CUOCERE,CONDIRE,BASTA
2230 DATA FINE,SERVIRE,BASTA
2255 REM
2260 REM
2265 RETURN
2270 REM
2275 REM
2280 REM SUBROUTINE CHE VERIFICA IL NOME
2285 REM INPUT: J,J$(J),P$
2290 REM OUTPUT: E$,K
2295 REM
2300 E$="OK"
2305 FOR K=1 TO J-1
2310     IF P$=J$(K) THEN 2330
2315 NEXT K
2320 PRINT P$:" NON PUO' PRECEDERE ";J$(K)
2325 E$="NON OK"
2330 REM
2335 RETURN
2340 REM
2345 REM
3000 REM SUBROUTINE CHE CALCOLA I TEMPI DI INIZIO MINIMI
3005 REM INPUT: L(),N,P()
3010 REM OUTPUT: S(J,1)
3015 REM
3020 REM IL TEMPO DI INIZIO MINIMO S(J,1) DELL'OPERAZIONE SI SITUA SUBITO
3025 REM DOPO IL TEMPO MASSIMO IN CUI E' STATO COMPLETATO IL SUO
3030 REM PREDECESSORE.
3035 REM
3040 REM L(J).....DURATA DELL'OPERAZIONE J
3045 REM S(J,1).....TEMPO DI INIZIO MINIMO
3050 REM E(J).....TEMPO FINALE
3055 REM E.....TEMPO FINALE MASSIMO
3060 REM
3065     FOR J=1 TO N
3070         E=0

```

```

3075     FOR K=1 TO N
3080     IF P(J,K)=1 AND E<E(K) THEN 3090 ELSE 3095
3090     E=E(K)
3095 REM
3100     NEXT K
3105     S(J,1)=E
3110     E(J)=E+L(J)
3115     NEXT J
3120 REM
3125 RETURN
3130 REM
3135 REM
4000 REM SUBROUTINE CHE CALCOLA I TEMPI DI INIZIO MASSIMI
4005 REM INPUT: L(),N,P(),C
4010 REM OUTPUT: S(J,2)
4015 REM
4020 REM
4025 REM IL TEMPO MASSIMO IN CUI UN'OPERAZIONE PUO' ESSERE COMPLETATA
4030 REM SI TROVA ESATTAMENTE PRIMA DELL'INIZIO DEL SUO SUCCESSORE.
4035 REM
4040 REM S(J,2).....TEMPO DI INIZIO MASSIMO DELL'OPERAZIONE J
4045 REM L.....TEMPO DI INIZIO MINIMO
4050 REM
4055     FOR K=N TO 1 STEP -1
4060     L=C
4065     FOR J=1 TO N
4070     IF P(J,K)<>0 AND L>S(J,2) THEN 4080 ELSE 4085
4080     L=S(J,2)
4085 REM
4090     NEXT J
4095     S(K,2)=L-L(K)
4100     NEXT K
4105 REM
4110 RETURN
4115 REM
4120 REM
7000 REM SUBROUTINE CHE STAMPA IL DIAGRAMMA DI GANTT
7005 REM INPUT: J*( ),L(),N,S()
7010 REM OUTPUT: --
7015 REM
7020 REM STAMPA UN DIAGRAMMA DI GANTT IN CUI LE "X" RAPPRESENTANO LE OPERAZIONI
7025 REM REGOLARI, LE "C" I PERCORSI CRITICI E I "." I PERIODI DI INATTIVITA'.
7030 REM
7035     PRINT TAB(35);"TEMPO (MINUTI)"
7040     PRINT
7045     PRINT"OPERAZIONE";TAB(11);"0"           50           100           150           200"
7050     PRINT TAB(11);"+-----+-----+-----+-----"
7055     FOR J=2 TO N-1
7060     PRINTJ*(J);TAB(11);"I";
7065     K=S(J,1)
7070     IF S(J,1)-S(J,2)<>0 THEN 7075 ELSE 7130
7075     IF K>S(J,1)+L(J)-1 THEN 7095
7080     PRINT TAB(11+K);"X";
7085     K=K+1
7090     GOTO 7075
7095 REM
7100     IF K>S(J,2)+L(J)-1 THEN 7120
7105     PRINT TAB(11+K);". ";
7110     K=K+1
7115     GOTO 7100
7120 REM
7125     GOTO 7160
7130 REM
7135     IF K>S(J,1)+L(J)-1 THEN 7155
7140     PRINT TAB(11+K);"C";
7145     K=K+1

```

```

7150          GOTO 7135
7155 REM
7160 REM
7165          PRINT
7170          NEXT J
7175          PRINT
7180          PRINT
7185 RETURN
7190 REM
8000 REM SUBROUTINE CHE STAMPA UNA TABELLA DI VALORI
8005 REM INPUT: C, J*( ), L( ), N, S( )
8010 REM OUTPUT: --
8015 REM
8020 REM
8025          PRINT"TEMPO COMPLESSIVO=";C*S
8030          PRINT
8034          PRINT"                                TEMPI DI INIZIO"
8035          PRINT"OPERAZIONE    DURATA    MINIMO    MASSIMO    SCARTO"
8040          PRINT
8045          FOR J=2 TO N-1
8050              PRINT J*(J);TAB(14);L(J);TAB(22);S(J,1);
8052              PRINT TAB(30);S(J,2);TAB(39);S(J,2)-S(J,1)
8055          NEXT J
8060 REM
8065 RETURN
8070 REM
8075 REM
8080          END

```


Il grosso del programma è costituito dalle subroutine richiamate dal programma principale. Ogni subroutine effettua un'operazione più o meno come si è detto nel testo. Per analizzare i dettagli di ogni pezzo di programma, bisognerebbe esaminarlo riga per riga. Per prevedere il funzionamento delle varie subroutine, basta prendere un pezzo di carta e seguirlo il programma passo per passo.

ESERCIZI

1. Eseguite CPM-I sul vostro computer e verificate che i risultati siano uguali a quelli del testo.
2. Tracciate il grafo di un progetto diverso e passate la tabella dei predecessori e le durate delle operazioni a CPM-I. Provate ad eseguirlo.

PIANIFICAZIONE CON SCARSITÀ DI RISORSE

A mezzogiorno tutti avevano costruito con il computer il diagramma di Gantt per un progetto di produzione della Deer Division. I tre manager di produzione erano entusiasti dei loro elaborati. Dopo pranzo si rincontrarono nell'aula delle conferenze. "Questo pomeriggio", esordì George Lee, "aiuteremo due giganti di nome Tread e Truss a costruire un ponte. Il loro problema è simile a quello che abbiamo analizzato stamattina, con una particolarità: hanno scarsità di risorse. Il diagramma di Gantt perfezionato per il loro progetto è questo:



“Per ora il progetto sembra realizzabile. Ma c'è un problema. Ci vogliono due giganti per arginare il fiume. Uno solo per costruire un pilastro. Due per disporre le travi di legno sui pilastri e due per fissare le tavole sulle travi. Un gigante per togliere l'argine. In questo caso la scarsità di risorse è scarsità di personale. Tutto è riassunto in questa tabella:

OPERAZIONE	DURATA	RISORSE	INIZIO	SCARTO
PIL. DES.	3	1	9	4
PIL. SIN.	3	1	12	1
ARGINE	8	2	1	0
PIL. CENTR.	7	1	9	0
TRAVI	2	2	16	0
TAVOLE	5	2	18	0
VIA L' ARG.	1	1	23	0

“Il relativo istogramma del personale utilizzato è così:



“Questo piano richiede quattro giganti per i primi tre giorni. Ma i nostri giganti sono solo due. Ecco la scarsità di risorse. Tread e Truss hanno esaminato la situazione ed hanno spostato i lavori di erezione dei pilastri destro e sinistro ai giorni 9 e 12.

“Eseguito CPM-II per la seconda volta videro di aver risolto il problema:



TEMPO COMPLESSIVO= 24

OPERAZIONE	DURATA	RISORSE	INIZIO	SCARTO
PIL. DES.	3	1	9	4
PIL. SIN.	3	1	12	1
ARGINE	8	2	1	0
PIL. CENTR.	7	1	9	0
TRAVI	2	2	16	0
TAVOLE	5	2	18	0
VIA L'ARG.	1	1	23	0

“Questa volta la richiesta di personale rispettava le loro disponibilità. In nessun momento c’era bisogno di più di due giganti. Data la loro disponibilità di personale, quel piano era una soluzione attuabile. Si misero al lavoro.”

“Cosa succede,” chiese Grant, “se parte di un lavoro richiede due giganti e l’altra parte ne richiede uno solo?”

“Ci sono due modi per risolvere questo caso,” disse Lee. “Innanzitutto, si potrebbe dire che quella operazione richiede 1.5 unità di personale. Se questo modo di trattare le cose non vi piace, potete dividere l’operazione in due, una che richiede due giganti e una che ne richiede uno.”

“È giusto dire,” chiese Grasso, “che pensi di eseguire il programma più volte prima di ottenere il risultato giusto?”

“Sì”, ammise Lee.

“Allora perché non fai in modo che il programma stesso controlli le approssimazioni?” proseguì Grasso.

“Lo si può fare,” convenne Lee, “ma il programma è più flessibile se noi ne manteniamo il controllo. Si perde qualche minuto a manipolare le operazioni, ma si ottiene esattamente la soluzione desiderata. I giganti possono avere un’amica ed invitarla a collaborare perché il ponte venga costruito più alla svelta. Se lei accettasse, si potrebbe ridurre la durata di alcuni lavori ed aumentare a tre la disponibilità di personale. Per esempio, l’applicazione delle tavole sulle travi potrebbe finire prima. Il fatto

è che i giganti hanno diverse possibilità di scelta grazie al programma che li tiene sotto controllo."

"Mi sembra che tu voglia dire che è meglio partire da lontano," disse Grasso. "Voglio fare una prova. Forse non è una cattiva idea. So che i miei progetti devono sempre essere attentamente valutati. Voglio vedere quanti tentativi mi ci vogliono prima di ottenere la soluzione giusta."

"È quello che faremo adesso", disse Lee. "Dobbiamo rivedere tutti i progetti che abbiamo esaminato stamattina perché tengano conto di un qualche tipo di risorsa. Può trattarsi di dollari per settimana, giournuomo, ore-uomo, o qualsiasi altra cosa. Bisogna assegnare delle risorse ad ogni lavoro, inserire la durata ed il tempo d'inizio, ed esaminare i risultati. Se si superano alcuni limiti ragionevoli per le risorse, bisogna ripianificare alcune operazioni ed eseguire CPM-II di nuovo finché tutto va bene."

IL PROGRAMMA CPM-II

```

1000 REM ***CPM-II***
1005 REM METODO DEL PERCORSO CRITICO (II)
1006 REM
1007 REM QUESTO PROGRAMMA PUO' ESSERE USATO PER EQUILIBRARE L'USO DELLE RISORSE
1008 REM DI UNA RETE DI PERCORSI CRITICI MODIFICANDO I TEMPI DI INIZIO.
1009 REM
1010 REM DATA UNA LISTA DI OPERAZIONI, LA LORO DURATA, I LORO TEMPI DI INIZIO,
1015 REM LE RISORSE RICHIESTE E I LORO PREDECESSORI STAMPA UN DIAGRAMMA DI GANTT
*
1020 REM UN ISTOGRAMMA DELLE RISORSE E UNA TABELLA DI VALORI PER LA RETE.
1030 REM
1035 REM VARIABILI:
1036 REM A.....VALORE DI OGNI ASTERISCO DELL'ISTOGRAMMA
1040 REM C.....TEMPO COMPLESSIVO DEL PROGETTO
1045 REM E.....TEMPO FINALE MASSIMO
1050 REM E().....TEMPO FINALE PER LE OPERAZIONI
1055 REM E*.....FLAG DI ERRORE
1056 REM G*().....GRAFICO PER L'ISTOGRAMMA
1057 REM H().....PASSI DELL'ISTOGRAMMA
1060 REM I,J,K.....VARIABILI INDICE
1065 REM J$().....NOME DELL'OPERAZIONE
1070 REM L.....TEMPO MINIMO DI INIZIO
1075 REM L().....DURATA DI UN'OPERAZIONE
1076 REM M.....PASSO MASSIMO DELL'ISTOGRAMMA
1080 REM N.....NUMERO DI OPERAZIONI
1085 REM P$().....NOME DEL PREDECESSORE
1090 REM P().....TABELLA DEI PREDECESSORI
1091 REM R().....RISORSE RICHIESTE
1095 REM S().....TEMPO DI INIZIO
1096 REM X.....VARIABILE DI LAVORO
1097 REM
1098 REM COSTANTI:
1099 H9=4
1100 S9=11
1101 T9=40
1102 T$="GIORNI"
1103 REM

```

```

1105 REM DIMENSIONI:
1110 DIM E(25),H(40),G*(10,40),J*(25)
1111 DIM L(25),P(25,25),R(25),S(25,2)
1115 REM
1120 REM
1125 REM PROGRAMMA PRINCIPALE
1130 REM
1135 GOSUB 2000
1140 IF E*((">"OK" THEN 1142 ELSE 1144
1142 GOTO 1180
1144 REM
1145 GOSUB 3000
1146 IF E*((">"OK" THEN 1147 ELSE 1148
1147 GOTO 1180
1148 REM
1150 C=S(N,1)+L(N)
1155 GOSUB 4000
1160 GOSUB 7000
1162 GOSUB 7200
1165 GOSUB 8000
1170 REM
1175 REM
1180 STOP
1185 REM
1190 REM
2000 REM SUBROUTINE CHE LEGGE I DATI
2005 REM INPUT: --
2010 REM OUTPUT: E*,J*( ),L( ),N,R( ),P( ),S( )
2015 REM
2020 REM LEGGE IL NUMERO DI OPERAZIONI N, LE DURATE DELLE OPERAZIONI L(J) E
2025 REM I NOMI DELLE OPERAZIONI CON I NOMI DELLE OPERAZIONI IMMEDIATAMENTE
2030 REM PRECEDENTI. CONTRASSEGNA LA TABELLA DEI PREDECESSORI CON UN 1 PER
2035 REM OGNI COLONNA K RELATIVA AD UN'OPERAZIONE CHE PRECEDA UNA DATA RIGA
2040 REM I RELATIVA AD UN'OPERAZIONE.
2045 REM
2050 REM IL LETTORE FA DIVERSI TENTATIVI E RIEMPIE LA TABELLA P() VARIE VOLTE
2055 REM PER VEDERNE I RISULTATI.
2060 REM
2065 E*="OK"
2070 READ N
2075 DATA 9
2080 REM
2085 FOR J=1 TO N
2087 READ X*,S(J,1),L(J),R(J)
2088 NEXT J
2090 DATA INIZIO,0,1,0
2091 DATA PIL. DES.,9,3,1
2092 DATA PIL. SIN.,12,3,1
2093 DATA ARGINE,1,8,2
2094 DATA PIL. CENTR.,9,7,1
2095 DATA TRAVI,16,2,2
2096 DATA TAVOLE,18,5,2
2097 DATA VIA L'ARG.,23,1,1
2098 DATA FINE,24,0,0
2101 REM
2102 READ X
2103 DATA -.9999
2104 IF X(<-.9999 THEN 2105 ELSE 2109
2105 PRINT"NUMERO ERRATO DI DATI DI PARTENZA."
2106 PRINT"CONTROLLATE DI AVERE INCLUSO I DATI DI PARTENZA PER LE"
2107 PRINT"OPERAZIONI 'INIZIO' E 'FINE'."
2108 E*="NO OK"
2109 REM
2110 FOR J=1 TO N
2115 READ J*(J),P*
2120 IF P*="BASTA" THEN 2160

```



```

2125      GOSUB 2280
2130      IF E$(C)<>"OK" THEN 2135 ELSE 2140
2135      GOTO 2265
2140 REM
2145      P(J,K)=1
2150      READ P#
2155      GOTO 2120
2160 REM
2165      NEXT J
2170 REM
2175      IF J$(N)<>"FINE" THEN 2180 ELSE 2190
2180      PRINT"OPERAZIONE NON FINITA"
2185      E$="NON OK"
2190 REM
2195 REM
2200      DATA INIZIO,BASTA
2205      DATA PIL. DES.,INIZIO,BASTA
2210      DATA PIL. SIN.,INIZIO,BASTA
2215      DATA ARGINE,INIZIO,BASTA
2220      DATA PIL. CENTR.,ARGINE,BASTA
2225      DATA TRAVI,PIL. SIN.,PIL. DES.,PIL. CENTR.,BASTA
2230      DATA TAVOLE,TRAVI,BASTA
2235      DATA VIA L'ARG.,TAVOLE,BASTA
2240      DATA FINE,TAVOLE,BASTA
2255 REM
2260 REM
2265 RETURN
2270 REM
2275 REM
2280 REM SUBROUTINE CHE VERIFICA IL NOME E TROVA IL PREDECESSORE
2285 REM INPUT: E$,J,J$( ),P#
2290 REM OUTPUT: E$,K
2295 REM
2305      FOR K=1 TO J-1
2310          IF J$(K)=P# THEN 2330
2315      NEXT K
2320      PRINT P#;" NON PUO' PRECEDERE ";J$(K)
2325      E$="NON OK"
2330 REM
2335 RETURN
2340 REM
2345 REM
3000 REM SUBROUTINE CHE CALCOLA I TEMPI DI INIZIO MINIMI
3005 REM INPUT: E$,L( ),N,P( ),S(J,1)
3010 REM OUTPUT: E$
3015 REM
3020 REM IL TEMPO DI INIZIO D(J,1) DELL'OPERAZIONE SI TROVA
3025 REM DOPO IL TEMPO MASSIMO IN CUI E' STATO COMPLETATO IL SUO
3030 REM PREDECESSORE.
3035 REM
3040 REM L(J).....DURATA DELL'OPERAZIONE J
3045 REM S(J,1).....TEMPO DI INIZIO MINIMO
3050 REM E(J).....TEMPO FINALE
3055 REM E.....TEMPO FINALE MASSIMO
3060 REM
3065      FOR J=1 TO N
3070          E=0
3075          FOR K=1 TO N
3080              IF P(J,K)=1 AND E<E(K) THEN 3090 ELSE 3095
3090              E=E(K)
3095 REM
3100          NEXT K
3102          IF S(J,1)<E THEN 3103 ELSE 3109
3103          PRINT"IL TEMPO DI PARTENZA";S(J,1);" DELL'OPERAZIONE #";
3104          PRINT J;" DEVE ESSERE MAGGIORE DI ";E;"."
3105          PRINT"RIBATTERE I TEMPI DI INIZIO."

```

```

3106 PRINT
3107 E$="NON OK"
3108 GOTO 3125
3109 REM
3110 E(J)=S(J,1)+L(J)
3115 NEXT J
3120 REM
3125 RETURN
3130 REM
3135 REM
4000 REM SUBROUTINE CHE CALCOLA I TEMPI DI INIZIO MASSIMI
4005 REM INPUT: L(),N,P(),C
4010 REM OUTPUT: S(J,2)
4020 REM
4025 REM IL TEMPO MASSIMO IN CUI UN'OPERAZIONE PUO' ESSERE COMPLETATA
4030 REM SI TROVA ESATTAMENTE PRIMA DELL'INIZIO DEL SUO SUCCESSORE.
4035 REM
4040 REM S(J,2).....TEMPO DI INIZIO MASSIMO DELL'OPERAZIONE J
4045 REM L.....TEMPO DI INIZIO MINIMO
4050 REM
4055 FOR K=N TO 1 STEP -1
4060 L=C
4065 FOR J=1 TO N
4070 IF P(J,K)=1 AND L>S(J,2) THEN 4080 ELSE 4085
4080 L=S(J,2)
4085 REM
4090 NEXT J
4095 S(K,2)=L-L(K)
4100 NEXT K
4105 REM
4110 RETURN
4115 REM
4120 REM
7000 REM SUBROUTINE CHE STAMPA IL DIAGRAMMA DI GANTT
7005 REM INPUT: J$( ),L(),N,S()
7010 REM OUTPUT: --
7015 REM
7020 REM STAMPA UN DIAGRAMMA DI GANTT IN CUI LE "X" RAPPRESENTANO LE OPERAZIONI
7025 REM REGOLARI, LE "C" I PERCORSI CRITICI E I "." I PERIODI DI INATTIVITA'.
7026 REM PER OGNI PERIODO K ACCUMULA LE RISORSE DELLE OPERAZIONI R(J) NEL
7027 REM PASSO DI ISTOGRAMMA H(K).
7029 REM
7030 FOR K=1 TO T9
7031 H(K)=0
7032 NEXT K
7033 REM
7035 PRINT TAB(35);"TEMPO-";T$
7045 PRINT"OPERAZIONE";TAB(S9);"0" 10 20 30 40"
7050 PRINT TAB(11);"----->----->----->----->"
7052 REM
7055 FOR J=2 TO N-1
7060 PRINTJ$(J);TAB(S9);"I";
7065 K=S(J,1)
7070 IF S(J,1)-S(J,2)<>0 THEN 7075 ELSE 7130
7075 IF K>S(J,1)+L(J)-1 THEN 7095
7080 PRINT TAB(S9+K);"X";
7083 H(K)=H(K)+R(J)
7085 K=K+1
7090 GOTO 7075
7095 REM
7100 IF K>S(J,2)+L(J)-1 THEN 7120
7105 PRINT TAB(S9+K);".";
7110 K=K+1
7115 GOTO 7100
7120 REM

```



```

8000 REM SUBROUTINE CHE STAMPA UNA TABELLA DI VALORI
8005 REM INPUT: C, J*( ), L( ), N, S( )
8010 REM OUTPUT: --
8015 REM
8020 REM
8025 PRINT "TEMPO COMPLESSIVO=" : C
8030 PRINT
8035 PRINT "OPERAZIONE DURATA RISORSE INIZIO SCARTO"
8040 PRINT
8045 FOR J=2 TO N-1
8050 PRINT J*(J);TAB(14);L(J);TAB(23);R(J);
8052 PRINT TAB(33);S(J,1);TAB(43);S(J,2)-S(J,1)
8055 NEXT J
8060 REM
8065 RETURN
8070 REM
8075 REM
8080 END

```

Il secondo programma di calcolo del percorso critico funziona più o meno come il primo per quanto riguarda le varie subroutine. Qualcosa di diverso, tuttavia, c'è. La differenza più ovvia è che ad ogni lavoro è associata una quantità di risorse. Ciò comporta una certa modifica nella logica di programmazione. Il programma produce, come prima, un diagramma di Gantt. Ma ora, una volta tracciato il diagramma il programma effettua delle somme per vedere qual è la quantità totale di risorse utilizzata in un dato momento. I risultati di questo calcolo sono riassunti nell'istogramma delle risorse utilizzate, chiamato talvolta diagramma di densità delle risorse, che appare subito sotto il diagramma di Gantt. La seconda differenza tra questo programma ed il precedente è che i tempi d'inizio sono inseriti dall'utente. I tempi d'inizio devono essere sotto controllo dell'utente perché questi sia in grado di organizzare il progetto in modo che l'istogramma delle risorse rispetti i limiti. La parte di programma CPM-I che calcolava i tempi d'inizio è stata sostituita con una fase di controllo dei tempi d'inizio in CPM-II. Questi controlli sono necessari perché l'utente può sbagliarsi e far iniziare un lavoro prima che i suoi predecessori siano finiti.

ESERCIZI

1. Modificate il progetto che avete costruito per CPM-I affinché tenga conto, per ogni lavoro, sia di una quantità di risorse che di un tempo d'inizio. Provatelo con CPM-II.
2. Talvolta è interessante vedere come si accumulano le risorse al procedere del progetto. Scrivete un'altra routine che stampi la quantità cumulativa delle risorse utilizzate. Costruite un grafico da 0 a 100 per cento delle risorse sull'asse delle ascisse e da 0 a 100 per cento delle risorse sull'asse delle ordinate.

(SUGG.: Guardate il programma PLOT nell'Appendice B per farvi delle idee sulla costruzione di grafici in Basic.)

IL CONTROLLO DEL PROGETTO

Jean Grant, Robert Grasso e Martin Graves completarono il perfezionamento dei loro piani e guardarono ammirati i rispettivi risultati quando si trovarono per l'ultima lezione del pomeriggio. Lee notò l'entusiasmo con cui si apprestavano a lavorare col computer.

"Avevi ragione, George", disse Grasso. "Ho dovuto adattare la durata di un lavoro e modificare i tempi d'inizio e le risorse per fare in modo che tutto vada bene. È stato un cambiamento un po' complesso. Mi sono divertito a cercare di controllare come si configurava il piano finale. "Sono contento che ti sia divertito," disse Lee, "perché ora faremo l'ultimo passo di pianificazione di questo progetto. Stiamo per completare il piano.


"I nostri due giganti si misero al lavoro e pochi giorni più tardi si fermarono per fare delle scorte. Avevano cominciato a disporre le travi. Gran parte del lavoro precedente era andato bene, ma le travi non erano facili da trovare e i due giganti avevano ritardato l'inizio dell'opera di montaggio di due giorni. Quando le operazioni ripresero, avevano segnato l'inizio e la fine effettivi di ogni operazione:

```

3240 REM SUBROUTINE CHE LEGGE I TEMPI REALI
3250 REM INPUT: N
3260 REM OUTPUT: A(),F(),T1
3270 REM
3290 REM LEGGE IL TEMPO DI PARTENZA REALE A(J,1) E IL TEMPO FINALE REALE F(J)
3300 REM
3310 REM
3320 REM LEGGE IL TEMPO ATTUALE T1
3330 READ T1
3340 DATA 19
3350 REM
3360 FOR J=1 TO N
3370 READ X#,A(J,1),F(1)
3380 NEXT J
3390 DATA INIZIO,0,0
3400 DATA PIL. DES.,9,11
3410 DATA PIL. SIN.,12,14
3420 DATA ARGINE,1,8
3430 DATA PIL. CENTR.,9,14
3440 DATA TRAVI,18,0
3450 DATA TAVOLE,0,0
3460 DATA VIA L'ARG.,0,0
3465 DATA FINE,0,0
3470 REM

```

"Un diagramma di confronto tra il loro piano finale ed il lavoro effettivo appariva così:



RISORSE UTILIZZATE



OGNI ASTERISCO RAPPRESENTA 1 UNITA' DI RISORSE.

OPERAZIONE	VALORI REALI (O PREVISTI)			
	DURATA	RISORSE	INIZIO	SCARTO
PIL. DES.	3	1	9	4
PIL. SIN.	3	1	12	1
ARGINE	8	2	1	0
PIL. CENTR.	7	1	9	0
TRAVI	2	2	18	-2
TAVOLE	5	2	20	-2
VIA L'ARG.	1	1	25	-2

IL TEMPO ATTUALE E' 19

LA STIMA DEL TEMPO PER COMPLETARE IL PROGETTO E' 24 GIORNI

"Truss e Tread guardarono come erano andate le cose e scoprirono con piacere di aver rispettato il piano fino al punto di disporre le travi. Il ritardo nell'approvvigionamento delle travi fece sì che la nuova data di completamento del progetto andasse oltre quella prevista, ma videro esattamente quale era la nuova data, e quindi non si preoccuparono molto per il ritardo."

"Sembra che quell'output richieda molti calcoli al programma", intervenne Graves. "È così difficile come sembra?"

"Non direi", rispose Lee. "La parte di pianificazione è effettuata come in CPM-II, a parte i tempi d'inizio. L'inizio e la fine effettiva rendono fattibile la storia del lavoro effettivo raccolta nella seconda riga di ogni

operazione. Finora questa previsione è stata realizzata senza tener conto del lavoro effettivo."

"Come si fa a confrontare il lavoro effettivo con un piano perfezionato?" chiese Grant.

"Bisogna ripristinare i dati e rieseguire il programma. Assicurarsi però di chiamare chiaramente l'output 'piano perfezionato'", disse Lee.

"Così abbiamo un piano alla volta?" si accertò Grant.

"Sì, un piano e una storia del lavoro effettivo", disse Lee. "Se si fanno le cose sofisticate, l'esperienza mostra che tendono a risultare molto confuse."

"Okay," disse Graves, "CPM-II mi ha convinto di questo. Ora possiamo provare quest'altro?"

"Certo", disse Lee. "CPM-III è l'ultimo, per oggi. Dovreste immaginare di essere a metà del vostro progetto ed aver completato alcune operazioni. Se non ci sono altre domande, torniamo al lavoro. Stiamo controllando il progetto."

 IL PROGRAMMA CPM-III

```

1000 REM ***CPM-III***
1010 REM
1020 REM QUESTO PROGRAMMA PUO' ESSERE USATO PER VEDERE COME LAVORA UN
1030 REM PROGETTO REALE, PIANIFICATO MEDIANTE CPM-I E CPM-II.
1040 REM IL PROGRAMMA LEGGE IL TEMPO ATTUALE T1, I TEMPI DI INIZIO REALI A(J,I)
1050 REM E I TEMPI FINALI REALI F(J) DELLE OPERAZIONI.
1060 REM
1062 REM IL PROGRAMMA STAMPA UN DIAGRAMMA DI GANTT CHE CONFRONTA L'OPERAZIONE
1064 REM PIANIFICATA CON QUELLA REALE (O PREVISTA).
1065 REM
1070 REM L'ISTOGRAMMA DELLE RISORSE MOSTRA LE RISORSE REALMENTE USATE
1072 REM (O PREVISTE).
1075 REM
1076 REM VARIABILI:
1080 REM   A().....TEMPI DI PARTENZA REALI (O PREVISTI)
1090 REM   A.....VALORE DI OGNI ASTERISCO NELL'ISTOGRAMMA
1100 REM   C.....TEMPO COMPLESSIVO DEL PROGETTO
1110 REM   E.....TEMPO FINALE MASSIMO
1120 REM   E().....TEMPO FINALE PER LE OPERAZIONI
1130 REM   E*.....FLAG DI ERRORE
1135 REM   F().....TEMPO FINALE REALE
1140 REM   G*().....GRAFICO PER L'ISTOGRAMMA
1150 REM   H().....PASSI DELL'ISTOGRAMMA
1160 REM   I,J,K.....VARIABILI INDICE
1170 REM   J*().....NOME DELL'OPERAZIONE
1180 REM   L.....TEMPO MINIMO DI INIZIO
1190 REM   L().....DURATA DI UN'OPERAZIONE
1200 REM   M.....PASSO MASSIMO DELL'ISTOGRAMMA
1210 REM   N.....NUMERO DI OPERAZIONI
1220 REM   P*.....NOME DEL PREDECESSORE
1230 REM   P().....TABELLA DEI PREDECESSORI
1240 REM   R().....RISORSE RICHIESTE
1250 REM   S().....TEMPO DI INIZIO
1260 REM   T1.....TEMPO REALE
1270 REM   X.....VARIABILE DI LAVORO
  
```

```

1280 REM
1370 REM COSTANTI:
1380 H9=4
1390 N=11
1400 S9=11
1410 T9=40
1420 T9="GIORNI"
1430 REM
1440 REM DIMENSIONI:
1450 DIM A(12,2),E(12),F(12),G*(4,40),H(40)
1460 DIM J*(12),L(12),P(12,12),R(12),S(12,2)
1470 REM
1490 REM PROGRAMMA PRINCIPALE
1500 REM
1510 GOSUB 1680
1520 IF E*((">"OK" THEN 1522 ELSE 1524
1522 GOTO 1640
1524 REM
1530 GOSUB 2610
1540 IF E*((">"OK" THEN 1542 ELSE 1544
1542 GOTO 1680
1544 REM
1550 C=S(N,1)+L(N)-1
1560 GOSUB 2990
1570 GOSUB 3240
1580 GOSUB 3510
1590 GOSUB 4170
1600 GOSUB 4420
1610 GOSUB 4690
1620 GOSUB 5220
1640 REM
1650 STOP
1660 REM
1670 REM
1680 REM SUBROUTINE CHE LEGGE I DATI
1690 REM INPUT: --
1700 REM OUTPUT: F*,J*( ),L( ),N,P( ),R( ),S( )
1710 REM
1720 REM PER I COMMENTI VEDERE IL PROGRAMMA CPM-II
1780 REM
1800 REM PER I PRIMI CASI, PER VEDERE COME LAVORA
1810 REM
1820 E*="OK"
1830 READ N
1840 DATA 9
1850 REM
1860 FOR J=1 TO N
1870 READ X*,S(J,1),L(J),R(J)
1880 NEXT J
1890 DATA INIZIO,0,1,0
1900 DATA PIL. DES.,9,3,1
1910 DATA PIL. SIN.,12,3,1
1920 DATA ARGINE,1,8,2
1930 DATA PIL. CENTR.,9,7,1
1940 DATA TRAVI,16,2,2
1950 DATA TAVOLE,18,5,2
1960 DATA VIA L'ARG.,23,1,1
1970 DATA FINE,24,0,0
2000 REM
2010 READ X
2020 DATA -.9999
2030 IF X(<)-.9999 THEN 2040 ELSE 2080
2040 PRINT"NUMERO ERRATO DI DATI."
2050 PRINT"CONTROLLATE DI AVER INCLUSO I DATI DI"
2060 PRINT"" INIZIO" E "FINE".
2065 E*="NON OK"

```



```

2070      GOTO 2410
2080 REM
2090      FOR J=1 TO N
2100          READ J*(J),P#
2110          IF P#="BASTA" THEN 2190
2120          GOSUB 2440
2130          IF E#<>"OK" THEN 2140 ELSE 2150
2140          GOTO 2410
2150 REM
2160          P(J,K)=1
2170          READ P#
2180          GOTO 2110
2190 REM
2200      NEXT J
2210 REM
2220      IF J*(N)<>"FINE" THEN 2230 ELSE 2260
2230      PRINT"OPERAZIONE NON FINITA"
2240      E#="NON OK"
2250      GOTO 2410
2260 REM
2270 REM
2280      DATA INIZIO,BASTA
2290      DATA PIL. DES.,INIZIO,BASTA
2300      DATA PIL. SIN.,INIZIO,BASTA
2310      DATA ARGINE,INIZIO,BASTA
2320      DATA PIL. CENTR.,ARGINE,BASTA
2340      DATA TRAVI,PIL. SIN.,PIL. DES.,PIL. CENTR.,BASTA
2350      DATA TAVOLE,TRAVI,BASTA
2360      DATA VIA L'ARG.,TAVOLE,BASTA
2370      DATA FINE,TAVOLE,BASTA
2390 REM
2410 RETURN
2420 REM
2430 REM
2440 REM SUBROUTINE CHE TROVA IL PREDECESSORE E VERIFICA IL NOME
2450 REM   INPUT: E#,J,J*(J),P#
2460 REM   OUTPUT: E#,K
2470 REM
2480      FOR K=1 TO J-1
2490          IF J*(K)=P# THEN 2500 ELSE 2510
2500          GOTO 2550
2510 REM
2520      NEXT K
2530      PRINT P#;" NON PRECEDE ";J*(K)
2540      E#="NON OK"
2550 REM
2560 RETURN
2570 REM
2580 REM
2610 REM SUBROUTINE CHE CALCOLA I TEMPI DI INIZIO
2620 REM   INPUT: E#,L(),N,P(),S(J,1)
2630 REM   OUTPUT: E#
2640 REM
2650 REM PER I COMMENTI VEDERE IL PROGRAMMA CPM-II
2720 REM
2730      FOR J=1 TO N
2740          E(J)=0
2750      NEXT J
2760 REM
2770      FOR J=1 TO N
2780          E=-1
2790          FOR K=1 TO N
2800              IF P(J,K)<>0 AND E(K)>E THEN 2810 ELSE 2820
2810              E=E(K)
2820 REM
2830          NEXT K

```

```

2840     IF S(J,1)<=E THEN 2850 ELSE 2920
2850     PRINT"IL TEMPO DI PARTENZA";S(J,1);" DELL'OPERAZIONE #":
2870     PRINT J;" DEVE ESSERE MAGGIORE DI "E;"."
2880     PRINT"RIBATTERE I TEMPI DI INIZIO."
2890     PRINT
2900     E$="NON OK"
2910     GOTO 2960
2920 REM
2930     E(J)=S(J,1)+L(J)-1
2940     NEXT J
2950 REM
2960 RETURN
2970 REM
2980 REM
2990 REM SUBROUTINE CHE CALCOLA I TEMPI DI INIZIO MASSIMI
3000 REM INPUT: L(),N,P(),C
3010 REM OUTPUT: S(J,2)
3020 REM
3040 REM PER I COMMENTI VEDERE IL PROGRAMMA CPM-II.
3090 REM
3100     FOR K=N TO 1 STEP -1
3110         L=C+1
3120         FOR J=1 TO N
3130             IF P(J,K)<>0 AND S(J,2)<L THEN 3140 ELSE 3150
3140             L=S(J,2)
3150 REM
3170         NEXT J
3180         S(K,2)=L-L(K)
3190     NEXT K
3200 REM
3210 RETURN
3220 REM
3230 REM
3240 REM SUBROUTINE CHE LEGGE I TEMPI REALI
3250 REM INPUT: N
3260 REM OUTPUT: A(),F(),T1
3270 REM
3290 REM LEGGE IL TEMPO DI PARTENZA REALE A(J,1) E IL TEMPO FINALE REALE F(J)
3300 REM
3310 REM
3320 REM LEGGE IL TEMPO ATTUALE T1
3330     READ T1
3340     DATA 19
3350 REM
3360     FOR J=1 TO N
3370         READ X$,A(J,1),F(1)
3380     NEXT J
3390     DATA INIZIO,0,0
3400     DATA PIL. DES.,9,11
3410     DATA PIL. SIN.,12,14
3420     DATA ARGINE,1,8
3430     DATA PIL. CENTR.,9,14
3440     DATA TRAVI,18,0
3450     DATA TAVOLE,0,0
3460     DATA VIA L'ARB.,0,0
3465     DATA FINE,0,0
3470 REM
3480 RETURN
3490 REM
3500 REM
3510 REM SUBROUTINE CHE CALCOLA I TEMPI DI PARTENZA REALI E PREVISTI
3520 REM INPUT: A(),E$,F(),L(),N,P()
3530 REM OUTPUT: A(),E$,F()
3540 REM
3550 REM L'INIZIO DI OGNI OPERAZIONE A(J,1) HA LUOGO DOPO CHE E' STATO
3560 REM COMPLETATO L'ULTIMO DEI SUOI PREDECESSORI.
3570 REM

```

```

3580 REM A(J,1).....TEMPO DI INIZIO REALE (0 PREVISTO)
3590 REM E(J).....TEMPO FINALE
3600 REM F(J).....TEMPO FINALE REALE (0 PREVISTO)
3610 REM E.....TEMPO FINALE MASSIMO
3620 REM
3630   FOR J=1 TO N
3640     E(J)=0
3650   NEXT J
3660 REM
3670   FOR J=1 TO N
3680     E=-1
3690     FOR K=1 TO N
3700       IF P(J,K)=1 AND E(K)>E THEN 3710 ELSE 3730
3710       E=E(K)
3730   REM
3740   NEXT K
3750   IF A(J,1)<>0 AND A(J,1)<=E THEN 3760 ELSE 3800
3760   PRINT"IL TEMPO DI PARTENZA REALE:A(J,1):" DELL'OPERAZIONE # ";
3770   PRINT J;"E" SCATTATO PRIMA CHE UNO DEI SUOI PREDECESSORI"
3780   PRINT"FOSSE FINITO."
3790   PRINT"VERIFICARE:"
3800 REM
3810 REM DOPO AVER TROVATO L'ULTIMO DEI PREDECESSORI, IL TEMPO E,
3820 REM E VERIFICATO L'ESISTENZA DI ERRORI, CALCOLA I TEMPI DI PARTENZA
3830 REM E DI FINE PREVISTI PER I PROSSIMI 'REALI' (ORA DI VALORE 0).
3840 REM
3850   IF A(J,1)<>0 AND F(J)=0 THEN 3860 ELSE 3900
3860   E(J)=A(J,1)+L(J)-1
3870   F(J)=A(J,1)+L(J)-1
3880   GOTO 4100
3900 REM
3910   IF A(J,1)<>0 AND F(J)<>0 THEN 3920 ELSE 3950
3920   E(J)=F(J)
3930   GOTO 4100
3950 REM
3960   IF A(J,1)=0 AND F(J)=0 THEN 3970 ELSE 4020
3970   A(J,1)=E+1
3980   E(J)=A(J,1)+L(J)-1
3990   F(J)=A(J,1)+L(J)-1
4000   GOTO 4100
4020 REM
4030   IF A(J,1)=0 AND F(J)<>0 THEN 4040 ELSE 4090
4040   PRINT"IL TEMPO FINALE "F(J):" PER L'OPERAZIONE ";
4050   PRINT J;" NON HA UN TEMPO DI INIZIO."
4060   PRINT"CORREGGERE I DATI."
4070   E#="NON OK"
4080   GOTO 4100
4090 REM
4100 REM
4110   IF E#<>"OK" THEN 4130
4120   NEXT J
4130 REM
4140 RETURN
4150 REM
4160 REM
4170 REM SUBROUTINE CHE CALCOLA I TEMPI MASSIMI DI PARTENZA REALI
4180 REM   INPUT: L(),N,F(),C
4190 REM   OUTPUT: A(J,2)
4200 REM
4220 REM L'ULTIMA OPERAZIONE PUO' ESSERE COMPLETATA APPENA PRIMA CHE
4230 REM DEBBA PARTIRE IL SUO SUCCESSORE.
4240 REM
4250 REM A(J,2).....TEMPO MASSIMO DI INIZIO DELL'OPERAZIONE J
4260 REM L.....TEMPO MINIMO DI INIZIO
4270 REM
4280   FOR K=N TO 1 STEP -1

```

```

4290      L=C+1
4300      FOR J=1 TO N
4310          IF P(J,K)=1 AND A(J,2)<L THEN 4320 ELSE 4330
4320              L=A(J,2)
4330      REM
4340          NEXT J
4350          A(K,2)=L-(F(K)-A(K,1)+1)
4360      NEXT K
4380      REM
4390      RETURN
4400      REM
4410      REM
4420      REM SUBROUTINE CHE STAMPA IL DIAGRAMMA DI GANTT
4430      REM INPUT: J(),L(),N,R(),S(),T1
4440      REM OUTPUT: H()
4450      REM
4460      REM STAMPA UN DIAGRAMMA DI GANTT IN CUI LE OPERAZIONI REGOLARI
4470      REM SONO RAPPRESENTATE CON "X", LE OPERAZIONI CRITICHE CON "C",
4480      REM E I PERIODI DI INATTIVITA' CON ".". L'OPERAZIONE REALE E'
4490      REM RAPPRESENTATA DA "*" O "C".
4491      REM
4492      REM MENTRE STAMPA IL DIAGRAMMA, PER OGNI PERIODO K ACCUMULA
4493      REM LE RISORSE ATTUALI R(J) NEI PASSI DI ISTOGRAMMA H(K).
4500      REM
4510          FOR K=1 TO T9
4520              H(K)=0
4530          NEXT K
4540      REM
4550          PRINT TAB(35):"TEMPO-";T#
4560          PRINT TAB(S9+T1):"+-TEMPO ATTUALE"
4570          PRINT"OPERAZIONE";TAB(S9):"0"          10          20          30
4580          PRINT TAB(S9):"-----+-----+-----+-----"
4590      REM
4600          FOR J=2 TO N-1
4610              GOSUB 5420
4620              GOSUB 5700
4630          NEXT J
4632          PRINT"LEGENDA:";TAB(15):"XXX.. LAVORO E INATTIVITA'"
4633          PRINT TAB(15):"CCC LAVORO CRITICO"
4635          PRINT TAB(15):"*** LAVORO REALE"
4637          PRINT TAB(15):"--AAA LAVORO RITARDATO E PREVISTO"
4640          PRINT
4650          PRINT
4660      RETURN
4670      REM
4680      REM
4690      REM SUBROUTINE CHE STAMPA L'ISTOGRAMMA DELLE RISORSE
4700      REM INPUT: H(),T1
4710      REM OUTPUT: --
4720      REM
4730      REM PER I COMMENTI VEDERE IL PROGRAMMA CPM-II.
4750      REM
4760          FOR I=1 TO H9
4770              FOR K=1 TO T9
4780                  G*(I,K)=" "
4790              NEXT K
4800          NEXT I
4810      REM
4830          M=H(1)
4840          FOR K=2 TO T9
4850              IF H(K)>M THEN 4860 ELSE 4870
4860                  M=H(K)
4870      REM
4880          NEXT K
4910      REM
4920          A=1

```



```

5630      GOTO 5600
5640 REM
5650 REM
5660      PRINT
5670 RETURN
5680 REM
5690 REM
5700 REM SUBROUTINE CHE STAMPA LE OPERAZIONI REALI (O PREVISTE)
5710 REM   INPUT: J,J*( ),S(),A(),F(),T1
5720 REM   OUTPUT: H()
5730 REM
5750      PRINT TAB(S9);"I";
5760      K=S(J,1)
5770      IF A(J,2)-A(J,1)>0 THEN 5780 ELSE 6020
5780      IF K>F(J) THEN 5950
5790      IF K>=A(J,1) THEN 5830
5800      PRINT TAB(S9+K);"-";
5810      K=K+1
5820      GOTO 5790
5830 REM
5840      IF K<T1 THEN 5850 ELSE 5890
5850      PRINT TAB(S9+K);"*";
5860      H(K)=H(K)+R(J)
5870      K=K+1
5880      GOTO 5930
5890 REM
5900      PRINT TAB(S9+K);"A";
5910      H(K)=H(K)+R(J)
5920      K=K+1
5930 REM
5940      GOTO 5780
5950 REM
5960      IF K>A(J,2)+F(J)-A(J,1) THEN 6000
5970      PRINT TAB(S9+K);".";
5980      K=K+1
5990      GOTO 5960
6000 REM
6010      GOTO 6140
6020 REM
6030      IF K>F(J) THEN 6130
6040      IF K>=A(J,1) THEN 6080
6050      PRINT TAB(S9+K);"-";
6060      K=K+1
6070      GOTO 6040
6080 REM
6081      IF K<T1 THEN 6082 ELSE 6089
6082      PRINT TAB(S9+K);"*";
6083      H(K)=H(K)+R(J)
6084      K=K+1
6085      GOTO 6115
6089 REM
6090      PRINT TAB(S9+K);"A";
6100      H(K)=H(K)+R(J)
6110      K=K+1
6115 REM
6120      GOTO 6030
6130 REM
6140 REM
6150      PRINT
6160 RETURN
6170 REM
6190      END

```

Il terzo programma sul percorso critico è utile al dirigente impegnato per attuare il controllo del progetto, dopo averlo pianificato. A questo fine si devono conoscere gli effettivi orari di inizio e di fine di ogni operazione. Inoltre bisogna sapere il tempo attuale. I risultati vengono stampati perché tutti li possano vedere. La logica di fondo della subroutine per il diagramma di Gantt è un po' più complicata, ma non eccessivamente. L'istogramma delle risorse utilizzate è riferito al lavoro effettivamente svolto, non a quello programmato. Così anche per la tabella dei valori.

ESERCIZI

1. Modificate il vostro progetto come hanno fatto i tre manager di produzione e sottoponetelo a CPM-III.
2. Notate che i dati per i tre programmi CPM (*Critical Path Method*) possono essere letti dallo stesso file se CPM-I e CPM-II ignorano alcune delle informazioni. Create tale file e modificate i tre programmi perché siano in grado di leggerlo. (Avvertenza: questo esercizio presuppone che abbiate imparato come inserire i dati tramite file. Per i dettagli consultate il vostro manuale di Basic.)
(sugg.: La nuova riga di lettura di CPM-I leggerà ma non utilizzerà molti dei dati. Servitevi in questo caso della variabile X o X\$. Ad esempio, la nuova riga potrebbe essere:

```
READ #1: X$, X, L(J), X, X, X
```

dove l'unica variabile utilizzata è L().)

STIME DEI TEMPI

Martin Graves si accorse che il corso sulla pianificazione dei progetti gli serviva effettivamente per programmare e controllare i suoi progetti. Qualche mese più tardi vide sulla rivista della società l'annuncio di una conferenza sulla stima dei tempi di realizzazione e decise di seguirla. Il relatore era un esperto ricercatore del centro scientifico della Hawk Division. Il suo gruppo seguiva normalmente lo sviluppo di più progetti di ricerca e, in parecchi anni, aveva potuto sperimentare molte tecniche di pianificazione. Il contenuto della sua lezione era proprio ciò che interessava a Graves. La sala delle conferenze era gremita. "Grazie per essere venuti. Oggi vorrei cercare una risposta ad una delle vostre domande più difficili: 'Quando sarà pronto?'. Voi sapete che se la previsione è a breve termine, potreste lavorare duramente per qualche notte e portare a termine l'impegno. Ma se la previsione è a termine troppo lungo, è facile che vi perdiate. Come rispondere a questo problema?"

“Come la maggior parte dei progetti, anche i vostri sono composti di più piccole operazioni, alcune delle quali sono molto poco prevedibili. Qui sta il problema. Se non c'è modo di sapere quanto tempo richiederà ogni componente, come potete stimare l'intero progetto se non con una previsione del tutto estemporanea? La risposta è data da tre elementi: la distribuzione beta, un teorema di base di statistica, ed un programmino per computer che traduca le cose in pratica.”

LA PREVISIONE DEI LAVORI

“Più un progetto può essere diviso in passi, meglio è. Esaminate attentamente ogni compito e prevedete la quantità minima di tempo necessaria per portarlo a termine. Poi pensate alla quantità massima, nel caso che si frappongano tutte le difficoltà. Infine, indovinate la quantità di tempo più probabile per svolgere quella operazione. Per esempio, prendete un lavoro che può durare sei giorni. Il tempo minimo, se tutto va perfettamente, è di quattro giorni. Se qualcosa non va, mancano fondi, si rompe l'attrezzatura, ecc., il lavoro potrebbe protrarsi fino a 18 giorni. Poi pensate al tempo probabilmente necessario, che è di otto giorni. “Se chiamiamo A la previsione per difetto, B quella per eccesso ed M la più probabile, possiamo tracciare una curva che lega i tre valori in questo modo.” (fig. 6.3)

“Questa curva è detta funzione di densità probabilistica. Illustra come i tempi probabili si distribuiscono tra A e B; nel nostro esempio, tra quattro e diciotto.



Fig. 6.3 Distribuzione beta del tempo di realizzazione di un lavoro

“La curva mostra anche che la maggior parte del tempo si addensa attorno ad M; nel nostro caso, a otto. Vuol dire che M è la stima migliore? Non necessariamente. Notate che la stima per eccesso è più distante dal centro di quella per difetto. Per ottenere la stima migliore del punto centrale, troviamo il punto di mezzo tra gli estremi, $(A + B)/2$. Nel no-

stro caso si tratta di $(4 + 18)/2 = 11$. Poi facciamo la media tra il punto di mezzo ed il doppio del valore di massima verosimiglianza ($2 \cdot M$) per avere una stima del valore atteso, E. L'equazione è:

$$E = (2 \cdot M + (A+B) / 2) / 3$$

o, nel nostro caso:

$$E = (2 \cdot 8 + (4+18) / 2) / 3$$

$$E = 9$$

“Ripeto, E è il valore atteso. Potete supporre che la vostra funzione di densità sia una distribuzione beta e che E sia la stima migliore della media di distribuzione.”

“Perché il valore atteso è diverso dal valore di massima verosimiglianza?” chiese qualcuno dalla prima fila.

Il relatore commentò: “È una buona domanda. Il valore atteso è migliore perché, oltre che di M, tiene conto anche degli estremi A e B. Così la risposta all'apparente paradosso è che il valore atteso differisce dal valore di massima verosimiglianza perché cerca di riassumere tutte le informazioni in un valore centrale.

“Supponiamo che ripetiate le stime con tutti gli altri otto lavori che formano il vostro progetto.” (fig. 6.4)

“Uno dei teoremi più interessanti della statistica ci assicura che *indipendentemente da quali siano le singole distribuzioni*, la somma di queste distribuzioni tenderà alla distribuzione a campana che è detta ‘normale’ o di Gauss. Ecco come appare la somma delle precedenti distribuzioni.” (fig. 6.5)

“Questa è la curva relativa all'intero progetto. Il valore atteso, E, è la somma degli E dei singoli lavori, e quindi noi conosciamo il valore atteso dell'intero progetto. Ma forse la caratteristica più interessante della curva del progetto è che ha eliminato gli elementi d'incertezza. I tempi scendono simmetricamente a partire dal valore atteso. La curva ha assorbito le lunghe code dei singoli lavori in una stima complessiva. Ha funzionato la legge di compensazione, che soccorre spesso gli statistici. “Il calcolo dell'apertura della curva è semplice. Poniamo che, per ogni lavoro, la distanza dal valore minimo al valore massimo sia di sei deviazioni standard. Così:

$$D = (B-A) / 6$$

nel nostro caso:

$$D = (18-4) / 6 = 2.333$$




Fig. 6-4 Le distribuzioni di otto lavori

“La deviazione standard è una misura di come la curva si allarga. Il lavoro 5 ha una curva larga e una deviazione standard grande. Il lavoro 4 ha una curva stretta ed una piccola deviazione standard. La deviazione standard dell'intero progetto è più piccola di quanto ci si potrebbe aspettare a causa degli effetti di compensazione nella somma delle singole deviazioni.




Fig. 6-5 Distribuzione del progetto

“In effetti non si sommano le deviazioni, ma i quadrati delle deviazioni. La deviazione standard del progetto complessivo è la radice quadrata della somma dei quadrati. Per spiegarmi meglio, ho fatto la seguente tabella:

TABELLA DEI LAVORI

#	MIN	PROB.	MAX	ATTESO	D*D
1	4	8	18		
2	5	6	12		
3	2	9	11		
4	8	12	16		
5	9	14	32		
6	3	4	9		
7	2	10	14		
8	2	7	18		

“Notate che due colonne sono vuote. Per riempirle ho scritto un piccolo programma per computer. Avrete già capito da quali parti è composto il programma.

1. Inserimento di A, B e M per ogni lavoro.
2. Calcolo di E e D*D.
3. Somma E e D*D.
4. Quando tutto è fatto, stampa i risultati.

“Ecco ciò che si è ottenuto dal programma:

IL VALOR MEDIO DEL PROGETTO E': 74.1667
 LA DEVIAZIONE STANDARD E': 6.13505

TEMPO:	61.8966	69.8721	74.1667	78.4612	86.4368
PROB. DI COMPLET.:	2%	24%	50%	76%	98%

TABELLA DEI LAVORI

#	MIN	PROB.	MAX	ATTESO	D*D
1	4	8	18	9	5.44445
2	5	6	12	6.83333	1.36111
3	2	9	11	8.16667	2.25
4	8	12	16	12	1.77778
5	9	14	32	16.1667	14.6944
6	3	4	9	4.66667	1
7	2	10	14	9.33333	4
8	2	7	18	8	7.11111
TOTALI DEL PROGETTO				74.1667	37.6389

Il cuore del programma è questo:

```

310 REM ESEGUE I CALCOLI
320 REM
322 S1=0
324 S2=0
330 FOR R=1 TO R9
340 E=(2*T(R,2)+(T(R,1)+T(R,3))/2)/3
350 T(R,4)=E
360 D=(T(R,3)-T(R,1))/6
370 T(R,5)=D*D
380 S1=S1+E
390 S2=S2+D*D
400 NEXT R

```

"Come potete vedere, ho memorizzato tutti i risultati in una tabella T(R,C). Le prime tre colonne della tabella contengono le stime dei valori minimo, di massima verosimiglianza e massimo. La quarta colonna è il valore atteso, E. La quinta è la deviazione standard al quadrato."

"A cosa servono S1 e S2?" chiese una persona seduta in fondo alla sala.

"Sono le mie due somme, una per il valore atteso ed una per la deviazione al quadrato", spiegò il relatore.

"Perché si fa il quadrato della deviazione standard?" intervenne qualcun'altro.

"È una necessità matematica. Non voglio entrare nei dettagli ora, ma se non si facesse il quadrato di D il risultato non sarebbe giusto. Cioè, si otterrebbe un valore di stima dell'apertura della curva, ma non sarebbe corretto chiamarla deviazione standard.

"Quando ho finito con i singoli lavori, utilizzo le somme per avere i due numeri che mi interessano di più: la media del progetto o valore atteso, E, e la deviazione standard del progetto, D."

```

410 E=S1
420 D=SQR(S2)
430 REM

```

“Torniamo a guardare la nostra risposta. Sappiamo che si tratta di una curva a campana e che ha i valori contenuti in fig. 6.5.


“Come possiamo rispondere alla domanda ‘Quando sarà pronto?’ che si fanno i dirigenti d’azienda? Se prendiamo la seconda deviazione standard al di sopra della media, vicino a 86, possiamo dire ‘probabilmente’, mentre se prendiamo il valore 74, diremo che c’è una probabilità del 50 per cento. È sorprendente quanto ciò rassicuri l’amministrazione, oltre a mettere tranquillo te stesso.

“La fig. 6.6 è una tabella di valori che mostra come cresce la probabilità (l’area) quando aumenta la deviazione standard.

“La ragione per cui sono stato alto è che mi sembra di avere ancora dei problemi nel considerare tutti i lavori da includere nel mio progetto. Tra i lavori che ho trascurato ci sono:

- La pianificazione del progetto
- Gli incontri per riferire sull’andamento
- L’aggiornamento degli utenti della nuova procedura
- La gestione ed il controllo del progetto

“Così le mie stime dei tempi mi sembrano un po’ basse. Tuttavia, il mio ultimo progetto si è mantenuto del 4 per cento sopra il valore atteso, perciò credo che tutto sommato ci siamo.”



d	A(d)	d	A(d)	d	A(d)	d	A(d)
.0	.000	1.1	.364	2.1	.482	3.1	.4990
.1	.040	1.2	.385	2.2	.486	3.2	.4993
.2	.079	1.3	.403	2.3	.489	3.3	.4995
.3	.118	1.4	.419	2.4	.492	3.4	.4997
.4	.155	1.5	.433	2.5	.494	3.5	.4998
.5	.191	1.6	.445	2.6	.495	3.6	.4998
.6	.226	1.7	.455	2.7	.497	3.7	.4999
.7	.258	1.8	.464	2.8	.497	3.8	.49993
.8	.288	1.9	.471	2.9	.498	3.9	.49995
.9	.316	2.0	.477	3.0	.4987	4.0	.49997
1.0	.341					5.0	.4999997

Fig. 6.6 Area sottesa a una curva normale

Martin Graves aveva preso appunti per tutta la conferenza. Era riuscito ad avere anche una copia del programma che il relatore aveva distribuito al termine. Gli venne un'idea. Avrebbe potuto includere queste stime dei tempi nella sua pianificazione col percorso critico.

IL PROGRAMMA TEMPI

```

100 REM ***TEMPI***
110 REM
120 REM LEGGE UNA TABELLA DI STIME DI TEMPO PER SINGOLI LAVORI.
130 REM CALCOLA LA MEDIA DI LAVORO, LA DEVIAZIONE STANDARD E LA DEVIAZIONE
140 REM STANDARD AL QUADRATO. SOMMA LE MEDIE E I QUADRATI DELLE DEVIAZIONI.
150 REM STAMPA I RISULTATI.
160 REM
165 REM VARIABILI:
170 REM   A.....STIMA MINIMA DEL LAVORO
174 REM   B.....STIMA MASSIMA DEL LAVORO
176 REM   C.....INDICE DI COLONNA
178 REM   D.....DEVIAZIONE STANDARD DEL LAVORO
180 REM   E.....TEMPO PREVISTO PER IL LAVORO (MEDIA:
182 REM   R.....INDICE DI RIGA
184 REM   S1,S2.....VARIABILI SOMMA
186 REM   T().....TABELLA DEI VALORI DI LAVORO
188 REM
198 REM COSTANTI:
199   R9=8
200 REM
205 REM LEGGE LA TABELLA
210 REM
220   FOR R=1 TO R9
230     READ A,M,B
240     T(R,1)=A
250     T(R,2)=M
260     T(R,3)=B
270   NEXT R
280   DATA 4,8,18
282   DATA 5,6,12
284   DATA 2,9,11
286   DATA 8,12,16
288   DATA 9,14,32
290   DATA 3,4,9
292   DATA 2,10,14
294   DATA 2,7,18
300 REM
310 REM ESEGUE I CALCOLI
320 REM
322   S1=0
324   S2=0
330   FOR R=1 TO R9
340     E=(2*T(R,2)+(T(R,1)+T(R,3))/2)/3
350     T(R,4)=E
360     D=(T(R,3)-T(R,1))/6
370     T(R,5)=D*D
380     S1=S1+E
390     S2=S2+D*D
400   NEXT R
410   E=S1
420   D=SQR(S2)
430 REM

```

```

500 REM STAMPA I RISULTATI
510 REM
520 PRINT"IL VALOR MEDIO DEL PROGETTO E' :";E
530 PRINT"LA DEVIAZIONE STANDARD E' :";D
540 PRINT
550 PRINT"          TEMPO: ";
560 PRINT TAB(12);E-2*D;TAB(18);E-.7*D;TAB(24);E;
570 PRINT TAB(30);E+.7*D;TAB(36);E+2*D
580 PRINT"PROB. DI COMPLET.:    2%      24%      50%      76%      98%"
590 REM
592 PRINT
593 PRINT"TABELLA DEI LAVORI"
595 PRINT" #          MIN          PROB.          MAX          ATTESO          D*D"
600 FOR R=1 TO R9
605   PRINT R;
610   FOR C=1 TO 5
620     PRINT TAB(C*9);T(R,C);
630   NEXT C
640   PRINT
650 NEXT R
652 PRINT
660 PRINT"TOTALI DEL PROGETTO";TAB(4*9);E;
665 PRINT TAB(5*9);S2
999   END

```

Il programma TEMPI separa nettamente le sue tre funzioni in tre parti diverse: lettura dei dati, calcolo e stampa dei risultati. Ogni parte ha la sua specifica funzione.

Un interessante trucco di questo programma è come la funzione TAB(C*9) viene usata per stampare i nove spazi bianchi. (Vedi un'altra versione dello stesso espediente in "Paragrafi di programma" nell'Appendice A.)

ESERCIZI

1. Usate TEMPI per stimare un vostro progetto, del quale inserite i dati. I risultati vi sembrano apprezzabili?
2. Modificate l'output di TEMPI perché si stampino solo i valori che effettivamente vi servono.

La gestione delle scorte

Louis Mason aveva lavorato alla pianificazione delle scorte e degli approvvigionamenti per 20 anni. Aveva una cura particolare per l'istruzione dei nuovi dipendenti. Negli ultimi cinque anni la pianificazione degli approvvigionamenti era stata completamente computerizzata. Mason aveva imparato tutto quel che ci voleva. Quando Simon Wilson lanciò un piccolo computer per l'ufficio ad un prezzo favorevole, Mason fu uno dei primi a comperarlo. Ora lo usava per istruire i nuovi manager della gestione delle scorte. Aveva notato che imparavano di più se potevano prima sperimentare su un computer i concetti con un modello. La giornata di introduzione gli consentiva di stare a contatto con un gran numero di giovani e di raccogliere parecchi suggerimenti, oltre che di darli. Il suo gruppo era composto da cinque persone.

“A lato della produzione si accumulano le scorte”, esordì. “Lo si nota quando arrivano le materie prime, nei passaggi da una fase intermedia all'altra, e quando i prodotti finiti attendono la distribuzione. I costi di magazzino accompagnano tutto il ciclo produttivo; l'arrivo dei materiali, il processo di produzione e la spedizione dei prodotti finiti. Il problema delle scorte è che sembra che crescano spontaneamente e che quelle in eccesso sprechano risorse preziose.

“Per gestire opportunamente le scorte bisogna conoscere parecchie cose sull'attività dell'azienda: il valore del capitale, il valore di un normale flusso di prodotti, il valore di un efficiente servizio al cliente, ed il margine di profitto del prodotto. Inoltre dovete avere qualche idea dell'an-

damento futuro dell'approvvigionamento di materie prime e della domanda di prodotti finiti. Infine dovete tener presente che qualsiasi sistema usiate per raccogliere le informazioni e per controllare le scorte, rappresenterà anch'esso in certa misura una spesa. Perciò dovete sapere quanto verrà a costare l'intero sistema.

“Ci sono grosso modo tre diversi tipi di scorte in una azienda — le materie prime, i prodotti intermedi e quelli finiti — ma tutti hanno le stesse proprietà di fondo. Si riforniscono tramite approvvigionamento, il loro immagazzinamento costa qualcosa e la loro mancanza è penalizzante.

UN MODELLO DI MAGAZZINO

“Il programma che costruiremo oggi rappresenta un modello di gestione delle scorte. Dovrebbe poi risultarvi abbastanza facile costruire dei modelli simili per le scorte di cui siete responsabili. Il primo problema è quello di focalizzare l'attenzione su come le scorte vengono gestite in realtà. Iniziamo con una scorta di 55 unità:

$$I=55$$

Supponiamo che la domanda sia di 5 unità a settimana:

$$D=5$$

Consideriamo un periodo di rinnovamento di 12 settimane:

$$P=12$$

Quando facciamo nuove ordinazioni, riempiamo il magazzino con 60 unità:

$$R=60$$

Così cominciamo ad intravedere cosa succede se gestiamo la scorta ogni settimana. Innanzitutto, togliamo la domanda settimanale dalla scorta:

$$I=I-D$$

Controlliamo se è il caso di fare rifornimento, e, se lo è, riempiamo il magazzino:”

```

360      IF INT(T/P)=T/P THEN 370 ELSE 380
370          I=I+R
380 REM

```

“Mi scusi”, disse uno degli studenti. “Cos'è T?”

“T è l'unità di tempo considerata. Va dalla settimana 1 alla settimana 52. L'istruzione IF...THEN...ELSE chiede se il tempo T è un multiplo del periodo di rinnovamento, P. Se lo è, allora si fanno nuove ordinazioni. “Alla fine stampiamo la scorta esistente:

```
PRINT I;
```

“Per coprire tutto l'anno, dobbiamo ripetere questo processo 52 volte. Così facciamo in modo che il tempo T, nel programma, vada da 1 a 52 con queste istruzioni:

```
FOR T=1 TO 52
  *
  *
  *
NEXT T
```

“Se riassumiamo tutto, otteniamo:

```
300 I=55
310 D=5
320 P=12
330 R=60
340 FOR T=1 TO 52
350   I=I-D
360   IF INT(T/P)=T/P THEN 370 ELSE 380
370     I=I+R
380 REM
390   PRINT I;
400 NEXT T
410 END
```

Ed eseguendo il programma abbiamo una serie di numeri come questa:

```
50 45 40 35 30 25 20 15 10 5 0 55 50 45 40 35 30 25 20
15 10 5 0 55 50 45 40 35 30 25 20 15 10 5 0 55 50 45
40 35 30 25 20 15 10 5 0 55 50 45 40 35
```

“Va tutto bene, ma sarebbe meglio avere una rappresentazione grafica dell'andamento delle scorte. Se sostituiamo l'istruzione PRINT I con queste righe:

```
FOR K=1 TO I
  PRINT"*";
NEXT K
PRINT
```

Otteniamo un diagramma che rappresenta le nostre scorte:


```

270 REM R.....RIFORMIMENTI
280 REM T.....PERIODO DI TEMPO
290 REM
300 I=55
310 D=5
320 P=13
330 R=60
340 FOR T=1 TO 52
350 I=I-D
360 IF INT(T/P)=T/P THEN 370 ELSE 380
370 I=I+R
380 REM
390 GOSUB 440
400 NEXT T
410 STOP
420 REM
430 REM
440 REM SUBROUTINE CHE STAMPA GLI ASTERISCHI
450 REM INPUT: I
460 REM OUTPUT:
470 REM
480 IF I<-10 THEN 490 ELSE 510
490 PRINT"<-10"
500 GOTO 680
510 REM
520 IF I>50 THEN 530 ELSE 550
530 PRINT">50"
540 GOTO 680
550 REM
560 IF I<0 THEN 570 ELSE 620
570 FOR K=I TO -1
580 PRINT TAB(10+I);"*";
590 NEXT K
600 PRINT"+"
610 GOTO 680
620 REM
630 PRINT TAB(10);"+";
640 FOR K=1 TO I
650 PRINT"*";
660 NEXT K
670 REM
680 REM
690 RETURN
700 REM
710 END

```

Questo programma è scaturito dalle idee descritte nel testo. L'istruzione IF...THEN...ELSE controlla se è tempo di fare nuove ordinazioni, guardando se il tempo T è un multiplo del periodo di rifornimento, P. Controlla cioè se T/P è un numero intero:

```
IF INT(T/P)=T/P THEN...ELSE...
```

La fantasiosa routine di stampa rappresenta la scorta tra -10 e +50. Si noti che si può eliminare la stampa togliendo l'istruzione GOSUB dal programma principale.

ESERCIZI

1. Provate il programma MAGAZZINO con una scorta iniziale diversa, diversi il periodo di rifornimento e la quantità d'approvvigionamento. Vedete secondo quali regole interagiscono questi valori?
2. Modificate MAGAZZINO in modo che l'approvvigionamento riporti la scorta ad un determinato livello.
3. Cambiate MAGAZZINO in modo che si faccia una nuova ordinazione quando le scorte si abbassano sotto un livello fissato.

SCORTA PER IL RIFORNIMENTO DEL MAGAZZINO

“Ora che abbiamo visto una prima rappresentazione delle scorte, possiamo porci la prima di una serie di questioni sul modo migliore per gestirle. La prima domanda è perché abbiamo un andamento annuale compreso tra questi estremi.” (fig. 7.1)




Fig. 7.1 Estremi della scorta annuale

“La risposta è che stiamo tentando un bilanciamento tra due tipi di costi, quelli di gestione e quelli di rifornimento. I costi di gestione comprendono le spese di approvvigionamento, l'interesse sul denaro investito nel magazzino, l'assicurazione dei beni, il deterioramento di una certa percentuale di merce, e l'invecchiamento di un'altra parte. I costi di gestione sono molto spesso considerati come una percentuale del valore delle scorte.”

“Quale percentuale?” chiese un altro dei nuovi manager.

“Di tutte le percentuali generalmente si fa la media, per ottenere una rappresentazione complessiva. Di solito il più importante dei costi di gestione è quello del capitale investito nella gestione del magazzino. Più denaro va al magazzino, meno si spende per il resto della produzione, più capitale si investe e minore è l'indice di rendimento dell'investimento (ROI) per un dato livello di resa. Mantenendo la minor quantità di scorte possibile, si hanno in definitiva:

1. Minori spese
2. Minor indebitamento
3. Denaro riservato ad altri scopi
4. Minori attività correnti investite
5. Incremento della resa sull'investimento dell'intera impresa

"L'altro costo è quello di rifornimento del magazzino. Vi sono inclusi i costi d'ufficio e di mano d'opera necessari per registrare e immagazzinare nuova merce. Di solito questo costo di riempimento è considerato una costante. Ci vuole la stessa quantità di inchiostro, carta e tempo per registrare un'ordinazione di 10 unità che una di 10 000 unità e più o meno la stessa quantità di risorse per trattare un'ordinazione grande o una piccola."

"È proprio vero?" obiettò uno dei manager.

"Mi rendo conto che in prima analisi sia difficile crederlo, ma nella maggior parte dei casi è proprio così. Ovviamente un'ordinazione grande richiederà più tempo di un piccola ed il controllo di qualità sarà più dispendioso, ma in molti casi la dimensione dei costi è un dato complessivo. Che si creda o no, è così.

"Proseguendo," continuò Mason, "possiamo aggiungere con poco sforzo questi due costi al nostro modello di magazzino. Il costo di gestione, C1, è il 20 per cento del valore annuale, così il costo di gestione settimanale è $.20/52$. Il costo di ogni articolo è di 54 dollari. Possiamo includere i costi di gestione nel nostro programma con queste righe:

```

430          IF I>=0 THEN 440 ELSE 450
440          C1=C1+.20/52*D1*I
450 REM

```

dove D1 è il prezzo di un articolo della scorta, 54 dollari. Il costo di rifornimento, C2, nel nostro esempio è di 80 dollari. Inseriamo una nuova riga di programma per la nuova ordinazione:

```
C2=C2+80
```

"Naturalmente dobbiamo iniziare con C1 al valore zero e C2 all'attuale frazione di riempimento del magazzino. Vogliamo stampare ogni volta i valori:

```

C1=0
C2=I/R*80
PRINT T;C1;C2,

```

Il cuore del programma ora è questo:

```

400      C1=0
410      C2=I/R*B0
420      FOR T=1 TO 52
430          IF I>=0 THEN 440 ELSE 450
440          C1=C1+.20/52*D1*I
450      REM
460          I=I-D
470          IF T/P-INT(T/P)<=.001 THEN 480 ELSE 500
480          C2=C2+B0
490          I=I+R
500      REM
510      NEXT T
520      C2=C2-(I/R*B0)
530      PRINT T;C1;C2

```

“Se eseguiamo il nostro programma per un anno, troviamo che i costi risultano essere:

52 313.615 346.667

“Ora notiamo che possiamo esprimere la nostra scorta iniziale e la nostra quantità di rifornimento in termini del nostro periodo di rinnovamento e della domanda settimanale:

```

D=5
P=12
I=(P-1)*D
R=P*D

```

“In effetti adesso siamo in grado di trovare il miglior periodo di rifornimento e la relativa quantità, R, variando P. Modificando in questo senso il programma, otteniamo:”

IL PROGRAMMA MAG

```

100 REM ***MAG***
110 REM
120 REM QUESTO PROGRAMMA CALCOLA QUANTO PUO' COSTARE UNA DIVERSA
130 REM POLITICA DI MAGAZZINO.
140 REM
180 REM VARIABILI:
190 REM C.....COSTO TOTALE
200 REM C1.....COSTO DI GESTIONE
210 REM C2.....COSTO DI RIFORNIMENTO
220 REM D.....DOMANDA
230 REM D1.....COSTO DI UN ARTICOLO DI MAGAZZINO ($)
240 REM I.....SCORTE
250 REM K.....VARIABILE INDICE
260 REM P.....PERIODO DI IMMAGAZZINAMENTO (SETTIMANE)
270 REM R.....RIFORNIMENTO
280 REM T.....INDICE DI TEMPO (SETTIMANE)
290 REM
300 REM PROGRAMMA PRINCIPALE

```



```

310 REM
320 PRINT"RIFORNIMENTO"
330 PRINT"PER. QUANT.," C. GEST.",
340 PRINT" C. IMM.," C. TOT."
350 D1=54
360 D=5
370 FOR P=5 TO 25 STEP 5
380 I=(P-1)*D
390 R=P*D
400 C1=0
410 C2=I/R*B0
420 FOR T=1 TO 52
430 IF I>=0 THEN 440 ELSE 450
440 C1=C1+.20/52*D1*I
450 REM
460 I=I-D
470 IF T/P-INT(T/P)<=.001 THEN 480 ELSE 500
480 C2=C2+B0
490 I=I+R
500 REM
510 NEXT T
520 C2=C2-(I/R*B0)
530 PRINT P;" ";R,C1,C2,C1+C2
540 NEXT P
550 REM
560 END

```

“Come si effettuano gli altri calcoli con C2?” chiese uno dei nuovi manager.

“La prima equazione assegna la parte dei costi di rifornimento relativi alla scorta attuale al valore iniziale di C2. La seconda toglie i costi di rifornimento delle scorte rimaste alla fine dell'anno. Il risultato è che C2 riflette i costi di rifornimento esattamente per il periodo di 52 settimane.”

“Un'esecuzione del programma dà i seguenti risultati:

RIFORNIMENTO				
PER.	QUANT.	C. GEST.	C. IMM.	C. TOT.
5	25	111.115	832	943.115
10	50	251.308	416	667.308
15	75	407.077	277.333	684.41
20	100	562.846	208	770.846
25	125	671.885	166.4	838.285

“Notiamo che il costo più basso è attorno ad un periodo di rifornimento di 15 settimane. Perciò rieseguiamo il nostro programma per P che va da 9 a 15 e troviamo:

RIFORNIMENTO				
PER.	QUANT.	C. GEST.	C. IMM.	C. TOT.
9	45	223.269	462.222	685.492
10	50	251.308	416	667.308
11	55	282.462	378.182	660.643
12	60	313.615	346.667	660.282
13	65	324	320	644
14	70	371.769	297.143	668.912
15	75	407.077	277.333	684.41

“Ora se analizziamo i risultati, notiamo che il costo totale minimo si ha se il periodo è di 13 settimane. A quel punto C1 raggiunge C2. Non è un caso. C'è un motivo ben preciso. Abbiamo iniziato tentando di bilanciare costi di rifornimento e costi di gestione.

“Così ad una domanda costante di cinque unità alla settimana si può gestire al meglio il magazzino facendo le ordinazioni della giusta quantità al momento giusto perché si bilancino i costi di rifornimento e di gestione. La giusta quantità è detta ‘quantità economica d'ordinazione’.

“Dai nostri risultati possiamo trarre due considerazioni. Per prima cosa, il miglior tempo e la miglior quantità di rifornimento sono rispettivamente 13 settimane e 65 unità. Secondo, quel magazzino verrà a costare almeno 644 dollari all'anno. Non lo si può gestire per meno se si vuol soddisfare la domanda.”

ESERCIZI

1. Considerate MAGAZZINO e modificalo secondo le indicazioni di Louis Mason. Eseguite le versioni intermedie per assicurarvi che dia- no gli stessi risultati. Quando avete finito, il vostro programma dovrebbe assomigliare a MAG.

RITARDO NEI RIFORNIMENTI

“Finora il nostro modello dava per scontato che un'ordinazione venisse soddisfatta immediatamente. Se invece è richiesta una certa quantità di tempo, possiamo modificare il programma in questo senso. Ad esempio se ci vogliono due settimane perché un'ordinazione venga portata a termine, il programma diventa così:

```

IF INT((T+2)/P)=(T+2)/P THEN...ELSE...
  R$="RIORDINO"
REM
REM
REM
IF R$="RIORDINO" AND INT(T/P)=T/P THEN...ELSE...
  I=I+R
  R$="ATTENDERE"
REM

```

“Quando poi eseguiamo il programma, tutto funzionerà allo stesso modo, ma il flag dell'ordinazione, R\$, ci ricorda che il rifornimento in questione era stato fatto due settimane prima.”

“La grande incognita nella gestione delle scorte è la domanda dei clienti. Ci sono parecchi modi per prevedere quale sarà la domanda. Se l'anno precedente aveva avuto un certo andamento, possiamo supporre che anche quest'anno sarà più o meno così. Per esempio, supponiamo che l'anno scorso la domanda fosse questa:




Fig. 7.2 Andamento della domanda


“Possiamo inserire questo andamento della domanda nel nostro programma aggiungendo un ciclo che lo assegni alla variabile D(T):

```
FOR T=1 TO 52
  READ D(T)
  DATA 4,3,2,5,4,1,5,4,3,5
  DATA 3,2,5,4,6,4,7,2,6,4
  DATA 3,5,6,4,5,3,6,2,6,7
  DATA 4,5,8,3,6,7,5,8,6,5
  DATA 7,9,5,7,8,10,6,9,11,4
  DATA 1,3
```

“Usiamo questo andamento quando consideriamo la nostra domanda, D, nel ciclo delle settimane, FOR T=1 TO 52, e la modifichiamo:

```
D=D(T)
```

“Ora eseguiamo il nostro programma e vediamo come appare il piano di rifornimento di 65 unità ogni 13 settimane:



“Potremmo migliorare il nostro piano di rifornimento riferendolo alla domanda dell'anno precedente e ordinando ciò che pensiamo ci servirà nelle prossime 13 settimane:

$$R = D(T+1) + D(T+2) + D(T+3) + \dots + D(T+13)$$

“Se notiamo che per caso esauriamo la scorta, possiamo provare a riordinarne un altro 5 per cento:

$$R = D(T+1) + \dots + D(T+13)$$

$$R = 1,05 * R$$

“Un altro modo per prevedere quale sarà la domanda dei clienti consiste nel supporre che sia casuale ed in questo senso si adatta il programma. Per esempio, se la stessa domanda di prima fosse così:




Fig. 7-3 Domanda casuale

“La domanda casuale può essere manipolata perché ci mostri la sua distribuzione:




Fig. 7-4 Istogramma della domanda casuale

“In effetti potremmo modificare l'andamento della domanda perché renda idea della sua distribuzione. Non è necessario farlo, ma è un buon modo per ricordarci che pensiamo all'andamento della domanda come ad una domanda da cui estrarremo un valore a caso:

```
FOR T=1 TO 52
  READ D(T)
NEXT T
DATA 1, 1
DATA 2, 2, 2, 2
DATA 3, 3, 3, 3, 3, 3, 3
DATA 4, 4, 4, 4, 4, 4, 4, 4
```



```

>50
+*****+
+*****+
+*****+
+*****+
+*****+
+*****+
+****
+++
*****
+*****+
<-10
<-10
+*****+

```

“Notate che, ancora una volta, abbiamo periodi in cui la scorta si esaurisce. Questo ci richiama l’importante questione di quanto costi alle società la mancanza di un articolo nel magazzino.”

IL PROGRAMMA MAGAZZ

```

100 REM ***MAGAZZ***
110 REM
120 REM QUESTO PROGRAMMA SIMULA UN MAGAZZINO CON DOMANDA CASUALE
130 REM E UN RIFORNIMENTO PERIODICO REGOLARE.
140 REM
150 REM VARIABILI:
160 REM C1.....COSTI DI GESTIONE
170 REM C2.....COSTI DI RIFORMIMENTO
180 REM C3.....COSTI DOVUTI A CARENZA DI SCORTE
190 REM D1.....VALORE DI OGNI UNITA' (%)
200 REM D.....DOMANDA
210 REM D().....DISTRIBUZIONE DELLA DOMANDA
220 REM I.....SCORTE
230 REM P.....PERIODO DI IMMAGAZZINAMENTO
240 REM R.....RIFORMIMENTI
250 REM R9.....NUMERO CASUALE
260 REM T.....PERIODO DI TEMPO
270 REM
310 REM DIMENSIONI:
320 DIM D(52)
330 REM
340 REM LEGGE L'ANDAMENTO DELLA DOMANDA
350 REM
360 FOR T=1 TO 52
370 READ D(T)
380 NEXT T
390 DATA 1,1
392 DATA 2,2,2,2
394 DATA 3,3,3,3,3,3
396 DATA 4,4,4,4,4,4,4,4
398 DATA 5,5,5,5,5,5,5,5,5
400 DATA 6,6,6,6,6,6,6,6
402 DATA 7,7,7,7,7
404 DATA 8,8,8
406 DATA 9,9
408 DATA 10
410 DATA 11
432 REM
435 REM

```

```

440 D1=54
450 P=13
460 R=65
470 I=(P-I)/P*R
480 C1=0
490 C2=I/R*80
500 C3=0
503 REM
505 REM ESEGUE L'INVENTARIO DI UN ANNO
507 REM
510 FOR T=1 TO 52
520 IF I>=0 THEN 530 ELSE 550
530 C1=C1+.20/52*D1*I
540 GOTO 570
550 REM
560 C3=C3+.05*D1*(-I)
570 REM
575 R9=RND(52)
580 D=D(R9)
590 I=I-D
600 IF INT(T/P)=T/P THEN 610 ELSE 630
610 I=I+R
620 C2=C2+80
630 REM
640 GOSUB 700
650 NEXT T
660 C2=C2-(I/R*80)
670 STOP
680 REM
690 REM
700 REM SUBROUTINE CHE STAMPA GLI ASTERISCHI
710 REM INPUT: I
720 REM OUTPUT:
730 REM
740 IF I<-10 THEN 750 ELSE 770
750 PRINT"<-10"
760 GOTO 940
770 REM
780 IF I>50 THEN 790 ELSE 810
790 PRINT">50"
800 GOTO 940
810 REM
820 IF I<0 THEN 830 ELSE 880
830 FOR K=I TO -1
840 PRINT TAB(10+I);"*";
850 NEXT K
860 PRINT"+ "
870 GOTO 940
880 REM
890 PRINT TAB(10);"+ ";
900 FOR K=1 TO I
910 PRINT"*";
920 NEXT K
930 PRINT
940 REM
950 RETURN
960 REM
970 END

```

Il programma è una versione rinnovata di **MAGAZZINO**. Al suo interno usiamo il nostro miglior periodo, 13, e la miglior quantità, 65, di rifornimento, per vedere cosa succede quando la domanda **D** è distribuita in modo casuale con una media di 5. (La distribuzione effettiva nel pro-

gramma approssima una distribuzione di Poisson con una media di 5.) Leggendo una distribuzione come questo programma ha letto D(T), e usando un numero casuale come lo ha letto il programma, si può simulare qualsiasi distribuzione della domanda.

ESERCIZI

1. Eseguite il programma più volte per vedere quali sono i vari risultati.
2. Stampate i costi associati alle diverse esecuzioni. Quanto costa esaurire le scorte?

ESAURIMENTO DELLE SCORTE

“Ci sono molti modi per considerare il costo dell'esaurimento di un articolo del magazzino. In tale situazione il cliente può disdire l'ordinazione o annullarla. Se i clienti annullano le loro ordinazioni, il costo dell'esaurimento della scorta è dato dalla perdita dell'utile e della fiducia del cliente. Se tutti disdicono un articolo, rimane il malanimo per aver atteso invano la merce. Perciò l'esaurimento della scorta si paga sempre.

“Il costo d'esaurimento può essere inteso come costo per articolo o costo per articolo per settimana. In quest'ultimo caso, se è calcolato come percentuale, comparirà nel nostro programma più o meno come il costo di gestione.

560

$$CS = CS + .05 * D1 * (-1)$$

“Qui il costo d'esaurimento è il 5 per cento del costo settimanale per articolo.”

COSTI BILANCIATI

“Il costo totale del magazzino è la somma del costo di gestione, di rifornimento e del costo d'esaurimento delle scorte. Questi sono sempre i tre costi fondamentali di qualsiasi problema di gestione del magazzino. Il metodo per trovare il giusto livello delle scorte al minimo costo sta nell'includere le opportune considerazioni nel modello del problema e nel variare le quantità ed i tempi che in realtà possono essere variati.

“Con questo, per questa settimana, ho concluso la spiegazione”, disse

Mason. "Ora analizzeremo varie politiche di gestione tramite il nostro modello. Come vedrete, situazioni diverse richiedono politiche diverse."

SCONTI SULLA QUALITÀ E VARIAZIONI NELLE FORNITURE

Nel pomeriggio Louis Mason riunì i suoi nuovi manager per una breve introduzione prima che tornassero al lavoro coi loro modelli di gestione delle scorte.

"Qualcuno di voi ha chiesto delucidazioni a proposito degli sconti sulla quantità e dei problemi relativi ai cambiamenti di prezzo delle materie prime. Ne parlerò brevemente.

"Capita spesso ai responsabili d'acquisto di avere la possibilità di comperare grosse quantità di merce a prezzi scontati. Il problema è 'quando conviene approfittare di queste occasioni?'. La risposta è semplice. Vale la pena se lo sconto supera il costo di gestione della quantità acquistata in sovrappiù.

"Per rispondere a questa questione nella pratica, si sommano le scorte al loro prezzo scontato e si controlla quale sarà il costo di gestione della merce extra. I costi di gestione cresceranno. Se aumentano a più del risparmio totale ottenuto, l'offerta non è vantaggiosa.

"Un altro problema comune ai responsabili di magazzino è quanto comperare adesso in previsione di un futuro aumento dei prezzi della merce. Esaminando varie possibilità con il modello, potete decidere se e quanta merce acquistare e calcolare quale sarà la spesa totale prevista.

"Ho finito", disse Mason. "Spenderemo il resto della giornata a lavorare con i vari modelli. Se pensate che agire coi modelli sia difficile, pensate a quanto lo era 30 anni fa quando tutto si imparava per esperienza diretta con le scorte reali."

ESERCIZI

1. Ripassate i primi esercizi fatti con MAGAZZINO. Provateli con MAGAZZ e guardate se potete trovare il metodo meno dispendioso per la gestione delle scorte. (Suggerimento: di sicuro dovete cambiare la distribuzione casuale con un andamento nel tempo. Per trovarne uno significativo leggete prima il testo.)
2. Immaginate come fare le ordinazioni se la metà dei clienti si ritira se un articolo è esaurito. (Considerate il costo di esaurimento come percentuale delle disdette settimanali più tutti i profitti delle vendite non effettuate.)
3. Analizzate l'andamento della domanda, fate un controllo casuale del periodo medio e determinate le ordinazioni del periodo di domanda massima.

4. Fate un controllo casuale delle domande passate di un certo periodo e ordinate merce per il 95 per cento della domanda.
5. Pensate ad uno spazio di magazzino limitato. Ciò può significare aumento di costi se dovete affittare un locale per il deposito delle scorte.

Il problema della dieta: programmazione lineare

La Hawk Division organizzò una serie di seminari di una giornata per i suoi manager di produzione. Tra gli argomenti in programma c'era la programmazione lineare, il metodo matematico per la pianificazione della produzione. Relatore era una giovane manager di produzione di nome Helen Anderson.

Cinque anni prima la Anderson aveva coordinato con la sezione elaborazione dati un progetto per la definizione di procedure di controllo della produzione di pezzi per aerei militari. Per il suo seminario ai nuovi manager di produzione ritenne utile ripassarsi l'argomento.

“La tecnica di cui oggi vi parlerò”, esordì, “può essere usata per la soluzione di problemi ben diversi dall'organizzazione della produzione, per esempio per programmare un sistema economico nazionale, per pianificare la distribuzione dai magazzini ai venditori al dettaglio, per assegnare uomini a progetti. Per scopi didattici tuttavia, inizieremo con una versione particolarmente semplice: il problema della dieta.

“Supponiamo che, tra tutte le vitamine, i minerali e gli alimenti di cui si ha bisogno, la vostra dieta richieda solo tre sostanze: proteine, carboidrati e grassi. Supponiamo inoltre che possiate acquistare solo cinque tipi di alimenti:

	Grammi di sostanza		
	Proteine	Carboidrati	Grassi
Latte (1/4)	32	48	40
Hamburger (1 libbra)	112	—	91
Tonno (10 once)	83	—	21
Patate (5 libbre)	45	500	—
Margarina (28 once)	126	95	262

“Il vostro bisogno settimanale di queste sostanze (in grammi) è:

Proteine	490
Carboidrati	1841
Grassi	392

“La questione è che quantità di quale alimento comperare per soddisfare le vostre esigenze dietetiche.

“Il primo modo di risolvere questo problema consiste nel provare a caso e vedere se funziona. Con 10 libbre di hamburger, 4 sacchetti di patate e 4 vasetti di margarina ogni esigenza di questa dieta immaginaria sarà soddisfatta. Finora, tuttavia, la nostra soluzione non ha tenuto conto di una caratteristica a cui la maggior parte di noi è molto attenta: il prezzo. Gli alimenti costano.

“Il prezzo è un dato importante per l'economia familiare, per quella aziendale e per quella nazionale. Una riformulazione comune del problema della dieta è: che quantità di quale alimento comperare per soddisfare la dieta *al minor prezzo possibile?*

“Supponiamo che i prezzi di questi generi (in centesimi di dollaro) siano:

1 quarto di latte	58
1 libbra di hamburger	139
1 barattolo da 10 once di tonno	189
1 sacchetto di patate da 5 libbre	98
1 vasetto di margarina da 28 once	159

Si potrebbe tentare ancora a caso, ma non è facile come prima.

Ora dobbiamo confrontare tutte le soluzioni nutritive soddisfacenti per trovare la più conveniente. Per semplificare il nostro compito è utile avere un programma che risolva il problema per noi.

“Un programma per la ricerca della soluzione deve tener conto di tre equazioni per trovare qual è il valore nutritivo di ogni singola prova. Se N1 rappresenta le proteine, N2 i carboidrati ed N3 i grassi, allora le equazioni saranno:

```

N1=32*M+112*H+83*T+45*P+126*B
N2=48*M+500*P+95*B
N3=40*M+91*H+23*T+262*B

```

“Ora sappiamo inoltre che vogliamo una combinazione solo se le proteine sono più di 490, i carboidrati più di 1841, ed i grassi più di 392. Quindi possiamo scegliere i nostri valori utilizzando alcune istruzioni IF...THEN...ELSE:

```

IF N1>=490 AND N2>=1841 AND N3>=392 THEN...ELSE...
C=C+1
REM

```

“Decidiamo di provare con 5 quarti di latte, 8 libbre di hamburger, 3 scatolette di tonno, 4 sacchetti di patate e 4 vasetti di margarina. Effettueremo il nostro controllo nutritivo su tutte le possibili combinazioni di questi alimenti, inserendo il test al centro di cinque cicli:

```

345 FOR M=0 TO 5
350 FOR H=0 TO 8
355 FOR T=0 TO 3
360 FOR P=0 TO 4
365 FOR B=0 TO 4
380 N1=32*M+112*H+83*T+45*P+126*B
390 N2=48*M+500*P+95*B
400 N3=40*M+91*H+23*T+262*B
410 IF N1>=490 AND N2>=1841 THEN 420 ELSE 520
420 IF N3>=392 THEN 430 ELSE 520
430 C=C+1
520 REM
530 NEXT B
540 NEXT P
550 NEXT T
560 NEXT H
570 NEXT M
580 REM

```

“Quando eseguiamo il programma scopriamo che si ottengono 1411 diverse combinazioni soddisfacenti le nostre richieste minime. Ora aggiungiamo un piccolo test per vedere se la scelta in questione è la più conveniente e, se lo è, la consideriamo come miglior soluzione corrente:

```

440 P9=58*M+139*H+189*T+98*P+159*B
450 IF P9<M9 THEN 460 ELSE 515
460 M9=P9
470 M1=M
480 H1=H
490 T1=T
500 P1=P
510 B1=B
515 REM

```

“Quando eseguiamo questo programma e stampiamo i valori finali degli alimenti ed il loro prezzo complessivo, sappiamo ciò che dobbiamo comparare.”

SCELTE GIUSTE: 1411
 LA SCELTA PIU' CONVENIENTE E': 8.26
 COMPRA:
 2 QUARTI DI LATTE
 0 POUND DI HAMBURGER
 0 SCATOLE DI TONNO DA 10 ONCE
 4 SACCHI DI PATATE DA 5 LIBBRE
 2 SCATOLE DI BURRO DI ARACHIDI DA 28 ONCE

IL PROGRAMMA DIETA

```

100 REM ***DIETA***
110 REM
120 REM QUESTO PROGRAMMA CALCOLA LA QUANTITA' DI CIBO CHE SODDISFA IL
130 REM FABBISOGNO NUTRIZIONALE MINIMO AL PREZZO PIU' CONVENIENTE.
140 REM
160 REM VARIABILI:
170 REM B.....BURRO DI ARACHIDI
180 REM B1.....QUANTITA' OTTIMALE DI BURRO DI ARACHIDI
190 REM C.....CONTATORE DELLE SCELTE GIUSTE
200 REM H.....HAMBURGER
210 REM H1.....QUANTITA' OTTIMALE DI HAMBURGER
220 REM N1.....PROTEINE
230 REM N2.....CARBOIDRATI
240 REM N3.....GRASSI
250 REM M.....LATTE
260 REM M1.....QUANTITA' OTTIMALE DI LATTE
270 REM M9.....PREZZO MINIMO
280 REM P.....PATATE
290 REM P1.....QUANTITA' OTTIMALE DI PATATE
300 REM P9.....PREZZO DELL'ALIMENTO
310 REM T.....TONNO IN SCATOLA
320 REM T1.....QUANTITA' OTTIMALE DI TONNO
325 REM
330 M9=10000
335 C=0
340 REM
345 FOR M=0 TO 5
350 FOR H=0 TO B
355 FOR T=0 TO 3
360 FOR P=0 TO 4
365 FOR B=0 TO 4
380 N1=32*M+112*H+83*T+45*P+126*B
390 N2=4B*M+500*P+95*B
400 N3=40*M+91*H+23*T+262*B
410 IF N1>=490 AND N2>=1841 THEN 420 ELSE 520
420 IF N3>=392 THEN 430 ELSE 520
430 C=C+1
440 P9=5B*M+139*H+189*T+9B*P+159*B
450 IF P9<M9 THEN 460 ELSE 515
460 M9=P9
470 M1=M
480 H1=H
490 T1=T
500 P1=P
510 B1=B
515 REM
520 REM
530 NEXT B
540 NEXT P
550 NEXT T
560 NEXT H

```

```

570 NEXT M
580 REM
590 PRINT "SCELTE GIUSTE:"; C
600 PRINT "LA SCELTA PIU' CONVENIENTE E' :"; M9/100
610 PRINT "COMPRA:"
620 PRINT M1; " QUARTI DI LATTE"
630 PRINT H1; " POUND DI HAMBURGER"
640 PRINT T1; " SCATOLE DI TONNO DA 10 ONCE"
650 PRINT P1; " SACCHI DI PATATE DA 5 LIBBRE"
660 PRINT B1; " SCATOLE DI BURRO DI ARACHIDI DA 28 ONCE"
670 REM
680 END

```

DIETA mostra come un programma per computer può servirsi di metodi semplici e brutali per risolvere un problema. Nessuno tenterebbe di calcolare queste combinazioni a mano. Inoltre è facile stabilire il metodo con cui il computer può fare il lavoro per noi.

La variabile M9 tien conto del minimo prezzo in modo che alla fine abbiamo sia questo che la quantità di ogni alimento che vogliamo compere.

Le quantità del programma sono espresse in unità angloamericane. Ricordiamo che 1 oncia = 28,35 gr; 1 libbra = 16 once; 1 quarto = 1,136 litri; 1 pound = 453,6. [N.d.R.]

ESERCIZI

1. Cambiate il limite superiore alle patate con 10 e rieseguite il programma. Va meglio?
2. Eliminate le patate dalla dieta e guardate cosa succede. Il risultato è più dispendioso? Perché?

RIFLESSIONE

"Analizziamo ciò che abbiamo fatto. Dicendo FOR M = 0 TO 5 quarti di litro, abbiamo passato in rassegna sei diversi possibilità per quanto riguarda il latte. Per ognuna di queste abbiamo tentato nove possibilità per gli hamburger. Tra questi due generi, abbiamo esaminato $(5+1) \cdot (8+1) = 6 \cdot 9 = 54$ possibilità. Con tutti cinque gli alimenti otteniamo 5400 possibilità.

"Il problema del nostro metodo è che funziona per un piccolo numero di sostanze. Se aggiungiamo solo cinque alimenti in più e facciamo sì che ognuno di essi vari tra 0 e 9, abbiamo aumentato il numero di casi da 5000 a 500 milioni! In attesa che il computer trovi la soluzione del problema della dieta con dimensione qualsiasi, potremmo benissimo morire di fame."

LA TABELLA DI TUCKER

“Quando cominciamo la ricerca di un metodo efficiente per risolvere il nostro problema con numeri molto grossi, dobbiamo aggiungere un elemento critico alle nostre considerazioni. Dobbiamo essere in grado di poter comperare mezzo sacchetto di patate. In altre parole, dobbiamo poter acquistare una *frazione dell'unità* di un certo alimento. Se questa modifica spaventa qualcuno, possiamo arrotondare i nostri numeri agli interi ed esaminare i risultati quando abbiamo trovato la soluzione esatta decimale.

“Il tipo particolare di tabella che dobbiamo costruire per risolvere il nostro problema della dieta, in matematica, è detto *n*-simpleso. Il nome è importante solo perché identifica anche l'insieme di regole che seguiremo per risolvere il nostro problema: il metodo del simpleso. Questo metodo si applica più facilmente ad una diversa versione della nostra tabella, la tabella di Tucker. Dunque, per prima cosa, dobbiamo organizzare i nostri dati in una tabella di Tucker. Quando poi l'abbiamo costruita, applicheremo ad essa il metodo del simpleso. Nell'applicare il metodo del simpleso, daremo vita alle parti di un programma Basic che effettua i calcoli. Infine esploreremo diversi aspetti interessanti dei risultati.

“Il nostro problema iniziale, contenuto in una tabella di Tucker, è quello della figura 8.1, dove V_1, \dots, V_5 sono quantità di cibo; U_1, U_2 e U_3 sono sostanze in eccedenza; X_1, X_2 e X_3 sono i costi relativi ad ogni sostanza; e Y_1, \dots, Y_5 sono scarti nei prezzi. Spiegherò più avanti questi termini.

“Dopo aver trovato i risultati con il metodo del simpleso, la nostra tabella sarà così.” (fig. 8.2)

“Ciò che ci offre il metodo del simpleso è una serie di valori per le variabili al contorno. Le variabili più importanti sono le V , i valori di tutti

		Sostanze (grammi)			Costo (€)	
		Proteine	Carb.	Grassi		
		x_1	x_2	x_3	-1	
	Latte v_1	32	48	40	58	$= y_1$
	Hamburger v_2	112	0	91	139	$= y_2$
Alimenti	Tonno v_3	83	0	23	189	$= y_3$
	Patate v_4	45	500	0	98	$= y_4$
	Margarina v_5	126	95	262	159	$= y_5$
	Bisogno minimo -1	490	1841	392		
		u_1	u_2	u_3		

Fig. 8.1 La tabella di Tucker iniziale

		Sostanze (grammi)			Costo (€)
		Proteine	Carb.	Grassi	
		1.20	.08	0	-1
Latte	0	32	48	40	58 = 15
Hamburger	0	112	0	91	139 = 5
Alimenti Tonno	0	83	0	23	189 = 90
Patate	3.16	45	500	0	98 = 0
Margarina	2.76	126	95	262	159 = 0
-1		490	1841	392	748
		0	0	331	

Fig. 8.2 La tabella di Tucker finale

gli alimenti. In questo esempio comperiamo 3.16 sacchetti di patate e 2.76 vasetti di margarina e basta. Se facciamo così, avremo esattamente la giusta quantità di proteine (colonna uno), di carboidrati (colonna due) e 331 grammi di grassi in eccedenza (colonna tre). Il costo sarà di 7.48 dollari (colonna quattro). Poiché otteniamo queste risposte con il metodo del semplice, siamo sicuri che si tratta del *minor costo possibile*.

“Per controllare una colonna si moltiplica ogni suo elemento per il valore V e si fanno le somme. Se ci sono delle imprecisioni, sono dovute ad errori di arrotondamento del computer. Queste imprecisioni dovrebbero ricordarvi che l'accuratezza dei calcoli, nell'applicazione al computer del metodo del semplice, è molto importante.

“Il problema rimane ‘Come si arriva ai risultati?’ Li otteniamo con gli otti passi del metodo del semplice. Questa tecnica modifica la tabella di Tucker cambiando righe e colonne finché si arriva al risultato. Ad ogni passo si cambia una riga ed una colonna della tabella. Ciò significa che le nostre variabili al contorno si sposteranno. Le U e le V sopra e sotto e le X e Y ai lati hanno i valori della tabella modificata. Il prezzo totale appare nell'angolo a destra in basso nella tabella finale.

“Prendiamo i risultati della tabella finale e li scriviamo nei loro posti d'origine nella tabella iniziale per vedere qual è la soluzione.”

IL METODO DEL SIMPLESSO A OTTO PASSI

“1. Costruiamo una tabella di Tucker. Per il nostro problema sarà così:

	x_1	x_2	x_3	-1	
v_1	$t_{1,1}$	$t_{1,2}$	$t_{1,3}$	c_1	$= v_1$
v_2	$t_{2,1}$	$t_{2,2}$	$t_{2,3}$	c_2	$= v_2$
v_3	$t_{3,1}$	$t_{3,2}$	$t_{3,3}$	c_3	$= v_3$
v_4	$t_{4,1}$	$t_{4,2}$	$t_{4,3}$	c_4	$= v_4$
v_5	$t_{5,1}$	$t_{5,2}$	$t_{5,3}$	c_5	$= v_5$
	b_1	b_2	b_3		b_4
	u_1	u_2	u_3		

Fig. 8.3 Tabella di Tucker simbolica

“Se M è il numero di righe, ed N è il numero di colonne, in Basic la costruzione si effettuerà così:

```

1820   FOR I=1 TO M
1830     FOR J=1 TO N
1840       READ T(I,J)
1850     NEXT J
1860     READ C(I)
1870   NEXT I
1880   FOR J=1 TO N
1890     READ B(J)
1900   NEXT J
1910   DATA 32,48,40,58
1920   DATA 112,0,91,139
1930   DATA 83,0,23,189
1940   DATA 45,500,0,98
1950   DATA 126,95,262,159
1970 REM
1980   DATA 490,1841,392
1990 REM

```

“I $B(J)$ sono detti indicatori. Saranno molto importanti più avanti.

“2. Troviamo la colonna con l'indicatore positivo più grande e la chiamiamo $P2$, la colonna pivot. (Se nessun indicatore è positivo si è giunti alla soluzione e ci si ferma).

“In Basic si fa così:

```

2270 REM 2. TROVA LA COLONNA PIVOT O SI FERMA
2280 REM
2290   M1=0
2300   P2=0
2310 REM
2320   FOR J=1 TO N
2330     IF B(J)>M1 THEN 2340 ELSE 2360
2340     M1=B(J)
2350     P2=J
2360 REM
2370   NEXT J
2380   IF M1<=0 THEN 2390 ELSE 2400
2390   GOTO 2690
2400 REM
2410 REM

```

“Quando il programma ha terminato il passo 2, o si è arrivati ad un risultato o si è scelta una colonna pivot, P2. Il termine pivot è usato qui per denotare l'elemento che determinerà quale riga e quale colonna dovranno scambiarsi. Questo scambio è chiamato passo del pivot.

“3. Procedendo da T(I,P2) a T(M,P2) controlliamo la colonna pivot. Se tutti gli elementi sono negativi o nulli, il problema non ha soluzione. Troviamo la riga con T(I,P2) positivo il cui costo per unità (C(I)/T(I,P2)) è minore e la chiamiamo P1, la riga pivot.

```

2420 REM 3. TROVA LA RIGA PIVOT O SI FERMA
2430 REM
2440     M2=1000000
2450     P1=0
2460     FOR I=1 TO M
2470         S(I)=T(I,P2)
2480         IF T(I,P2)>.00001 THEN 2490 ELSE 2520
2490         IF C(I)/T(I,P2)<=M2 THEN 2500 ELSE 2520
2500         M2=C(I)/T(I,P2)
2510         P1=I
2520 REM
2530     NEXT I
2540     S(M+1)=B(P2)
2550     IF P1=0 THEN 2560 ELSE 2580
2560     PRINT"NON C'E' SOLUZIONE"
2570 REM
2580 REM
2590 REM
2600     P=T(P1,P2)
2610 REM

```

“Ora abbiamo trovato una riga pivot, P1, una colonna pivot, P2, e un elemento pivot, T(P1,P2). Assegnamo il valore dell'elemento pivot alla variabile chiamata P.

“4. Dividiamo la riga pivot per l'elemento pivot.

```

2770 REM 4. DIVIDE LA RIGA PIVOT PER L'ELEMENTO PIVOT
2780 REM
2790     FOR J=1 TO N
2800         T(P1,J)=T(P1,J)/P
2810     NEXT J
2820     C(P1)=C(P1)/P
2830 REM

```

“5. Prendiamo ogni altra riga e sottraiamo l'elemento della colonna pivot, T(I,P2), moltiplicato per la riga pivot.

```

2840 REM 5. RIADATTA LE RIGHE ALLA NUOVA RIGA PIVOT
2850 REM
2860     FOR I=1 TO M
2870         IF I<>P1 THEN 2880 ELSE 2930
2880         X=T(I,P2)
2890         FOR J=1 TO N

```

```

2900      T(I,J)=T(I,J)-X*T(P1,J)
2910      NEXT J
2920      C(I)=C(I)-X*C(P1)
2930 REM
2940      NEXT I
2950 REM
2960      X=B(P2)
2970      FOR J=1 TO N
2980          B(J)=B(J)-X*T(P1,J)
2990      NEXT J
3000      B(N+1)=B(N+1)-X*C(P1)
3010 REM

```

“6. Sostituiamo ad ogni elemento della colonna pivot, $T(i,P2)$, il negativo dell'elemento originale diviso per il pivot, $-T(i,P2)/P$, ed al pivot stesso il suo inverso, $1/P$.

```

3020 REM 6. RICOSTRUISCE LA COLONNA PIVOT
3030 REM
3040      FOR I=1 TO M
3050          T(I,P2)=-S(I)/P
3060      NEXT I
3070      B(P2)=-S(M+1)/P
3080      T(P1,P2)=1/P
3090 REM

```

“Nel nostro programma Basic abbiamo memorizzato una copia di scorta, $S(i)$, della nostra colonna pivot per potercene poi servire.

“7. Scambiamo gli indicatori orizzontali e verticali, le righe e le colonne. Se all'inizio sono così:



Fig. 8.4 Indicatori iniziali di riga (verticali) e colonna (orizzontali)

Possiamo, ad esempio, scambiare la riga quattro e la colonna due così:



Fig. 8.5 Indicatori di riga (verticali) e colonna (orizzontali) dopo uno scambio

```

3100 REM 7. CAMBIA GLI INDICATORI DI RISPOSTA
3110 REM
3120 X=H(P2)
3130 H(P2)=V(P1)
3140 V(P1)=X
3150 REM

```

“8. Ripetiamo i passi da 2 a 8 finché è possibile. Se eseguiamo effettivamente il programma in questo particolare caso, il ciclo verrà effettuato quattro volte prima di arrivare ad una soluzione. Pur essendo complicato, il metodo è molto più veloce di quello brutale che richiedeva 5000 tentativi.”

```

2000 REM CICLO PRINCIPALE
2010 REM
2020 FOR L=1 TO 10000
    *
    *
    *
3160 REM PASSO 8. RIPETE DAL PASSO 2 AL PASSO 8
3170 NEXT L
3175 PRINT"NON TROVA UNA RISPOSTA"
3180 STOP
3185 REM

```

I RISULTATI

“I risultati sono stampati dal programma. Non ci resta che disporli correttamente al bordo della nostra tabella iniziale. Le V e le U ci dicono molto di ciò che volevamo sapere. Le V indicano la quantità ed il genere di alimento da comperare. In questa dieta i risultati sono 3.16 sacchetti di patate e 2.76 vasetti di margarina, e nient'altro. Le U ci dicono che non avremo proteine né carboidrati in eccedenza, mentre ci saranno 331 grammi di grassi in più.

“Le Y sono lo scarto nei prezzi dei vari generi. Gli alimenti che abbiamo nella soluzione, patate e margarina, non hanno scarto nei loro prezzi. Avevamo un prezzo abbastanza basso da poter rientrare nella dieta. Y(1) a 15.5 significa che un quarto di latte dovrebbe scendere di 15.5 centesimi di prezzo per essere così conveniente da essere incluso nella nostra dieta. Una libbra di hamburger dovrebbe costare 5 centesimi di meno; una scatoletta di tonno 83 centesimi di meno.”

“La tabella relativa al problema della dieta può essere ridotta a questa (fig. 8.6) dove il problema è quanto di ogni riga utilizzare per soddisfare le richieste minime al minor costo totale.

		Costi
v_1		
v_2		
v_3		
v_4		
v_5		
Bisogno minimo		Costo minimo

Fig. 8.6 Tabella di Tucker ridotta per il problema del costo minore (Dieta)

“C’è un problema simile a quello della dieta, in cui la tabella è così:

	x_1	x_2	x_3	Capacità massima
Utili				Utile massimo

Fig. 8.7 Tabella di Tucker ridotta per il problema dell’utile massimo

“Si tratta del problema di quanto utilizzare di ogni colonna per rimanere nelle capacità massime ed ottenere il massimo utile totale.”

“Immaginate per un attimo di aver una macchina speciale per regolare la produzione agricola di un’isola. La nostra macchina ha tre misure; per le proteine, i carboidrati ed i grassi. Se fissiamo le proteine al valore 1 otteniamo:

32 mucche
 112 giovenche
 83 pesci
 45 ortaggi
 126 noci

e 49 unità di benefici politici.

“Se regoliamo a 1 la misura dei carboidrati, otteniamo:

48 mucche
 0 giovenche
 0 pesci
 500 ortaggi
 95 noci

e 184.1 unità di benefici politici.

“Se poniamo a 1 la misura dei grassi, abbiamo:

40 mucche
 91 giovenche
 23 pesci
 0 ortaggi
 262 noci

e 39.2 unità di benefici politici.

“Se vogliamo ottenere il massimo vantaggio politico aumentiamo il più possibile tutte le misure. Ma ciò in realtà non si può fare perché la nostra isola ha dei limiti. E sono piuttosto severi:

mucche	$< = 58$
giovenche	$< = 139$
pesci	$< = 189$
ortaggi	$< = 98$
noci	$< = 159$

“Così il nostro problema diventa come regolare le misure della nostra macchina in modo da ottenere il massimo beneficio politico, nel rispetto dei limiti imposti dalla condizione dell'isola e dalle caratteristiche stesse della macchina. Ecco come possiamo riassumere i nostri risultati.” (fig. 8.8)

“Naturalmente, se sappiamo come regolare le misure, sapremo anche quale dimensione può raggiungere il vantaggio politico totale. Ma come risolviamo questo problema?

“Le questioni emergenti sono parecchie. Quale metodo possiamo usare?

	Quantità			Limiti	
	#1	#2	#3	dell'isola	
	x_1	x_2	x_3	-1	
Mucche v_1	32	48	40	58	$= y_1$
Giovenche v_2	112	0	91	139	$= y_2$
Pesci v_3	83	0	23	189	$= y_3$
Ortaggi v_4	45	500	0	98	$= y_4$
Noci v_5	126	95	262	159	$= y_5$
Beneficio politico	49	184.1	39.2		
	u_1	u_2	u_3		

Fig. 8.8 Tabella di Tucker per il problema duale

Ci darà i risultati giusti? Quanto assomiglia a quello che abbiamo usato per il problema della dieta?

“La risposta a queste domande in effetti è piuttosto sorprendente: useremo il metodo del simplesso proprio come prima. Infatti, dato che le cifre del nuovo problema sono simili a quelle del problema della dieta, i risultati saranno ancora numeri di quel tipo, ma con un'interpretazione diversa. Regoleremo così le nostre quantità:

<i>Proteine</i>	<i>Carboidrati</i>	<i>Grassi</i>
1.19	.08	0

Il beneficio politico totale che otteniamo è:

$$1.19 \times 49 + .08 \times 184.1 + 0 \times 39.2 = 74.8$$

“Ci siamo già imbattuti in una delle più interessanti proprietà matematiche del metodo del simplesso. Per ogni problema di minimizzazione (il problema della dieta) c'è un problema di massimizzazione duale e, se esiste una soluzione, *entrambi i problemi hanno la stessa soluzione!*

“Nel nuovo problema le variabili al contorno hanno un'interpretazione diversa. Le X rappresentano il valore che devono assumere le misure perché la macchina agricola della nostra isola produca il massimo utile. Le Y sono variabili di scarto della soluzione. Scopriamo che non c'è posto per altri ortaggi e altre piante di noci, mentre ne abbiamo per circa 15 mucche, 5 giovenche e 90 pesci in più.

“Le V rappresentano i vari contributi alla soluzione. Gli ortaggi contribuiscono con 3.16 unità di beneficio politico ognuno. Ogni albero di no-

ce dà 2.76 unità di vantaggio politico. Gli altri settori danno un contributo, ma nella nostra soluzione hanno ancora una gran quantità di scarto, cosicché il loro apporto relativo alla soluzione è zero.

“Le U indicano il beneficio extra. Le misure 1 e 2 non ne hanno perché vengono utilizzate nella soluzione. I contributi alla soluzione spingono la misura 3 a produrre 33.1 unità di beneficio più di quanto potrebbe, cosicché è esclusa dalla soluzione. La misura 3 è posta a zero.”

APPLICAZIONI

“Questi problemi hanno molte applicazioni?” chiese un manager.

“Certo”, disse Anderson. “Ne abbiamo citate alcune all’inizio. Ecco cosa significa quel che abbiamo appena detto per alcune di esse.

“Per il problema della dieta: un sistema economico deve produrre determinate quantità di cibi, capi di vestiario e abitazioni. Tra settore pubblico e privato, tutti tre gli elementi vengono provvisti in diverse proporzioni. Il settore pubblico è più dispendioso di quello privato. Che quantità di ciascuno di questi beni è necessaria per soddisfare il fabbisogno al minor costo?

“Per il problema duale: coltivazioni diverse richiedono combinazioni diverse di lavoro e d’investimento di capitale e danno utili netti differenti. L’agricoltore è in grado d’investire solo una certa quantità di capitale e di fornire solo una quantità limitata di prestazioni lavorative. Che quantità e quali coltivazioni gli converrà piantare?

“Un altro problema duale: tre prodotti richiedono fasi di lavorazione in quattro macchine diverse. Ad ognuno di essi è associato un utile diverso. Le macchine sono a disposizione solo per un certo numero di ore alla settimana. Quali sono le quantità dei vari prodotti che massimizzano l’utile totale?”

“Grazie, sono esempi molto interessanti.”

“Sì, è vero. Il nostro semplice metodo è in grado di risolvere una gran varietà di problemi. Dopo pranzo vi mostrerò parecchi esempi, ora voglio presentarvi un limite del nostro metodo.”

UGUAGLIANZE

“Finora abbiamo sempre lavorato con sistemi ben definiti. Dicendo ben definiti intendiamo che la colonna a destra (i costi o i limiti superiori) contiene sempre numeri positivi. In effetti le cose non vanno proprio così.

“Perché dovremmo avere valori negativi in quella colonna? Consideriamo per un attimo la seconda limitazione del nostro problema duale. Non potevamo avere più di 139 giovenche. Possiamo esprimere questa limitazione guardando nella seconda riga della tabella di Tucker del problema e ricavandone la seguente equazione:

$$112 \times X_1 + 0 \times X_2 + 91 \times X_3 = < 139$$

Ciò significa che, qualsiasi siano i valori finali di X_1 , X_2 e X_3 , non potranno superare il limite dell'isola, 139 giovenche.

“Ma cosa accadrebbe se volessimo avere esattamente 139 giovenche? Come potremmo regolarci? La risposta consiste nell'aggiungere un'altra restrizione alla tabella. Lo faremo in tre passi.

“Il primo è la restrizione stessa. Se vogliamo 139 giovenche, ed abbiamo già specificato che la soluzione non deve essere superiore a 139, dobbiamo aggiungere che non dovrà neppure essere inferiore. Lo esprimiamo così:

$$112 \times X_1 + 0 \times X_2 + 91 \times X_3 > = 139$$

“Per poter inserire questa disuguaglianza nella tabella, dobbiamo invertirla, lasciandole però lo stesso significato:

$$-112 \times X_1 - 0 \times X_2 - 91 \times X_3 = < -139$$

“Il terzo passo consiste nell'introdurre questa equazione in fondo alla nostra tabella, così:” (fig. 8.9)

		x_1	x_2	x_3		
		·	·	·		
		·	·	·		
		·	·	·		
Noci v_5	126	95	262	98	=	y_5
Giovenche v_6	-112	0	-91	-139	=	y_6
Beneficio politico	49	184.1	39.2			
		u_1	u_2	u_3		

Fig. 8.9 Modifica alla tabella di Tucker

“Se risolviamo questo problema troviamo dei risultati leggermente diversi:

Quantità di proteine	Quantità di carboidrati	Quantità di grassi
1.24	.03	0.00

Abbiamo aumentato la quantità di proteine, X1, a sufficienza per avere esattamente 139 giovenche. La nostra nuova soluzione soddisfa anche le altre condizioni perché abbiamo ridotto la quantità di carboidrati.

“Ma ora torniamo al programma di cui parlavamo all'inizio. Nella colonna a destra c'è un numero negativo. Modifichiamo il nostro metodo del semplice in modo che ne tenga conto, che cerchi il numero negativo e, se lo trova, scelga in quella riga l'elemento pivot. Quando sono stati eliminati tutti i numeri negativi, si usa il metodo come prima. In Basic questa ricerca si esprime così:

```

2030 REM SUBROUTINE DI CONTROLLO DELLA TABELLA
2040 REM INPUT: B(),C(),H(),M,N,T(),V()
2050 REM OUTPUT: B(),C(),H(),T(),V()
2060 REM
2070 G#="OK"
2080 FOR I=M TO 1 STEP -1
2090 IF C(I)<-.00001 THEN 2110 ELSE 2130
2110 GOSUB 3620
2120 G#="RIPROVA"
2130 REM
2140 NEXT I
2150 IF G#="OK" THEN 2170
2160 GOTO 2070
2170 REM
2180 RETURN
2190 REM
2200 REM

```

“La scelta dell'elemento pivot invece:

```

3620 REM SUBROUTINE CHE RIELABORA LA TABELLA
3630 REM INPUT: B(),C(),H(),I,M,N,T(),V()
3640 REM OUTPUT: B(),C(),H(),T(),V()
3650 REM
3660 REM
3670 P1=I
3680 FOR J=1 TO N
3690 IF T(P1,J)<-.00001 THEN 3700 ELSE 3710
3700 GOTO 3760
3710 REM
3720 NEXT J
3730 PRINT"IL PROBLEMA MASSIMO NON HA SOLUZIONI POSSIBILI."
3740 PRINT"IL PROBLEMA MINIMO HA UNA SOLUZIONE INFINITA."
3750 STOP
3760 REM
3770 P2=J
3780 REM
3790 FOR I=1 TO M
3800 S(I)=T(I,P2)
3810 NEXT I
3820 S(M+1)=B(P2)
3830 REM

```

```

3840 M2=C(P1)/T(P1,P2)
3850 FOR I=P1 TO M
3860 IF T(I,P2)>.00001 THEN 3870 ELSE 3900
3870 IF (C(I)/T(I,P2))<=M2 THEN 3880 ELSE 3900
3880 P1=I
3890 M2=C(I)/T(I,P2)
3900 REM
3910 NEXT I
3920 P=T(P1,P2)
3930 GOSUB 2730
3940 RETURN
3950 REM
3960 REM

```

“Con queste modifiche il programma SEMPLISSO risolverà tutti i problemi di questo tipo che hanno una soluzione. Si noti che non dobbiamo preoccuparci di aggiungere altre colonne ai problemi ‘della dieta’. Tutti i valori negativi della riga in fondo vanno bene.”

 IL PROGRAMMA SEMPLISSO

```

1200 REM ***SEMPLISSO***
1210 REM
1220 REM QUESTO PROGRAMMA RISOLVE UN PROBLEMA DI PROGRAMMAZIONE LINEARE
1230 REM MEDIANTE UNA TABELLA DI TUCKER E IL METODO DEL SEMPLISSO.
1240 REM
1250 REM VARIABILI:
1260 REM B().....RIGA DEI BENEFICI (O BISOGNI)
1270 REM C().....COLONNA DELLE CAPACITA' (O COSTI)
1280 REM G#.....FLAG 'AVANTI'
1290 REM H().....GUIDA ALLA RISPOSTA ORIZZONTALE
1300 REM I,J.....VARIABILI INDICE DI CICLO
1310 REM L.....INDICE DI CICLO DEL PROGR. PRINCIPALE
1320 REM M1.....ELEMENTO MASSIMO
1330 REM M2.....ELEMENTO MINIMO
1340 REM N.....NUMERO DI RIGHE DELLA TABELLA
1350 REM N.....NUMERO DI COLONNE DELLA TABELLA
1360 REM P.....VALORE DELL'ELEMENTO PIVOT
1370 REM P1.....RIGA PIVOT
1380 REM P2.....COLONNA PIVOT
1390 REM S().....COPIA SALVATA DELLA COLONNA PIVOT
1400 REM T().....TABELLA DI TUCKER
1410 REM V().....GUIDA ALLA RISPOSTA VERTICALE
1420 REM X.....VARIABILE DI TEMPO
1430 REM
1440 REM PROGRAMMA PRINCIPALE
1450 REM
1460 GOSUB 1610
1470 GOSUB 2030
1480 GOSUB 2210
1490 GOSUB 3190
1500 STOP
1510 REM
1520 REM
1530 REM SUBROUTINE 'PASSO 1'
1540 REM INPUT: --
1550 REM OUTPUT: B(),C(),H(),M,N,T(),V()
1560 REM

```

```

1650 REM PREPARA LE GUIDE ALLA RISPOSTA E LA TABELLA DI TUCKER
1660 REM
1670 REM OGNI RIGA CON LE ULTIME COLONNE NEGATIVE DEVE TROVARSI IN FONDO
1680 REM ALLA TABELLA.
1690 REM
1700 DIM B(4),C(6),H(3),T(6,3),V(6)
1710 REM
1720 M=6
1730 N=3
1740 REM
1750 FOR I=1 TO M
1760 V(I)=I
1770 NEXT I
1780 FOR J=1 TO N
1790 H(J)=-J
1800 NEXT J
1810 REM
1820 FOR I=1 TO M
1830 FOR J=1 TO N
1840 READ T(I,J)
1850 NEXT J
1860 READ C(I)
1870 NEXT I
1880 FOR J=1 TO N
1890 READ B(J)
1900 NEXT J
1910 DATA 32,48,40,58
1920 DATA 112,0,91,139
1930 DATA 83,0,23,189
1940 DATA 45,500,0,98
1950 DATA 126,95,262,159
1960 DATA -112,0,-91,-139
1970 REM
1980 DATA 490,1841,392
1990 REM
2000 RETURN
2010 REM
2020 REM
2030 REM SUBROUTINE DI CONTROLLO DELLA TABELLA
2040 REM INPUT: B(),C(),H(),M,N,T(),V()
2050 REM OUTPUT: B(),C(),H(),T(),V()
2060 REM
2070 G$="OK"
2080 FOR I=M TO 1 STEP -1
2090 IF C(I)<-.00001 THEN 2110 ELSE 2130
2110 GOSUB 3620
2120 G$="RIPROVA"
2130 REM
2140 NEXT I
2150 IF G$="OK" THEN 2170
2160 GOTO 2070
2170 REM
2180 RETURN
2190 REM
2200 REM
2210 REM SUBROUTINE "PASSI 2-8"
2220 REM INPUT: B(),C(),H(),M,N,T(),V()
2230 REM OUTPUT: B(),C(),H(),T(),V()
2240 REM
2250 FOR L=1 TO 10000
2260 REM
2270 REM 2. TROVA LA COLONNA PIVOT O SI FERMA
2280 REM
2290 M1=0
2300 P2=0
2310 REM

```

```

2320     FOR J=1 TO N
2330         IF B(J)>M1 THEN 2340 ELSE 2360
2340             M1=B(J)
2350             P2=J
2360     REM
2370     NEXT J
2380     IF M1<=0 THEN 2390 ELSE 2400
2390         GOTO 2690
2400     REM
2410     REM
2420     REM 3. TROVA LA RIGA PIVOT O SI FERMA
2430     REM
2440         M2=1000000
2450         P1=0
2460         FOR I=1 TO M
2470             S(I)=T(I,P2)
2480             IF T(I,P2)>.00001 THEN 2490 ELSE 2520
2490             IF C(I)/T(I,P2)<=M2 THEN 2500 ELSE 2520
2500                 M2=C(I)/T(I,P2)
2510                 P1=I
2520     REM
2530     NEXT I
2540     S(M+1)=B(P2)
2550     IF P1=0 THEN 2560 ELSE 2580
2560         PRINT"NON C'E' SOLUZIONE"
2570     REM
2580     REM
2590     REM
2600     P=T(P1,P2)
2610     REM
2620     GOSUB 2730
2630     REM
2640     REM B. RIPETE DA 2 A B
2650     REM
2660     NEXT L
2670     PRINT"NON SI TROVA RISPOSTA"
2680     STOP
2690     REM
2700     RETURN
2710     REM
2720     REM
2730     REM SUBROUTINE "PASSI 4-7"
2740     REM INPUT: B(),C(),H(),M,N,P,P1,P2,S(),T(),V()
2750     REM OUTPUT: B(),C(),H(),T(),V()
2760     REM
2770     REM 4. DIVIDE LA RIGA PIVOT PER L'ELEMENTO PIVOT
2780     REM
2790     FOR J=1 TO N
2800         T(P1,J)=T(P1,J)/P
2810     NEXT J
2820     C(P1)=C(P1)/P
2830     REM
2840     REM 5. RIADATTA LE RIGHE ALLA NUOVA RIGA PIVOT
2850     REM
2860     FOR I=1 TO M
2870         IF I<>P1 THEN 2880 ELSE 2930
2880             X=T(I,P2)
2890             FOR J=1 TO N
2900                 T(I,J)=T(I,J)-X*T(P1,J)
2910             NEXT J
2920             C(I)=C(I)-X*C(P1)
2930     REM
2940     NEXT I
2950     REM
2960     X=B(P2)
2970     FOR J=1 TO N

```

```

2980      B(J)=B(J)-X*T(P1,J)
2990      NEXT J
3000      B(N+1)=B(N+1)-X*C(P1)
3010      REM
3020      REM 6. RICOSTRUISCE LA COLONNA PIVOT
3030      REM
3040      FOR I=1 TO M
3050          T(I,P2)=-S(I)/P
3060      NEXT I
3070      B(P2)=-S(M+1)/P
3080      T(P1,P2)=1/P
3090      REM
3100      REM 7. CAMBIA GLI INDICATORI DI RISPOSTA
3110      REM
3120          X=H(P2)
3130          H(P2)=V(P1)
3140          V(P1)=X
3150      REM
3160      RETURN
3170      REM
3180      REM
3190      REM SUBROUTINE CHE STAMPA LE RISPOSTE
3200      REM
3210          PRINT"MASSIMO BENEFICIO E MINIMO COSTO: ";-B(N+1)
3220          PRINT
3230      REM
3240          PRINT
3250          FOR J=1 TO N
3260              IF H(J)>0 THEN 3270 ELSE 3290
3270                  PRINT"V(";H(J);")=";-B(J)
3280                  GOTO 3310
3290      REM
3300                  PRINT"U(";-H(J);")=";-B(J)
3310      REM
3320          NEXT J
3330          PRINT"TUTTI GLI ALTRI V E U SONO UGUALI A ZERO"
3340      REM
3350          PRINT
3360          PRINT
3370          FOR I=1 TO M
3380              IF V(I)<0 THEN 3390 ELSE 3410
3390                  PRINT"X(";-V(I);")=";C(I)
3400                  GOTO 3430
3410      REM
3420                  PRINT"Y(";V(I);")=";C(I)
3430      REM
3440          NEXT I
3450          PRINT"TUTTI GLI ALTRI X E Y SONO UGUALI A ZERO"
3460          PRINT
3470          PRINT"LA TABELLA FINALE E'"
3480          FOR I=1 TO M
3490              FOR J=1 TO N
3500                  PRINT T(I,J),
3510              NEXT J
3520          PRINT C(I)
3530          NEXT I
3540          FOR J=1 TO N+1
3550              PRINT B(J),
3560          NEXT J
3570          PRINT
3580      REM
3590      RETURN
3600      REM
3610      REM
3620      REM SUBROUTINE CHE RIELABORA LA TABELLA
3630      REM      INPUT: B(),C(),H(),I,M,N,T(),V.)

```



```

3640 REM OUTPUT: B(),C(),H(),T(),V()
3650 REM
3660 REM
3670 P1=I
3680 FOR J=1 TO N
3690 IF T(P1,J)<-.00001 THEN 3700 ELSE 3710
3700 GOTO 3760
3710 REM
3720 NEXT J
3730 PRINT"IL PROBLEMA MASSIMO NON HA SOLUZIONI POSSIBILI,"
3740 PRINT"IL PROBLEMA MINIMO HA UNA SOLUZIONE INFINITA."
3750 STOP
3760 REM
3770 P2=J
3780 REM
3790 FOR I=1 TO M
3800 S(I)=T(I,P2)
3810 NEXT I
3820 S(M+1)=B(P2)
3830 REM
3840 M2=C(P1)/T(P1,P2)
3850 FOR I=P1 TO M
3860 IF T(I,P2)>.00001 THEN 3870 ELSE 3900
3870 IF (C(I)/T(I,P2))<=M2 THEN 3880 ELSE 3900
3880 P1=I
3890 M2=C(I)/T(I,P2)
3900 REM
3910 NEXT I
3920 P=T(P1,P2)
3930 GOSUB 2730
3940 RETURN
3950 REM
3960 REM
3970 END

```

MASSIMO BENEFICIO E MINIMO COSTO: 658.995

V(6) = 17.4263
V(5) = 19.379
U(3) = 3099.49
TUTTI GLI ALTRI V E U SONO UGUALI A ZERO

Y(1) = 16.9594
Y(2) = 0
Y(3) = 85.9911
Y(4) = 28.336
X(2) = .0276316
X(1) = 1.24107
TUTTI GLI ALTRI X E Y SONO UGUALI A ZERO

LA TABELLA FINALE E'

- .282707	- .505263	-66.6526	16.9594
1	0	0	0
.741071	0	-44.4375	85.9911
-5.51927	-5.26316	-876.694	28.336
.0118421	.0105263	1.68026	.0276316
-8.92857E-03	0	.8125	1.24107
-17.4263	-19.379	-3099.49	-658.995

Siccome il programma è piuttosto lungo, lo abbiamo suddiviso in subroutine. Ognuna di queste non è più lunga di una pagina, perciò la si può leggere tutta intera senza voltar pagina. Le varie subroutine del programma sono organizzate nel seguente modo:

```
PASSO 1
CONTROLLO TABELLA
  RIELABORAZIONE TABELLA
    PASSI 2-7
PASSI 2-8
  PASSI 2-7
STAMPA LE RISPOSTE
```

Si può notare che la subroutine PASSI 2-7 è richiamata in diversi punti del programma. L'istruzione RETURN contenuta in PASSI 2-7 è in grado di tornare proprio dopo il GOSUB che ha richiamato quella subroutine.

SIMPLESSO contiene due cicli che saranno ripetuti un numero di volte imprecisato. Il ciclo che effettua i passi 2-8, FOR L = 1 TO 1000, di norma sarà ripetuto solo poche volte per trovare una soluzione soddisfacente. Nella subroutine CONTROLLO TABELLA c'è un ciclo determinato da un'istruzione GOTO. Per uscire da un ciclo chiuso da GOTO è necessaria un'istruzione IF...THEN. In questo caso si tratta di IF G\$ = "OKAY" THEN 2170.

I passi 2 e 3 insegnano come trovare il massimo e il minimo di un insieme di numeri positivi. L'idea vi può tornare utile in altri programmi. Alcune delle istruzioni IF...THEN...ELSE del programma SIMPLESSO sono un po' curiose. Invece di

```
IF C(I)<0 THEN 2110 ELSE 2130
```

il programma dice:

```
IF C(I)<-.00001 THEN 2110 ELSE 2130
```

La seconda istruzione IF...THEN...ELSE testa se il valore è molto prossimo allo zero. Con questa condizione si rintracciano tutti i valori vicini a zero, sia positivi che negativi. Questi sono considerati 'zeri approssimati', zeri a meno di un piccolo errore di arrotondamento. Ogni valore compreso tra $-.00001$ e $+.00001$ è considerato zero. Anche questo particolare ci conferma che la precisione numerica, nel metodo del semplice, è molto importante.

NOTA: Prima di usare questo programma con dati reali, provatelo con un insieme di dati di cui conoscete i risultati e prevedetene i limiti e le possibilità.

ESERCIZI

1. Cambiate i dati di SIMPLESSO per risolvere il seguente problema di economia:

	<i>Case</i>	<i>Cibi</i>	<i>Vestiti</i>	<i>Costi</i>
Settore pubblico	66	80	70	150
Settore privato	50	125	100	120
Fabbisogno nazionale	100	200	160	

Quanto costerà il soddisfacimento del fabbisogno nazionale? Quale settore contribuirà in maggior misura? Quali saranno i beni in eccedenza?

2. Modificate la situazione economica del problema 1 ed affrontate il nuovo problema.
3. Considerate tre prodotti che richiedono le seguenti quantità di ore di lavoro su tre macchine diverse:

	<i>Prodotto A</i>	<i>Prodotto B</i>	<i>Prodotto C</i>	<i>Disponibilità di macchina</i>
Macchina 1	22	33	44	100
Macchina 2	30	25	50	80
Macchina 3	48	42	6	160
Utile/prodotto	20	10	30	

In che quantità riusciamo a realizzare i prodotti? Quale sarà l'utile complessivo? Quali macchine lavorano ai limiti della loro disponibilità?

4. Progettate da voi un problema della dieta e risolvetelo con il programma SIMPLESSO.
5. Progettate anche un problema duale e provate a risolverlo.
6. Modificate qualcuno dei precedenti programmi in modo che tenga conto di un vincolo d'uguaglianza e controllate se ha ancora una soluzione e, se è così, trovatela.
7. Rappresentate il problema di un dirigente d'azienda che ha appena saputo che il governo ha reso illegale uno dei suoi prodotti, si tratti di una sostanza stupefacente o di un'automobile pericolosa, e deve scegliere uno di tre prodotti per la sostituzione.
8. Definite il problema di un manager che deve scegliere di quale dei tre metodi di spedizione — trasporto aereo, per posta o su strada — servirsi per i suoi quattro prodotti.

PROBLEMI DI TRASPORTO

“Tra i problemi di cui ho parlato stamattina,” disse Helen Anderson iniziando la conferenza pomeridiana, “ce n’è uno famoso col nome di problema di trasporto. Può essere formulato in molti modi ma le caratteristiche fondamentali sono quelle mostrate nel prossimo esempio. “Consideriamo tre magazzini e tre negozi. I numeri vicino alle linee che li collegano indicano il costo unitario per il trasporto della merce dal magazzino al negozio.




Fig. 8.10 Tragitti dai magazzini ai negozi

“Possiamo rappresentare la stessa informazione con una tabella come questa:

Magazzini	Negozi		
	S ₁	S ₂	S ₃
120 W ₁	2	3	4
60 W ₂	12	10	11
200 W ₃	5	11	9
	100	200	80

Fig. 8.11 Tabella dei percorsi dai magazzini ai negozi

“Gli oggetti da trasportare sono dei cacciavite. Nel magazzino 1 vi sono 120 cacciavite, nel 2 ce ne sono 60, e nel magazzino 3 altri 200. Il negozio 1 ne ha richiesti 100, il 2 200 ed il negozio 3 invece 80.

“Il problema consiste nell’effettuare il trasporto nel modo più conveniente. Una prima soluzione è quella di usare i mezzi più economici. In questo modo portiamo la merce del magazzino 1 al negozio 1 così:




Fig. 8.12 Trasporto magazzino-negozio

e così via. Un secondo passo verso la soluzione è rappresentato dal pensare il problema come un problema 'della dieta' e nel costruire la relativa tabella di Tucker:

	W_1	W_2	W_3	S_1	S_2	S_3	Costo
V_{11}	1	0	0	1	0	0	2
V_{12}	1	0	0	0	1	0	3
V_{13}	1	0	0	0	0	1	4
V_{21}	0	1	0	1	0	1	12
V_{22}	0	1	0	0	1	0	10
V_{23}	0	1	0	0	0	1	11
V_{31}	0	0	1	1	0	0	5
V_{32}	0	0	1	0	1	0	11
V_{33}	0	0	1	0	0	1	9
Richieste minime	-1	120	60	200	100	200	80

Fig. 8.13 Tabella di Tucker per il problema dei trasporti

Se applichiamo a questo problema il programma SIMPLESSO otteniamo risultati come questi." (fig. 8.14)

"La soluzione prodotta dal programma non è del tutto esatta. Perché? Il programma vorrebbe che spedissimo altri 100 unità dal magazzino e ne portassimo 100 in più al negozio 1. Ma non si può trasferire merce che non c'è! Però guardate un attimo la tabella di Tucker. Il problema

	S_1	S_2	S_3	
W_1	(2) ¹⁰⁰	(3) ¹²⁰	4	120+100
W_2	12	(10) ⁶⁰	11	60
W_3	(5) ¹⁰⁰	(11) ²⁰	(9) ⁸⁰	200
	100	200	80	
	+100			

Fig. 8.14 Prima soluzione del problema dei trasporti

'della dieta' dà la 'soluzione più economica che soddisfa le richieste minime'. Proprio ciò che ha fatto il programma.

"In effetti il risultato è interessante perché ci indica che possiamo trasportare tutta la nostra merce più 100 unità risparmiando di più che non spendendo la quantità giusta. Converrebbe cercare attentamente nel magazzino 1 le unità richieste in sovrappiù.

"Ciò che volevamo, ma non abbiamo scritto nella nostra tabella di Tucker, è che la soluzione sia *esattamente corrispondente* alle richieste. Per rappresentare questa condizione sono necessarie altre sei colonne. Ma preferiamo aggiungere una sola colonna che richieda che il magazzino 1 sia utilizzato pienamente. Quindi aggiungiamo una colonna come questa:

W_1
-1
-1
-1
0
0
0
0
0
0
-120

Fig. 8.15 Restrizione ulteriore per il problema dei trasporti

"Con una tabella estesa, otteniamo i risultati che volevamo. È stata usata esattamente la merce fornita.

"Ci sono inoltre altre due interessanti applicazioni del problema dei trasporti", prosegue Anderson.

"Considerate questo grafo:




Fig. 8.16 Grafo illustrativo

Se lo si vuole percorrere al minimo costo, si può pensare di dover spedire attraverso la rete un singolo articolo. La tabella corrispondente è questa:

	A	B	C	D	Fine	
Inizio	5	-	6	-	-	1
A	0	2	-	-	-	1
B	-	0	-	3	-	1
C	-	2	0	3	-	1
D	-	-	-	0	4	1
	1	1	1	1	1	

Fig. 8.17 Tabella dei cammini sul grafo

Notate che i valori al contorno sono tutti uno. Ciò significa che in ogni nodo può entrare ed uscire una sola unità. Nel nodo di partenza non ne entra alcuna. E così nessuna unità esce dal nodo finale. Il trasporto da un nodo intermedio a se stesso non costa nulla. La soluzione è rappresentata da quei nodi che non si servono da soli. Prima di risolvere questo problema, ogni trattino verrà rimpiazzato da un numero grande. In questo esempio potrebbe essere 100.

“Scopriamo che la soluzione di questo problema è il cammino minimo attraverso il grafo. Se rendessimo negativi tutti i costi, allora la soluzione sarebbe il cammino massimo (il percorso critico) sul grafo.”

“Potremmo usare questo programma per trovare il percorso critico?” chiese uno studente.

“No, per questo particolare problema è più utile il programma CPM di cui forse vi hanno già parlato”, rispose la Anderson.

“La seconda applicazione interessante del problema dei trasporti è l'abbinamento tra uomini e lavori: il problema d'assegnazione. Pensiamo alle persone A, B e C e ai lavori 1, 2 e 3. Si potrebbe costruire una tabella con i costi d'addestramento delle varie persone ai diversi lavori. Nel nostro esempio, addestrare la persona B al lavoro 3 costerebbe 470 dollari. La tabella sarebbe così:

	Uomini			
	#1	#2	#3	
A	300	100	500	1
Lavori B	280	250	470	1
C	600	500	200	1
	1	1	1	

Fig. 8.18 Tabella per il problema d'assegnazione

“Anche qui, una persona sarà assegnata ad un solo lavoro e perciò i valori al contorno sono tutti uno.

“Infine, vorrei dirvi che c'è un altro metodo per risolvere i problemi di trasporto, che si serve essenzialmente della rappresentazione in tabella. Que-

sto metodo è chiamato *stepping-stone*. A chi è interessato ad approfondire l'argomento posso suggerire un paio di libri."

Quindi Anderson rispose ad alcune domande e dichiarò concluso il seminario.

ESERCIZI

1. Usate SIMPLESSO per risolvere i vari problemi di trasporto.

Il problema dell'ombrello: analisi delle decisioni

George Lee vide che il tempo era instabile e sorrise pensando che stava per fare una lezione sull'analisi delle decisioni e che il tempo si prestava perfettamente come esempio del problema che intendeva trattare. Era probabile che cominciasse a piovere e lui doveva decidere se prendere o no l'ombrello. Era estate e sarebbe stato veramente scomodo girare con l'ombrello per niente. Ed avrebbe dovuto portarlo in giro per tutto il giorno. Scrutò il cielo di nuovo e decise di lasciarlo a casa. Quando l'aula si riempì di studenti, fuori pioveva.

"L'argomento del seminario di questa settimana è l'analisi delle decisioni", cominciò Lee. "Oggi vorrei illustrarvi il concetto con un paio di esempi ed un programma per computer. Durante la settimana, potreste migliorare i programmi includendo nuovi elementi. Più tardi cercherò di farvi lavorare in prima persona su un piccolo problema.

"L'esempio di cui ci serviremo è il problema dell'ombrello: decidere se portare con sé o no l'ombrello quando il tempo è incerto. È importante che il cielo sia nuvoloso. Notate che non ho detto piovoso. Non sappiamo se poverà o no. Diciamo che la probabilità che piova è del 50 per cento.

"Il tempo instabile significa futuro incerto. Ecco come si può rappresentare la previsione con un diagramma" (fig. 9.1):




Fig. 9.1 Diagramma della previsione del tempo

“Se guadagnaste un decimo di dollaro se cominciasse a piovere ed un dollaro se uscisse il sole, il valore atteso relativo alla situazione sarebbe dato dal totale della posta per la possibilità di ottenerla, cioè:

$$\begin{aligned}\text{Valore atteso} &= (1/2)*10 + (1/2)*100 \\ &= 55\end{aligned}$$

“Se le previsioni del tempo danno un terzo di probabilità alla pioggia, il valore atteso diventa:

$$\begin{aligned}\text{Valore atteso} &= (1/3)*10 + (2/3)*100 \\ &= 70\end{aligned}$$

“Vediamo che se la probabilità si sposta verso un esito favorevole, il valore atteso della previsione sale. Ciò risponde al comune buon senso.

“Questo è importante perché influenza la nostra decisione. Supponiamo per un attimo di guadagnare 10 punti se, al momento in cui comincia a piovere, abbiamo l'ombrello, di perdere 20 punti se portiamo con noi l'ombrello e c'è il sole, di perderne 50 se non lo abbiamo e comincia a piovere e di guadagnarne 80 se lo lasciamo a casa ed il tempo diventa bello.

“Assegnati i valori alle varie possibilità, siamo in grado di prendere la decisione con cognizione di causa. La situazione è illustrata da questo diagramma.” (fig. 9.2)




Fig. 9.2 Albero delle decisioni per il problema dell'ombrello

“Abbiamo aggiunto la nostra decisione prima dei due eventi naturali. Il motivo del fatto che ci sono due eventi è che gli esiti hanno valori diversi:

$$\begin{aligned}\text{Ombrello} &= (1/2)*10 + (1/2)*-20 \\ &= -5\end{aligned}$$

$$\begin{aligned}\text{Niente ombrello} &= (1/2)*-50 + (1/2)*80 \\ &= 15\end{aligned}$$

“Quindici punti sono preferibili a -5, perciò decidiamo di non prendere l'ombrello se le previsioni danno alla pioggia il 50 per cento di probabilità.”

UN PROGRAMMA

“Ora consideriamo il problema dell'ombrello con lo scopo di scrivere un programma per computer. Notate che nell'albero delle decisioni ogni nodo è numerato.

“Una delle prime osservazioni che possiamo fare circa l'albero è che ci sono tre tipi di nodi: nodi di decisione (1), nodi dell'evento naturale (2,3) e nodi terminali (4,5,6,7). Compito del programma è di calcolare ogni nodo. Se il nodo è terminale, il suo valore è un numero. Se è un nodo di evento, dobbiamo valutarlo proprio come prima. Se è un nodo di decisione, dobbiamo scegliere l'alternativa migliore.

“Per calcolare i vari nodi cominciamo dalla fine dell'albero e risaliamo fino all'inizio. Per questo i nodi devono essere numerati. Possiamo leggere le variabili iniziali così:

```

225 REM LETTURA DEI DATI
230 REM
235   N9=7
240 REM
241 REM PER I NODI DI DECISIONE 'DEC' LEGGE IL NUMERO DI SOTTONODI E
242 REM IL NUMERO DI NODI DI OGNI SOTTONODO.
243 REM PER I NODI DI EVENTO 'NAT' LEGGE IL NUMERO DI SOTTONODI E LA
244 REM PROBABILITA' E IL NUMERO DI NODI DI OGNI SOTTONODO.
245 REM PER I NODI FINALI 'END' LEGGE IL VALORE DEL NODO.
250 REM
255   FOR I=1 TO N9
260     READ X,N(I)
265     IF N(I)='DEC' THEN 270 ELSE 295
270     READ N(I,0)
275     FOR J=1 TO N(I,0)
280       READ N(I,J)
285     NEXT J
290     GOTO 365
295 REM
300     IF N(I)='NAT' THEN 305 ELSE 330
305     READ N(I,0)
310     FOR J=1 TO N(I,0)

```

```

315             READ P(I,J),N(I,J)
320             NEXT J
325             GOTO 365
330 REM
335             IF N$(I)="END" THEN 340 ELSE 345
340             READ V(I)
345             GOTO 365
350 REM
355             PRINT"ERRORE NELLA LETTURA DEI DATI"
360             STOP
365 REM
370             NEXT I
375 REM
380             DATA 1,DEC,2,2,3
385             DATA 2,NAT,2,.5,4,.5,5
390             DATA 3,NAT,2,.5,6,.5,7
395             DATA 4,END,10
400             DATA 5,END,-20
405             DATA 6,END,-50

```

“Il programma assegna ad ogni nodo l'etichetta DEC, NAT o END. Ad ogni etichetta sono associate le relative informazioni: per un nodo di decisione, quanti nodi gli sono attaccati ed i relativi numeri; per i nodi degli eventi naturali, quanti sono i nodi connessi, le loro probabilità e i loro numeri; per i nodi terminali, il valore corrente.

“Nel nostro programma la fase di lettura dei dati è divisa in tre parti e così pure la fase di calcolo dei nodi dell'albero. Il valore dei nodi terminali è stato introdotto in input, perciò, nel calcolo, li saltiamo:

```

450             IF N$(I)="END" THEN 455 ELSE 460
455             GOTO 605
460 REM
465 REM

```

“Per quanto riguarda i nodi degli eventi naturali, si sommano i valori di tutti i nodi ad essi subordinati:

```

470             IF N$(I)="NAT" THEN 475 ELSE 510
475             S=0
480             FOR J=1 TO N(I,0)
485                 V=P(I,J)*V(N(I,J))
490                 S=S+V
495             NEXT J
500             V(I)=S
505             GOTO 605
510 REM
515 REM

```

“Infine, ecco la parte relativa ai nodi di decisione:

```

520             IF N$(I)="DEC" THEN 525 ELSE 590
525             M=-1E36
530             FOR J=1 TO N(I,0)
535                 V=V(N(I,J))
540                 IF V>M THEN 545 ELSE 555

```

```

545             M=V
550             NI=N(I,J)
555 REM
560 REM
565             NEXT J
570             V(I)=M
575             PRINT"LA DECISIONE AL NODO":I:"E' NODO":NI
580             PRINT"IL VALORE DELLA DECISIONE E'":V(I)
582             PRINT
585             GOTO 605
590 REM
592 REM

```

“Ricordiamo, come abbiamo già detto, che, per calcolare il nostro diagramma, dobbiamo procedere a ritroso. Inseriamo le nostre tre routine in un unico ciclo che ci permette di risalire l'albero:

```

FOR I=N9 TO 1 STEP -1
.
.
.
NEXT I

```

“Ad ogni passaggio per un nodo di decisione, annunciamo il risultato, e alla fine sapremo quale sarà la decisione giusta. Con questi commenti ricostruiremo la storia del nostro programma a partire dal fondo dell'albero fino alla decisione finale.

```

575             PRINT"LA DECISIONE AL NODO":I:"E' NODO":NI
580             PRINT"IL VALORE DELLA DECISIONE E'":V(I)
582             PRINT

```

“Se mettiamo assieme i vari pezzi, il programma darà:

```

LA DECISIONE AL NODO 1 E' NODO 3
IL VALORE DELLA DECISIONE E' 15

```

```

VALORI DEI NODI
1 15           2 -5           3 15           4 10           5 -20
6 -50         7 80

```

“Il risultato conferma i nostri calcoli fatti a mano.”

IL PROGRAMMA DECISIONE

```

100 REM ***DECISIONE***
105 REM
110 REM SCEGLIE L'ALTERNATIVA MIGLIORE MEDIANTE UN ALBERO DELLE DECISIONI.
115 REM
150 REM VARIABILI:
155 REM I,J.....VARIABILI INDICE

```

```

160 REM M.....VARIABILE MASSIMA
165 REM N$(.....TIPO DI NODO
170 REM N1.....NUMERO DI NODI SCELTO
175 REM N(I,0).....NUMERO DI NODI LEGATI A I
180 REM N(I,J).....NUMERO DI NODI LEGATI
185 REM P().....PROBABILITA' DI NODO
190 REM S.....VARIABILE SOMMA
195 REM V().....VARIABILE DEL VALORE TEMPO
200 REM V().....VALORE DI UN NODO
205 REM X().....VARIABILE DI TEMPO
210 REM
215 DIM N$(24),N(24,10),P(24,10),V(24)
220 REM
225 REM LETTURA DEI DATI
230 REM
235 N9=7
240 REM
241 REM PER I NODI DI DECISIONE 'DEC' LEGGE IL NUMERO DI SOTTONODI E
242 REM IL NUMERO DI NODI DI OGNI SOTTONODO.
243 REM PER I NODI DI EVENTO 'NAT' LEGGE IL NUMERO DI SOTTONODI E LA
244 REM PROBABILITA' E IL NUMERO DI NODI DI OGNI SOTTONODO.
245 REM PER I NODI FINALI 'END' LEGGE IL VALORE DEL NODO.
250 REM
255 FOR I=1 TO N9
260 READ X,N$(I)
265 IF N$(I)="DEC" THEN 270 ELSE 295
270 READ N(I,0)
275 FOR J=1 TO N(I,0)
280 READ N(I,J)
285 NEXT J
290 GOTO 365
295 REM
300 IF N$(I)="NAT" THEN 305 ELSE 330
305 READ N(I,0)
310 FOR J=1 TO N(I,0)
315 READ P(I,J),N(I,J)
320 NEXT J
325 GOTO 365
330 REM
335 IF N$(I)="END" THEN 340 ELSE 345
340 READ V(I)
345 GOTO 365
350 REM
355 PRINT"ERRORE NELLA LETTURA DEI DATI"
360 STOP
365 REM
370 NEXT I
375 REM
380 DATA 1,DEC,2,2,3
385 DATA 2,NAT,2,.5,4,.5,5
390 DATA 3,NAT,2,.5,6,.5,7
395 DATA 4,END,10
400 DATA 5,END,-20
405 DATA 6,END,-50
410 DATA 7,END,80
415 REM
420 REM CALCOLO DEI NODI
425 REM
430 REM CALCOLA IL VALORE TOTALE DI UN NODO 'NAT', LA SCELTA MIGLIORE DI
435 REM UN NODO 'DEC' E SCAVALCA I NODI 'END'.
440 REM
445 FOR I=N9 TO 1 STEP -1
450 IF N$(I)="END" THEN 455 ELSE 460
455 GOTO 605
460 REM
465 REM

```

```

470      IF N$(I)="NAT" THEN 475 ELSE 510
475      S=0
480      FOR J=1 TO N(I,0)
485          V=F(I,J)+V(N(I,J))
490          S=S+V
495      NEXT J
500      V(I)=S
505      GOTO 605
510 REM
515 REM
520      IF N$(I)="DEC" THEN 525 ELSE 590
525      M=-1E36
530      FOR J=1 TO N(I,0)
535          V=V(N(I,J))
540          IF V>M THEN 545 ELSE 555
545              M=V
550              N1=N(I,J)
555 REM
560 REM
565      NEXT J
570      V(I)=M
575      PRINT"LA DECISIONE AL NODO":I:"E' NODO":N1
580      PRINT"IL VALORE DELLA DECISIONE E'":V(I)
582      PRINT
585      GOTO 605
590 REM
592 REM
595      PRINT"ERRORE NEL CALCOLO DELL'ALBERO"
600      PRINT
605 REM
610      NEXT I
615 REM
620      PRINT"VALORI DEI NODI"
625      FOR I=1 TO N9
630          PRINT I;V(I),
635      NEXT I
640 REM
645      END

```

Il programma si serve della variabile stringa N\$(I) per decidere che tipo di nodo sta trattando. In questi casi le variabili stringa offrono un paio di vantaggi rispetto a quelle numeriche. Consentono di documentare le istruzioni DATA e di scrivere le IF...THEN...ELSE in modo che la funzione del programma sia chiara in ogni suo punto.

Il programma è completamente dipendente dalla numerazione dei nodi. Ogni successore dev'essere calcolato prima del nodo padre. Entrambi i cicli "lettura dei dati" e "calcolo dei nodi" hanno dispositivi di controllo per l'identificazione dei nodi non corretti. Ci assicuriamo così che il programma si fermi se si trova in questa eventualità.

Il nodo di decisione cerca di individuare il massimo valore, M, delle scelte possibili. Ogni volta la variabile M inizia al valore -1E36, meno dieci alla 36-esima potenza, il numero più piccolo che si possa rappresentare. Così ad M verrà assegnato il valore della prima scelta e poi quello di ogni scelta successiva che sia maggiore.

ESERCIZI

1. Portate a .33 la probabilità che inizi a piovere (ed a .67 la probabilità che esca il sole). Rieseguite DECISIONE ed esaminate attentamente i risultati. In che modo le probabilità influenzano le vostre decisioni? C'è un punto nel quale non avete una decisione migliore?
2. Modificate DECISIONE perché rintracci eventuali errori nella sequenza dei nodi. Cominciate con l'inizializzare tutte le V a -.11111 all'inizio del programma. Nel ciclo di calcolo dei nodi dell'albero, inserite un controllo IF V(I) = -.11111 ... Quando c'è un errore stampate un messaggio d'errore e fermate il programma.
3. Confrontate i programmi DECISIONE e RISCHIO (cap. 4). Quando li usereste?

UNA DECISIONE DI MARKETING

"Il nostro prossimo esempio sarà un po' più impegnativo per il computer e lascerà intravedere meglio le potenzialità dell'analisi delle decisioni", disse Lee.

"Un responsabile di marketing ha la possibilità di mettere sul mercato un nuovo prodotto o di continuare con la produzione solita. In una ricerca sulle alternative possibili, è arrivato ad un albero delle decisioni come questo" (fig. 9.3).

"Il costo di lancio del nuovo prodotto è di 50 000 dollari. L'utile finale atteso, se il mercato è favorevole, è di 80 000 dollari. La produzione corrente garantisce 200 000 dollari. La stima delle possibilità di soccorso del nuovo prodotto è di 3/4, il 75 per cento.

"Potremmo trattare il problema col nostro programma, ma non abbiamo ancora descritto completamente le possibilità del manager. Potrebbe effettuare una ricerca di mercato ed avere la certezza al 90 per cento sulla risposta del mercato al nuovo prodotto. Ora le sue possibilità sono queste." (fig. 9.4)




Fig. 9.3 Albero delle decisioni per il nuovo prodotto




Fig. 9.4 Albero delle decisioni per il nuovo prodotto con ricerca di mercato

“Traducendo il diagramma in dati di input abbiamo:

```

380 DATA 1,DEC,2,2,3
382 DATA 2,DEC,2,4,5
384 DATA 3,NAT,2,.75,6,.25,7
386 DATA 4,NAT,2,.75,8,.25,9
388 DATA 5,END,200
390 DATA 6,DEC,2,10,11
392 DATA 7,DEC,2,12,13
394 DATA 8,END,280
396 DATA 9,END,150
402 DATA 10,NAT,2,.9,14,.1,15
404 DATA 11,END,200
406 DATA 12,NAT,2,.1,16,.9,17
408 DATA 13,END,200
410 DATA 14,END,280
412 DATA 15,END,150
414 DATA 16,END,280
416 DATA 17,END,150
418 REN

```

Eseguendo il programma si ha:

LA DECISIONE AL NODO 7 E' NODO 13
IL VALORE DELLA DECISIONE E' 200

LA DECISIONE AL NODO 6 E' NODO 10
IL VALORE DELLA DECISIONE E' 267

LA DECISIONE AL NODO 2 E' NODO 4
IL VALORE DELLA DECISIONE E' 247.5

LA DECISIONE AL NODO 1 E' NODO 3
IL VALORE DELLA DECISIONE E' 250.25

VALORI DEI NODI

1	250.25	2	247.5	3	250.25	4	247.5	5	200
6	267	7	200	8	280				
9	150	10	267	11	200	12	163	13	200
14	280	15	150	16	280				
17	150								

“Così il nostro manager si convince che il test di mercato è una buona idea. Il gruppo per la ricerca di mercato valuta in 1000 dollari la spesa per il test. Il responsabile del marketing riesamina i suoi risultati. La sua decisione senza test vale 247 500 dollari; con il test invece vale 250 250 dollari. Il test gli consente di guadagnare 2750 dollari, perciò decide di effettuarlo.”
“Questo esempio fornisce un modo per stabilire se un'informazione ulteriore è proficua?” chiese uno degli studenti. “Sembra di sì.”
“Sì, è così. In effetti questo esempio è caro agli analisti di mercato.”

ESERCIZI

1. Se la ricerca richiede 3000 dollari, che percentuale di certezza ci vuole perché risulti vantaggiosa?
2. Ai costi di ricerca più alti, quali modifiche ai risultati la renderebbero vantaggiosa?

UN GIOCHETTO FINALE

“Nel problema dell'ombrello in verità non ci era molto comodo avere valori numerici per i vari risultati delle decisioni. La questione è come assegnare valori migliori ai risultati.

“Un modo migliore consisterebbe nell'assegnare zero al risultato peggiore e uno a quello migliore. Quindi potremmo assegnare una certa frazione ad ognuno di quelli intermedi. Nel nostro esempio la soluzione peggiore era 'pioggia senza ombrello', e le abbiamo assegnato il valore zero, 'Sole senza ombrello' era la migliore, e il suo valore è dunque 1. Il problema diventa più complesso nei casi 'pioggia e ombrello' e 'sole e ombrello'.

“Ora facciamo un giochetto con i nodi terminali. Per ottenere il valore 'pioggia e ombrello', chiediamo di quale tipo di possibilità del miglior risultato abbiamo bisogno per scambiare il nostro risultato attuale con una decisione di rischiare ancora. Bene, avere l'ombrello quando piove è certo molto meglio di 'pioggia senza ombrello', ma non è meglio di 'sole senza ombrello'. Decidiamo che se avessimo una probabilità del 65 per cento per il miglior esito, potremmo deciderci a rischiare ancora.

“Avere l'ombrello quando c'è il sole non è piacevole, perciò avremmo bisogno di un 25 per cento di probabilità del miglior risultato per metterci a rischiare di nuovo.

“In un problema più complicato ripeteremmo questo giochetto per ogni altro evento compreso tra il migliore ed il peggiore. Al termine di questo esercizio, abbiamo un valore chiamato 'utilità' del risultato. Un interessante effetto collaterale di questa utilità è che tiene conto delle stime di rischio ad ogni evento. Nel prendere le nostre decisioni abbiamo dunque solo bisogno di scegliere l'utilità maggiore. Il rischio è già stato calcolato.

“I valori di utilità per il nostro problema dell'ombrello sono:”

— Pioggia e ombrello	.65
— Sole e ombrello	.25
— Pioggia senza ombrello	0.00
— Sole senza ombrello	1.00

Quella sera, dopo la lezione, quando sali sul treno per tornare a casa George Lee era bagnato fino alle ossa.

ESERCIZI

1. Riconsiderate il problema dell'ombrello con i valori di utilità. Le decisioni cambiano?
2. Fate lo stesso con il problema di marketing. Assegnate i valori di utilità in stretto accordo con il valore monetario del risultato.
3. Supponete di essere licenziato se, con la vostra decisione di marketing, perdetevi più di 60 000 dollari. Rifate la stima dei valori d'utilità alla luce di questa informazione. I nuovi valori che ottenete modificano la vostra decisione?

Previsioni: vendite del prossimo mese, riunioni della prossima settimana

Peter Bates sospirò. Dopo aver combattuto per mezz'ora con un testo di statistica, trovò ciò che cercava: l'equazione della miglior interpolazione tra un insieme di punti. L'equazione era divisa in due pezzi, M, l'inclinazione della retta, e B, il punto dove la linea incrociava l'asse delle Y. L'inclinazione era data da:

$$M = \frac{\sum_{i=1}^n (Y_i - \bar{Y})(X_i - \bar{X})}{\sum_{i=1}^n (X_i - \bar{X})^2}$$

dove X_i , Y_i sono le coordinate dei punti i e dove \bar{X} è la media degli X_i e \bar{Y} è la media degli Y_i .

L'intersezione B era data da una formula più semplice:

$$B = \bar{Y} - M \cdot \bar{X}$$

L'equazione della curva che interpola i punti nel modo migliore era:

$$Y = M \cdot X + B$$

In questa equazione Bates poteva inserire nuovi valori X, come i dati del mese seguente, e ricavare il valore Y che sarebbe la proiezione della miglior interpolazione relativa alle vendite del mese seguente.

Per esempio, i primi otto dati delle vendite mensili erano 27, 24, 21, 25, 30, 37, 35, 32. L'equazione della miglior interpolazione era $Y = 2.1 X + 18.7$ dove X era il numero del mese. Per trovare cosa potrebbe accadere nel nono mese, Bates inserì 9 nella sua equazione ed ottenne 37.8.

Guardando il risultato, decise che avrebbe voluto spesso le proiezioni per i periodi seguenti. Inoltre, gli sarebbe piaciuto che i risultati fossero rappresentati graficamente. Così scrisse un piccolo programma a questo scopo. Si convinse che la sua linea retta era una interpretazione che diventava sempre meno buona con l'estendersi della previsione ma che per brevi periodi andava bene. Alla fine, i risultati erano come quelli che seguono.

 IL PROGRAMMA REGRESSIONE

```

100 REM ***REGRESSIONE***
110 REM
120 REM CALCOLA LA MIGLIOR RETTA PASSANTE PER UN INSIEME DI PUNTI.
130 REM IN QUESTO ESEMPIO I PUNTI FORMANO UNA SERIE TEMPORALE.
140 REM
160 REM LEGGENDO SIA X(I) CHE Y(I) OGNI INSIEME DI PUNTI PUO'
170 REM ESSERE APPROSSIMATO DA UNA RETTA.
180 REM
190 REM PER PUNTI DIVERSI SI DEVE CAMBIARE IL GRAFICO.
200 REM
210 REM
220 REM VARIABILI:
230 REM B.....L'INTERCETTA Y DELLA MIGLIOR RETTA DI APPROSSIMAZIONE
240 REM G*(I).....LA STRINGA DI CARATTERI CHE DISEGNA LA RETTA
250 REM I,J.....VARIABILI INDICE
260 REM L.....L'INCLINAZIONE DELLA RETTA SUL GRAFICO
270 REM M.....PENDENZA DELLA MIGLIOR RETTA DI APPROSSIMAZIONE
280 REM N.....NUMERO DI PUNTI
290 REM S1,S2,S3,S4.....VARIABILI SOMMA
300 REM X(),Y().....COORDINATE DEI PUNTI
310 REM X.....MEDIA DELLE COORDINATE X
320 REM Y.....MEDIA DELLE COORDINATE Y
330 REM
340 REM DIMENSIONI:
350 DIM G*(50),X(12),Y(12)
352 REM
355 REM VALORI INIZIALI E DATI
360 FOR I=1 TO 40
370   G*(I)=" "
380 NEXT I
390 REM
400 N=8
410 DATA 22,24,21,25,30,37,35,32
420 REM
425 REM CALCOLA LA MIGLIOR RETTA DI APPROSSIMAZIONE
430 S1=0
440 S2=0
450 FOR I=1 TO N
460   X(I)=I
470   READ Y(I)
480   S1=S1+X(I)
490   S2=S2+Y(I)
500 NEXT I
510 REM
  
```


```

520 X=S1/N
530 Y=S2/N
540 REM
550 S3=0
560 S4=0
570 FOR I=1 TO N
580 S3=S3+(Y(I)-Y)*(X(I)-X)
590 S4=S4+(X(I)-X)*(X(I)-X)
600 NEXT I
610 REM
620 M=S3/S4
630 B=Y-M*X
640 PRINT"LA MIGLIOR RETTA DI APPROSSIMAZIONE E' Y=";M;"X";B
650 PRINT
660 REM
662 REM DISEGNA LA MIGLIOR RETTA (+) E I PUNTI (*).
664 REM PROIETTA LA LINEA OLTRE I PUNTI.
668 REM
670 PRINT"L'ASSE DELLE X VA VERSO IL BASSO, L'ASSE DELLE Y VERSO DESTRA"
680 PRINT
690 REM
700 PRINT"0          10          20          30          40 PUNTI"
720 PRINT"+-----+-----+-----+-----+ Y, X(I), Y(I)"
740 FOR I=1 TO N*4
744 X(I)=I
750 L=M*X(I)+B
760 G$(L)="+"
765 IF I<=N THEN 770 ELSE 775
770 G$(Y(I))="*"
775 REM
780 PRINT"I"
790 PRINT"I"
800 PRINT"+";
810 FOR J=1 TO 40
820 PRINT G$(J);
830 G$(J)=" "
840 NEXT J
845 IF I<=N THEN 850 ELSE 855
850 PRINT TAB(46):X(I);" "Y(I)
852 GOTO 859
855 REM
856 PRINT TAB(46):L
859 REM
860 NEXT I
870 PRINT"X"
872 PRINT
874 PRINT"*--- PUNTO"
876 PRINT"+--- RETTA DI APPROSSIMAZIONE"
880 END

```

LA MIGLIOR RETTA DI APPROSSIMAZIONE E' $Y = 2.11905 X + 18.714$

L'ASSE DELLE X VA VERSO IL BASSO, L'ASSE DELLE Y VERSO DESTRA



*-- PUNTO

+-- RETTA DI APPROSSIMAZIONE

Il programma trova la miglior interpolazione servendosi della formula che Bates trovò nel libro di statistica. L'espressione complessa

$$M = \frac{\sum_{i=1}^n (Y_i - \bar{Y})(X_i - \bar{X})}{\sum_{i=1}^n (X_i - \bar{X})^2}$$

si traduce in sole otto righe di Basic:

```

550      S3=0
560      S4=0
570      FOR I=1 TO N
580          S3=S3+(Y(I)-Y)*(X(I)-X)
590          S4=S4+(X(I)-X)*(X(I)-X)
600      NEXT I
610      REM
620      M=S3/S4

```

La routine grafica disegna sia i punti che la retta. Prima vengono stampati i punti e poi la linea.

Il programma carica una linea di 40 caratteri servendosi della variabile stringa con indici G\$(I). Quando i caratteri giusti sono stati inseriti nelle rispettive locazioni di G\$(I), tutta la linea viene stampata, poi cancellata, ed il processo ripetuto.

L'approssimazione di un insieme di punti con una linea è anche chiamata regressione lineare. Il metodo utilizzato in REGRESSIONE è detto "regressione lineare dei minimi quadrati".

ESERCIZI

1. Utilizzate REGRESSIONE con un vostro insieme di dati.
2. Modificate REGRESSIONE per leggere X(I), Y(I) ed eseguitelo con un diverso insieme di punti. Cambiate la routine grafica come volete voi.

PREVISIONE DEI MESI SEGUENTI

Quando aveva bisogno di previsioni dettagliate sulle vendite di una certa gamma di prodotti, Bates si serviva sempre di un modello econometrico. Il modello era l'equivalente del programma REGRESSIONE in più dimensioni. Approssimava un insieme di punti rappresentati da più variabili indipendenti (vari tipi di X) con una superficie piana, per giungere alla previsione di Y.

Bates sapeva che ottenere una proiezione dal modello econometrico richiedeva tempo e denaro. Pensò che sarebbe stato abbastanza facile dare un rapido sguardo ai dati stessi e vedere se proprio c'era bisogno del modello econometrico.

Ciò che veramente gli interessava sapere erano le vendite dei mesi seguenti. Quando approssimò con una linea le vendite degli ultimi dodici mesi, la sua linea gli apparve come una leggera salita. Quando limitò la linea a nove e poi a sei mesi, trovò previsioni a breve termine sempre più precise che trascuravano la tendenza di lungo periodo.

Se da un lato le previsioni cambiavano rapidamente quando l'orizzonte della proiezione diminuiva da 12 a 2, allora Bates si accorgeva che accadeva qualcosa di significativo e pensava di doversi servire di un modello più

grande. Dall'altro, se i suoi valori per il mese oggetto della proiezione si accumulavano attorno ad un valore centrale quando la proiezione si restringeva da 12 mesi a 2, allora poteva capire da solo che la proiezione andava bene.

Il programma per fare queste proiezioni ripeteva più volte la parte centrale di REGRESSIONE per ottenere una serie di proiezioni dalla linea a 12 mesi fino a quella a 2 mesi.


Modificò il suo vecchio programma perché tenesse conto del numero base di mesi ed effettuasse la previsione per il mese seguente. Aggiunse gli ultimi quattro mesi dell'ultimo anno ai suoi dati: 30, 27, 20, 24. Rappresentò la base di previsione con una variabile chiamata F ed approntò il ciclo FOR $F = 12$ TO 2 STEP -1. I risultati sono quelli che seguono. Sembravano scendere verso un valore minore di 29. Tuttavia, il cambiamento improvviso di direzione rendeva le ultime previsioni un po' problematiche, perciò Bates decise di avere una proiezione completa di ciò che stava per accadere dall'ufficio statistiche. Scopri così che le vendite del prossimo mese sarebbero state di 28.8, in accordo con la sua proiezione.

Bates tenne conto di quel valore nel suo piano di richiesta del materiale per avere delle scorte in quantità adatta alle esigenze del mese seguente. (Si veda la discussione sulla domanda del capitolo 7.) A sua volta il piano avrebbe costruito una stima per il modello finanziario del flusso di cassa del mese seguente.

PREVISIONI: VENDITE DEL PROSSIMO MESE

CON UNA BASE DI PREVISIONE DI 12 MESI, LA
MIGLIOR RETTA DI APPROSSIMAZIONE È $Y = .87063 X + 21.5909$

L'ASSE DELLE X VA VERSO IL BASSO, L'ASSE DELLE Y VERSO DESTRA



*-- PUNTO
+-- RETTA DEL MESE 12

BASE (MESI)	PREVISIONE	CARATTERI
12	32.9091	A
11	34.6909	1
10	36.4	0
9	36.6111	9
8	37.7857	8
7	38.2857	7
6	39.2	6
5	37.5	5
4	34.5	4
3	29.6667	3
2	29	2

NEL GRAFICO ALCUNI PUNTI POSSONO RISULTARE SOVRAPPosti.

IL PROGRAMMA PROSMESE

```

100 REM ***PROSMESE***
105 REM
110 REM CALCOLA LE PREVISIONI PER IL PROSSIMO MESE DALLE RETTE CHE
115 REM APPROSSIMAND GLI ULTIMI 12 MESI, GLI ULTIMI 11, GLI ULTIMI 10 E
120 REM COSI' VIA FINO AGLI ULTIMI 2 MESI.
125 REM
150 REM IN QUESTO ESEMPIO I PUNTI FORMANO UNA SERIE TEMPORALE, MA
155 REM LEGGENDO X(I) E Y(I) OGNI INSIEME DI PUNTI SI PUO' APPROSSIMARE
160 REM CON UNA RETTA.
165 REM
170 REM PER PUNTI DIVERSI SI DEVE CAMBIARE IL GRAFICO.
175 REM
185 REM VARIABILI:
190 REM B().....L'INTERCETTA Y SULLA MIGLIOR RETTA DI APPROSS.
195 REM F.....LA BASE DI PREVISIONE IN MESI
200 REM F9.....LA BASE DI PREVISIONE STAMPATA
205 REM G*(I).....STRINGA DI CARATTERI CHE DISEGNA UNA RETTA
215 REM I,J.....VARIABILI INDICE
220 REM L.....LOCAZIONE DELLA RETTA SUL GRAFICO
225 REM M().....PENDENZA DELLA MIGLIOR RETTA DI APPROSS.
230 REM N.....NUMERO DI PUNTI
235 REM P().....CARATTERI PER I PUNTI DI PREVISIONE
240 REM S1,S2,S3,S4.....VARIABILI SOMMA DIVERSE
245 REM X(),Y().....COORDINATE DEI PUNTI
250 REM X.....MEDIA DELLE COORDINATE X
255 REM Y.....MEDIA DELLE COORDINATE Y
260 REM
265 REM
270 REM DIMENSIONI:
275 DIM B(25),G*(50),M(25),P*(12),P(12),X(25),Y(25)
280 REM
285 REM VALORI INIZIALI E DATI
290 REM
295 F9=12
300 FOR I=1 TO 40
305 G*(I)=" "
310 NEXT I
315 FOR I=2 TO 12
320 READ P*(I)
325 NEXT I
330 DATA "2","3","4","5","6","7","8","9","0","1","A"
335 REM
340 N=12
345 FOR I=1 TO N
350 X(I)=I
355 READ Y(I)
360 NEXT I
365 X(N+1)=N+1
370 DATA 30,27,20,24
375 DATA 22,24,21,25,30,37,35,32
380 REM
385 REM CALCOLA LE MIGLIORI RETTE DI APPROSSIMAZIONE.
390 FOR F=12 TO 2 STEP -1
395 S1=0
400 S2=0
405 FOR I=(N-F+1) TO N
410 S1=S1+X(I)
415 S2=S2+Y(I)
420 NEXT I
425 REM

```

```

430      X=S1/F
435      Y=S2/F
440  REM
445      S3=0
450      S4=0
455      FOR I=(N-F+1) TO N
460          S3=S3+(Y(I)-Y)*(X(I)-X)
465          S4=S4+(X(I)-X)*(X(I)-X)
470      NEXT I
475  REM
480      M(F)=S3/S4
485      B(F)=Y-M(F)*X
490      P(F)=M(F)*X(N+1)+B(F)
495  NEXT F
500  PRINT
505  REM
510  REM DISEGNA IL GRAFICO PRIVILEGIATO (VEDI F9)
515  REM
520  PRINT"CON UNA BASE DI PREVISIONE DI"+F9+"MESI, LA"
525  PRINT"MIGLIOR RETTA DI APPROSSIMAZIONE E' Y=";M(F9);"X+";B(F9)
530  PRINT
535  REM
540  REM DISEGNA LA MIGLIOR RETTA DI APPROSSIMAZIONE (+), E I PUNTI (*).
542  REM PROIETTA LA LINEA OLTRE I PUNTI.
550  REM
555  PRINT"L'ASSE DELLE X VA VERSO IL BASSO, L'ASSE DELLE Y VERSO DESTRA"
560  PRINT
565  REM
570  PRINT"O          10          20          30          40 PUNTI"
580  PRINT"+-----+-----+-----+-----+ Y, X(I), Y(I)"
590  FOR I=(N-F9+1) TO N
595      X(I)=I
600      L=M(F9)*X(I)+B(F9)
605      L=INT(L+.5)
610      G$(L)="+"
615      G$(INT(Y(I)+.5))="*"
620  REM
625      PRINT"I"
630      PRINT"I"
635      PRINT"+":
640      FOR J=1 TO 40
645          PRINT G$(J);
650          G$(J)=" "
655      NEXT J
660      PRINT TAB(46);X(I);"      "Y(I)
665  NEXT I
670  REM
675  REM CONSIDERA LE ULTIME 12 PREVISIONI
680  REM
685  FOR F=F9 TO 2 STEP -1
690      L=INT(P(F)+.5)
695      G$(L)=P$(F)
700  NEXT F
705  PRINT"I"
710  PRINT"I"
715  PRINT"+":
720  FOR J=1 TO 40
725      PRINT G$(J);
730      G$(J)=" "
735  NEXT J
740  PRINT TAB(46);"<PREVISIONI"
745  REM
750  PRINT"X"
755  PRINT
760  PRINT"*-- PUNTO"
765  PRINT"+-- RETTA DEL MESE":F9

```

```

770 PRINT
775 REM
780 PRINT"BASE (MESI) PREVISIONE CARATTERI"
785 FOR F=12 TO 2 STEP -1
790 PRINT F;TAB(12);P(F);TAB(25);P*(F)
795 NEXT F
800 PRINT"NEL GRAFICO ALCUNI PUNTI POSSONO *";
805 PRINT"RISULTARE SOVRAPPosti."
810 REM
815 END

```

PROMESE è una modifica di REGRESSIONE. Al posto degli asterischi e dei segni più, stampa dei numeri. I risultati mostrano che le previsioni sono raggruppate tra 29 e 39. Tuttavia, le previsioni più recenti (rappresentate da 4, 3, e 2) fanno pensare a un calo. Si può dedurre che probabilmente il prossimo mese sarà sotto il valore 29. La parte grafica del programma serve all'utente per poter vedere la retta relativa a qualsiasi base, F9. La tabella dei valori mostra le previsioni di tutte le dodici linee d'interpolazione. Il grafico e la tabella, assieme, offrono una serie di previsioni che ci mettono in condizione di prevedere quale sarà il prossimo valore probabile.

ESERCIZI

1. Provate PROMESE con i vostri dati. Assicuratevi di avere valori per tutto l'anno. Modificate la routine grafica, se vi sembra il caso.

RIUNIONI DELLA PROSSIMA SETTIMANA

"Per poter dire che rivoluziona davvero la vita dell'ufficio," disse Carolyn Grimes, la segretaria di Steven Cauldwell, "questo computer dovrebbe essere in grado di pedinare te e i tuoi quattro manager di settore. Sto perdendo troppo tempo per sapere quando ognuno è libero per programmare le riunioni."

Cauldwell alzò le sopracciglia. "Vuoi dire che ti serve un modo per sapere quando siamo liberi?"

"Sì", disse la Grimes "e anche quando sono liberi solo due o tre di voi, per delle riunioni più piccole."

"Quanto tempo ci vorrebbe per avere gli orari degli impegni dalle rispettive segretarie?" chiese Cauldwell.

"Per telefono, dieci o quindici minuti, credo", rispose la Grimes. "Perché?"

"Se puoi avere i loro orari della settimana, ti posso fornire un metodo per organizzare facilmente le riunioni", disse Cauldwell.

"Okay", disse la Grimes. "Avrò gli orari oggi prima che tu te ne vada." Steven Cauldwell doveva alla sua segretaria più di un favore. Per parecchi mesi non aveva avuto né il tempo né il motivo per usare il suo computer.

Ma quella sera scrisse un programma chiamato TEMPOLIBERO ed il giorno dopo lo portò con sé in ufficio.

"Provalo un po'," disse alla Grimes che sedeva nel suo ufficio davanti al computer. "Segui le indicazioni."

La Grimes era scettica, ma cominciò a rispondere "Si" alla domanda se voleva organizzare una riunione, un 1, 3 e 5 per indicare chi avrebbe dovuto parteciparvi, ed uno zero per dire che aveva finito. Il computer stampò la seguente lista:

VUOI UNA RIUNIONE COLLETTIVA (SI O NO)? NO
VUOI ORGANIZZARE UNA RIUNIONE? SI

PARTECIPANTI:
1. STEVE CAULDMELL
2. FRANK BRADSHAW
3. EILEEN RANDALL
4. ROSE THOMPSON
5. TOM DELLER

BATTI UN NUMERO TRA 1 E 5
PER OGNI PARTECIPANTE.
PER FINIRE BATTI 0.
? 1
? 3
? 5
? 0

LUN 5:00
LUN 5:30

MER 11:30

MER 5:30

VEN 3:30
VEN 4:00
VEN 4:30
VEN 5:00
VEN 5:30

La Grimes era ancora scettica, nonostante la sorpresa. "Come ha fatto?" chiese.

"Il programma ha esaminato gli orari per vedere quali erano le ore libere. Ecco com'erano gli orari che mi hai dato:

	VUOI UNA RIUNIONE COLLETTIVA (SI O NO)? SI				
	STEVE	FRANK	EILEEN	ROSE	TOM
LUN 8:00	*	*	*		
LUN 8:30	*	*	*		
LUN 9:00	*	*	*	*	
LUN 9:30	*	*	*	*	
LUN 10:00			*		*
LUN 10:30			*		*
LUN 11:00					*
LUN 11:30					*
LUN 12:00					*

LUN 12:30				*
LUN 1:00				*
LUN 1:30				*
LUN 2:00	*	*	*	
LUN 2:30	*	*	*	
LUN 3:00	*		*	
LUN 3:30	*		*	
LUN 4:00			*	
LUN 4:30			*	
LUN 5:00				*
LUN 5:30				*
MAR 8:00	*	*	*	*
MAR 8:30	*	*	*	*
MAR 9:00	*	*	*	*
MAR 9:30	*	*	*	*
MAR 10:00	*	*	*	*
MAR 10:30	*	*	*	*
MAR 11:00	*	*	*	*
MAR 11:30	*	*	*	*
MAR 12:00	*	*	*	*
MAR 12:30	*	*	*	*
MAR 1:00	*	*	*	*
MAR 1:30	*	*	*	*
MAR 2:00	*	*	*	*
MAR 2:30	*	*	*	*
MAR 3:00	*	*	*	*
MAR 3:30	*	*	*	*
MAR 4:00	*	*	*	*
MAR 4:30	*	*	*	*
MAR 5:00	*	*	*	*
MAR 5:30	*	*	*	*
MER 8:00		*	*	*
MER 8:30		*	*	*
MER 9:00	*	*	*	*
MER 9:30	*	*	*	*
MER 10:00	*	*	*	*
MER 10:30	*	*	*	*
MER 11:00	*	*	*	*
MER 11:30	*	*	*	*
MER 12:00	*	*	*	*
MER 12:30	*	*	*	*
MER 1:00	*	*	*	*
MER 1:30	*	*	*	*
MER 2:00	*	*	*	*
MER 2:30	*	*	*	*
MER 3:00	*	*	*	*
MER 3:30	*	*	*	*
MER 4:00	*	*	*	*
MER 4:30	*	*	*	*
MER 5:00	*	*	*	*
MER 5:30	*	*	*	*
GIO 8:00	*		*	*
GIO 8:30	*		*	*
GIO 9:00	*		*	*
GIO 9:30	*		*	*
GIO 10:00	*		*	*
GIO 10:30	*		*	*
GIO 11:00	*		*	*
GIO 11:30	*		*	*
GIO 12:00	*		*	*
GIO 12:30	*		*	*
GIO 1:00	*		*	*
GIO 1:30	*		*	*

GIO 2:00	*			*	*
GIO 2:30	*			*	*
GIO 3:00	*			*	*
GIO 3:30	*			*	*
GIO 4:00	*		*	*	*
GIO 4:30	*		*	*	*
GIO 5:00	*		*	*	*
GIO 5:30	*		*	*	*
VEN 8:00	*				*
VEN 8:30	*				*
VEN 9:00	*			*	*
VEN 9:30	*			*	*
VEN 10:00	*		*	*	*
VEN 10:30	*		*	*	*
VEN 11:00	*		*	*	*
VEN 11:30	*		*	*	*
VEN 12:00	*		*	*	*
VEN 12:30	*		*	*	*
VEN 1:00	*		*	*	*
VEN 1:30	*		*	*	*
VEN 2:00		*	*	*	*
VEN 2:30		*	*	*	*
VEN 3:00		*	*	*	*
VEN 3:30			*	*	*
VEN 4:00					
VEN 4:30					
VEN 5:00					
VEN 5:30					

VUOI ORGANIZZARE UNA RIUNIONE? NO

"Come fa il computer a sapere che gli orari sono questi?" chiese la Grimes.
 "Ho inserito le informazioni che mi hai dato ieri. Nel programma entravano così:

```

1992 REM
2000 DATA 1,LUN,8,10,2,4,7,7
2010 DATA 1,MAR,8,6,7,7
2020 DATA 1,MER,9,11,12,1,5,4,5,7,7
2030 DATA 1,GIO,8,6,7,7
2040 DATA 1,VEN,8,2,7,7
2050 DATA 2,LUN,8,10,2,3,7,7
2060 DATA 2,MAR,8,6,7,7
2070 DATA 2,MER,8,6,7,7
2080 DATA 2,VEN,2,3,5,7,7
2090 DATA 3,LUN,8,11,2,5,7,7
2100 DATA 3,MAR,8,9,10,12,2,4,7,7
2110 DATA 3,MER,8,9,10,11,5,12,2,3,5,5,7,7
2120 DATA 3,GIO,8,10,4,6,7,7
2130 DATA 3,VEN,10,3,5,7,7
2140 DATA 4,LUN,9,10,5,6,7,7
2150 DATA 4,MAR,8,6,7,7
2160 DATA 4,MER,8,6,7,7
2170 DATA 4,GIO,8,6,7,7
2180 DATA 4,VEN,9,11,2,4,7,7
2190 DATA 5,LUN,10,2,7,7
2200 DATA 5,MAR,8,10,2,5,5,7,7
2210 DATA 5,MER,9,11,12,1,5,2,3,5,7,7
2220 DATA 5,GIO,9,9,5,2,2,5,7,7
2230 DATA 5,VEN,8,8,5,11,11,5,12,2,7,7
2240 DATA 77
2242 REM
  
```


"Cosa significa 10.5?" chiese la Grimes.

"Le dieci e mezza. L'orario è preciso fino alle mezz'ore."

Grimes annui e guardò di nuovo i dati. "E perché quel 7,7 alla fine di ogni riga? A cosa serve?"

"Dice al programma che quella riga è finita", rispose Cauldwell.

"Quindi anche il 77 in fondo segnala la fine dei dati?"

"Proprio così", disse Cauldwell.

"Cosa succede se all'orario viene aggiunta una riunione?" chiese la Grimes.

"Basta introdurre una nuova riga di dati in qualche punto", disse Cauldwell.

"Ma se viene modificato l'orario di qualcuno?" insistette la Grimes.

"Bisogna riscrivere quel giorno con le modifiche", disse Cauldwell.

Grimes fece un segno d'approvazione. "Potrebbe andare. Con qualche modifica potrei avere una nuova copia degli orari."

"Certo," disse Cauldwell, "puoi avere sempre una copia aggiornata. Per realizzare quest'idea, dovrai avere gli orari di tutti all'inizio della settimana ed il loro impegno d'informarti di tutte le modifiche. Potresti corrispondere a questo impegno inviando loro copie dell'orario modificato."

"Va bene", disse Grimes. "Finora passavo la metà del mio tempo al telefono per organizzare le riunioni. Questo programma mi consentirà di dedicarmi anche alle altre cose. Grazie."

Sei settimane dopo, la Grimes confidò a Cauldwell che era stata invitata a cena da due delle segretarie dei manager di settore.

"Per quale occasione?" chiese Cauldwell.

"Festeggiamo il miglioramento nell'organizzazione delle riunioni grazie a TEMPOLIBERO. Tutte abbiamo guadagnato delle ore per merito di quel programma."

Cauldwell sorrise. "Bene. Riconosci dunque anche tu che i computer sono utili in ufficio, anche se non proprio rivoluzionari?"

La Grimes sorrise: "Sì, è giusto riconoscerlo."

Erano innovazioni necessarie, pensò Cauldwell. Una piccola vittoria contro il tempo. La società, col passare del tempo, era cambiata. Tutti erano diventati esperti di computer. Li usavano nelle riunioni, nelle conversazioni informali e nel prendere le decisioni. I piccoli computer stavano a poco a poco cambiando le cose.

IL PROGRAMMA TEMPOLIBERO

1000 REM ***TEMPOLIBERO***

1010 REM

1020 REM PROGRAMMA GLI APPUNTAMENTI SETTIMANALI DI CINQUE PERSONE.

1030 REM A RICHIESTA STAMPA LA SCHEDA.

1040 REM

```

2310          IF T1>13 THEN 2320 ELSE 2340
2320          PRINT T1:"E' TROPPO GRANDE. PERSONA #":P:"AL ":ID#
2330          STOP
2340 REM
2350          IF T2>13 THEN 2360 ELSE 2380
2360          PRINT T2:"E' TROPPO GRANDE. PERSONA #":P:"AL ":ID#
2370          STOP
2380 REM
2390          IF T1=7 THEN 2540
2400          X=T1
2410          GOSUB 2680
2420          T1=X
2430          X=T2-.5
2440          GOSUB 2680
2450          T2=X
2460          FOR K=T1 TO T2
2470             S*(K,P)="*"
2480          NEXT K
2490          NEXT J
2500          PRINT"PIU' DI 15 RIUNIONI AL"
2510          PRINT D#:" PER IL PARTECIPANTE":P:". "
2520          PRINT"CONTROLLARE I DATI."
2530          STOP
2540 REM
2550          IF T2=7 THEN 2590
2560          PRINT"I TEMPI NON VANNO BENE PER IL PARTECIPANTE":
2570          PRINT P:"AL ":ID#
2580          STOP
2590 REM
2600          NEXT I
2610          PRINT"PIU' DI 150 LINEE NELLA SCHEDA."
2620          PRINT"CONTROLLARE I DATI."
2630          STOP
2640 REM
2650 RETURN
2660 REM
2670 REM
2680 REM SUBROUTINE CHE CONVERTE I GIORNI IN NUMERI
2690 REM INPUT: D#,X
2700 REM OUTPUT: X
2710 REM
2720 REM CONVERTE IL GIORNO E L'INTERVALLO DI TEMPO (8-6)
2730 REM IN UN NUMERO COMPRESO TRA 1 E 100.
2740 REM
2750          T3=X
2760          FOR K=1 TO 5
2770             IF D*(K)=D# THEN 2820
2780          NEXT K
2790          PRINT D#:" NON E' UN GIORNO PERMESSO."
2800          PRINT"CONTROLLARE I DATI."
2810          STOP
2820 REM
2830          X=(K-1)*20
2840          IF T3<7 THEN 2850 ELSE 2860
2850          T3=T3+12
2860 REM
2870          X=X+((T3-8)*2+1)
2880 REM
2890 RETURN
2900 REM
2910 REM
2920 REM SUBROUTINE CHE STAMPA LA SCHEDA
2930 REM INPUT: D*( ),D( ),S*( ),T*( ),T( )
2940 REM OUTPUT: --
2950 REM
2952          PRINT
2954          PRINT

```

```

2960 PRINT TAB(10);"STEVE FRANK EILEEN ROSE TOM"
2970 PRINT TAB(10);"-----"
2980 REM
2990 FOR I=1 TO 100
3000 PRINT D$(D(I));" ";T$(T(I));TAB(10);
3010 FOR J=1 TO P9
3020 PRINT TAB(10+(J-1)*8);S$(I,J);
3030 NEXT J
3040 PRINT
3050 IF INT(I/20)=I/20 THEN 3060 ELSE 3070
3060 PRINT
3070 REM
3080 NEXT I
3090 REM
3100 RETURN
3110 REM
3120 REM
3130 REM SUBROUTINE CHE FISSA LA RIUNIONE
3140 REM INPUT: D$( ),D(),S$( ),T$( ),T()
3150 REM OUTPUT: --
3160 REM
3170 PRINT
3180 PRINT"PARTECIPANTI:"
3190 PRINT"1. STEVE CAULDWELL"
3200 PRINT"2. FRANK BRADSHAW"
3210 PRINT"3. EILEEN RANDALL"
3220 PRINT"4. ROSE THOMPSON"
3230 PRINT"5. TOM DELLER"
3240 PRINT
3250 PRINT"BATTI UN NUMERO TRA 1 E 5"
3260 PRINT"PER OGNI PARTECIPANTE."
3270 PRINT"PER FINIRE BATTI 0."
3275 REM
3280 FOR I=1 TO P9+1
3282 INPUT N(I)
3284 IF N(I)=0 THEN 3340
3290 IF N(I)<0 OR N(I)>5 THEN 3294 ELSE 3298
3294 PRINT"IL NUMERO DEVE ESSERE COMPRESO TRA 0 E 5."
3295 PRINT"RIFARE."
3296 I=I-1
3298 REM
3310 NEXT I
3320 PRINT"NON PIU' DI";P9;"INPUT."
3330 STOP
3340 REM
3350 P1=I-1
3360 I1=0
3370 FOR I=1 TO 100
3380 FOR J=1 TO P1
3390 IF S$(I,N(J))="*" THEN 3460
3400 NEXT J
3410 IF I-1<>I1 THEN 3420 ELSE 3430
3420 PRINT
3430 REM
3440 PRINT D$(D(I));" ";T$(T(I))
3450 I1=I
3460 REM
3470 NEXT I
3480 REM
3490 RETURN
3500 END

```

Il programma TEMPOLIBERO è diviso in subroutine. Ognuna ha uno scopo specifico e ripassa i risultati alla riga di programma che l'aveva richiamata. Ecco lo schema del programma:

ROUTINE PRINCIPALE
SUDDIVISIONE DEL TEMPO
INTRODUZIONE DEGLI ORARI
CONVERSIONE IN NUMERI
STAMPA GLI ORARI
FISSA LA RIUNIONE

Il programma si serve dell'istruzione INPUT per permettere all'utente di decidere che funzione dovrà svolgere. Cauldwell aveva scritto questo programma per la sua segretaria e sapeva per cosa lei avrebbe potuto servirsene, ed in questo senso lo aveva progettato.

Un'altra caratteristica utile all'utente è la serie di istruzioni IF...THEN...ELSE che evitano che dati scorretti entrino nella subroutine INTRODUZIONE DEGLI ORARI. Il programma sa che le righe DATA possono contenere degli errori e cerca di identificarli. Programmi di questo tipo sono difficili da scrivere, ma facili da usare.

CONVERSIONE IN NUMERI traduce i dati in numeri da 1 a 100 e SUDDIVISIONE DEL TEMPO converte uno di questi numeri nel corrispondente giorno ed ora. Il secondo passaggio è contenuto come una tabella nelle variabili D() e T(), per cui FISSA LA RIUNIONE e STAMPA GLI ORARI possono servirsene tranquillamente.

ESERCIZI

1. Usate TEMPOLIBERO per organizzare alcune riunioni nel vostro ufficio.
2. Migliorate la subroutine INTRODUZIONE DEGLI ORARI perché controlli che P corrisponda ad una persona realmente esistente e D\$ ad un giorno valido.
3. (Difficile.) Cambiate il programma perché funzioni con 9 persone e con ore intere.

Appendici

Introduzione al Basic ed alla strutturazione dei programmi

Questa appendice è composta di quattro parti. I principianti dovrebbero leggerle tutte. I lettori esperti possono saltare la prima, ma è bene che leggano le altre tre. Le quattro parti sono: "Il Basic in azione", "Stile e struttura", "Paragrafi di programma" e "Il programma più grande".

"Il Basic in azione" introduce il linguaggio Basic ed alcune sue caratteristiche. "Stile e struttura" discute le idee fondamentali della realizzazione dei programmi. "Paragrafi di programma" svela diversi trucchi di programmazione in Basic molto utili. "Il programma più grande" spiega come si assemblano tanti programmini per formarne uno grande.

I programmi che seguono si spiegano da soli. La lista del programma (LIST) assieme alla sua esecuzione (RUN) mostrerà non solo com'era il programma in Basic, ma anche come viene eseguito dal computer. Leggendo questi programmi, potrete imparare cosa significano le istruzioni Basic PRINT, END, LET, REM, FOR e NEXT, READ e DATA, INPUT, IF...THEN. Se avete qualche dubbio su uno dei programmi, provatelo su un computer e guardate cosa fa.

LIST

```
10 PRINT"SI PUO' SCRIVERE QUALCOSA TRA VIRGOLETTE"
20 END
```

RUN

SI PUO' SCRIVERE QUALCOSA TRA VIRGOLETTE

LIST

```
10 PRINT"USANDO DIVERSE LINEE SI PUO' SCRIVERE"
20 PRINT"UN MESSAGGIO ESTESO."
30 PRINT"USANDO IL PUNTO E VIRGOLA (:)":
40 PRINT"IL TESTO"
50 PRINT"VIENE STAMPATO DI SEGUITO SULLA"
60 PRINT"STESSA RIGA."
70 END
```

RUN

```
USANDO DIVERSE LINEE SI PUO' SCRIVERE
UN MESSAGGIO ESTESO.
USANDO IL PUNTO E VIRGOLA (:)IL TESTO
VIENE STAMPATO DI SEGUITO SULLA
STESSA RIGA.
```

LIST

```
10 PRINT"12345678911234567891123456789112345678911234567891123456789"
20 PRINT TAB(13);"SI PUO' SCRIVERE A PARTIRE DA UNA POSIZIONE SCELTA."
30 END
```

RUN

```
12345678911234567891123456789112345678911234567891123456789
      SI PUO' SCRIVERE A PARTIRE DA UNA POSIZIONE SCELTA.
```

LIST

```
10 PRINT"12345678911234567891123456789112345678911234567891123456789"
20 PRINT TAB(5);"MEDIANTE";TAB(15);"L' ISTRUZIONE TAB";
30 PRINT TAB(35);"SI PUO' SCRIVERE";TAB(55);"DOVE"
40 PRINT"SI VUOLE";TAB(20);"SULLA RIGA."
50 END
```

RUN

```
12345678911234567891123456789112345678911234567891123456789
      MEDIANTE L' ISTRUZIONE TAB      SI PUO' SCRIVERE      DOVE
SI VUOLE                          SULLA RIGA.
```

LIST

```
10 PRINT"SI POSSONO SCRIVERE ANCHE NUMERI."
20 PRINT 333,1.68243,38.2,54321
30 END
```


RUN

```
SI POSSONO SCRIVERE ANCHE NUMERI.  
333          1,68243          38.2          54321
```

LIST

```
10 PRINT"SI PUO' SCRIVERE IL RISULTATO DI UN CALCOLO."  
20 PRINT 666/3,4+18,(3*4)/6,2-4  
30 END
```

RUN

```
SI PUO' SCRIVERE IL RISULTATO DI UN CALCOLO.  
222          22          2          -2
```

LIST

```
10 PRINT"SI POSSONO SCRIVERE SIA MESSAGGI"  
20 PRINT"CHE CALCOLI.",33+44  
30 END
```

RUN

```
SI POSSONO SCRIVERE SIA MESSAGGI  
CHE CALCOLI.      77
```

LIST

```
10 PRINT"LE ENTRATE SONO ":275  
20 PRINT"I COSTI SONO   ":216  
30 PRINT"IL GUADAGNO E' ":275-216  
40 END
```

RUN

```
LE ENTRATE SONO 275  
I COSTI SONO    216  
IL GUADAGNO E' 59
```

LIST

```
10 PRINT"SI PUO' STAMPARE UNA RIGA BIANCA"  
20 PRINT  
30 PRINT"USANDO UN'ISTRUZIONE PRINT VUOTA."  
40 END
```

RUN

```
SI PUO' STAMPARE UNA RIGA BIANCA  
USANDO UN'ISTRUZIONE PRINT VUOTA.
```

LIST

```
10 PRINT"SI PUD' STAMPARE IL VALORE DI UNA VARIABILE."
20 V=1776
30 PRINT"V HA VALORE",V
40 END
```

RUN

```
SI PUD' STAMPARE IL VALORE DI UNA VARIABILE.
V HA VALORE      1776
```

LIST

```
10 PRINT"SI PUD' STAMPARE IL RISULTATO DI UN CALCOLO"
20 PRINT"ASSEGNOTO A UNA VARIABILE."
30 V=(17*3)/6
40 PRINT V
50 END
```

RUN

```
SI PUD' STAMPARE IL RISULTATO DI UN CALCOLO
ASSEGNOTO A UNA VARIABILE.
      8.5
```

LIST

```
10 PRINT"SI POSSONO STAMPARE DIVERSE VARIABILI."
15 PRINT
20 R=2.6
30 D=2*R
40 C=3.14159*D
50 A=3.14159*R*R
60 PRINT"PER UN CERCHIO DI RAGGIO":R
70 PRINT"IL DIAMETRO E'":D
80 PRINT"LA CIRCONFERENZA E'":C
90 PRINT"L'AREA E'":A
99 END
```

RUN

```
SI POSSONO STAMPARE DIVERSE VARIABILI.

PER UN CERCHIO DI RAGGIO 2.6
IL DIAMETRO E' 5.2
LA CIRCONFERENZA E' 16.3363
L'AREA E' 21.2371
```

LIST

```
10 PRINT"SI PUD' PASSARE DAL VECCHIO VALORE"
20 PRINT"DI UNA VARIABILE AL NUOVO,"
30 S=100
40 PRINT"ALLA LINEA 40 S E'":S
50 S=1.15*S
60 PRINT"ALLA LINEA 60 S E'":S
70 END
```

RUN

```
SI PUD' PASSARE DAL VECCHIO VALORE
DI UNA VARIABILE AL NUOVO.
ALLA LINEA 40 S E' 100
ALLA LINEA 60 S E' 115
```

LIST

```
10 PRINT"SI PUD' MODIFICARE PIU' VOLTE UNA VARIABILE."
20 PRINT"SI PUD' PENSARE S COME UNA SOMMA"
30 PRINT"COMPOSTA ANNUALMENTE AL 15%."
40 S=100
50 PRINT S
60 S=1.15*S
70 PRINT S
80 S=1.15*S
90 PRINT S
100 S=1.15*S
110 PRINT S
120 END
```

RUN

```
SI PUD' MODIFICARE PIU' VOLTE UNA VARIABILE.
SI PUD' PENSARE S COME UNA SOMMA
COMPOSTA ANNUALMENTE AL 15%.
100
115
132.25
152.088
```

LIST

```
10 REM SIGNIFICA COMMENTO (REMARK). NON VIENE MAI STAMPATO.
20 PRINT"QUESTA E' L'UNICA LINEA CHE VIENE STAMPATA."
30 END
```

RUN

QUESTA E' L'UNICA LINEA CHE VIENE STAMPATA.

LIST

```
100 REM CON LA COPPIA DI ISTRUZIONI 'FOR...NEXT'
110 REM SI PUD' RIPETERE PIU' VOLTE LA STESSA
120 REM OPERAZIONE.
130 REM
140 S=100
150 FOR Y=1 TO 4
160 S=1.15*S
170 PRINT Y,S
180 NEXT Y
190 PRINT"DOPD 4 ANNI LA SOMMA E'":S
200 END
```

APPENDICE A

RUN

```

1             ,15
2             152.25
3             152.088
4             174.901
DOPO 4 ANNI LA SOMMA E' 174.901

```

LIST

```

110 REM SI POSSONO LEGGERE DEI DATI USANDO
120 REM LE ISTRUZIONI 'READ' E 'DATA'.
130 REM
140 DATA 275,216
150 READ R,C
160 P=R-C
170 PRINT"ENTRATE","COSTI","GUADAGNO"
180 PRINT R,C,P
190 END

```

RUN

ENTRATE	COSTI	GUADAGNO
275	216	59

LIST

```

100 REM SI POSSONO CALCOLARE I DATI DI DIVERSE DIVISIONI
110 REM MEDIANTE UN CICLO E PIU' LINEE DI ISTRUZIONI
120 REM 'DATA'.
130 REM
140 DATA 33,22
150 DATA 600,423
160 DATA 500,275
170 DATA 208,106
180 REM
190 PRINT"DIVISIONE","ENTRATE","COSTI","PROFITTO"
200 FOR D=1 TO 4
210 READ R,C
220 P=R-C
230 PRINT D,R,C,P
240 NEXT D
250 END

```

RUN

DIVISIONE	ENTRATE	COSTI	PROFITTO
1	33	22	11
2	600	423	177
3	500	275	225
4	208	106	102

LIST

```

10 REM QUESTO PROGRAMMA POTRA' LAVORARE SOLO NEL 1984.
20 PRINT"INSERITE I DATI NEL PROGRAMMA MEDIANTE"
30 PRINT"L' ISTRUZIONE INPUT."
40 PRINT
50 PRINT"QUANTI ANNI HAI COMPIUTO QUEST'ANNO"
60 INPUT A

```

```

70 Y=1984-A
80 PRINT
90 PRINT"CIO' SIGNIFICA CHE SEI NATO NEL";Y
100 END

```

RUN

INSERITE I DATI NEL PROGRAMMA MEDIANTE L'ISTRUZIONE INPUT.

```

QUANTI ANNI HAI COMPIUTO QUEST'ANNO
? 37
CIO' SIGNIFICA CHE SEI NATO NEL 1947

```

LIST

```

100 REM SI POSSONO CONTROLLARE VARIE ALTERNATIVE
110 REM MEDIANTE L'ISTRUZIONE 'IF...THEN'.
120 REM
130 PRINT"QUANTI ANNI HAI";
140 INPUT A
150 PRINT
160 IF A>30 THEN 190
170 PRINT"SEI UNO SBARBATELLO."
180 GOTO 210
190 REM
200 PRINT"SEI DECREPITO."
210 REM
220 END

```

RUN

```

QUANTI ANNI HAI? 44
SEI DECREPITO.

```

RUN

```

QUANTI ANNI HAI? 22
SEI UNO SBARBATELLO.

```

STILE E STRUTTURA

Ogni programma ha tre componenti essenziali: l'input, l'elaborazione e l'output. Per l'input si usa l'istruzione LET, INPUT oppure READ. Per l'elaborazione ci vuole un'apposita istruzione LET. L'output è prodotto dall'istruzione PRINT. Quando il programma legge i dati, li elabora e stampa i risultati, il suo lavoro può procedere in tre modi:

1. Sequenza (numeri di riga),
2. Decisione (istruzione IF...THEN...ELSE),
3. Iterazione (coppia di istruzioni FOR-NEXT).

Esempio (con NOT ma senza ELSE)

```

100 IF NOT (Y=8) THEN 150
110 PRINT"Y E' UGUALE A OTTO"
150 REM

```

Esempio (ANSI minimale)

```

100 IF Y<>8 THEN 150
110 PRINT"Y E' UGUALE A OTTO"
150 REM

```

Un'altra versione particolare di IF...THEN...ELSE è la struttura CASE. Si tratta semplicemente di una sequenza di istruzioni IF...THEN...ELSE.




Fig. A.4 Diagramma di flusso di una decisione dilatata

Esempio (Basic Level II del TRS-80)

```

100 IF B=1 THEN 110 ELSE 130
110 PRINT"GEORGE"
120 GOTO 230
130 IF B=2 THEN 140 ELSE 160
140 PRINT"PAUL"
150 GOTO 230
160 IF B=3 THEN 170 ELSE 190
170 PRINT"JOHN"
180 GOTO 230
190 IF B=4 THEN 200 ELSE 220
200 PRINT"RINGO"
210 GOTO 230
220 PRINT"BEATLE FUORI DAI LIMITI"
230 REM

```




Fig. A.5 Diagramma di flusso di iterazione

Iterazione

Sicuramente la struttura più complessa di un programma per computer è il ciclo. Con esso il programma torna indietro ad una certa riga e fa sì che il programma ripeta le istruzioni seguenti.

Esempio

```
100 FOR X=1 TO 10
110 PRINT X
120 NEXT X
130 REM
```

```
100 IF C>600 THEN 150
110 C=C+33
120 PRINT C
130 GOTO 100
150 REM
```

Entrambi questi cicli vengono ripetuti finché è soddisfatta una certa condizione. La coppia FOR-NEXT viene ripetuta dieci volte (finché X non supera 10). Il secondo ciclo continua finché il valore di C non supera 600. Da un punto di vista tecnico, un ciclo FOR-NEXT può essere realizzato con IF...THEN e un'istruzione GOTO. In pratica, tuttavia, è meglio tradurre tutti i cicli in costruzioni FOR-NEXT. Il secondo esempio potrebbe essere riscritto così.

Esempio

```
100 FOR I=1 TO 100000
110 IF C>600 THEN 150
120 C=C+33
130 PRINT C
140 NEXT I
150 REM
```

Anche se in Basic è possibile seguire flussi logici diversi, le tre strutture citate — sequenza, decisione e iterazione — sono le uniche necessarie per scrivere un programma. In effetti, i vostri programmi scritti usando solo que-

ste tre strutture saranno i migliori. È la buona struttura che distingue i programmi chiari da quelli confusi, quelli corretti da quelli sbagliati.

Alcune regole sullo stile e la struttura in Basic

1. I numeri di riga che seguono THEN e ELSE non devono mai far riferimento a righe precedenti la stessa istruzione IF...THEN...ELSE.
2. Quando possibile, usate cicli FOR-NEXT. È il modo più sicuro per costruire i cicli in Basic.
3. Non usate mai istruzioni GOTO fuori da strutture IF...THEN...ELSE.
4. Fate largo uso delle rientranze.

PARAGRAFI DI PROGRAMMA

Nei programmi di tipo commerciale alcuni concetti si ripetono frequentemente. Tra i più comuni c'è un semplice ciclo per la lettura di una variabile con indice.

```
100  FOR I=1 TO 4
110  READ X(I)
120  NEXT I
130  DATA 10,11,9,7
```

Questo ciclo è equivalente alle due righe seguenti.

```
100  READ X(1),X(2),X(3),X(4)
110  DATA 10,11,9,7
```

Quando un programma ha i valori di una variabile con indici, spesso gli serve la loro somma.

```
100  S=0
110  FOR I=1 TO 4
120      S=S+X(I)
130  NEXT I
140  PRINT"LA SOMMA E' ";S
```

Per aumentare un valore del 12 per cento si usa la seguente istruzione.

```
100  X=1.12*X
```

Per diminuirlo del 20 per cento:

```
100  X=.80*X
```


Per arrotondarlo:

```
100 X=INT(X+.5)
```

Per arrotondare i dollari ai penny:

```
100 X=(INT(X*100+.5))/100
```

Per stampare nove numeri su una riga di 72 caratteri:

```
100 FOR C=1 TO 9
110 PRINT TAB((C-1)*8);X(C);
120 NEXT C
```

Per inserire una riga bianca ogni cinque righe di stampa:

```
100 FOR R=1 TO 11
110 PRINT"QUESTA E' LA LINEA":R
120 IF (R/5)=INT(R/5) THEN 130 ELSE 140
130 PRINT
140 REM
150 NEXT R
```

Una variabile a due indici può rappresentare una tabella di dati. Negli esempi che seguono la variabile a due indici è T(R,C). R è l'indice di riga e C è l'indice di colonna. Per leggere la prima riga della tabella usate:

```
100 R=1
110 READ T(R,1),T(R,2),T(R,3),T(R,4)
120 DATA 10,22,13,24
```

Per leggere una lunga riga provate questo. (Legge 15 elementi.)

```
100 LET R=1
110 FOR C=1 TO 15
120 READ T(R,C)
130 NEXT C
140 DATA 10,9,8,14,-77,3,8,999,12,-2,88,.34,3.25,6.90,3
```

Per leggere più righe in una tabella. (Legge 12 variabili.)

```
100 FOR R=1 TO 4
110 FOR C=1 TO 3
120 READ T(R,C)
130 NEXT C
140 NEXT R
150 DATA 14,13,12
152 DATA 3,2,1
154 DATA 33,44,55
156 DATA -100,-200,-333
```

Per costruire una nuova colonna a partire da due vecchie (per ogni riga).

```

100   FOR R=1 TO 4
110       T(R,7)=T(R,1)+T(R,3)
120   NEXT R

```

Per fare una riga nuova a partire da due vecchie (per ogni colonna).

```

100   FOR C=1 TO 3
110       T(5,C)=(T(1,C)+T(3,C))*1.15
120   NEXT C

```

Per sommare gli elementi della quinta riga:

```

100   R=5
110   S=0
120   FOR C=1 TO 3
130       S=S+T(R,C)
140   NEXT C
150   PRINT"LA SOMMA DELLA RIGA":R;"E'":S

```

Per sommare lungo una colonna

```

100   FOR C=1 TO 3
110       FOR R=1 TO 5
120           S(C)=S(C)+T(R,C)
130       NEXT R
140   PRINT"LA SOMMA DELLA COLONNA":C;"E'":S(C)
150   NEXT C

```

IL PROGRAMMA PIU GRANDE

I pezzi di un programma finito non dovrebbero essere più lunghi di una pagina. Le parti che effettuano una funzione comune dovrebbero essere riunite in uno stesso punto. Le parti usate ripetutamente possono essere scritte una volta sola. Per questi motivi i programmi vengono divisi in subroutine. Un programma di media grandezza può avere la seguente struttura (fig. A-6).




Fig. A.6 Struttura di un programma di media grandezza

Il programma principale richiama quattro subroutine principali SUB-A, SUB-B, SUB-C e SUB-D. Queste subroutine a loro volta richiamano altre subroutine. Nell'esempio, le subroutine in fondo sono due routine d'utilità chiamate UTIL-1 e UTIL-2.

Per richiamare una subroutine dovreste usare una istruzione GOSUB. Di per sé, il GOSUB non sembra differire dal GOTO. Tuttavia, al GOSUB è associata un'altra istruzione, la RETURN passa il controllo alla riga che segue il GOSUB, che ha chiamato la subroutine l'ultima volta. Il RETURN ha tenuto conto di qual è questo GOSUB.

Se la subroutine ha dei parametri in input, sarebbe bene elencarli su una riga che inizia con REM IN: . I parametri in output — anche quelli che sono stati modificati dalla subroutine — dovrebbero essere scritti dopo REM OUT: . Ecco come appare il programma strutturato:

```

100 REM ***FORMA***
110 REM
120 REM E' UN PROGRAMMA DIMOSTRATIVO DELL'USO
130 REM DELLE SUBROUTINE.
140 REM
150 REM VARIABILI:
160 REM N.....IL NUMERO
170 REM
250 REM PROGRAMMA PRINCIPALE
255 REM
260 PRINT"PROGRAMMA PRINCIPALE"
270 GOSUB 500
280 IF (N/2)=INT(N/2) THEN 290 ELSE 310
290 GOSUB 600
300 GOTO 330
310 REM
320 GOSUB 700
330 REM
  
```

```
340 GOSUB 800
350 STOP
360 REM
370 REM
500 REM SUBROUTINE 'SUB-A'
510 REM INPUT: --
520 REM OUTPUT: N
530 REM
535 REM INTRODUCE UN NUMERO N.
537 REM
540 PRINT TAB(3):"SUB-A"
545 PRINT TAB(25):"DUAL E' IL TUD NUMERO?";
550 INPUT N
560 GOSUB 900
570 RETURN
580 REM
590 REM
600 REM SUBROUTINE 'SUB-B'
610 REM INPUT: N
620 REM OUTPUT: --
630 REM
635 REM DEFINISCE UN NUMERO PARI.
637 REM
640 PRINT TAB(3):"SUB-B"
645 PRINT TAB(25):"N E' UN NUMERO PARI"
650 GOSUB 900
680 RETURN
682 REM
684 REM
700 REM SUBROUTINE 'SUB-C'
710 REM INPUT: N
720 REM OUTPUT: N
730 REM
732 REM DEFINISCE E MOLTIPLICA PER DUE UN NUMERO DISPARI.
733 REM
740 PRINT TAB(3):"SUB-C"
745 PRINT TAB(25):"N E' UN NUMERO DISPARI"
750 GOSUB 900
760 GOSUB 950
770 GOSUB 900
780 RETURN
782 REM
784 REM
800 REM SUBROUTINE 'SUB-D'
810 REM INPUT: N
820 REM OUTPUT: --
830 REM
832 REM MOLTIPLICA PER QUATTRO IL NUMERO FINALE.
834 REM
840 PRINT TAB(3):"SUB-D"
850 GOSUB 900
860 GOSUB 950
870 GOSUB 950
880 GOSUB 900
882 RETURN
884 REM
886 REM
900 REM SUBROUTINE 'UTIL-1'
910 REM INPUT: N
920 REM OUTPUT: --
930 REM
932 REM STAMPA IL NUMERO N.
934 REM
940 PRINT TAB(6):"UTIL-1"
942 PRINT TAB(25):"N E' ORA"IN
```

```

944 RETURN
946 REM
948 REM
950 REM SUBROUTINE 'UTIL-2'
960 REM INPUT: N
970 REM OUTPUT: N
980 REM
982 REM MOLTIPLICA PER DUE IL NUMERO N.
984 REM
990 PRINT TAB(6);"UTIL-2"
992 N=2*N
994 RETURN
996 REM
997 REM
998 END

```

RUN

PROGRAMMA PRINCIPALE

SUB-A

UTIL-1

QUAL E' IL TUD NUMERO? 4

N E' ORA 4

SUB-B

N E' UN NUMERO PARI

UTIL-1

N E' ORA 4

SUB-D

UTIL-1

N E' ORA 4

UTIL-2

UTIL-2

UTIL-1

N E' ORA 16

RUN

PROGRAMMA PRINCIPALE

SUB-A

UTIL-1

QUAL E' IL TUD NUMERO? 5

N E' ORA 5

SUB-C

N E' UN NUMERO DISPARI

UTIL-1

N E' ORA 5

UTIL-2

UTIL-1

N E' ORA 10

SUB-D

UTIL-1

N E' ORA 10

UTIL-2

UTIL-2

UTIL-1

N E' ORA 40

Quando costruite i vostri programmi, affrontate prima le parti principali e poi i dettagli. Perfezionare i dettagli in modo che ogni parte non sia più lunga di una pagina. Continuate così finché il programma è finito. Procedendo in questo modo manterrete il controllo dei vostri concetti e del vostro programma.

Programmi d'utility

I programmi d'utility di questa appendice sono facilmente convertibili in subroutine e sono quindi molto utili come parti standard per la costruzione di grossi programmi. Per poterne usare uno come subroutine in uno dei vostri programmi, dovrete apportare le seguenti modifiche:

1. Togliete le istruzioni READ e DATA perché le variabili vengono passate alla subroutine come parametri.
2. Togliete le istruzioni PRINT perché le variabili in output vengono ripassate al programma principale.
3. Assicuratevi che le variabili della subroutine siano elencate tra le altre all'inizio del programma principale.
4. Cambiate l'intestazione di programma con una intestazione di subroutine seguita dalle righe IN e OUT.
5. Sostituite RETURN all'istruzione END.

Gli ultimi due programmi dell'appendice illustrano come queste regole si applicano in pratica.

```
2000 REM ***LETTURA***
2010 REM
2020 REM ASSEGNA FINO A 100 NUMERI ALLA VARIABILE CON INDICE X()
2030 REM E CONTA I NUMERI. -.999999 INDICA LA FINE DELLA LISTA.
2040 REM
2060 REM VARIABILI:
2070 REM I.....VARIABILE INDICE DI CICLO
```

```

2080 REM N.....NUMERO DI ELEMENTI
2090 REM X().....LISTA DI DATI
2100 REM
2110 REM DIMENSIONI:
2120 DIM X(100)
2130 REM
2140 REM INPUT: --
2150 REM OUTPUT: N, X()
2160 REM
2170 FOR I=1 TO 100
2180 READ X(I)
2190 IF X(I)=-.999999 THEN 2210
2200 NEXT I
2210 REM
2220 N=I-1
2230 REM
2240 DATA 12,92,27,45,36
2250 REM
2260 REM TUTTI I DATI DEVONO ESSERE INSERITI PRIMA DI QUESTA LINEA.
2270 DATA -.999999
2280 REM
2282 FOR I=1 TO N
2284 PRINT X(I);
2286 NEXT I
2288 PRINT
2289 PRINT"N E'";N
2290 REM
2300 END

```

```

12 92 27 45 36
N E' 5

```

```

2400 REM ***PERCENTO***
2410 REM
2440 REM STAMPA UNA TABELLA DI NUMERI X() E LE LORO PERCENTUALI.
2450 REM
2470 REM VARIABILI:
2480 REM I.....INDICE DEL CICLO
2490 REM N.....NUMERO DI ELEMENTI DELLA LISTA
2500 REM P.....PERCENTUALE
2510 REM S().....SOMME
2520 REM X().....NUMERI
2530 REM
2540 REM DIMENSIONI:
2550 DIM X(100)
2560 REM
2570 REM INPUT: N, X()
2580 REM OUTPUT: --
2590 REM
2591 N=5
2592 FOR I=1 TO N
2593 READ X(I)
2594 NEXT I
2595 DATA 17,2,9,6,21
2597 REM
2600 S(1)=0
2610 FOR I=1 TO N
2620 S(I)=S(I)+X(I)
2630 NEXT I
2640 REM
2650 PRINT"NUMERO", "% DEL TOTALE"
2660 PRINT
2670 REM
2680 S(2)=0

```



```

2690   FOR I=1 TO N
2700     P=INT(X(I)/S(1)*100+.5)
2710     PRINT X(I),P
2720     S(2)=S(2)+P
2730   NEXT I
2740 REM
2750   PRINT"-----", "-----"
2760   PRINT S(1),S(2); " *"
2770   PRINT
2780   PRINT
2790   PRINT"* LA SOMMA DELLE PERCENTUALI NON PUO' ESSERE 100";
2800   PRINT"A CAUSA DEGLI ERRORI DI ARROTONDAMENTO."
2810 REM
2820   END

```

NUMERO	% DEL TOTALE
17	31
2	4
9	16
6	11
21	38
----	----
55	100 *

* LA SOMMA DELLE PERCENTUALI NON PUO' ESSERE 100A CAUSA DEGLI ERRORI DI ARROTONDAMENTO.

```

2900 REM ***MASSIMO***
2910 REM
2950 REM TROVA IL MASSIMO DI UNA LISTA DI NUMERI.
2960 REM
2970 REM VARIABILI:
2980 REM I.....INDICE DEL CICLO
2990 REM M.....NUMERO MASSIMO
3000 REM N.....NUMERO DI ELEMENTI
3010 REM X().....NUMERI
3020 REM
3030 REM DIMENSIONI:
3040   DIM X(100)
3050 REM
3060 REM INPUT: N, X()
3070 REM OUTPUT: M
3080 REM
3081   N=S
3082   FOR I=1 TO N
3083     READ X(I)
3084   NEXT I
3085   DATA 17,9,2,6,21
3086 REM
3090   M=X(1)
3100 REM
3110   FOR I=2 TO N
3120     IF X(I)>M THEN 3130 ELSE 3140
3130     M=X(I)
3140 REM
3150   NEXT I
3160 REM
3170   PRINT"IL MASSIMO E'";M
3180 REM
3190   END

```

IL MASSIMO E' 21

```

3200 REM ***SORT***
3210 REM
3240 REM DISPONE IN ORDINE CRESCENTE UN INSIEME DISORDINATO DI NUMERI.
3250 REM E' EFFICIENTE PER INSIEMI CONTENENTI MENO DI 50 NUMERI.
3270 REM
3280 REM VARIABILI:
3290 REM I.....VARIABILE INDICE
3300 REM L.....ELEMENTO GUIDA DELLA LISTA CORRENTE
3310 REM N.....NUMERO DI ELEMENTI DELL'INSIEME
3320 REM X().....INSIEME DEI NUMERI
3330 REM X.....VARIABILE DI SCAMBIO
3340 REM
3350 REM DIMENSIONI:
3360 DIM X(100)
3370 REM
3380 REM INPUT: N, X()
3390 REM OUTPUT: X() <ORDINATO>
3400 REM
3401 N=5
3402 FOR I=1 TO N
3403 READ X(I)
3404 PRINT X(I);
3405 NEXT I
3406 PRINT
3407 DATA 17,2,6,9,21
3408 REM
3420 FOR L=1 TO N-1
3430 FOR I=L+1 TO N
3440 IF X(I)<X(L) THEN 3450 ELSE 3480
3450 X=X(I)
3460 X(I)=X(L)
3470 X(L)=X
3480 REM
3490 NEXT I
3500 NEXT L
3510 REM
3511 FOR I=1 TO N
3512 PRINT X(I);
3513 NEXT I
3514 PRINT
3515 REM
3520 END

```

```

17 2 6 9 21
2 6 9 17 21

```

```

3600 REM ***SHELL SORT***
3610 REM
3650 REM METTE IN ORDINE CRESCENTE UN INSIEME DISORDINATO DI NUMERI.
3660 REM E' EFFICIENTE PER INSIEMI CONTENENTI DA 50 A 100 NUMERI.
3670 REM
3720 REM VARIABILI:
3730 REM E#.....INDICATORE DI SCAMBIO
3740 REM G.....INTERVALLO
3750 REM I.....VARIABILE INDICE
3760 REM N.....NUMERO DI ELEMENTI NELLA LISTA
3770 REM S.....PASSO ATTRAVERSO L'INTERVALLO
3780 REM T.....
3790 REM X().....INSIEME DEI NUMERI
3800 REM X.....VARIABILE DI SCAMBIO
3810 REM
3820 REM DIMENSIONI:
3830 DIM X(100)
3840 REM

```

```

3850 REM INPUT: N, X()
3860 REM OUTPUT: X() <ORDINATO>
3870 REM
3871 N=N
3872 FOR I=1 TO N
3873 READ X(I)
3874 PRINT X(I);
3875 NEXT I
3876 PRINT
3877 DATA 17,2,6,9,21
3878 REM
3890 REM CONFRONTA GLI ELEMENTI NELL'INTERVALLO DI GRANDEZZA N/2
3892 REM FINCHE' NON CI SONO PIU' SCAMBI, POI TABLIA A META'
3900 REM L'INTERVALLO E RIPETE L'OPERAZIONE.
3910 REM
3920 G=N
3930 REM
3940 IF G<=1 THEN 4110
3950 G=INT(G/2)
3960 T=N-G
3970 E$="NESSUND SCAMBIO"
3980 FOR I=1 TO T
3990 S=I+G
4000 IF X(I)>X(S) THEN 4010 ELSE 4050
4010 X=X(I)
4020 X(I)=X(S)
4030 X(S)=X
4040 E$="SCAMBIO"
4050 REM
4060 NEXT I
4070 IF E$="NESSUND SCAMBIO" THEN 4090
4080 GOTO 3970
4090 REM
4100 GOTO 3940
4110 REM
4111 FOR I=1 TO N
4112 PRINT X(I);
4113 NEXT I
4114 PRINT
4115 REM
4120 END

```

```

17 2 6 9 21
2 6 9 17 21

```

```

4200 REM ***PLOT***
4210 REM
4240 REM DISEGNA TUTTI I PUNTI NELLA REGIONE X1-X2,Y1-Y2.
4250 REM SU CARTA, LA REGIONE E' LARGA W9 CARATTERI (Y)
4260 REM E LUNGA L9 LINEE (X).
4270 REM
4280 REM VARIABILI:
4290 REM G().....GRAFICO SU CUI SONO SEGNATI I PUNTI
4300 REM I,J.....VARIABILI INDICE
4310 REM N.....NUMERO DI PUNTI
4320 REM O.....CONTATORE DEI PUNTI ESTERNI
4330 REM X(),Y()....COORDINATE DEI PUNTI
4340 REM X1.....LIMITE SINISTRO DEL GRAFICO
4350 REM X2.....LIMITE DESTRO DEL GRAFICO
4360 REM Y1.....LIMITE INFERIORE DEL GRAFICO
4370 REM Y2.....LIMITE SUPERIORE DEL GRAFICO
4380 REM
4390 REM COSTANTI:
4400 L9=20
4410 W9=36


```

```

4420 REM
4430 REM DIMENSIONI:
4440 DIM G(20,36),X(100),Y(100)
4450 REM
4460 REM INPUT: N, X(), Y(), X1, X2, Y1, Y2
4470 REM OUTPUT: --
4480 REM
4481 X1=0
4482 X2=50
4483 Y1=0
4484 Y2=50
4485 N=20
4486 FOR I=1 TO N
4487   READ X(I),Y(I)
4488 NEXT I
4489 DATA 1,1,5,5,15,15,30,30,50,50
4490 DATA 1,4,3,8,8,10,44,10,37,11,38,6,3,9
4491 DATA 23,-7,-12,-55,16,0,0,48,45,45
4492 DATA 2,4,3,9,4,16,5,25,6,36
4493 REM
4500 O=0
4510 FOR I=1 TO L9
4520   FOR J=1 TO M9
4530     G(I,J)=0
4540   NEXT J
4550 NEXT I
4560 REM
4570 FOR I=1 TO N
4580   IF (X(I)>=X1) AND (X(I)<=X2) THEN 4590 ELSE 4680
4590   IF (Y(I)>=Y1) AND (Y(I)<=Y2) THEN 4620 ELSE 4680
4620   X(I)=X(I)/(X2-X1)
4630   Y(I)=Y(I)/(Y2-Y1)
4640   X(I)=INT(X(I)*(L9-1)+1,5)
4650   Y(I)=INT(Y(I)*(M9-1)+1,5)
4660   G(X(I),Y(I))=G(X(I),Y(I))+1
4670   GOTO 4695
4680 REM
4690   O=O+1
4695 REM
4700 NEXT I
4710 REM
4720 PRINT"PUNTI:";"I:" INTERNI:";"N-O;" ESTERNI:";O
4730 PRINT"X DA"X1;"A"X2;"DALL'ALTO IN BASSO"
4740 PRINT"Y DA"Y1;"A"Y2;"DA SINISTRA A DESTRA"
4750 PRINT"* INDICA 1 (O PIU') PUNTI"
4760 PRINT
4770 PRINT"I:"
4780 FOR J=1 TO M9+1
4790   PRINT"-:";
4800 NEXT J
4810 PRINT"Y"
4820 FOR I=1 TO L9
4830   PRINT"I:";
4840   FOR J=1 TO M9
4850     IF G(I,J)=0 THEN 4860 ELSE 4880
4860     PRINT" ";
4870     GOTO 4895
4880 REM
4890     PRINT"*:";
4895 REM
4900   NEXT J
4910   PRINT
4920 NEXT I
4930 PRINT"X"
4940 REM
4950 END

```

PUNTI: 20 INTERNI: 18 ESTERNI: 2
 X DA 0 A 50 DALL'ALTO IN BASSO
 Y DA 0 A 50 DA SINISTRA A DESTRA
 * INDICA I (O PIU') PUNTI



REM ***ISTOGRAMMA***

```

5010 REM
5040 REM STAMPA UN ISTOGRAMMA DELLA DISTRIBUZIONE DI UNA LISTA
5050 REM DI NUMERI X().
5060 REM
5070 REM VARIABILI:
5080 REM H().....LUNGHEZZA DI OGNI BARRA DELL'ISTOGRAMMA
5090 REM I.....INTERVALLO DELL'ISTOGRAMMA
5100 REM J,K.....VARIABILI INDICE
5110 REM M.....H() MASSIMO
5120 REM N.....NUMERO DI ELEMENTI DELLA LISTA
5130 REM X().....NUMERI CONTENUTI NELLA LISTA
5140 REM X1.....LIMITE MINIMO DELL'ISTOGRAMMA
5150 REM X2.....LIMITE MASSIMO DELL'ISTOGRAMMA
5160 REM
5170 REM COSTANTI:
5180 H9=20
5190 L9=35
5200 REM
5220 REM DIMENSIONI:
5230 DIM H(20),X(100)
5240 REM
5250 REM INPUT: N, X(), X1, X2
5260 REM OUTPUT: --
5270 REM
5271 X1=0
5272 X2=100
5273 N=40
5274 FOR J=1 TO N
5275 READ X(J)
5276 NEXT J
5277 DATA 77,26,88,86,75,76,83,86,90,68
5278 DATA 71,55,83,77,71,66,74,76,86,60
5279 DATA 85,70,93,91,80,83,88,94,96,74
5280 DATA 66,78,79,88,86,84,90,78,75,79
5281 REM
5290 REM DISPONE LE X() NELLA BARRA CORRISPONDENTE DELL'ISTOGRAMMA

```

```

5300 REM H(K) E RICERCA UN NUOVO MASSIMO.
5310 REM
5320 I=(X2-X1)/H9
5330 M=L9
5340 REM
5350 FOR J=1 TO N
5360 K=INT(H9*X(J)/(X2-X1)+1)
5370 H(K)=H(K)+1
5380 IF H(K)>M THEN 5390 ELSE 5400
5390 M=H(K)
5400 REM
5410 NEXT J
5420 REM
5430 REM STAMPA L'ISTOGRAMMA
5440 REM
5450 PRINT"DA":X1;"A":X2;"AD INTERVALLI DI":I
5460 PRINT"LA MASSIMA ALTEZZA POSSIBILE E'":M;"PUNTI"
5470 PRINT"OGNI PUNTO E'":M/L9;"UNITA'"
5480 FOR J=1 TO H9
5490 PRINT"I";
5495 IF H(J)=0 GOTOS550
5500 FOR K=1 TO INT(H(J)/M*(L9)+.5)
5510 PRINT"*";
5520 NEXT K
5530 PRINT
5540 NEXT J
5550 REM
5560 END

```

DA 0 A 100 AD INTERVALLI DI 5
LA MASSIMA ALTEZZA POSSIBILE E' 35 PUNTI
OGNI PUNTO E' 1 UNITA'

```

I
I
I
I
I
I*
I
I
I
I
I
I*
I*
I***
I*****
I*****
I*****
I*****
I*

```

```

5600 REM ***TORTA***
5610 REM
5640 REM DISEGNA UN GRAFICO CIRCOLARE PER UNA LISTA DI NUMERI X().
5650 REM PER ADATTARE IL GRAFICO ALLA STAMPANTE O ALLO SCHERMO SPOSTA
5660 REM IL CENTRO DEL CERCHIO (R1,C1) E MODIFICARE IL RAGGIO R2.
5670 REM PER ELIMINARE L'EFFETTO FOOTBALL, MODIFICARE F7.
5680 REM
5710 REM VARIABILI:
5720 REM C.....INDICE DI COLONNA
5730 REM C1.....COLONNA DEL CENTRO
5740 REM C9.....COLONNA PIU' LARGA

```

```

5750 REM F7.....FATTORE 'FOOTBALL'
5760 REM G*(.....GRAFICO SU CUI E' DISEGNATA LO SCHEMA
5770 REM I.....VARIABILE INDICE
5780 REM N.....NUMERO DI ELEMENTI DELLA LISTA
5790 REM P1.....PI GRECO
5800 REM P().....PERCENTUALI DI X()
5810 REM R.....INDICE DI RIGA
5820 REM R1.....RIGA DEL CENTRO
5830 REM R9.....RIGA PIU' LARGA
5840 REM R2.....RAGGIO DEL CERCHIO
5850 REM R3.....RAGGIO PARZIALE
5860 REM T.....VARIABILE TOTALE
5870 REM X().....LISTA DI NUMERI
5880 REM
5890 REM INPUT: N, X()
5900 REM OUTPUT: --
5910 REM
5911 N=4
5912 FOR I=1 TO N
5913 READ X(I)
5914 NEXT I
5915 DATA 1,3,5,7
5916 REM
5920 C1=30
5930 R1=20
5940 R2=20
5950 P1=3.1415926
5960 F7=.75
5970 C9=INT((C1+(1/F7)*R2)+1)
5980 R9=INT((R1+F7*R2)+1)
5990 REM
6000 REM
6010 PRINT"LA TORTA VA DALLA COLONNA":C1-(1/F7)*R2;"ALLA":C9
6020 PRINT"LA TORTA VA DALLA RIGA":R1-F7*R2;"ALLA":R9
6030 PRINT
6040 REM
6050 REM DIMENSIONI: ASSICURARSI CHE G*( ) SIA (R9,C9)
6060 DIM G*(40,60)
6070 REM
6080 REM AZZERA LO SCHEMA
6090 REM
6100 FOR R=1 TO R9
6110 FOR C=1 TO C9
6120 G*(R,C)=" "
6130 NEXT C
6140 NEXT R
6150 REM
6160 REM CALCOLA LA SOMMA E LE PERCENTUALI
6170 REM
6180 T=0
6190 FOR I=1 TO N
6200 T=T+X(I)
6210 NEXT I
6220 FOR I=1 TO N
6230 P(I)=X(I)/T
6240 NEXT I
6250 REM
6260 REM DEFINISCE IL CENTRO DELLA TORTA
6270 REM
6280 G*(R1,C1)="*"
6290 REM
6300 REM DEFINISCE IL BORDO DELLA TORTA
6310 REM
6320 FOR I=0 TO 2*P1 STEP P1/10
6330 R=F7*R2*SIN(I)
6340 C=(1/F7)*R2*COS(I)


```

```

6350      C=C+C1
6360      R=R+R1
6370      G$(INT(R+.5),INT(C+.5))="*"
6380      NEXT I
6390 REM
6400 REM DISEGNA LE FETTE DI TORTA
6410 REM
6420      T=0
6430      FOR I=1 TO N
6440          T=(P(I)*2*P1)+T
6450          FOR J=1 TO 5
6460              R3=J/5*R2
6470              R=R1+F7*R3*SIN(T)
6480              C=C1+(1/F7)*R3*COS(T)
6490              G$(INT(R+.5),INT(C+.5))="*"
6500          NEXT J
6510      NEXT I
6520 REM
6530 REM STAMPA IL DISEGNO
6540 REM
6550      FOR R=R9 TO 1 STEP -1
6560          FOR C=1 TO C9
6570              PRINT G$(R,C);
6580          NEXT C
6590          PRINT
6600      NEXT R
6610      END

```

LA TORTA VA DALLA COLONNA 3.3333 ALLA 57
 LA TORTA VA DALLA RIGA 5 ALLA 36




```

6700 REM ***AMMORTAMENTO***
6710 REM
6750 REM DATO L'AMMONTARE DI UN PRESTITO X, IL NUMERO DI ANNI IN CUI
6760 REM VIENE PAGATO Y, E L'INTERESSE ANNUALE I, STAMPA UNA TABELLA
6770 REM DELL'INTERESSE E DEL CAPITALE DI OGNI PAGAMENTO MENSILE.
6780 REM
6800 REM VARIABILI:
6810 REM I.....TASSO DI INTERESSE ANNUALE
6820 REM I1.....INTERESSE MENSILE
6830 REM M.....VARIABILE INDICE DEL MESE
6840 REM P.....PAGAMENTO MENSILE
6850 REM X.....AMMONTARE DEL PRESTITO
6860 REM Y.....DURATA DEL PRESTITO IN ANNI
6870 REM
6880 REM INPUT: I, X, Y
6890 REM OUTPUT: --
6900 REM
6901 I=.105
6902 X=1000
6903 Y=2
6904 REM
6910 REM PAGAMENTI MENSILI
6920 P=X*((I/12)/(1-(1/(1+(I/12))^(Y*12))))
6930 REM
6940 PRINT"AMMONTARE DEL PRESTITO:";X
6950 PRINT"TASSO DI INTERESSE ANNUALE:";I
6960 PRINT"NUMERO DI PAGAMENTI MENSILI:";Y*12
6970 PRINT"AMMONTARE DEI PAG. MENSILI:";P
6980 PRINT
6990 REM
7000 PRINT"MESE INTERESSE", "CAPITALE", "SALDO PASSIVO"
7010 FOR M=1 TO (Y*12)
7020 I1=(I/12)*X
7030 X=X-(P-I1)
7040 PRINT M;I1, (P-I1), X
7050 NEXT M
7060 END

```

AMMONTARE DEL PRESTITO: 1000
TASSO DI INTERESSE ANNUALE: .105
NUMERO DI PAGAMENTI MENSILI: 24
AMMONTARE DEI PAG. MENSILI: 46.3763

MESE	INTERESSE	CAPITALE	SALDO PASSIVO
1	8.75	37.6263	962.374
2	8.42077	37.9555	924.418
3	8.08866	38.2876	886.131
4	7.75364	38.6227	847.508
5	7.41569	38.9606	808.547
6	7.07479	39.3015	769.246
7	6.7309	39.6454	729.6
8	6.384	39.9923	689.608
9	6.03407	40.3422	649.266
10	5.68108	40.6952	608.571
11	5.32499	41.0513	567.519
12	4.96579	41.4105	526.109
13	4.60345	41.7728	484.336
14	4.23794	42.1384	442.198
15	3.86923	42.5071	399.691
16	3.49729	42.879	356.812
17	3.1221	43.2542	313.557
18	2.74363	43.6327	269.925
19	2.36184	44.0145	225.91
20	1.97671	44.3996	181.511
21	1.58822	44.7881	136.723
22	1.19632	45.18	91.5426
23	.800998	45.5753	45.9673
24	.402214	45.9741	-6.77109E-03

```

7100 REM ***DEPA***
7110 REM DEPREZZAMENTO-A
7120 REM
7150 REM CALCOLA LA RETTA DI DEPREZZAMENTO DELLA CIFRA X
7160 REM DEPREZZATA SU Y ANNI.
7170 REM
7180 REM VARIABILI:
7190 REM D().....DEPREZZAMENTO ANNUALE
7200 REM I.....VARIABILE INDICE PER IL CICLO ANNUALE
7210 REM X.....AMMONTARE INIZIALE
7220 REM Y.....ANNI DI DEPREZZAMENTO (VITA DELL'INVESTIMENTO)
7230 REM
7240 REM INPUT: X, Y
7250 REM OUTPUT: D()
7260 REM
7261 X=1000
7262 Y=8
7263 REM
7270 PRINT"RETTA DI DEPREZZAMENTO DI *":X;"IN":
7280 PRINT Y;"ANNI."
7290 PRINT
7300 PRINT"ANNO", "DEPREZZAMENTO"
7310 FOR I=1 TO Y
7320 D(I)=X/Y
7330 PRINT I, D(I)
7340 NEXT I
7350 END

```

RETTA DI DEPREZZAMENTO DI * 1000 IN 8 ANNI.

ANNO	DEPREZZAMENTO
1	125
2	125
3	125
4	125
5	125
6	125
7	125
8	125

```

7400 REM ***DEPB***
7410 REM DEPREZZAMENTO-B
7420 REM
7450 REM CALCOLA IL SALDO DI DEPREZZAMENTO DECLINANTE DOPPIO
7460 REM CON CONVERSIONE ALLA RETTA PER UNA CIFRA X CON UNA
7470 REM VITA DI Y ANNI.
7480 REM
7490 REM VARIABILI:
7500 REM B.....SALDO RESIDUO
7510 REM D1.....DEPREZZAMENTO
7520 REM D2.....LINEA DEL DEPREZZAMENTO
7530 REM D().....DEPREZZAMENTO SCELTO
7540 REM I.....VARIABILE INDICE PER IL CICLO ANNUALE
7550 REM X.....AMMONTARE INIZIALE
7560 REM Y.....ANNI DI DEPREZZAMENTO (VITA DELL'INVESTIMENTO)
7570 REM
7580 REM INPUT: X, Y
7590 REM OUTPUT: D()
7600 REM
7601 X=1000
7602 Y=8
7603 REM
7610 PRINT"SALDO DI DEPREZZAMENTO DECLINANTE DOPPIO ":
7620 PRINT"D1";X;" IN";Y;"ANNI."
7630 PRINT

```

```

7640 PRINT"ANNO", "DECL.DOPPIO", "RETTA",
7645 PRINT"SALDO"
7650 B=X
7660 FOR I=1 TO Y
7670   D1=B*2/Y
7680   D2=B/(Y-I+1)
7690   IF D1>D2 THEN 7700 ELSE 7740
7700   D(I)=D1
7710   B=B-D(I)
7720   PRINT I,D(I),,B
7730   GOTO 7780
7740 REM
7750   D(I)=D2
7760   B=B-D(I)
7770   PRINT I,,D(I),B
7780 REM
7790 NEXT I
7800 END

```

SALDO DI DEPREZZAMENTO DECLINANTE DOPPIO DI 1000 * IN 8 ANNI.

ANNO	DECL.DOPPIO	RETTA	SALDO
1	250		750
2	187.5		562.5
3	140.625		421.875
4	105.469		316.406
5		79.1016	237.305
6		79.1016	158.203
7		79.1016	79.1016
8		79.1016	0

```

7900 REM ***DEPC***
7910 REM DEPREZZAMENTO-C
7920 REM
7950 REM CALCOLA IL DEPREZZAMENTO CUMULATIVO DI
7960 REM UNA SOMMA X SU Y ANNI.
7970 REM
7980 REM VARIABILI:
7990 REM D().....DEPREZZAMENTO ANNUALE
8000 REM I.....VARIABILE INDICE PER IL CICLO ANNUALE
8010 REM X.....AMMONTARE INIZIALE
8020 REM Y.....ANNI DI DEPREZ. (VITA DELL'INVESTIMENTO)
8030 REM
8040 REM INPUT: X, Y
8050 REM OUTPUT: D()
8060 REM
8061 X=1000
8062 Y=8
8063 REM
8070 PRINT"DEPREZZAMENTO CUMULATIVO ";
8080 PRINT"D(I);X;"* IN";Y;"ANNI."
8090 PRINT
8100 REM
8110 REM FORMULA PER LA SOMMA DI 1 A Y
8120 S=Y*(Y+1)/2
8130 REM
8140 PRINT"ANNO", "DEPREZZAMENTO"
8150 FOR I=1 TO Y
8160   D(I)=X*(Y-I+1)/S
8170   PRINT I,D(I)
8180 NEXT I
8190 END

```

DEPREZZAMENTO CUMULATIVO DI 1000 \$ IN 8 ANNI.

ANNO	DEPREZZAMENTO
1	222.222
2	194.444
3	166.667
4	138.889
5	111.111
6	83.333
7	55.556
8	27.778

```

8200 REM ***LIVELLARE***
8210 REM
8240 REM APPLICA IL LIVELLAMENTO 'TRE SU TRE' A UNA
8250 REM SERIE DI NUMERI X().
8260 REM LA X(I) RISULTANTE E'
8270 REM (1/9)*X(I-2)+(2/9)*X(I-1)+(3/9)*X(I)+(2/9)*X(I+1)+(1/9)*X(I+2)
8280 REM PER VALORI DI I COMPRESI TRA 3 E (N-2).
8290 REM
8300 REM VARIABILI:
8310 REM I.....VARIABILE INDICE PER IL CICLO
8320 REM N.....NUMERO DI ELEMENTI DELLA SERIE
8330 REM X().....NUMERI DELLA SERIE
8340 REM Y().....SERIE TEMPORALI
8350 REM
8360 REM DIMENSIONI:
8370 DIM X(100),Y(100)
8380 REM
8390 REM INPUT: N, X()
8400 REM OUTPUT: X(O) <APPIATTITO>
8410 REM
8411 N=25
8412 FOR I=1 TO N
8413 READ X(I)
8414 PRINT I;X(I),
8415 NEXT I
8416 PRINT
8417 PRINT
8418 DATA 1,2,3,4,5,5,4,3,2,1,1,2,3,4,5,5,4,3,2,1,1,2,3,4,5
8419 REM
8420 FOR I=2 TO N-1
8430 Y(I)=(X(I-1)+X(I)+X(I+1))/3
8440 NEXT I
8450 Y(1)=X(1)
8460 Y(N)=X(N)
8470 REM
8480 FOR I=2 TO N-1
8490 X(I)=(Y(I-1)+Y(I)+Y(I+1))/3
8500 NEXT I
8510 REM
8511 FOR I=1 TO N
8512 PRINT I;X(I),
8513 NEXT I
8514 PRINT
8515 REM
8520 END

```

1	1	2	2	3	3	4	4	5	5
6	5	7	4	8	3	9	2	10	1
11	1	12	2	13	3	14	4	15	5
16	5	17	4	18	3	19	2	20	1
21	1	22	2	23	3	24	4	25	5

1	1	2	2	3	3	4	3.88889	5	4.44445
6	4.44444	7	3.88889	8	3	9	2.11111	10	1.55556
11	1.55556	12	2.11111	13	3	14	3.88889	15	4.44445
16	4.44444	17	3.88889	18	3	19	2.11111	20	1.55556
21	1.55556	22	2.11111	23	3	24	4	25	5

```

100 REM ***GRAFICO A BARRE***
110 REM
130 REM STAMPA UN GRAFICO A BARRE DI UNA LISTA CONTENENTE
140 REM FIND A 100 NUMERI.
150 REM
160 REM VARIABILI:
170 REM I.....VARIABILE INDICE
180 REM L.....LUNGHEZZA DELLA BARRA
190 REM M.....IL NUMERO PIU' GRANDE DELLA LISTA
200 REM N.....NUMERO DI ELEMENTI DELLA LISTA
210 REM P.....PERCENTUALE SUL TOTALE DI OGNI NUMERO
220 REM S.....SOMMA DEI NUMERI
230 REM X().....NUMERI DELLA LISTA
240 REM
250 REM DIMENSIONI:
260 DIM X(100)
270 REM
280 REM PROGRAMMA PRINCIPALE
290 REM
300 GOSUB 370
310 GOSUB 530
320 GOSUB 680
330 REM
340 STOP
350 REM
360 REM
370 REM SUBROUTINE CHE LEGGE I DATI
380 REM INPUT: ---
390 REM OUTPUT: X(), N
400 REM
410 FOR I=1 TO 100
420 READ X(I)
430 IF X(I)=-.999999 THEN 450
440 NEXT I
450 REM
460 N=I-1
470 REM
480 DATA 12,92,27,45,36
490 DATA -.999999
500 RETURN
510 REM
520 REM
530 REM SUBROUTINE CHE TROVA IL MASSIMO
540 REM INPUT: X(), N
550 REM OUTPUT: M
560 REM
570 M=X(1)
580 REM
590 FOR I=2 TO N
600 IF X(I)>M THEN 610 ELSE 620
610 M=X(I)
620 REM
630 NEXT I
640 REM
650 RETURN
660 REM
670 REM
680 REM SUBROUTINE CHE STAMPA IL GRAFICO
690 REM INPUT: X(), M, N

```

```

700 REM OUTPUT: ---
710 REM
720 REM CALCOLA LA PERCENTUALE SUL TOTALE, LA LUNGHEZZA DELLA BARRA
730 REM E POI STAMPA LA BARRA E LA PERCENTUALE.
740 REM
750   FOR I=1 TO N
760     L=INT(X(I)/M*50+.5)
770 REM
780     PRINT TAB(8);",."
790     PRINT X(I);TAB(8);"+";
800 REM
810     FOR J=1 TO L
820       PRINT"*";
830     NEXT J
840     PRINT
850 REM
860   NEXT I
870 REM
880 RETURN
890 REM
900 REM
910   END

```

```

12   .
    ++++++
    .
92   ++++++++++++++++++++++++++++++++++++++
    .
27   ++++++
    .
45   ++++++
    .
36   ++++++

```

L'acquisto

Quando avete comperato la vostra prima automobile, avevate qualche idea di come la volevate, perché sapevate a cosa doveva servirvi. Comperare il primo computer è più o meno la stessa cosa: l'acquisto sarà ben fatto nella misura in cui si sa per cosa verrà utilizzato.

VERIFICA DELLE ESIGENZE

Un modo per approfondire l'argomento computer è quello d'imparare a programmare quello di un vostro amico. Se pensate di non avere il tempo per farlo, ricordate almeno il vecchio e saggio detto "Chiedi a chi ha provato" ed andate a consultare due o tre persone che stanno risolvendo i problemi che volete affrontare col computer. Possono suggerirvi cose a cui non avevate pensato. Senz'altro vi daranno dei pareri sulle caratteristiche della loro macchina e sull'efficienza del servizio di manutenzione. Si tratta dunque di riflettere su cosa vorrete fare col vostro computer.

Se vi servirà per hobby, sarà bene avere buone possibilità grafiche a colori. Se lo adopererete per calcoli finanziari, ci vorrà la stampante. Invece per gestione di archivi avrete bisogno di molta memoria su disco.

Proprio come quando si compera una macchina, bisogna trovare la giusta combinazione dei requisiti richiesti per le particolari esigenze. Comperare accessori che non servono non è saggio. Sfortunatamente, il paragone con le automobili vale anche in altri sensi. Ci sono così tante combinazioni di

prodotti e possibilità che l'unico modo per capire come vanno realmente le cose è l'esperienza diretta. Al momento dell'acquisto, insistete perché possiate provare proprio la configurazione che intendete comperare. Ricordate che è l'ultimo momento in cui potete scoprire le eventuali lacune, prima di dover pagare.

BUONA ASSISTENZA

Qualche volta, dopo l'acquisto, il computer si guasta. Quando ciò accade, salta subito alla mente il paragone con l'acquisto dell'automobile. Sia l'uno che l'altra, se non vanno bene, devono essere riparati al più presto. Perciò, anche in questo caso, l'assistenza gioca un ruolo importante nella scelta d'acquisto. A seconda delle varie esigenze, un servizio d'assistenza può essere giudicato "buono" se è rapido, affidabile o a buon mercato. Se il computer servirà per lavoro, si preferirà la rapidità e l'affidabilità. Assicuratevi comunque della bontà del servizio d'assistenza, prima di scegliere.

BUONA MARCA

I computer sono adesso allo stadio in cui erano le automobili nel 1910. Sul mercato ci sono tantissime case produttrici e non si sa bene chi sopravviverà alla concorrenza durante i prossimi cinque anni. Fino al 1980 in America due case si sono distinte: Radio Shack e Apple. Ma altre grosse marche di personal computer sono Heathkit (tra i suoi c'è anche un computer della Digital Equipment Corporation), Hewlett-Packard, e Texas Instruments. E poi Atari, Commodore Pet, CompuColor, Cromenco, Exidy ed Ohio Scientific.

Tutti i programmi di questo libro funzioneranno sul personal computer più venduto negli Stati Uniti: il TRS-80 Radio Shack. La versione di Basic in questione è Level II. La quantità di memoria principale richiesta è di 16 000 caratteri (16K) ed il supporto della memoria di massa può essere una cassetta magnetica. Il Basic della Radio Shack ha una caratteristica sorprendente: conserva la struttura dei programmi. La maggior parte dei Basic per personal computer elimina le rientranze nel testo dei programmi. Per chi vuol scrivere programmi eleganti e ben strutturati (Appendice A) le rientranze sono essenziali.

Negli Stati Uniti la maggior concorrente della Radio Shack è la Apple. La sua grafica ad alta risoluzione ed a colori è molto interessante. Con l'Apple, per conservare le rientranze nei testi, potete usare un trucco che trasforma in un buon servizio una caratteristica negativa. Il Basic dell'Apple

permette di scrivere:

```
100 LET A = 2 : LET B = 4
```

I due punti (:) separano due istruzioni scritte su una stessa riga. Mettere due istruzioni su una stessa riga non è una buona norma di programmazione, ma ci si può servire dei due punti per realizzare delle pseudo-rientranze. Per esempio:

```
100 FOR I = 1 TO 5  
110 : PRINT I  
120 : PRINT I*3.1415926  
130 : LET B = I * 3  
140 NEXT I
```

Questo trucco andrà bene per molti Basic che eliminano le rientranze. Si usa il carattere che separa le istruzioni per dar luogo a delle pseudo-rientranze.

Comperare un computer è proprio come comperare una macchina. Non dovrebbe sorprendervi il fatto che più cura mettete nella scelta dell'acquisto, più probabile è che poi siate soddisfatti.

Il VisiCalc

Pensate di essere davanti ad un registro aperto con le righe numerate 1, 2, 3, ... e le colonne rappresentate dalle lettere A, B, C, ... Supponete di essere in grado di scrivere un numero o una etichetta in qualsiasi punto della pagina. Immaginate di poter fare in modo che la colonna C sia la somma della A e della B e che la colonna C possa apparire come per magia. Pensate di poter correggere un errore nella colonna A e di aver aggiornato automaticamente tutte le cifre che derivavano da quel valore.

Supponete di dover inserire altre tre righe e che nel registro appaiano le tre righe bianche proprio nel punto giusto. Immaginate di dover confrontare la colonna A e la colonna Z e che il registro vi aiuti spostandole una proprio vicino all'altra.

Pensate di poter cambiare un singolo valore per vedere "cosa succede se...?" e di vedere improvvisamente tutti i calcoli rifatti sulla base del nuovo valore.

Tutto questo è possibile con VisiCalc.

VisiCalc fa tutte queste cose (e molte altre) sul video del vostro computer. Se avete una stampante, potete stampare i risultati. Se disponete di un disco di memoria, potete memorizzare tutto per altre occasioni. Le tabelle costruite con VisiCalc possono essere lette da programmi Basic e viceversa. Non tutti i piccoli calcolatori hanno a disposizione il VisiCalc. Apple, Atari, Hewlett-Packard, Pet e Radio Shack sicuramente sì. E prossimamente ci sarà anche su altri. Ogni programma di questo libro che può essere scritto facilmente in VisiCalc è segnato con un asterisco nell'indice dei programmi.

La traduzione da Basic a VisiCalc è semplicissima. Ad esempio, il primo programma del libro, in VisiCalc è così:

A1 = ANNO
 B1 = CIFRA
 C1 = TASSO
 A2 = 0
 B2 = 100
 C2 = 1.15
 A3 = A2 + 1
 A4 = A3 + 1
 A5 = A4 + 1
 A6 = A5 + 1
 B3 = C2 * B2
 B4 = C2 * B3
 B5 = C2 * B4
 B6 = C2 * B5

La tabella del VisiCalc sarebbe questa:

	A	B	C
1	ANNO	CIFRA	TASSO
2	0	100.00	1.15
3	1	115.00	
4	2	132.25	
5	3	152.09	
6	4	174.90	

Cambiando la quantità iniziale, B2, o il tasso d'interesse, C2, si cambierà l'intera tabella.

Il secondo programma del libro, INFLAZIONE, in VisiCalc assomiglierà alla fotografia che segue.

Se possedete il VisiCalc, potete provare a tradurre alcuni dei programmi di questo libro.

F10

	A	B	C	D	E	F
1	GROWTH RT	1.03	1.15		1.12	
2						
3	YEAR	MAT	LABOR TOTAL	CST	PRICE MARGIN(%)	
4	0.00	56.00	21.00	77.00	100.00	23.00
5	1.00	57.68	24.15	81.83	112.00	26.94
6	2.00	59.41	27.77	87.18	125.44	30.50
7	3.00	61.19	31.94	93.13	140.49	33.71
8	4.00	63.03	36.73	99.76	157.35	36.60
9	5.00	64.92	42.24	107.16	176.23	39.20
10						
11						
12						

Fig. D.1 Un programma in VisiCalc



*Finito di stampare il 31 ottobre 1983 presso Lito Velox - Trento
Fotocomposto e impaginato da Lito Velox - Trento
Printed in Italy.*

